# Calculemus (Vol. 2: Demostraciones con Lean4)

José A. Alonso Jiménez

Grupo de Lógica Computacional Dpto. de Ciencias de la Computación e Inteligencia Artificial Universidad de Sevilla

Sevilla, 10 de julio de 2023 (versión del 12 de agosto de 2023)

Esta obra está bajo una licencia Reconocimiento-NoComercial-Compartirlgual 2.5 Spain de Creative Commons.

#### Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

#### **Bajo las condiciones siguientes:**



**Reconocimiento**. Debe reconocer los créditos de la obra de la manera especificada por el autor.



**No comercial**. No puede utilizar esta obra para fines comerciales.



**Compartir bajo la misma licencia**. Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- Alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <a href="http://creativecommons.org/licenses/by-nc-sa/2">http://creativecommons.org/licenses/by-nc-sa/2</a>. 5/es/ o envie una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

### Índice general

1.	Intr	oducción	5
2.	Den	nostraciones de una propiedad de los números enteros	7
	2.1.	$\forall$ m n $\in$ N, Even n $\rightarrow$ Even (m * n)	7
3.	Prop	piedades elementales de los números reales	11
	3.1.	En $\mathbb{R}$ , (ab)c = b(ac)	11
	3.2.	En $\mathbb{R}$ , (cb)a = b(ac)	12
	3.3.	En $\mathbb{R}$ , a(bc) = b(ac)	13
	3.4.	En $\mathbb{R}$ , si ab = cd y e = f, entonces a(be) = c(df)	14
	3.5.	En $\mathbb{R}$ , si bc = ef, entonces ((ab)c)d = ((ae)f)d	16
	3.6.	En $\mathbb{R}$ , si c = ba-d y d = ab, entonces c = 0	17
	3.7.	En $\mathbb{R}$ , $(a+b)(a+b) = aa+2ab+bb$	18
	3.8.	En $\mathbb{R}$ , $(a+b)(c+d) = ac+ad+bc+bd \dots \dots \dots \dots$	20
	3.9.	En $\mathbb{R}$ , $(a+b)(a-b) = a^2-b^2$	22
	3.10.	En $\mathbb{R}$ , si c = da+b y b = ad, entonces c = 2ad	24
	3.11.	En $\mathbb{R}$ , si a+b = c, entonces (a+b)(a+b) = ac+bc	26
4.	Prop	piedades elementales de los anillos	29
	4.1.	Si R es un anillo y a $\in$ R, entonces a + 0 = a	29
	4.2.	Si R es un anillo y a $\in$ R, entonces a + -a = 0	30
	4.3.	Si R es un anillo y a, $b \in R$ , entonces -a + (a + b) = b	32
	4.4.	Si R es un anillo y a, b $\in$ R, entonces (a + b) + -b = a	33
	4.5.	Si R es un anillo y a, b, $c \in R$ tales que $a+b=a+c$ , entonces $b=c$	34
	4.6.	Si R es un anillo y a, b, $c \in R$ tales que $a+b=c+b$ , entonces $a=c$	37
	4.7.	Si R es un anillo y a $\in$ R, entonces a.0 = 0	39
	4.8.	Si R es un anillo y a $\in$ R, entonces $0.a = 0 \dots \dots$	41
	4.9.	Si R es un anillo y a, $b \in R$ tales que $a+b=0$ , entonces $-a=b$	42
	4.10.	Si R es un anillo y a, $b \in R$ tales que $a+b=0$ , entonces $a=-b$	44

4	Índice general

Bibliografía		
4.13. Si R es un anillo y a, b $\in$ R, entonces a - b = a + -b		. 48
4.12. Si R es un anillo y a $\in$ R, entonces -(-a) = a		47
4.11. Si R es un anillo, entonces $-0 = 0 \dots$		. 46

### Capítulo 1

#### Introducción

Este libro es una recopilación de los ejercicios de demostración con Lean4 que se han ido publicando, desde el 10 de julio de 20023, en el blog Calculemus.

La ordenación de los ejercicios es simplemente temporal según su fecha de publicación en Calculemus y el orden de los ejercicios en Calculemus responde a los que me voy encontrando en mis lecturas.

En cada ejercicio, se comienza proponiendo soluciones en lenguaje natural y, a continuación, se exponen distintas demostraciones con Lean4 ordenadas desde las más detalladas a las más automáticas. Al final de cada ejercicio hay un enlace para interactuar con sus soluciones en Lean4 Web.

Las soluciones del libro están en este repositorio de GitHub.

El libro se irá actualizando periódicamente con los nuevos ejercicios que se proponen diariamente en Calculemus.

Este libro es una continuación de

- DAO (Demostración Asistida por Ordenador) con Lean que es una introducción a la demostración con Lean3 y
- Calculemus (Vol. 1: Demostraciones con Isabelle/HOL y Lean3) que es la recopilación de la primera parte de los ejercicios del blog con demostraciones en Isabelle/HOL y Lean3.

### Capítulo 2

### Demostraciones de una propiedad de los números enteros

#### 2.1. $\forall$ m n $\in$ $\mathbb{N}$ , Even n $\rightarrow$ Even (m \* n)

```
-- Demostrar que los productos de los números naturales por números
-- pares son pares.
-- Demostración en lenguaje natural
-- Si n es par, entonces (por la definición de 'Even') existe un k tal que
-- \qquad n = k + k \tag{1}
-- Por tanto,
-- mn = m(k + k) (por (1))
    = mk + mk (por la propiedad distributiva)
-- Por consiguiente, mn es par.
-- Demostraciones en Lean4
import Mathlib.Data.Nat.Basic
import Mathlib.Data.Nat.Parity
import Mathlib.Tactic
open Nat
```

```
-- 1ª demostración
-- ==========
example : \forall m n : \mathbb{N}, Even n \rightarrow Even (m * n) := by
  rintro m n \langle k, hk \rangle
  use m * k
  rw [hk]
  ring
-- 2ª demostración
-- ===========
example : \forall m n : \mathbb{N}, Even n \rightarrow Even (m * n) := by
  rintro m n (k, hk)
  use m * k
  rw [hk]
  rw [mul_add]
-- 3ª demostración
-- ===========
example : \forall m n : \mathbb{N}, Even n \rightarrow Even (m * n) := by
  rintro m n (k, hk)
  use m * k
  rw [hk, mul_add]
-- 4ª demostración
-- ===========
example : \forall m n : Nat, Even n \rightarrow Even (m * n) := by
  rintro m n (k, hk); use m * k; rw [hk, mul add]
-- 5ª demostración
-- ==========
example : \forall m n : \mathbb{N}, Even n \rightarrow Even (m * n) := by
  rintro m n (k, hk)
  exact (m * k, by rw [hk, mul_add])
-- 6ª demostración
-- ===========
example : ∀ m n : Nat, Even n → Even (m * n) :=
fun m n \langle k, hk \rangle \mapsto \langle m * k, by rw [hk, mul add] \rangle
```

```
-- 7º demostración
-- ==========
example : \forall m n : \mathbb{N}, Even n \rightarrow Even (m * n) := by
  rintro m n (k, hk)
  use m * k
  rw [hk]
  exact mul add m k k
-- 8ª demostración
-- ===========
example : \forall m n : \mathbb{N}, Even n \rightarrow Even (m * n) := by
  intros m n hn
  unfold Even at *
  cases hn with
  | intro k hk =>
   use m * k
    rw [hk, mul add]
-- 9ª demostración
-- ==========
example : \forall m n : \mathbb{N}, Even n \rightarrow Even (m * n) := by
  intros m n hn
  unfold Even at *
  cases hn with
  | intro k hk =>
    use m * k
    calc m * n
       = m * (k + k) := by exact congrArg (HMul.hMul m) hk
      = m * k + m * k := by exact mul add m k k
-- 10ª demostración
-- ============
example : \forall m n : Nat, Even n \rightarrow Even (m * n) := by
  intros; simp [*, parity_simps]
```

10 Capítulo 2. Demostraciones de una propiedad de los números enteros

### Capítulo 3

### Propiedades elementales de los números reales

#### 3.1. En $\mathbb{R}$ , (ab)c = b(ac)

```
-- Demostrar que los números reales tienen la siguiente propiedad
-- (a * b) * c = b * (a * c)
-- Demostración en lenguaje natural
- - ------
-- Por la siguiente cadena de igualdades
-- (ab)c = (ba)c [por la conmutativa]
      = b(ac) [por la asociativa]
-- Demostraciones con Lean4
-- ================
import Mathlib.Tactic
import Mathlib.Data.Real.Basic
-- 1ª demostración
example
 (abc:\mathbb{R})
 : (a * b) * c = b * (a * c) :=
 (a * b) * c = (b * a) * c := by rw [mul\_comm a b]
           \_ = b * (a * c) := by rw [mul_assoc b a c]
```

```
-- 2ª demostración

example (a b c : R) : (a * b) * c = b * (a * c) :=

by

rw [mul_comm a b]

rw [mul_assoc b a c]

-- 3ª demostración

example (a b c : R) : (a * b) * c = b * (a * c) :=

by ring
```

#### 3.2. En $\mathbb{R}$ , (cb)a = b(ac)

```
-- Demostrar que los números reales tienen la siguiente propiedad
-- (c * b) * a = b * (a * c)
-- Demostración en lenguaje natural
-- Por la siguiente cadena de igualdades:
-- (c * b) * a
-- = (b * c) * a [por la conmutativa]

-- = b * (c * a) [por la asociativa]

-- = b * (a * c) [por la conmutativa]
-- Demostraciones con Lean4
- - -----
import Mathlib.Tactic
import Mathlib.Data.Real.Basic
-- 1ª demostración
example
 (abc:\mathbb{R})
  : (c * b) * a = b * (a * c) :=
calc
 (c * b) * a
   = (b * c) * a := by rw [mul\_comm c b]
   _ = b * (c * a) := by rw [mul_assoc]
  \underline{\phantom{a}} = b * (a * c) := by rw [mul\_comm c a]
```

```
-- 2ª demostración

example
    (a b c : R)
    : (c * b) * a = b * (a * c) :=

by
    rw [mul_comm c b]
    rw [mul_assoc]
    rw [mul_comm c a]

-- 3ª demostración

example
    (a b c : R)
    : (c * b) * a = b * (a * c) :=

by ring
```

#### 3.3. En $\mathbb{R}$ , a(bc) = b(ac)

```
-- Demostrar que los números reales tienen la siguiente propiedad
-- a * (b * c) = b * (a * c)
-- Demostración en lenguaje natural
-- Por la siguiente cadena de igualdades:
-- a(bc)
-- = (ab)c [por la asociativa]
-- = (ba)c [por la conmutativa]
-- = b(ac) [por la asociativa]
-- Demostraciones en Lean4
import Mathlib.Tactic
import Mathlib.Data.Real.Basic
-- 1ª demostración
example
 (a b c : \mathbb{R}) : a * (b * c) = b * (a * c) :=
calc
```

```
a * (b * c)
= (a * b) * c := by rw [←mul_assoc]
_ = (b * a) * c := by rw [mul_comm a b]
_ = b * (a * c) := by rw [mul_assoc]

-- 2ª demostración

example
(a b c : ℝ) : a * (b * c) = b * (a * c) := by
rw [←mul_assoc]
rw [mul_comm a b]
rw [mul_assoc]

-- 3ª demostración

example
(a b c : ℝ) : a * (b * c) = b * (a * c) := by ring
```

## 3.4. En $\mathbb{R}$ , si ab = cd y e = f, entonces a(be) = c(df)

```
import Mathlib.Tactic
import Mathlib.Data.Real.Basic
-- 1ª demostración
example
 (abcdef:\mathbb{R})
  (h1 : a * b = c * d)
  (h2 : e = f)
  : a * (b * e) = c * (d * f) :=
calc
  a * (b * e)
   = a * (b * f) := by rw [h2]
  \underline{\phantom{a}} = (a * b) * f := by rw [\leftarrow mul_assoc]
  \underline{\ } = (c * d) * f := by rw [h1]
  \underline{\phantom{a}} = c * (d * f) := by rw [mul_assoc]
-- 2ª demostración
example
 (abcdef:\mathbb{R})
  (h1 : a * b = c * d)
  (h2 : e = f)
  : a * (b * e) = c * (d * f) :=
  rw [h2]
  rw [←mul_assoc]
  rw [h1]
  rw [mul_assoc]
-- 3ª demostración
example
  (abcdef:\mathbb{R})
  (h1 : a * b = c * d)
  (h2 : e = f)
  : a * (b * e) = c * (d * f) :=
by
 simp [*, ←mul_assoc]
```

#### 3.5. En $\mathbb R$ , si bc = ef, entonces ((ab)c)d = ((ae)f)d

```
-- Demostrar que si a, b, c, d, e y f son números reales tales que
-- b * c = e * f
-- entonces
   ((a * b) * c) * d = ((a * e) * f) * d
-- Demostración en lenguaje natural
-- Por la siguiente cadena de igualdades
     ((ab)c)d
-- = (a(bc))d [por la asociativa]

-- = (a(ef))d [por la hipótesis]

-- = ((ae)f)d [por la asociativa]
-- Demostraciones con Lean4
import Mathlib.Data.Real.Basic
import Mathlib.Tactic
-- 1ª demostración
example
 (abcdef: \mathbb{R})
 (h : b * c = e * f)
  : ((a * b) * c) * d = ((a * e) * f) * d :=
 ((a * b) * c) * d
  = (a * (b * c)) * d := by rw [mul_assoc a]
  \underline{\ } = (a * (e * f)) * d := by rw [h]
  \underline{\phantom{a}} = ((a * e) * f) * d := by rw [\leftarrow mul_assoc a]
-- 2ª demostración
example
 (abcdef: \mathbb{R})
  (h : b * c = e * f)
  : ((a * b) * c) * d = ((a * e) * f) * d :=
  rw [mul_assoc a]
  rw [h]
  rw [←mul_assoc a]
```

```
-- 3ª demostración

example
  (a b c d e f : ℝ)
  (h : b * c = e * f)
  : ((a * b) * c) * d = ((a * e) * f) * d :=

by
  rw [mul_assoc a, h, ←mul_assoc a]
```

#### 3.6. En $\mathbb{R}$ , si c = ba-d y d = ab, entonces c = 0

```
-- Demostrar que si a, b, c y d son números reales tales que
-- \qquad c = b * a - d
     d = a * b
-- entonces
-- c = 0
-- Demostración en lenguaje natural
-- -----
-- Por la siguiente cadena de igualdades
-- c = ba - d [por la primera hipótesis]

-- = ab - d [por la conmutativa]

-- = ab - ab [por la segunda hipótesis]
       = 0
-- Demostraciones en Lean4
- - -----
import Mathlib.Data.Real.Basic
import Mathlib.Tactic
-- 1ª demostración
example
 (abcd:\mathbb{R})
  (h1 : c = b * a - d)
 (h2 : d = a * b)
 : c = 0 :=
calc
```

```
c = b * a - d := by rw [h1]
  = a * b - d := by rw [mul_comm]
 _{-} = a * b - a * b := by rw [h2]
                   := by rw [sub self]
-- 2ª demostración
example
  (abcd:\mathbb{R})
  (h1 : c = b * a - d)
  (h2 : d = a * b)
  : c = 0 :=
by
  rw [h1]
  rw [mul_comm]
  rw [h2]
  rw [sub_self]
-- 3ª demostración
example
 (abcd:\mathbb{R})
 (h1 : c = b * a - d)
 (h2 : d = a * b)
 : c = 0 :=
  rw [h1, mul_comm, h2, sub_self]
```

#### 3.7. En $\mathbb{R}$ , (a+b)(a+b) = aa+2ab+bb

```
= aa + ba + ab + bb
                               [por la asociativa]
-- = aa + (ba + ab) + bb
                               [por la asociativa]
    = aa + (ab + ab) + bb
                               [por la conmutativa]
-- = aa + 2(ab) + bb
                               [por def. de doble]
-- Demostraciones con Lean4
import Mathlib.Data.Real.Basic
import Mathlib.Tactic
variable (a b c : \mathbb{R})
-- 1ª demostración
example:
 (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
 (a + b) * (a + b)
   = (a + b) * a + (a + b) * b := by rw [mul_add]

= a * a + b * a + (a + b) * b := by rw [add_mul]
 _{-} = a * a + b * a + (a * b + b * b) := by rw [add mul]
 \_ = a * a + b * a + a * b + b * b := by rw [\leftarrowadd_assoc]
  _{-} = a * a + (b * a + a * b) + b * b := by rw [add_assoc (a * a)]
 = a * a + (a * b + a * b) + b * b := by rw [mul_comm b a]
  \_ = a * a + 2 * (a * b) + b * b := by rw [\leftarrowtwo mul]
-- 2ª demostración
example:
 (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
calc
 (a + b) * (a + b)
  = a * a + b * a + (a * b + b * b) := by rw [mul_add, add_mul, add_mul]
   = a * a + (b * a + a * b) + b * b := by rw [-add assoc, add assoc (a * a)]
  \_ = a * a + 2 * (a * b) + b * b := by rw [mul_comm b a, \( +two_mul \} \)
-- 3ª demostración
example:
 (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
calc
 (a + b) * (a + b)
   = a * a + b * a + (a * b + b * b) := by ring
   = a * a + (b * a + a * b) + b * b := by ring
 _{-} = a * a + 2 * (a * b) + b * b := by ring
-- 4º demostración
```

```
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by ring
-- 5ª demostración
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by
  rw [mul add]
  rw [add mul]
  rw [add_mul]
  rw [←add assoc]
  rw [add_assoc (a * a)]
  rw [mul_comm b a]
  rw [←two_mul]
-- 6ª demostración
example:
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
  rw [mul add, add mul, add mul]
  rw [←add_assoc, add_assoc (a * a)]
  rw [mul comm b a, ←two mul]
-- 7ª demostración
example:
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by linarith
```

#### 3.8. En $\mathbb{R}$ , (a+b)(c+d) = ac+ad+bc+bd

```
= a(c + d) + b(c + d) [por la distributiva]
-- = ac + ad + b(c + d)
                               [por la distributiva]
-- = ac + ad + (bc + bd) [por la distributiva]

-- = ac + ad + bc + bd [por la asociativa]
-- Demostraciones con Lean4
import Mathlib.Data.Real.Basic
import Mathlib.Tactic
variable (a b c d : \mathbb{R})
-- 1ª demostración
example
 : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
 (a + b) * (c + d)
  = a * (c + d) + b * (c + d) := by rw [add_mul]
= a * c + a * d + b * (c + d) := by rw [mul_add]
  \_ = a * c + a * d + (b * c + b * d) := by rw [mul add]
  \_ = a * c + a * d + b * c + b * d := by rw [\leftarrowadd_assoc]
-- 2ª demostración
example
 : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
 (a + b) * (c + d)
  = a * (c + d) + b * (c + d) := by ring

= a * c + a * d + b * (c + d) := by ring
  _{-} = a * c + a * d + (b * c + b * d) := by ring
  \_ = a * c + a * d + b * c + b * d := by ring
-- 3ª demostración
example: (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
by ring
-- 4ª demostración
example
 : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
by
   rw [add mul]
   rw [mul_add]
   rw [mul add]
   rw [← add assoc]
```

```
-- 5<sup>a</sup> demostración

example : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=

by rw [add_mul, mul_add, mul_add, ←add_assoc]
```

#### 3.9. En $\mathbb{R}$ , (a+b)(a-b) = a^2-b^2

```
-- Demostrar que si a y b son números reales, entonces
-- (a + b) * (a - b) = a^2 - b^2
-- Demostración en lenguaje natural
-- Por la siguiente cadena de igualdades:
-- (a + b)(a - b)
     = a(a - b) + b(a - b)
                                      [por la distributiva]
    = (aa - ab) + b(a - b)
                                      [por la distributiva]
   = (a^2 - ab) + b(a - b)
                                      [por def. de cuadrado]
     = (a^2 - ab) + (ba - bb)
                                      [por la distributiva]
   = (a^2 - ab) + (ba - b^2) [por def. de cuadrace]

= (a^2 + -(ab)) + (ba - b^2) [por def. de resta]

= a^2 + (-(ab) + (ba - b^2)) [por la asociativa]
                                      [por def. de cuadrado]
                                      [por def. de resta]
    = a^2 + (-(ab) + (ba + -b^2))
     = a^2 + ((-(ab) + ba) + -b^2) [por la asociativa]
    = a^2 + ((-(ab) + ab) + -b^2) [por la conmutativa]
     = a^2 + (0 + -b^2)
                                       [por def. de opuesto]
    = (a^2 + 0) + -b^2
                                       [por asociativa]
    = a^2 + -b^2
                                       [por def. de cero]
                                       [por def. de resta]
     = a^2 - b^2
-- Demostraciones con Lean4
- - ==============
import Mathlib.Data.Real.Basic
import Mathlib.Tactic
variable (a b : ℝ)
-- 1º demostración
```

```
-- ==========
example: (a + b) * (a - b) = a^2 - b^2 :=
calc
 (a + b) * (a - b)
 _{-} = (a^2 - a * b) + (b * a - b * b) := by rw [mul_sub]
 = (a^2 - a * b) + (b * a - b^2) := by rw [\leftarrow pow_two]
 = (a^2 + -(a * b)) + (b * a - b^2) := by ring
  = a^2 + (-(a * b) + (b * a - b^2)) := by rw [add assoc]
 = a^2 + (-(a * b) + (b * a + -b^2)) := by ring
 _{-} = a^2 + ((-(a * b) + b * a) + -b^2) := by rw [\leftarrow add_assoc
                                           (-(a * b)) (b * a) (-b^2)
 = a^2 + ((-(a * b) + a * b) + -b^2) := by rw [mul_comm]
 = a^2 + (0 + -b^2)
                                    := by rw [neg add self (a * b)]
  = (a^2 + 0) + -b^2 
                                     := by rw [← add assoc]
  = a^2 + -b^2
                                    := by rw [add zero]
 _{-} = a^2 - b^2
                                     := by linarith
-- 2ª demostración
- - ===========
example: (a + b) * (a - b) = a^2 - b^2 :=
calc
 (a + b) * (a - b)
 = (a^2 - a * b) + (b * a - b * b) := by ring
  _{-} = (a^2 - a * b) + (b * a - b^2) := by ring
   = (a^2 + -(a * b)) + (b * a - b^2) := by ring 
 _{-} = a^2 + (-(a * b) + (b * a - b^2)) := by ring
 = a^2 + (-(a * b) + (b * a + -b^2)) := by ring
  = a^2 + ((-(a * b) + b * a) + -b^2) := by ring
 _{-} = a^2 + ((-(a * b) + a * b) + -b^2) := by ring
  = a^2 + (0 + -b^2)
                                    := by ring
  _{-} = (a^2 + 0) + -b^2
                                     := by ring
  _{-} = a^2 + -b^2
                                     := by ring
 _{-} = a^2 - b^2
                                     := by ring
-- 3ª demostración
-- ===========
```

```
example: (a + b) * (a - b) = a^2 - b^2 :=
by ring
-- 4ª demostración
-- El lema anterior es
lemma aux : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
by ring
-- La demostración es
example : (a + b) * (a - b) = a^2 - b^2 :=
  rw [sub_eq_add_neg]
  rw [aux]
  rw [mul_neg]
  rw [add_assoc (a * a)]
  rw [mul comm b a]
  rw [neg add self]
  rw [add zero]
  rw [← pow two]
  rw [mul neg]
  rw [← pow two]
  rw [- sub eq add neg]
```

### 3.10. En $\mathbb{R}$ , si c = da+b y b = ad, entonces c = 2ad

```
c = da + b [por la primera hipótesis]
      = da + ad [por la segunda hipótesis]
       = ad + ad [por la conmutativa]
      = 2(ad) [por la def. de doble]
       = 2ad
                   [por la asociativa]
-- Demostraciones con Lean4
-- ================
import Mathlib.Data.Real.Basic
import Mathlib.Tactic
variable (a b c d : \mathbb{R})
-- 1ª demostración
example
 (h1 : c = d * a + b)
 (h2 : b = a * d)
 : c = 2 * a * d :=
calc
 c = d * a + b := by rw [h1]
 \_ = d * a + a * d := by rw [h2]
 _{-} = a * d + a * d := by rw [mul_comm d a]
 _{-} = 2 * (a * d) := by rw [\leftarrow two_mul (a * d)]
 _{-} = 2 * a * d := by rw [mul_assoc]
-- 2ª demostración
example
 (h1 : c = d * a + b)
 (h2 : b = a * d)
 : c = 2 * a * d :=
by
  rw [h2] at h1
 clear h2
  rw [mul_comm d a] at h1
  rw [← two_mul (a*d)] at h1
  rw [← mul assoc 2 a d] at h1
 exact h1
-- 3ª demostración
example
 (h1 : c = d * a + b)
 (h2 : b = a * d)
 : c = 2 * a * d :=
by rw [h1, h2, mul_comm d a, ← two_mul (a * d), mul_assoc]
```

```
-- 4ª demostración
example
 (h1 : c = d * a + b)
 (h2 : b = a * d)
  : c = 2 * a * d :=
  rw [h1]
  rw [h2]
  ring
-- 5ª demostración
example
 (h1 : c = d * a + b)
  (h2 : b = a * d)
  : c = 2 * a * d :=
  rw [h1, h2]
  ring
-- 6ª demostración
example
 (h1 : c = d * a + b)
 (h2 : b = a * d)
  : c = 2 * a * d :=
by rw [h1, h2]; ring
-- 7ª demostración
example
  (h1 : c = d * a + b)
  (h2 : b = a * d)
 : c = 2 * a * d :=
by linarith
```

## 3.11. En $\mathbb{R}$ , si a+b = c, entonces (a+b)(a+b) = ac+bc

```
-- Demostrar que si a, b y c son números reales tales que
-- a + b = c,
-- entonces
```

```
-- (a + b) * (a + b) = a * c + b * c
-- Demostración en lenguaje natural
-- Por la siguiente cadena de igualdades
-- (a + b)(a + b)
    = (a + b)c [por la hipótesis]
= ac + bc [por la distributiva]
-- Demostraciones con Lean4
-- ================
import Mathlib.Data.Real.Basic
import Mathlib.Tactic
variable (a b c : \mathbb{R})
-- 1ª demostración
example
 (h : a + b = c)
  : (a + b) * (a + b) = a * c + b * c :=
calc
 (a + b) * (a + b)
   = (a + b) * c := by exact congrArg (HMul.hMul (a + b)) h
  _{-} = a * c + b * c := by rw [add_mul]
-- 2ª demostración
example
  (h : a + b = c)
  : (a + b) * (a + b) = a * c + b * c :=
bv
  nth rewrite 2 [h]
  rw [add mul]
```

### Capítulo 4

# Propiedades elementales de los anillos

## 4.1. Si R es un anillo y a ∈ R, entonces a + 0 = a

```
_ = a := by rw [zero_add]
-- 2ª demostración
example : a + 0 = a :=
by
  rw [add comm]
  rw [zero add]
-- 3ª demostración
example : a + 0 = a :=
by rw [add_comm, zero_add]
-- 4ª demostración
example : a + 0 = a :=
by exact add_zero a
-- 5ª demostración
example : a + 0 = a :=
  add zero a
-- 5ª demostración
example : a + 0 = a :=
by simp
```

### 4.2. Si R es un anillo y a ∈ R, entonces a + -a= 0

```
variable {R : Type _} [Ring R]
variable (a : R)
-- 1ª demostración
-- ==========
example : a + -a = 0 :=
calc a + -a = -a + a := by rw [add_comm]
         _ = 0 := by rw [add_left_neg]
-- 2ª demostración
-- ===========
example : a + -a = 0 :=
 rw [add_comm]
 rw [add_left_neg]
-- 3ª demostración
-- ===========
example : a + -a = 0 :=
by rw [add_comm, add_left_neg]
-- 4ª demostración
-- ==========
example : a + -a = 0 :=
by exact add_neg_self a
-- 5ª demostración
-- ==========
example : a + -a = 0 :=
 add neg self a
-- 6ª demostración
-- ==========
example : a + -a = 0 :=
by simp
```

## 4.3. Si R es un anillo y a, $b \in R$ , entonces -a + (a + b) = b

```
-- Demostrar en Lean4 que si R es un anillo, entonces
-- \forall a, b : R, -a + (a + b) = b
   -----
-- Demostración en lenguaje natural
-- Por la siguiente cadena de igualdades
-a + (a + b) = (-a + a) + b [por la asociativa]
                 = 0 + b [por inverso por la izquierda]
= b [por cero por la izquierda]
import Mathlib.Algebra.Ring.Defs
variable {R : Type } [Ring R]
variable (a b : R)
-- Demostraciones con Lean4
-- 1º demostración
example : -a + (a + b) = b :=
calc -a + (a + b) = (-a + a) + b := by rw [\leftarrow add_assoc]
              _ = 0 + b := by rw [add_left_neg]
_ = b := bv rw [zero add]
-- 2ª demostración
example : -a + (a + b) = b :=
  rw [←add_assoc]
  rw [add_left_neg]
  rw [zero add]
-- 3ª demostración
example : -a + (a + b) = b :=
by rw [←add_assoc, add_left_neg, zero_add]
-- 4ª demostración
example : -a + (a + b) = b :=
by exact neg add cancel left a b
```

```
-- 5<sup>a</sup> demostración

example : -a + (a + b) = b :=

neg_add_cancel_left a b

-- 6<sup>a</sup> demostración

example : -a + (a + b) = b :=

by simp
```

## 4.4. Si R es un anillo y a, b ∈ R, entonces (a + b) + -b = a

```
-- Demostrar en Lean4 que si R es un anillo, entonces
-- \forall a, b : R, (a + b) + -b = a
-- Demostración en lenguaje natural
-- Por la siguiente cadena de igualdades
-- (a + b) + -b = a + (b + -b) [por la asociativa]
              -- Demostraciones con Lean4
import Mathlib.Algebra.Ring.Defs
variable {R : Type _} [Ring R]
variable (a b : R)
-- 1ª demostración
example : (a + b) + -b = a :=
calc
 (a + b) + -b = a + (b + -b) := by rw [add assoc]
          _ = a + 0 := by rw [add_right_neg]
                        := by rw [add_zero]
-- 2ª demostración
```

```
example : (a + b) + -b = a :=
by
  rw [add_assoc]
  rw [add right neg]
  rw [add zero]
-- 3ª demostración
example : (a + b) + -b = a :=
by rw [add assoc, add right neg, add zero]
-- 4ª demostración
example : (a + b) + -b = a :=
  add_neg_cancel_right a b
-- 5ª demostración
example : (a + b) + -b = a :=
 add_neg_cancel_right _ _
-- 6ª demostración
example : (a + b) + -b = a :=
by simp
```

## 4.5. Si R es un anillo y a, b, c ∈ R tales que a+b=a+c, entonces b=c

```
= (-a + a) + b [por suma con opuesto]
      = -a + (a + b)  [por asociativa]
= -a + (a + c) [por hipótesis]
= (-a + a) + c [por asociativa]
       = 0 + c
                         [por suma con opuesto]
       = c
                          [por suma con cero]
-- 2ª demostración en LN
- - -----
-- Por la siguiente cadena de implicaciones
     a + b = a + c
    ==> -a + (a + b) = -a + (a + c) [sumando -a]
    ==> (-a+a)+b=(-a+a)+c [por la asociativa]
    ==> 0 + b = 0 + b
                                           [suma con opuesto]
    ==> b = c
                                           [suma con cero]
-- 3ª demostración en LN
-- ==============
-- Por la siguiente cadena de igualdades
-- b = -a + (a + b)
      = -a + (a + c) [por la hipótesis]
        = c
-- Demostraciones con Lean4
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]
variable {a b c : R}
-- 1ª demostración
example
 (h : a + b = a + c)
  : b = c :=
calc
 b = 0 + b := by rw [zero_add]
  _ = (-a + a) + b := by rw [add_left_neg]
 \underline{\phantom{a}} = -a + (a + b) := by rw [add_assoc]
 \underline{\ } = -a + (a + c) := by rw [h]
  \underline{\phantom{a}} = (-a + a) + c := by rw [\leftarrow add_assoc]
 _{-} = 0 + c := by rw [add_left_neg]
```

```
_ = c
                    := by rw [zero_add]
-- 2ª demostración
example
  (h : a + b = a + c)
  : b = c :=
by
 have h1 : -a + (a + b) = -a + (a + c) :=
   congrArg (HAdd.hAdd (-a)) h
  clear h
  rw [- add_assoc] at h1
  rw [add_left_neg] at h1
  rw [zero add] at h1
  rw [- add_assoc] at h1
  rw [add_left_neg] at h1
  rw [zero add] at h1
  exact h1
-- 3ª demostración
example
  (h : a + b = a + c)
  : b = c :=
calc
  b = -a + (a + b) := by rw [neg_add_cancel_left a b]
  \underline{\ } = -a + (a + c) := by rw [h]
                   := by rw [neg_add_cancel_left]
-- 4º demostración
example
 (h : a + b = a + c)
  : b = c :=
  rw [← neg add cancel left a b]
  rw [h]
  rw [neg add cancel left]
-- 5ª demostración
example
  (h : a + b = a + c)
  : b = c :=
  rw [~ neg_add_cancel_left a b, h, neg_add_cancel_left]
-- 6ª demostración
example
```

```
(h : a + b = a + c)
: b = c :=
add_left_cancel h
```

# 4.6. Si R es un anillo y a, b, c ∈ R tales que a+b=c+b, entonces a=c

```
-- Demostrar que si R es un anillo y a, b, c ∈ R tales que
-- a + b = c + b
-- entonces
-- a = c
-- Demostraciones en lenguaje natural (LN)
-- 1ª demostración en LN
-- ==============
-- Por la siguiente cadena de igualdades
[por suma con cero]
-- 2ª demostración en LN
-- ============
-- Por la siguiente cadena de igualdades
-- a = (a + b) + -b
     = (c + b) + -b [por hipótesis]
     = c
-- Demostraciones con Lean4
```

```
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]
variable {a b c : R}
-- 1ª demostración con Lean4
example
 (h : a + b = c + b)
 : a = c :=
calc
 a = a + 0 := by rw [add_zero]
 \_ = a + (b + -b) := by rw [add_right_neg]
 _{-} = (a + b) + -b := by rw [add_assoc]
 \underline{\ } = (c + b) + -b := by rw [h]
 \underline{\phantom{a}} = c + (b + -b) := by rw [\leftarrow add_assoc]
 \_ = c + 0 := by rw [\leftarrow add_right_neg]
 _ = c
                := by rw [add zero]
-- 2ª demostración con Lean4
example
 (h : a + b = c + b)
 : a = c :=
calc
 a = (a + b) + -b := (add_neg_cancel_right a b).symm
 \underline{\ } = (c + b) + -b := by rw [h]
                := add_neg_cancel_right c b
-- 3ª demostración con Lean4
example
 (h : a + b = c + b)
 : a = c :=
  rw [~ add_neg_cancel_right a b]
  rw [h]
  rw [add_neg_cancel_right]
-- 4ª demostración con Lean4
```

### 4.7. Si R es un anillo y a $\in$ R, entonces a.0 = 0

```
-- Demostrar que si R es un anillo y a ∈ R, entonces
-- a * 0 = 0
-- Demostración en lenguaje natural
-- -----
-- Basta aplicar la propiedad cancelativa a
a.0 + a.0 = a.0 + 0
-- que se demuestra mediante la siguiente cadena de igualdades
-- a.0 + a.0 = a.(0 + 0) [por la distributiva]
              = a.0 [por suma con cero]
= a.0 + 0 [por suma con cero]
             = a.0
-- Demostraciones con Lean4
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]
variable (a : R)
```

```
-- 1ª demostración
-- ===========
example : a * 0 = 0 :=
by
 have h : a * 0 + a * 0 = a * 0 + 0 :=
    calc a * 0 + a * 0 = a * (0 + 0) := by rw [mul_add a 0 0]
                     _ = a * 0 := by rw [add_zero 0]
                      = a * 0 + 0 := by rw [add_zero (a * 0)]
  rw [add_left_cancel h]
-- 2ª demostración
-- ==========
example : a * 0 = 0 :=
by
 have h : a * 0 + a * 0 = a * 0 + 0 :=
    calc a * 0 + a * 0 = a * (0 + 0) := by rw [\leftarrow mul_add]
                     \underline{\phantom{a}} = a * 0 := by rw [add_zero]
                      = a * 0 + 0 := by rw [add zero]
  rw [add left cancel h]
-- 3ª demostración
-- ===========
example : a * 0 = 0 :=
by
 have h : a * 0 + a * 0 = a * 0 + 0 :=
   by rw [← mul add, add zero, add zero]
  rw [add left cancel h]
-- 4º demostración
-- ==========
example : a * 0 = 0 :=
by
 have : a * 0 + a * 0 = a * 0 + 0 :=
   calc a * 0 + a * 0 = a * (0 + 0) := by simp
                     \underline{\phantom{a}} = a * 0 := by simp
                     _{-} = a * 0 + 0 := by simp
  simp
-- 5ª demostración
-- ===========
```

### 4.8. Si R es un anillo y a $\in$ R, entonces 0.a = 0

```
-- Demostrar que si R es un anillo y a ∈ R, entonces
-- 0 * a = 0
-- Demostración en lenguaje natural
-- Basta aplicar la propiedad cancelativa a
     0.a + 0.a = 0.a + 0
-- que se demuestra mediante la siguiente cadena de igualdades
   0.a + 0.a = (0 + 0).a [por la distributiva]
              = 0.a [por suma con cero]
= 0.a + 0 [por suma con cero]
-- Demostraciones con Lean4
- - -----
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type } [Ring R]
variable (a : R)
-- 1ª demostración
example : 0 * a = 0 :=
 have h : 0 * a + 0 * a = 0 * a + 0 :=
   calc 0 * a + 0 * a = (0 + 0) * a := by rw [add_mul]
                 _ = 0 * a := by rw [add_zero]
```

```
= 0 * a + 0 := by rw [add zero]
  rw [add_left_cancel h]
-- 2ª demostración
example : 0 * a = 0 :=
 have h : 0 * a + 0 * a = 0 * a + 0 :=
   by rw [←add mul, add zero, add zero]
  rw [add left cancel h]
-- 3ª demostración
example : 0 * a = 0 :=
 have : 0 * a + 0 * a = 0 * a + 0 :=
    calc 0 * a + 0 * a = (0 + 0) * a := by simp
                     \underline{\phantom{a}} = 0 * a := by simp
                       = 0 * a + 0 := by simp 
  simp
-- 4ª demostración
example : 0 * a = 0 :=
 have : 0 * a + 0 * a = 0 * a + 0 := by simp
  simp
-- 5ª demostración
example : 0 * a = 0 :=
by simp
-- 6ª demostración
example : 0 * a = 0 :=
zero_mul a
```

### 4.9. Si R es un anillo y a, b ∈ R tales que a+b=0, entonces -a=b

```
-- Demostrar que si es un anillo y a, b \in R tales que

-- a + b = 0

-- entonces
```

```
-- -a = b
-- Demostraciones en lenguaje natural (LN)
-- 1º demostración en LN
-- ------
-- Por la siguiente cadena de igualdades
-- -a = -a + \theta [por suma cero]
     = -a + (a + b) [por hipótesis]
       = b
                       [por cancelativa]
-- 2ª demostración en LN
-- Sumando -a a ambos lados de la hipótesis, se tiene
-- -a + (a + b) = -a + 0
-- El término de la izquierda se reduce a b (por la cancelativa) y el de
-- la derecha a -a (por la suma con cero). Por tanto, se tiene
   b = -a
-- Por la simetría de la igualdad, se tiene
     -a = b
-- Demostraciones con Lean 4
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]
variable {a b : R}
-- 1º demostración (basada en la 1º en LN)
example
 (h : a + b = 0)
 : -a = b :=
calc
 -a = -a + 0 := by rw [add_zero]
  _{-} = -a + (a + b) := by rw [h]
                 := by rw [neg_add_cancel_left]
-- 2ª demostración (basada en la 1º en LN)
example
```

```
(h : a + b = 0)
 : -a = b :=
calc
 -a = -a + 0 := by simp
   \underline{\ } = -a + (a + b) := by rw [h]
                   := by simp
-- 3ª demostración (basada en la 2º en LN)
example
  (h : a + b = 0)
  : -a = b :=
  have h1 : -a + (a + b) = -a + 0 := congrArg (HAdd.hAdd (-a)) h
  have h2 : -a + (a + b) = b := neg_add_cancel_left a b
 have h3 : -a + 0 = -a := add_zero (-a)
  rw [h2, h3] at h1
  exact h1.symm
-- 4ª demostración
example
 (h : a + b = 0)
  : -a = b :=
neg_eq_iff_add_eq_zero.mpr h
```

## 4.10. Si R es un anillo y a, b ∈ R tales que a+b=0, entonces a=-b

```
-- Por la siguiente cadena de igualdades
-- a = (a + b) + -b [por la concelativa]
     = 0 + -b [por la hipótesis]
= -b [por la suma con ce
                       [por la suma con cero]
-- 2ª demostración en LN
-- ------
-- Sumando -a a ambos lados de la hipótesis, se tiene
-- (a + b) + -b = 0 + -b
-- El término de la izquierda se reduce a a (por la cancelativa) y el de
-- la derecha a -b (por la suma con cero). Por tanto, se tiene
-- a = -b
-- Demostraciones con Lean4
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]
variable {a b : R}
-- 1ª demostración (basada en la 1ª en LN)
example
 (h : a + b = 0)
 : a = -b :=
calc
 a = (a + b) + -b := by rw [add_neg_cancel_right]
  = 0 + -b  := by rw [h]
 _{-} = -b
                 := by rw [zero add]
-- 2ª demostración (basada en la 1ª en LN)
example
 (h : a + b = 0)
 : a = -b :=
calc
 a = (a + b) + -b := by simp
 _{-} = 0 + -b := by rw [h]
 _ = -b
                := by simp
-- 3ª demostración (basada en la 1ª en LN)
example
 (h : a + b = 0)
 : a = -b :=
```

```
by
   have h1 : (a + b) + -b = 0 + -b := by rw [h]
   have h2 : (a + b) + -b = a := add_neg_cancel_right a b
   have h3 : 0 + -b = -b := zero_add (-b)
   rwa [h2, h3] at h1

-- 4<sup>a</sup> demostración
example
   (h : a + b = 0)
        : a = -b :=
add_eq_zero_iff_eq_neg.mp h
```

### 4.11. Si R es un anillo, entonces -0 = 0

```
-- Demostrar que si R es un anillo, entonces
-- -\theta = \theta
-- Demostraciones en lenguaje natural (LN)
-- -----
-- 1º demostración en LN
-- Por la suma con cero se tiene
-- \qquad \theta + \theta = \theta
-- Aplicándole la propiedad
\neg \forall a b \in R, a + b = 0 \rightarrow \neg a = b
-- se obtiene
-- -\Theta = \Theta
-- 2ª demostración en LN
- - -----
-- Puesto que
\neg \forall a b \in R, a + b = 0 \rightarrow \neg a = b
-- basta demostrar que
-- \qquad 0 + 0 = 0
-- que es cierta por la suma con cero.
```

```
-- Demostraciones con Lean4
-- ===============
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]
-- 1ª demostración (basada en la 1ª en LN)
example : (-0 : R) = 0 :=
  have h1 : (0 : R) + 0 = 0 := add zero 0
  show (-0 : R) = 0
  exact neg_eq_of_add_eq_zero_left h1
-- 2ª demostración (basada en la 2ª en LN)
example : (-0 : R) = 0 :=
 apply neg_eq_of_add_eq_zero_left
  rw [add zero]
-- 3ª demostración
example : (-0 : R) = 0 :=
  neg_zero
-- 4º demostración
example : (-0 : R) = 0 :=
by simp
```

# 4.12. Si R es un anillo y a ∈ R, entonces -(-a) = a

```
-- Es consecuencia de las siguiente propiedades demostradas en
-- ejercicios anteriores:
     \forall a b \in R, a + b = 0 \rightarrow -a = b
      \forall a \in R, -a + a = 0
-- Demostraciones con Lean4
- - -----
import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]
variable {a : R}
-- 1ª demostración
example : -(-a) = a :=
 have h1 : -a + a = 0 := add left neg a
  show - (-a) = a
 exact neg eq of add eq zero right h1
-- 2ª demostración
example : -(-a) = a :=
  apply neg eq of add eq zero right
  rw [add_left_neg]
-- 3ª demostración
example : -(-a) = a :=
neg neg a
-- 4ª demostración
example : -(-a) = a :=
by simp
```

### 4.13. Si R es un anillo y a, b ∈ R, entonces a -b = a + -b

### Bibliografía

- [1] J. A. Alonso. Lean para matemáticos <sup>1</sup>, 2021.
- [2] J. A. Alonso. Matemáticas en Lean <sup>2</sup>, 2021.
- [3] J. A. Alonso. DAO (Demostración Asistida por Ordenador) con Lean <sup>3</sup>, 2021.
- [4] J. A. Alonso. Calculemus (Vol. 1: Demostraciones con Isabelle/HOL y Lean3) <sup>4</sup> , 2021.
- [5] J. Avigad, L. de Moura, and S. Kong. Theorem Proving in Lean4<sup>5</sup>, 2021.
- [6] J. Avigad, G. Ebner, and S. Ullrich. The Lean4 Manual <sup>6</sup>, 2021.
- [7] J. Avigad, M. J. H. Heule, and W. Nawrocki. Logic and mechanized reasoning <sup>7</sup>, 2023.
- [8] J. Avigad, R. Y. Lewis, and F. van Doorn. Logic and proof 8, 2021.
- [9] J. Avigad and P. Massot. Mathematics in Lean <sup>9</sup>, 2023.
- [10] A. Baanen, A. Bentkamp, J. Blanchette, J. Hölzl, and J. Limperg. The Hitchhiker's Guide to Logical Verification <sup>10</sup>, 2020.

https://github.com/jaalonso/Lean\_para\_matematicos

<sup>&</sup>lt;sup>2</sup>https://github.com/jaalonso/Matematicas en Lean

<sup>&</sup>lt;sup>3</sup>https://raw.githubusercontent.com/jaalonso/DAO con Lean/master/DAO con Lean.pdf

<sup>4</sup>https://raw.githubusercontent.com/jaalonso/Calculemus/master/Calculemus.pdf

<sup>&</sup>lt;sup>5</sup>https://leanprover.github.io/theorem proving in lean4/

<sup>6</sup>https://leanprover.github.io/lean4/doc/

<sup>&</sup>lt;sup>7</sup>https://avigad.github.io/lamr/logic\_and\_mechanized\_reasoning.pdf

<sup>8</sup>https://leanprover.github.io/logic\_and\_proof/logic\_and\_proof.pdf

<sup>9</sup>https://leanprover-community.github.io/mathematics\_in\_lean/

<sup>10</sup>https://raw.githubusercontent.com/blanchette/logical\_verification\_2020/master/ hitchhikers guide.pdf

52 Bibliografía

[11] M. Ballard. Transition to advanced mathematics (Thinking and communicating like a mathematician) <sup>11</sup>.

- [12] K. Buzzard. Sets and logic (in Lean) 12.
- [13] K. Buzzard. Functions and relations (in Lean) 13.
- [14] K. Buzzard. Course on formalising mathematics <sup>14</sup>, 2021.
- [15] K. Buzzard. Course on formalising mathematics <sup>15</sup>, 2023.
- [16] K. Buzzard and M. Pedramfar. The Natural Number Game, version 1.3.3
- [17] D. T. Christiansen. Functional programming in Lean <sup>17</sup>, 2023.
- [18] M. Dvořák. Lean 4 Cheatsheet 18.
- [19] S. Hazratpour. Introduction to proofs <sup>19</sup>, 2022.
- [20] S. Hazratpour. Introduction to proofs with Lean proof assistant <sup>20</sup>, 2022.
- [21] R. Lewis. Formal proof and verification, 2022 <sup>21</sup>, 2022.
- [22] R. Lewis. Discrete structures and probability <sup>22</sup>, 2023.
- [23] C. Löh. Exploring formalisation (A primer in human-readable mathematics in Lean 3 with examples from simplicial topology) <sup>23</sup>, 2022.
- [24] H. Macbeth. The mechanics of proof <sup>24</sup>, 2023.
- [25] P. Massot. Introduction aux mathématiques formalisées <sup>25</sup>.

```
11https://300.f22.matthewrobertballard.com/
12https://www.ma.imperial.ac.uk/~buzzard/M4000x_html/M40001_C1.html
13https://www.ma.imperial.ac.uk/~buzzard/M4000x_html/M40001_M40001_C2.html
14https://github.com/ImperialCollegeLondon/formalising-mathematics
15https://github.com/ImperialCollegeLondon/formalising-mathematics-2023
16https://www.ma.imperial.ac.uk/~buzzard/xena/natural_number_game/
17https://leanprover.github.io/functional_programming_in_lean/
18https://raw.githubusercontent.com/madvorak/lean4-cheatsheet/main/lean-tactics.pdf
19https://introproofs.github.io/s22/
20https://sinhp.github.io/teaching/2022-introduction-to-proofs-with-Lean
21https://github.com/BrownCS1951x/fpv2022
22https://github.com/brown-cs22/CS22-Lean-2023
23https://loeh.app.uni-regensburg.de/mapa/main.pdf
24https://hrmacbeth.github.io/math2001/index.html
25https://www.imo.universite-paris-saclay.fr/~pmassot/enseignement/math114/
```

Bibliografía 53

[26] F. L. Roux. Code Lean contenant les preuves d'un cours standard sur les espaces métriques <sup>26</sup>, 2020.

- [27] W. Schulze. Learning LeanProver <sup>27</sup>.
- [28] Varios. LFTCM 2020: Lean for the Curious Mathematician 2020 28.
- [29] D. J. Velleman. How to prove it with Lean <sup>29</sup>.

<sup>&</sup>lt;sup>26</sup>https://github.com/FredericLeRoux/LEAN ESPACES METRIQUES

<sup>27</sup>https://youtube.com/playlist?list=PLYwF9EIrl42RFQgbmcR\_LSCWRIx2WKbXs

<sup>&</sup>lt;sup>28</sup>https://leanprover-community.github.io/lftcm2020/schedule.html

<sup>&</sup>lt;sup>29</sup>https://djvelleman.github.io/HTPIwL/