

# CalcuIemus

## (Vol. 2: Demostraciones con Lean4)

José A. Alonso Jiménez

---

Grupo de Lógica Computacional  
Dpto. de Ciencias de la Computación e Inteligencia Artificial  
Universidad de Sevilla  
Sevilla, 10 de julio de 2023 (versión del 4 de febrero de 2024)

Esta obra está bajo una licencia Reconocimiento-NoComercial-CompartirIgual 2.5 Spain de Creative Commons.

**Se permite:**

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

**Bajo las condiciones siguientes:**

**Reconocimiento.** Debe reconocer los créditos de la obra de la manera especificada por el autor.



**No comercial.** No puede utilizar esta obra para fines comerciales.



**Compartir bajo la misma licencia.** Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- Algunas de estas condiciones pueden no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/2.5/es/> o envíe una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

# Índice general

<b>1. Introducción</b>	<b>9</b>
<b>2. Demostraciones de una propiedad de los números enteros</b>	<b>11</b>
2.1. $\forall m, n \in \mathbb{N}$ , Even $n \rightarrow$ Even $(m * n)$	11
<b>3. Propiedades elementales de los números reales</b>	<b>15</b>
3.1. En $\mathbb{R}$ , $(ab)c = b(ac)$	15
3.2. En $\mathbb{R}$ , $(cb)a = b(ac)$	16
3.3. En $\mathbb{R}$ , $a(bc) = b(ac)$	17
3.4. En $\mathbb{R}$ , si $ab = cd$ y $e = f$ , entonces $a(be) = c(df)$	19
3.5. En $\mathbb{R}$ , si $bc = ef$ , entonces $((ab)c)d = ((ae)f)d$	20
3.6. En $\mathbb{R}$ , si $c = ba-d$ y $d = ab$ , entonces $c = 0$	22
3.7. En $\mathbb{R}$ , $(a+b)(a+b) = aa+2ab+bb$	23
3.8. En $\mathbb{R}$ , $(a+b)(c+d) = ac+ad+bc+bd$	25
3.9. En $\mathbb{R}$ , $(a+b)(a-b) = a^2-b^2$	27
3.10. En $\mathbb{R}$ , si $c = da+b$ y $b = ad$ , entonces $c = 2ad$	30
3.11. En $\mathbb{R}$ , si $a+b = c$ , entonces $(a+b)(a+b) = ac+bc$	32
3.12. Si $x$ e $y$ son sumas de dos cuadrados, entonces $xy$ también lo es	33
3.13. En $\mathbb{R}$ , $x^2 + y^2 = 0 \leftrightarrow x = 0 \wedge y = 0$	36
3.14. En $\mathbb{R}$ , $x^2 = 1 \rightarrow x = 1 \vee x = -1$	40
3.15. En $\mathbb{R}$ , $x^2 = y^2 \rightarrow x = y \vee x = -y$	43
3.16. En $\mathbb{R}$ , $ a  =  a - b + b $	45
<b>4. Propiedades elementales de los anillos</b>	<b>47</b>
4.1. Si $R$ es un anillo y $a \in R$ , entonces $a + 0 = a$	47
4.2. Si $R$ es un anillo y $a \in R$ , entonces $a + -a = 0$	48
4.3. Si $R$ es un anillo y $a, b \in R$ , entonces $-a + (a + b) = b$	50
4.4. Si $R$ es un anillo y $a, b \in R$ , entonces $(a + b) + -b = a$	52
4.5. Si $R$ es un anillo y $a, b, c \in R$ tales que $a+b=a+c$ , entonces $b=c$	53

4.6.	Si $R$ es un anillo y $a, b, c \in R$ tales que $a+b=c+b$ , entonces $a=c$	56
4.7.	Si $R$ es un anillo y $a \in R$ , entonces $a \cdot 0 = 0$	58
4.8.	Si $R$ es un anillo y $a \in R$ , entonces $0 \cdot a = 0$	60
4.9.	Si $R$ es un anillo y $a, b \in R$ tales que $a+b=0$ , entonces $-a=b$	62
4.10.	Si $R$ es un anillo y $a, b \in R$ tales que $a+b=0$ , entonces $a=-b$	64
4.11.	Si $R$ es un anillo, entonces $-0 = 0$	66
4.12.	Si $R$ es un anillo y $a \in R$ , entonces $-(-a) = a$	68
4.13.	Si $R$ es un anillo y $a, b \in R$ , entonces $a - b = a + -b$	69
4.14.	Si $R$ es un anillo y $a \in R$ , entonces $a - a = 0$	70
4.15.	En los anillos, $1 + 1 = 2$	71
4.16.	Si $R$ es un anillo y $a \in R$ , entonces $2a = a+a$	71
<b>5.</b>	<b>Propiedades elementales de los grupos</b>	<b>73</b>
5.1.	Si $G$ es un grupo y $a \in G$ , entonces $aa^{-1} = 1$	73
5.2.	Si $G$ es un grupo y $a \in G$ , entonces $a \cdot 1 = a$	75
5.3.	Si $G$ es un grupo y $a, b \in G$ tales que $ab = 1$ entonces $a^{-1} = b$	76
5.4.	Si $G$ es un grupo y $a, b \in G$ , entonces $(ab)^{-1} = b^{-1}a^{-1}$	78
<b>6.</b>	<b>Propiedades de orden en los números reales</b>	<b>81</b>
6.1.	En $\mathbb{R}$ , si $a \leq b$ , $b < c$ , $c \leq d$ y $d < e$ , entonces $a < e$	81
6.2.	En $\mathbb{R}$ , si $2a \leq 3b$ , $1 \leq a$ y $d = 2$ , entonces $d + a \leq 5b$	84
6.3.	En $\mathbb{R}$ , si $1 \leq a$ y $b \leq d$ , entonces $2 + a + e^b \leq 3a + e^d$	85
6.4.	En $\mathbb{R}$ , si $a \leq b$ y $c < d$ , entonces $a + e^c + f \leq b + e^d + f$	87
6.5.	En $\mathbb{R}$ , si $d \leq f$ , entonces $c + e^{(a+d)} \leq c + e^{(a+f)}$	89
6.6.	En $\mathbb{R}$ , si $a \leq b$ , entonces $\log(1+e^a) \leq \log(1+e^b)$	91
6.7.	En $\mathbb{R}$ , si $a \leq b$ , entonces $c - e^b \leq c - e^a$	93
6.8.	En $\mathbb{R}$ , $2ab \leq a^2 + b^2$	94
6.9.	En $\mathbb{R}$ , $ ab  \leq (a^2+b^2)/2$	96
6.10.	En $\mathbb{R}$ , $\min(a,b) = \min(b,a)$	98
6.11.	En $\mathbb{R}$ , $\max(a,b) = \max(b,a)$	100
6.12.	En $\mathbb{R}$ , $\min(\min(a,b),c) = \min(a,\min(b,c))$	102
6.13.	En $\mathbb{R}$ , $\min(a,b)+c = \min(a+c,b+c)$	106
6.14.	En $\mathbb{R}$ , $ a  -  b  \leq  a - b $	110
6.15.	En $\mathbb{R}$ , $\{0 < \varepsilon, \varepsilon \leq 1,  x  < \varepsilon,  y  < \varepsilon\} \vdash  xy  < \varepsilon$	111
6.16.	En $\mathbb{R}$ , $a < b \rightarrow \neg(b < a)$	115
6.17.	Hay algún número real entre 2 y 3	116

6.18.	Si $(\forall \varepsilon > 0)[x \leq \varepsilon]$ , entonces $x \leq 0$	.117
6.19.	Si $0 < 0$ , entonces $a > 37$ para cualquier número $a$	.119
6.20.	$\{x \leq y, y \neq x\} \vdash x \leq y \wedge x \neq y$	.121
6.21.	$x \leq y \wedge x \neq y \vdash y \neq x$	.124
6.22.	$(\exists x \in \mathbb{R})[2 < x < 3]$	.127
6.23.	Si $(\exists z \in \mathbb{R})[x < z < y]$ , entonces $x < y$	.128
6.24.	En $\mathbb{R}$ , $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge y \neq x$	.130
6.25.	En $\mathbb{R}$ , si $x \leq y$ , entonces $y \neq x \leftrightarrow x \neq y$	.132
6.26.	Si $ x + 3  < 5$ , entonces $-8 < x < 2$	.136
6.27.	En $\mathbb{R}$ , $y > x^2 \vdash y > 0 \vee y < -1$	.138
6.28.	En $\mathbb{R}$ , $-y > x^2 + 1 \vdash y > 0 \vee y < -1$	.139
6.29.	En $\mathbb{R}$ , si $x <  y $ , entonces $x < y$ ó $x < -y$	.142
6.30.	En $\mathbb{R}$ , $x \leq  x $	.143
6.31.	En $\mathbb{R}$ , $-x \leq  x $	.145
6.32.	En $\mathbb{R}$ , $ x + y  \leq  x  +  y $	.147
6.33.	En $\mathbb{R}$ , si $x \neq 0$ entonces $x < 0$ ó $x > 0$	.150
6.34.	Si $(\exists x, y \in \mathbb{R})[z = x^2 + y^2 \vee z = x^2 + y^2 + 1]$ , entonces $z \geq 0$	.151
6.35.	En $\mathbb{R}$ , si $1 < a$ , entonces $a < aa$	.155
<b>7.</b>	<b>Divisibilidad</b>	<b>159</b>
7.1.	Si $x, y, z \in \mathbb{N}$ , entonces $x \mid yxz$	.159
7.2.	Si $x$ divide a $w$ , también divide a $y(xz) + x^2 + w^2$	.160
7.3.	Transitividad de la divisibilidad	.162
7.4.	Si $a$ divide a $b$ y a $c$ , entonces divide a $b + c$	.165
7.5.	Conmutatividad del máximo común divisor	.167
7.6.	Si $(m \mid n \wedge m \neq n)$ , entonces $(m \mid n \wedge \neg(n \mid m))$	.169
7.7.	Existen números primos $m$ y $n$ tales que $4 < m < n < 10$	.172
7.8.	3 divide al máximo común divisor de 6 y 15	.172
7.9.	Si $m$ divide a $n$ o a $k$ , entonces $m$ divide a $nk$	.174
7.10.	Existen infinitos números primos	.176
7.11.	Si $n^2$ es par, entonces $n$ es par	.179
7.12.	La raíz cuadrada de 2 es irracional	.181
<b>8.</b>	<b>Retículos</b>	<b>185</b>
8.1.	En los retículos, $x \sqcap y = y \sqcap x$	.185
8.2.	En los retículos, $x \sqcup y = y \sqcup x$	.187

8.3.	En los retículos, $(x \sqcap y) \sqcap z = x \sqcap (y \sqcap z)$ . . . . .	.189
8.4.	En los retículos, $(x \sqcup y) \sqcup z = x \sqcup (y \sqcup z)$ . . . . .	.193
8.5.	En los retículos, $x \sqcap (x \sqcup y) = x$ . . . . .	.198
8.6.	En los retículos, $x \sqcup (x \sqcap y) = x$ . . . . .	.201
8.7.	En los retículos, una distributiva del ínfimo implica la otra . . .	.203
8.8.	En los retículos, una distributiva del supremos implica la otra .	.204
<b>9.</b>	<b>Relaciones de orden</b>	<b>207</b>
9.1.	En los órdenes parciales, $a < b \leftrightarrow a \leq b \wedge a \neq b$ . . . . .	.207
9.2.	Si $\leq$ es un preorden, entonces $<$ es irreflexiva . . . . .	.212
9.3.	Si $\leq$ es un preorden, entonces $<$ es transitiva . . . . .	.213
<b>10.</b>	<b>Anillos ordenados</b>	<b>217</b>
10.1.	En los anillos ordenados, $a \leq b \rightarrow 0 \leq b - a$ . . . . .	.217
10.2.	En los anillos ordenados, $0 \leq b - a \rightarrow a \leq b$ . . . . .	.218
10.3.	En los anillos ordenados, $\{a \leq b, 0 \leq c\} \vdash ac \leq bc$ . . . . .	.220
<b>11.</b>	<b>Espacios métricos</b>	<b>223</b>
11.1.	En los espacios métricos, $\text{dist}(x,y) \geq 0$ . . . . .	.223
<b>12.</b>	<b>Funciones reales</b>	<b>227</b>
12.1.	La suma de una cota superior de $f$ y una cota superior de $g$ es una cota superior de $f+g$ . . . . .	.227
12.2.	La suma de una cota inferior de $f$ y una cota inferior de $g$ es una cota inferior de $f+g$ . . . . .	.229
12.3.	El producto de funciones no negativas es no negativo . . . . .	.231
12.4.	Si $a$ es una cota superior no negativa de $f$ y $b$ es es una cota superior de la función no negativa $g$ , entonces $ab$ es una cota superior de $fg$ . . . . .	.234
12.5.	La suma de dos funciones acotadas superiormente también lo está . . . . .	.237
12.6.	La suma de dos funciones acotadas inferiormente también lo está . . . . .	.239
12.7.	Si $a$ es una cota superior de $f$ y $c \geq 0$ , entonces $ca$ es una cota superior de $cf$ . . . . .	.241
12.8.	Si $c \geq 0$ y $f$ está acotada superiormente, entonces $c \cdot f$ también lo está . . . . .	.243

12.9. Si para cada $a$ existe un $x$ tal que $f(x) > a$ , entonces $f$ no tiene cota superior . . . . .	245
12.10. Si para cada $a$ existe un $x$ tal que $f(x) < a$ , entonces $f$ no tiene cota inferior . . . . .	247
12.11. La función identidad no está acotada superiormente . . . . .	248
12.12. Si $f$ no está acotada superiormente, entonces $(\forall a)(\exists x)[f(x) > a]$	249
12.13. Si $\neg(\forall a)(\exists x)[f(x) > a]$ , entonces $f$ está acotada superiormente	253
12.14. Suma de funciones monótonas . . . . .	255
12.15. Si $c$ es no negativo y $f$ es monótona, entonces $cf$ es monótona.	257
12.16. La composición de dos funciones monótonas es monótona . .	259
12.17. Si $f$ es monótona y $f(a) < f(b)$ , entonces $a < b$ . . . . .	261
12.18. Si $a, b \in \mathbb{R}$ tales que $a \leq b$ y $f(b) < f(a)$ , entonces $f$ no es monótona	263
12.19. No para toda $f : \mathbb{R} \rightarrow \mathbb{R}$ monótona, $(\forall a, b)[f(a) \leq f(b) \rightarrow a \leq b]$	265
12.20. Si $f$ no es monótona, entonces $\exists x \exists y[x \leq y \wedge f(y) < f(x)]$ . . . .	266
12.21. $f : \mathbb{R} \rightarrow \mathbb{R}$ no es monótona si $\exists x, y(x \leq y \wedge f(x) > f(y))$ . . . .	268
12.22. La función $x \mapsto -x$ no es monótona creciente . . . . .	270
12.23. La suma de dos funciones pares es par . . . . .	270
12.24. El producto de dos funciones impares es par . . . . .	272
12.25. El producto de una función par por una impar es impar . . . .	274
12.26. Si $f$ es par y $g$ es impar, entonces $(f \circ g)$ es par . . . . .	276
12.27. Para cualquier conjunto $s$ , $s \subseteq s$ . . . . .	278
12.28. Las funciones $f(x, y) = (x + y)^2$ y $g(x, y) = x^2 + 2xy + y^2$ son iguales . . . . .	279
<b>13. Teoría de conjuntos</b>	<b>281</b>
13.1. Si $r \subseteq s$ y $s \subseteq t$ , entonces $r \subseteq t$ . . . . .	281
13.2. Si $a$ es una cota superior de $s$ y $a \leq b$ , entonces $b$ es una cota superior de $s$ . . . . .	283
13.3. La función $(x \mapsto x + c)$ es inyectiva . . . . .	285
13.4. Si $c \neq 0$ , entonces la función $(x \mapsto cx)$ es inyectiva . . . . .	286
13.5. La composición de funciones inyectivas es inyectiva . . . . .	287
13.6. La función $(x \mapsto x + c)$ es suprayectiva . . . . .	290
13.7. Si $c \neq 0$ , entonces la función $(x \mapsto cx)$ es suprayectiva . . . .	292
13.8. Si $c \neq 0$ , entonces la función $(x \mapsto cx + d)$ es suprayectiva . .	293
13.9. Si $f : \mathbb{R} \rightarrow \mathbb{R}$ es suprayectiva, entonces $\exists x \in \mathbb{R}$ tal que $f(x)^2 = 9$	295
13.10. La composición de funciones suprayectivas es suprayectiva .	296

<b>14. Lógica</b>	<b>301</b>
14.1. Si $\neg(\exists x)P(x)$ , entonces $(\forall x)\neg P(x)$ . . . . .	.301
14.2. Si $(\forall x)\neg P(x)$ , entonces $\neg(\exists x)P(x)$ . . . . .	.304
14.3. Si $\neg(\forall x)P(x)$ , entonces $(\exists x)\neg P(x)$ . . . . .	.306
14.4. Si $(\exists x)\neg P(x)$ , entonces $\neg(\forall x)P(x)$ . . . . .	.307
14.5. $\neg\neg P \rightarrow P$ . . . . .	.309
14.6. $P \rightarrow \neg\neg P$ . . . . .	.311
14.7. $(P \rightarrow Q) \leftrightarrow \neg P \vee Q$ . . . . .	.312
<b>15. Límites de sucesiones</b>	<b>317</b>
15.1. La sucesión constante $s_n = c$ converge a $c$ . . . . .	.317
<b>Bibliografía</b>	<b>319</b>
<b>Lemas usados</b>	<b>325</b>



# Capítulo 1

## Introducción

Este libro es una recopilación de los ejercicios de demostración con Lean4 que se han ido publicando, desde el 10 de julio de 20023, en el blog [Calculemus](#).

La ordenación de los ejercicios es simplemente temporal según su fecha de publicación en Calculemus y el orden de los ejercicios en Calculemus responde a los que me voy encontrando en mis [lecturas](#).

En cada ejercicio, se comienza proponiendo soluciones en lenguaje natural y, a continuación, se exponen distintas demostraciones con Lean4 ordenadas desde las más detalladas a las más automáticas. Al final de cada ejercicio hay un enlace para interactuar con sus soluciones en [Lean4 Web](#).

Las soluciones del libro están en [este repositorio de GitHub](#).

El libro se irá actualizando periódicamente con los nuevos ejercicios que se proponen diariamente en [Calculemus](#).

Este libro es una continuación de

- [DAO \(Demostración Asistida por Ordenador\) con Lean](#) que es una introducción a la demostración con Lean3 y
- [Calculemus \(Vol. 1: Demostraciones con Isabelle/HOL y Lean3\)](#) que es la recopilación de la primera parte de los ejercicios del blog con demostraciones en Isabelle/HOL y Lean3.



## Capítulo 2

# Demostraciones de una propiedad de los números enteros

### 2.1. $\forall m n \in \mathbb{N}, \text{Even } n \rightarrow \text{Even } (m * n)$

```
-----  
-- Demostrar que los productos de los números naturales por números  
-- pares son pares.  
-----  
  
-- Demostración en lenguaje natural  
-- =====  
  
-- Si n es par, entonces (por la definición de 'Even') existe un k tal que  
--   n = k + k          (1)  
-- Por tanto,  
--   mn = m(k + k)      (por (1))  
--       = mk + mk       (por la propiedad distributiva)  
-- Por consiguiente, mn es par.  
  
-- Demostraciones en Lean4  
-- =====  
  
import Mathlib.Data.Nat.Basic  
import Mathlib.Data.Nat.Parity  
import Mathlib.Tactic  
  
open Nat
```

```

-- 1ª demostración
-- =====

example : ∀ m n : ℕ, Even n → Even (m * n) := by
  rintro m n ⟨k, hk⟩
  use m * k
  rw [hk]
  ring

-- 2ª demostración
-- =====

example : ∀ m n : ℕ, Even n → Even (m * n) := by
  rintro m n ⟨k, hk⟩
  use m * k
  rw [hk]
  rw [mul_add]

-- 3ª demostración
-- =====

example : ∀ m n : ℕ, Even n → Even (m * n) := by
  rintro m n ⟨k, hk⟩
  use m * k
  rw [hk, mul_add]

-- 4ª demostración
-- =====

example : ∀ m n : Nat, Even n → Even (m * n) := by
  rintro m n ⟨k, hk⟩; use m * k; rw [hk, mul_add]

-- 5ª demostración
-- =====

example : ∀ m n : ℕ, Even n → Even (m * n) := by
  rintro m n ⟨k, hk⟩
  exact ⟨m * k, by rw [hk, mul_add]⟩

-- 6ª demostración
-- =====

example : ∀ m n : Nat, Even n → Even (m * n) :=
fun m n ⟨k, hk⟩ ↦ ⟨m * k, by rw [hk, mul_add]⟩

```

```

-- 7ª demostración
-- =====

example :  $\forall m n : \mathbb{N}, \text{Even } n \rightarrow \text{Even } (m * n) := \text{by}$ 
  rintro m n ⟨k, hk⟩
  use m * k
  rw [hk]
  exact mul_add m k k

-- 8ª demostración
-- =====

example :  $\forall m n : \mathbb{N}, \text{Even } n \rightarrow \text{Even } (m * n) := \text{by}$ 
  intros m n hn
  unfold Even at *
  cases hn with
  | intro k hk =>
    use m * k
    rw [hk, mul_add]

-- 9ª demostración
-- =====

example :  $\forall m n : \mathbb{N}, \text{Even } n \rightarrow \text{Even } (m * n) := \text{by}$ 
  intros m n hn
  unfold Even at *
  cases hn with
  | intro k hk =>
    use m * k
    calc m * n
      = m * (k + k) := by exact congrArg (HMul.hMul m) hk
      _ = m * k + m * k := by exact mul_add m k k

-- 10ª demostración
-- =====

example :  $\forall m n : \mathbb{N}, \text{Even } n \rightarrow \text{Even } (m * n) := \text{by}$ 
  intros; simp [*, parity_simps]

-- Lemas usados
-- =====

-- #check (mul_add :  $\forall a b c : \mathbb{N}, a * (b + c) = a * b + a * c$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).



# Capítulo 3

## Propiedades elementales de los números reales

### 3.1. En $\mathbb{R}$ , $(ab)c = b(ac)$

```
-- Demostrar que los números reales tienen la siguiente propiedad
-- (a * b) * c = b * (a * c)
```

```
-- Demostración en lenguaje natural
-- =====
```

```
-- Por la siguiente cadena de igualdades
-- (ab)c = (ba)c [por la conmutativa]
--        = b(ac) [por la asociativa]
```

```
-- Demostraciones con Lean4
-- =====
```

```
import Mathlib.Tactic
import Mathlib.Data.Real.Basic
```

```
-- 1ª demostración
```

```
example
```

```
(a b c : ℝ)
: (a * b) * c = b * (a * c) :=
```

```
calc
```

```
(a * b) * c = (b * a) * c := by rw [mul_comm a b]
_ = b * (a * c) := by rw [mul_assoc b a c]
```

```

-- 2ª demostración
example (a b c : ℝ) : (a * b) * c = b * (a * c) :=
by
  rw [mul_comm a b]
  rw [mul_assoc b a c]

-- 3ª demostración
example (a b c : ℝ) : (a * b) * c = b * (a * c) :=
by ring

-- Lemas usados
-- =====

-- #check (mul_comm : ∀ (a b : ℝ), a * b = b * a)
-- #check (mul_assoc : ∀ (a b c : ℝ), (a * b) * c = a * (b * c))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 3.2. En $\mathbb{R}$ , $(cb)a = b(ac)$

```

-----
-- Demostrar que los números reales tienen la siguiente propiedad
--   (c * b) * a = b * (a * c)
-----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades:
--   (c * b) * a
--   = (b * c) * a    [por la conmutativa]
--   = b * (c * a)    [por la asociativa]
--   = b * (a * c)    [por la conmutativa]

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
import Mathlib.Data.Real.Basic

-- 1ª demostración
example
  (a b c : ℝ)

```



```

: (c * b) * a = b * (a * c) :=
calc
  (c * b) * a
    = (b * c) * a := by rw [mul_comm c b]
  _ = b * (c * a) := by rw [mul_assoc]
  _ = b * (a * c) := by rw [mul_comm c a]

-- 2ª demostración
example
  (a b c : ℝ)
  : (c * b) * a = b * (a * c) :=
by
  rw [mul_comm c b]
  rw [mul_assoc]
  rw [mul_comm c a]

-- 3ª demostración
example
  (a b c : ℝ)
  : (c * b) * a = b * (a * c) :=
by ring

-- Lemas usados
-- =====

-- #check (mul_comm : ∀ (a b : ℝ), a * b = b * a)
-- #check (mul_assoc : ∀ (a b c : ℝ), (a * b) * c = a * (b * c))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.3. En $\mathbb{R}$ , $a(bc) = b(ac)$

```

-- -----
-- Demostrar que los números reales tienen la siguiente propiedad
--   a * (b * c) = b * (a * c)
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades:
--   a(bc)
--   = (ab)c    [por la asociativa]

```

```

--      = (ba)c      [por la conmutativa]
--      = b(ac)      [por la asociativa]

-- Demostraciones en Lean4
-- =====

import Mathlib.Tactic
import Mathlib.Data.Real.Basic

-- 1ª demostración
example
  (a b c : ℝ) : a * (b * c) = b * (a * c) :=
calc
  a * (b * c)
    = (a * b) * c := by rw [mul_assoc]
  _ = (b * a) * c := by rw [mul_comm a b]
  _ = b * (a * c) := by rw [mul_assoc]

-- 2ª demostración
example
  (a b c : ℝ) : a * (b * c) = b * (a * c) :=
by
  rw [mul_assoc]
  rw [mul_comm a b]
  rw [mul_assoc]

-- 3ª demostración
example
  (a b c : ℝ) : a * (b * c) = b * (a * c) :=
by ring

-- Lemas usados
-- =====

-- #check (mul_comm : ∀ (a b : ℝ), a * b = b * a)
-- #check (mul_assoc : ∀ (a b c : ℝ), (a * b) * c = a * (b * c))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.4. En $\mathbb{R}$ , si $ab = cd$ y $e = f$ , entonces $a(be) = c(df)$

```
-- Demostrar que si a, b, c, d, e y f son números reales tales que
--   a * b = c * d y
--   e = f,
-- entonces
--   a * (b * e) = c * (d * f)
```

```
-- Demostración en lenguaje natural
```

```
-- =====
```

```
-- Por la siguiente cadena de igualdades
```

```
--   a(be)
--   = a(bf)    [por la segunda hipótesis]
--   = (ab)f    [por la asociativa]
--   = (cd)f    [por la primera hipótesis]
--   = c(df)    [por la asociativa]
```

```
-- Demostraciones en Lean4
```

```
-- =====
```

```
import Mathlib.Tactic
import Mathlib.Data.Real.Basic
```

```
-- 1ª demostración
```

```
example
```

```
(a b c d e f : ℝ)
(h1 : a * b = c * d)
(h2 : e = f)
: a * (b * e) = c * (d * f) :=
```

```
calc
```

```
a * (b * e)
= a * (b * f) := by rw [h2]
_ = (a * b) * f := by rw [mul_assoc]
_ = (c * d) * f := by rw [h1]
_ = c * (d * f) := by rw [mul_assoc]
```

```
-- 2ª demostración
```

```
example
```

```
(a b c d e f : ℝ)
(h1 : a * b = c * d)
```

```
(h2 : e = f)
: a * (b * e) = c * (d * f) :=
```

```
by
```

```
rw [h2]
rw [←mul_assoc]
rw [h1]
rw [mul_assoc]
```

```
-- 3ª demostración
```

```
example
```

```
(a b c d e f : ℝ)
(h1 : a * b = c * d)
(h2 : e = f)
: a * (b * e) = c * (d * f) :=
```

```
by
```

```
simp [*, ←mul_assoc]
```

```
-- Lemas usados
```

```
-- =====
```

```
-- #check (mul_assoc : ∀ (a b c : ℝ), (a * b) * c = a * (b * c))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.5. En $\mathbb{R}$ , si $bc = ef$ , entonces $((ab)c)d = ((ae)f)d$

```
-- -----
-- Demostrar que si a, b, c, d, e y f son números reales tales que
--   b * c = e * f
-- entonces
--   ((a * b) * c) * d = ((a * e) * f) * d
-- -----
```

```
-- Demostración en lenguaje natural
```

```
-- =====
```

```
-- Por la siguiente cadena de igualdades
```

```
--   ((ab)c)d
--   = (a(bc))d    [por la asociativa]
--   = (a(ef))d    [por la hipótesis]
--   = ((ae)f)d    [por la asociativa]
```

```
-- Demostraciones con Lean4
```

```

-- =====

import Mathlib.Data.Real.Basic
import Mathlib.Tactic

-- 1ª demostración
example
  (a b c d e f : ℝ)
  (h : b * c = e * f)
  : ((a * b) * c) * d = ((a * e) * f) * d :=
calc
  ((a * b) * c) * d
    = (a * (b * c)) * d := by rw [mul_assoc a]
_   = (a * (e * f)) * d := by rw [h]
_   = ((a * e) * f) * d := by rw [←mul_assoc a]

-- 2ª demostración
example
  (a b c d e f : ℝ)
  (h : b * c = e * f)
  : ((a * b) * c) * d = ((a * e) * f) * d :=
by
  rw [mul_assoc a]
  rw [h]
  rw [←mul_assoc a]

-- 3ª demostración
example
  (a b c d e f : ℝ)
  (h : b * c = e * f)
  : ((a * b) * c) * d = ((a * e) * f) * d :=
by
  rw [mul_assoc a, h, ←mul_assoc a]

-- Lemas usados
-- =====

-- #check (mul_assoc : ∀ (a b c : ℝ), (a * b) * c = a * (b * c))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.6. En $\mathbb{R}$ , si $c = ba - d$ y $d = ab$ , entonces $c = 0$

```

-----
-- Demostrar que si a, b, c y d son números reales tales que
--   c = b * a - d
--   d = a * b
-- entonces
--   c = 0
-----

```

```

-- Demostración en lenguaje natural
-- =====

```

```

-- Por la siguiente cadena de igualdades
--   c = ba - d      [por la primera hipótesis]
--     = ab - d      [por la conmutativa]
--     = ab - ab     [por la segunda hipótesis]
--     = 0

```

```

-- Demostraciones en Lean4
-- =====

```

```

import Mathlib.Data.Real.Basic
import Mathlib.Tactic

```

```

-- 1ª demostración

```

```

example

```

```

  (a b c d : ℝ)
  (h1 : c = b * a - d)
  (h2 : d = a * b)
  : c = 0 :=

```

```

calc

```

```

  c = b * a - d      := by rw [h1]
  _ = a * b - d      := by rw [mul_comm]
  _ = a * b - a * b := by rw [h2]
  _ = 0              := by rw [sub_self]

```

```

-- 2ª demostración

```

```

example

```

```

  (a b c d : ℝ)
  (h1 : c = b * a - d)
  (h2 : d = a * b)
  : c = 0 :=

```

```

by

```

```

rw [h1]
rw [mul_comm]
rw [h2]
rw [sub_self]

-- 3ª demostración
example
  (a b c d : ℝ)
  (h1 : c = b * a - d)
  (h2 : d = a * b)
  : c = 0 :=
by
  rw [h1, mul_comm, h2, sub_self]

-- Lemas usados
-- =====

-- #check (mul_comm : ∀ (a b : ℝ), a * b = b * a)
-- #check (sub_self : ∀ (a : ℝ), a - a = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 3.7. En $\mathbb{R}$ , $(a+b)(a+b) = aa+2ab+bb$

```

-----
-- Demostrar que si a y b son números reales, entonces
--   (a + b) * (a + b) = a * a + 2 * (a * b) + b * b
-----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades
--   (a + b)(a + b)
--   = (a + b)a + (a + b)b      [por la distributiva]
--   = aa + ba + (a + b)b      [por la distributiva]
--   = aa + ba + (ab + bb)     [por la distributiva]
--   = aa + ba + ab + bb       [por la asociativa]
--   = aa + (ba + ab) + bb     [por la asociativa]
--   = aa + (ab + ab) + bb     [por la conmutativa]
--   = aa + 2(ab) + bb         [por def. de doble]

-- Demostraciones con Lean4

```

```

-- =====

import Mathlib.Data.Real.Basic
import Mathlib.Tactic

variable (a b c : ℝ)

-- 1ª demostración
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
calc
  (a + b) * (a + b)
    = (a + b) * a + (a + b) * b      := by rw [mul_add]
  _ = a * a + b * a + (a + b) * b    := by rw [add_mul]
  _ = a * a + b * a + (a * b + b * b) := by rw [add_mul]
  _ = a * a + b * a + a * b + b * b   := by rw [←add_assoc]
  _ = a * a + (b * a + a * b) + b * b := by rw [add_assoc (a * a)]
  _ = a * a + (a * b + a * b) + b * b := by rw [mul_comm b a]
  _ = a * a + 2 * (a * b) + b * b     := by rw [←two_mul]

-- 2ª demostración
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
calc
  (a + b) * (a + b)
    = a * a + b * a + (a * b + b * b) := by rw [mul_add, add_mul, add_mul]
  _ = a * a + (b * a + a * b) + b * b := by rw [←add_assoc, add_assoc (a * a)]
  _ = a * a + 2 * (a * b) + b * b     := by rw [mul_comm b a, ←two_mul]

-- 3ª demostración
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
calc
  (a + b) * (a + b)
    = a * a + b * a + (a * b + b * b) := by ring
  _ = a * a + (b * a + a * b) + b * b := by ring
  _ = a * a + 2 * (a * b) + b * b     := by ring

-- 4ª demostración
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by ring

-- 5ª demostración
example :

```



```

(a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by
  rw [mul_add]
  rw [add_mul]
  rw [add_mul]
  rw [←add_assoc]
  rw [add_assoc (a * a)]
  rw [mul_comm b a]
  rw [←two_mul]

-- 6ª demostración
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by
  rw [mul_add, add_mul, add_mul]
  rw [←add_assoc, add_assoc (a * a)]
  rw [mul_comm b a, ←two_mul]

-- 7ª demostración
example :
  (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by linarith

-- Lemas usados
-- =====

-- #check (add_assoc : ∀ a b c : ℝ, (a + b) + c = a + (b + c))
-- #check (add_mul : ∀ a b c : ℝ, (a + b) * c = a * c + b * c)
-- #check (mul_add : ∀ a b c : ℝ, a * (b + c) = a * b + a * c)
-- #check (mul_comm : ∀ (a b : ℝ), a * b = b * a)
-- #check (two_mul : ∀ (a : ℝ), 2 * a = a + a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.8. En $\mathbb{R}$ , $(a+b)(c+d) = ac+ad+bc+bd$

```

-----
-- Demostrar que si a, b, c y d son números reales, entonces
--   (a + b) * (c + d) = a * c + a * d + b * c + b * d
-----

-- Demostración en lenguaje natural
-- =====

```

```

-- Por la siguiente cadena de igualdades
--   (a + b)(c + d)
--   = a(c + d) + b(c + d)    [por la distributiva]
--   = ac + ad + b(c + d)    [por la distributiva]
--   = ac + ad + (bc + bd)    [por la distributiva]
--   = ac + ad + bc + bd      [por la asociativa]

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
import Mathlib.Tactic

variable (a b c d : ℝ)

-- 1ª demostración
example
  : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
calc
  (a + b) * (c + d)
    = a * (c + d) + b * (c + d)      := by rw [add_mul]
  _ = a * c + a * d + b * (c + d)    := by rw [mul_add]
  _ = a * c + a * d + (b * c + b * d) := by rw [mul_add]
  _ = a * c + a * d + b * c + b * d   := by rw [←add_assoc]

-- 2ª demostración
example
  : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
calc
  (a + b) * (c + d)
    = a * (c + d) + b * (c + d)      := by ring
  _ = a * c + a * d + b * (c + d)    := by ring
  _ = a * c + a * d + (b * c + b * d) := by ring
  _ = a * c + a * d + b * c + b * d   := by ring

-- 3ª demostración
example : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
by ring

-- 4ª demostración
example
  : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
by
  rw [add_mul]

```

```

rw [mul_add]
rw [mul_add]
rw [← add_assoc]

-- 5ª demostración
example : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
by rw [add_mul, mul_add, mul_add, ←add_assoc]

-- Lemas usados
-- =====

-- #check (add_mul : ∀ (a b c : ℝ), (a + b) * c = a * c + b * c)
-- #check (mul_add : ∀ (a b c : ℝ), a * (b + c) = a * b + a * c)
-- #check (add_assoc : ∀ (a b c : ℝ), (a + b) + c = a + (b + c))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.9. En $\mathbb{R}$ , $(a+b)(a-b) = a^2 - b^2$

```

-----
-- Demostrar que si a y b son números reales, entonces
--   (a + b) * (a - b) = a^2 - b^2
-----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades:
--   (a + b)(a - b)
--   = a(a - b) + b(a - b)           [por la distributiva]
--   = (aa - ab) + b(a - b)         [por la distributiva]
--   = (a^2 - ab) + b(a - b)         [por def. de cuadrado]
--   = (a^2 - ab) + (ba - bb)        [por la distributiva]
--   = (a^2 - ab) + (ba - b^2)       [por def. de cuadrado]
--   = (a^2 + -(ab)) + (ba - b^2)    [por def. de resta]
--   = a^2 + (-(ab) + (ba - b^2))    [por la asociativa]
--   = a^2 + (-(ab) + (ba + -b^2))   [por def. de resta]
--   = a^2 + ((-(ab) + ba) + -b^2)   [por la asociativa]
--   = a^2 + ((-(ab) + ab) + -b^2)   [por la conmutativa]
--   = a^2 + (0 + -b^2)              [por def. de opuesto]
--   = (a^2 + 0) + -b^2              [por asociativa]
--   = a^2 + -b^2                    [por def. de cero]
--   = a^2 - b^2                     [por def. de resta]

```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
import Mathlib.Tactic

variable (a b : ℝ)

-- 1ª demostración
-- =====

example : (a + b) * (a - b) = a^2 - b^2 :=
calc
  (a + b) * (a - b)
    = a * (a - b) + b * (a - b)           := by rw [add_mul]
  _ = (a * a - a * b) + b * (a - b)       := by rw [mul_sub]
  _ = (a^2 - a * b) + b * (a - b)         := by rw [← pow_two]
  _ = (a^2 - a * b) + (b * a - b * b)      := by rw [mul_sub]
  _ = (a^2 - a * b) + (b * a - b^2)        := by rw [← pow_two]
  _ = (a^2 + -(a * b)) + (b * a - b^2)     := by ring
  _ = a^2 + (-(a * b) + (b * a - b^2))     := by rw [add_assoc]
  _ = a^2 + (-(a * b) + (b * a + -b^2))    := by ring
  _ = a^2 + ((-(a * b) + b * a) + -b^2)    := by rw [← add_assoc
    (-(a * b)) (b * a) (-b^2)]
  _ = a^2 + ((-(a * b) + a * b) + -b^2)    := by rw [mul_comm]
  _ = a^2 + (0 + -b^2)                    := by rw [neg_add_self (a * b)]
  _ = (a^2 + 0) + -b^2                    := by rw [← add_assoc]
  _ = a^2 + -b^2                          := by rw [add_zero]
  _ = a^2 - b^2                           := by linarith

-- 2ª demostración
-- =====

example : (a + b) * (a - b) = a^2 - b^2 :=
calc
  (a + b) * (a - b)
    = a * (a - b) + b * (a - b)           := by ring
  _ = (a * a - a * b) + b * (a - b)       := by ring
  _ = (a^2 - a * b) + b * (a - b)         := by ring
  _ = (a^2 - a * b) + (b * a - b * b)      := by ring
  _ = (a^2 - a * b) + (b * a - b^2)        := by ring
  _ = (a^2 + -(a * b)) + (b * a - b^2)     := by ring
  _ = a^2 + (-(a * b) + (b * a - b^2))     := by ring
  _ = a^2 + (-(a * b) + (b * a + -b^2))    := by ring

```

```

_ = a^2 + ((-(a * b) + b * a) + -b^2) := by ring
_ = a^2 + ((-(a * b) + a * b) + -b^2) := by ring
_ = a^2 + (0 + -b^2)                  := by ring
_ = (a^2 + 0) + -b^2                  := by ring
_ = a^2 + -b^2                        := by ring
_ = a^2 - b^2                         := by ring

-- 3ª demostración
-- =====

example : (a + b) * (a - b) = a^2 - b^2 :=
by ring

-- 4ª demostración
-- =====

-- El lema anterior es
lemma aux : (a + b) * (c + d) = a * c + a * d + b * c + b * d :=
by ring

-- La demostración es
example : (a + b) * (a - b) = a^2 - b^2 :=
by
  rw [sub_eq_add_neg]
  rw [aux]
  rw [mul_neg]
  rw [add_assoc (a * a)]
  rw [mul_comm b a]
  rw [neg_add_self]
  rw [add_zero]
  rw [← pow_two]
  rw [mul_neg]
  rw [← pow_two]
  rw [← sub_eq_add_neg]

-- Lemas usados
-- =====

-- #check (add_assoc : ∀ (a b c : ℝ), (a + b) + c = a + (b + c))
-- #check (add_zero : ∀ (a : ℝ), a + 0 = a)
-- #check (add_mul : ∀ (a b c : ℝ), (a + b) * c = a * c + b * c)
-- #check (mul_comm : ∀ (a b : ℝ), a * b = b * a)
-- #check (mul_neg : ∀ (a b : ℝ), a * -b = -(a * b))
-- #check (mul_sub : ∀ (a b c : ℝ), a * (b - c) = a * b - a * c)
-- #check (neg_add_self : ∀ (a : ℝ), -a + a = 0)

```

```
-- #check (pow_two : ∀ (a : ℝ), a ^ 2 = a * a)
-- #check (sub_eq_add_neg : ∀ (a b : ℝ), a - b = a + -b)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.10. En $\mathbb{R}$ , si $c = da + b$ y $b = ad$ , entonces $c = 2ad$

```
-- -----
-- Demostrar que si a, b, c y d son números reales tales que
--   c = d * a + b
--   b = a * d
-- entonces
--   c = 2 * a * d
-- -----
```

```
-- Demostración en lenguaje natural
```

```
-- =====
```

```
-- Por la siguiente cadena de igualdades
```

```
--   c = da + b      [por la primera hipótesis]
--   = da + ad      [por la segunda hipótesis]
--   = ad + ad      [por la conmutativa]
--   = 2(ad)         [por la def. de doble]
--   = 2ad           [por la asociativa]
```

```
-- Demostraciones con Lean4
```

```
-- =====
```

```
import Mathlib.Data.Real.Basic
```

```
import Mathlib.Tactic
```

```
variable (a b c d : ℝ)
```

```
-- 1ª demostración
```

```
example
```

```
  (h1 : c = d * a + b)
```

```
  (h2 : b = a * d)
```

```
  : c = 2 * a * d :=
```

```
calc
```

```
  c = d * a + b      := by rw [h1]
```

```
  _ = d * a + a * d := by rw [h2]
```

```

_ = a * d + a * d := by rw [mul_comm d a]
_ = 2 * (a * d)    := by rw [← two_mul (a * d)]
_ = 2 * a * d      := by rw [mul_assoc]

-- 2ª demostración
example
  (h1 : c = d * a + b)
  (h2 : b = a * d)
  : c = 2 * a * d :=
by
  rw [h2] at h1
  clear h2
  rw [mul_comm d a] at h1
  rw [← two_mul (a*d)] at h1
  rw [← mul_assoc 2 a d] at h1
  exact h1

-- 3ª demostración
example
  (h1 : c = d * a + b)
  (h2 : b = a * d)
  : c = 2 * a * d :=
by rw [h1, h2, mul_comm d a, ← two_mul (a * d), mul_assoc]

-- 4ª demostración
example
  (h1 : c = d * a + b)
  (h2 : b = a * d)
  : c = 2 * a * d :=
by
  rw [h1]
  rw [h2]
  ring

-- 5ª demostración
example
  (h1 : c = d * a + b)
  (h2 : b = a * d)
  : c = 2 * a * d :=
by
  rw [h1, h2]
  ring

-- 6ª demostración
example

```

```

(h1 : c = d * a + b)
(h2 : b = a * d)
: c = 2 * a * d :=
by rw [h1, h2] ; ring

-- 7ª demostración
example
  (h1 : c = d * a + b)
  (h2 : b = a * d)
  : c = 2 * a * d :=
by linarith

-- Lemas usados
-- =====

-- #check (mul_assoc : ∀ (a b c : ℝ), (a * b) * c = a * (b * c))
-- #check (mul_comm : ∀ (a b : ℝ), a * b = b * a)
-- #check (two_mul : ∀ (a : ℝ), 2 * a = a + a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 3.11. En $\mathbb{R}$ , si $a+b = c$ , entonces $(a+b)(a+b) = ac+bc$

```

-----
-- Demostrar que si a, b y c son números reales tales que
--   a + b = c,
-- entonces
--   (a + b) * (a + b) = a * c + b * c
-----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades
--   (a + b)(a + b)
--   = (a + b)c           [por la hipótesis]
--   = ac + bc           [por la distributiva]

-- Demostraciones con Lean4
-- =====

```



```

import Mathlib.Data.Real.Basic
import Mathlib.Tactic

variable (a b c : ℝ)

-- 1ª demostración
example
  (h : a + b = c)
  : (a + b) * (a + b) = a * c + b * c :=
calc
  (a + b) * (a + b)
    = (a + b) * c := by exact congrArg (HMul.hMul (a + b)) h
  _ = a * c + b * c := by rw [add_mul]

-- 2ª demostración
example
  (h : a + b = c)
  : (a + b) * (a + b) = a * c + b * c :=
by
  nth_rewrite 2 [h]
  rw [add_mul]

-- Lemas usados
-- =====

-- #check (add_mul : ∀ (a b c : ℝ), (a + b) * c = a * c + b * c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 3.12. Si $x$ e $y$ son sumas de dos cuadrados, entonces $xy$ también lo es

```

-----
-- Demostrar que si  $x$  e  $y$  son sumas de dos cuadrados, entonces  $xy$ 
-- también lo es.
-----

-- Demostración en lenguaje natural
-- =====

-- Puesto que  $x$  e  $y$  se pueden escribir como la suma de dos cuadrados,
-- existen  $a, b, c$  y  $d$  tales que

```

```

--       $x = a^2 + b^2$ 
--       $y = c^2 + d^2$ 
--      Entonces,
--       $xy = (ac - bd)^2 + (ad + bc)^2$ 
--      En efecto,
--       $xy = (a^2 + b^2)(c^2 + d^2)$ 
--             $= a^2c^2 + b^2d^2 + a^2d^2 + b^2c^2$ 
--             $= a^2c^2 - 2acbd + b^2d^2 + a^2d^2 + 2adbc + b^2c^2$ 
--             $= (ac - bd)^2 + (ad + bc)^2$ 
--      Por tanto,  $xy$  es la suma de dos cuadrados.

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable {α : Type _} [CommRing α]
variable {x y : α}

-- (suma_de_cuadrados x) afirma que x se puede escribir como la suma
-- de dos cuadrados.
def suma_de_cuadrados (x : α) :=
  ∃ a b, x = a^2 + b^2

-- 1ª demostración
example
  (hx : suma_de_cuadrados x)
  (hy : suma_de_cuadrados y)
  : suma_de_cuadrados (x * y) :=
by
  rcases hx with ⟨a, b, xeq : x = a^2 + b^2⟩
  -- a b : α
  -- xeq : x = a ^ 2 + b ^ 2
  rcases hy with ⟨c, d, yeq : y = c^2 + d^2⟩
  -- c d : α
  -- yeq : y = c ^ 2 + d ^ 2
  have h1: x * y = (a*c - b*d)^2 + (a*d + b*c)^2 :=
    calc x * y
      = (a^2 + b^2) * (c^2 + d^2) :=
        by rw [xeq, yeq]
      _ = a^2*c^2 + b^2*d^2 + a^2*d^2 + b^2*c^2 :=
        by ring
      _ = a^2*c^2 - 2*a*c*b*d + b^2*d^2 + a^2*d^2 + 2*a*d*b*c + b^2*c^2 :=
        by ring
      _ = (a*c - b*d)^2 + (a*d + b*c)^2 :=
        by ring

```

```

have h2 : ∃ f, x * y = (a*c - b*d)^2 + f^2 :=
  Exists.intro (a*d + b*c) h1
have h3 : ∃ e f, x * y = e^2 + f^2 :=
  Exists.intro (a*c - b*d) h2
show suma_de_cuadrados (x * y)
exact h3

-- 2ª demostración
example
  (hx : suma_de_cuadrados x)
  (hy : suma_de_cuadrados y)
  : suma_de_cuadrados (x * y) :=
by
  rcases hx with ⟨a, b, xeq : x = a^2 + b^2⟩
  -- a b : α
  -- xeq : x = a ^ 2 + b ^ 2
  rcases hy with ⟨c, d, yeq : y = c^2 + d^2⟩
  -- c d : α
  -- yeq : y = c ^ 2 + d ^ 2
  have h1: x * y = (a*c - b*d)^2 + (a*d + b*c)^2 :=
    calc x * y
      = (a^2 + b^2) * (c^2 + d^2)      := by rw [xeq, yeq]
      _ = (a*c - b*d)^2 + (a*d + b*c)^2 := by ring
  have h2 : ∃ e f, x * y = e^2 + f^2 :=
    by tauto
  show suma_de_cuadrados (x * y)
  exact h2

-- 3ª demostración
example
  (hx : suma_de_cuadrados x)
  (hy : suma_de_cuadrados y)
  : suma_de_cuadrados (x * y) :=
by
  rcases hx with ⟨a, b, xeq⟩
  -- a b : α
  -- xeq : x = a ^ 2 + b ^ 2
  rcases hy with ⟨c, d, yeq⟩
  -- c d : α
  -- yeq : y = c ^ 2 + d ^ 2
  rw [xeq, yeq]
  -- ⊢ suma_de_cuadrados ((a ^ 2 + b ^ 2) * (c ^ 2 + d ^ 2))
  use a*c - b*d, a*d + b*c
  -- ⊢ (a ^ 2 + b ^ 2) * (c ^ 2 + d ^ 2)
  --   = (a * c - b * d) ^ 2 + (a * d + b * c) ^ 2

```

```

ring

-- 4ª demostración
example
  (hx : suma_de_cuadrados x)
  (hy : suma_de_cuadrados y)
  : suma_de_cuadrados (x * y) :=
by
  rcases hx with ⟨a, b, rfl⟩
  -- ⊢ suma_de_cuadrados ((a ^ 2 + b ^ 2) * y)
  rcases hy with ⟨c, d, rfl⟩
  -- ⊢ suma_de_cuadrados ((a ^ 2 + b ^ 2) * (c ^ 2 + d ^ 2))
  use a*c - b*d, a*d + b*c
  -- ⊢ (a ^ 2 + b ^ 2) * (c ^ 2 + d ^ 2)
  --   = (a * c - b * d) ^ 2 + (a * d + b * c) ^ 2
ring

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

### 3.13. En $\mathbb{R}$ , $x^2 + y^2 = 0 \leftrightarrow x = 0 \wedge y = 0$

```

-----
-- Demostrar que si  $x, y \in \mathbb{R}$ , entonces
--  $x^2 + y^2 = 0 \leftrightarrow x = 0 \wedge y = 0$ 
-----

-- Demostración en lenguaje natural
-- =====

-- En la demostración usaremos el siguiente lema auxiliar
--  $(\forall x, y \in \mathbb{R})[x^2 + y^2 = 0 \rightarrow x = 0]$ 
--
-- Para la primera implicación, supongamos que
--  $x^2 + y^2 = 0$  (1)
-- Entonces, por el lema auxiliar,
--  $x = 0$  (2)
-- Además, aplicando la conmutativa a (1), se tiene
--  $y^2 + x^2 = 0$ 
-- y, por el lema auxiliar,
--  $y = 0$  (3)
-- De (2) y (3) se tiene
--  $x = 0 \wedge y = 0$ 
--

```

```

-- Para la segunda implicación, supongamos que
--    $x = 0 \wedge y = 0$ 
-- Por tanto,
--    $x^2 + y^2 = 0^2 + 0^2$ 
--            $= 0$ 
--
-- En la demostración del lema auxiliar se usarán los siguientes lemas
--    $(\forall x \in \mathbb{R})(\forall n \in \mathbb{N})[x^n = 0 \rightarrow x = 0]$  (L1)
--    $(\forall x, y \in \mathbb{R})[x \leq y \rightarrow y \leq x \rightarrow x = y]$  (L2)
--    $(\forall x, y \in \mathbb{R})[0 \leq y \rightarrow x \leq x + y]$  (L3)
--    $(\forall x \in \mathbb{R})[0 \leq x^2]$  (L4)
--
-- Por el lema L1, para demostrar el lema auxiliar basta demostrar
--    $x^2 = 0$  (1)
-- y, por el lema L2, basta demostrar las siguientes desigualdades
--    $x^2 \leq 0$  (2)
--    $0 \leq x^2$  (3)
--
-- La prueba de la (2) es
--    $x^2 \leq x^2 + y^2$  [por L3 y L4]
--    $= 0$  [por la hipótesis]
--
-- La (3) se tiene por el lema L4.

-- Demostraciones con Lean 4
-- =====

import Mathlib.Data.Real.Basic
variable {x y : ℝ}

-- 1ª demostración del lema auxiliar
-- =====

example
  (h : x^2 + y^2 = 0)
  : x = 0 :=
by
  have h' : x^2 = 0 := by
    { apply le_antisymm
      . show x ^ 2 ≤ 0
        calc x ^ 2 ≤ x^2 + y^2 := by simp [le_add_of_nonneg_right,
                                           pow_two_nonneg]
              _ = 0 := by exact h
      . show 0 ≤ x ^ 2
        apply pow_two_nonneg }

```

```

show x = 0
exact pow_eq_zero h'

-- 2ª demostración lema auxiliar
-- =====

example
  (h : x^2 + y^2 = 0)
  : x = 0 :=
by
  have h' : x^2 = 0 := by
  { apply le_antisymm
    . --  $\vdash x^2 \leq 0$ 
      calc x ^ 2 ≤ x^2 + y^2 := by simp [le_add_of_nonneg_right,
                                          pow_two_nonneg]
          _ = 0 := by exact h
    . --  $\vdash 0 \leq x^2$ 
      apply pow_two_nonneg }
  exact pow_eq_zero h'

-- 3ª demostración lema auxiliar
-- =====

lemma aux
  (h : x^2 + y^2 = 0)
  : x = 0 :=
  have h' : x ^ 2 = 0 := by linarith [pow_two_nonneg x, pow_two_nonneg y]
  pow_eq_zero h'

-- 1ª demostración
-- =====

example : x^2 + y^2 = 0 ↔ x = 0 ∧ y = 0 :=
by
  constructor
  . --  $\vdash x^2 + y^2 = 0 \rightarrow x = 0 \wedge y = 0$ 
    intro h
    --  $h : x^2 + y^2 = 0$ 
    --  $\vdash x = 0 \wedge y = 0$ 
    constructor
    . --  $\vdash x = 0$ 
      exact aux h
    . --  $\vdash y = 0$ 
      rw [add_comm] at h
      --  $h : x^2 + y^2 = 0$ 

```

```

    exact aux h
. --  $\vdash x = 0 \wedge y = 0 \rightarrow x^2 + y^2 = 0$ 
intro h1
--  $h1 : x = 0 \wedge y = 0$ 
--  $\vdash x^2 + y^2 = 0$ 
rcases h1 with ⟨h2, h3⟩
--  $h2 : x = 0$ 
--  $h3 : y = 0$ 
rw [h2, h3]
--  $\vdash 0^2 + 0^2 = 0$ 
norm_num

-- 2ª demostración
-- =====

example :  $x^2 + y^2 = 0 \leftrightarrow x = 0 \wedge y = 0 :=$ 
by
  constructor
. --  $\vdash x^2 + y^2 = 0 \rightarrow x = 0 \wedge y = 0$ 
  intro h
  --  $h : x^2 + y^2 = 0$ 
  --  $\vdash x = 0 \wedge y = 0$ 
  constructor
. --  $\vdash x = 0$ 
  exact aux h
. --  $\vdash y = 0$ 
  rw [add_comm] at h
  --  $h : x^2 + y^2 = 0$ 
  exact aux h
. --  $\vdash x = 0 \wedge y = 0 \rightarrow x^2 + y^2 = 0$ 
  rintro ⟨h1, h2⟩
  --  $h1 : x = 0$ 
  --  $h2 : y = 0$ 
  --  $\vdash x^2 + y^2 = 0$ 
  rw [h1, h2]
  --  $\vdash 0^2 + 0^2 = 0$ 
  norm_num

-- 3ª demostración
-- =====

example :  $x^2 + y^2 = 0 \leftrightarrow x = 0 \wedge y = 0 :=$  by
  constructor
. --  $\vdash x^2 + y^2 = 0 \rightarrow x = 0 \wedge y = 0$ 
  intro h

```

```

-- h : x ^ 2 + y ^ 2 = 0
-- ⊢ x = 0 ∧ y = 0
constructor
· -- x = 0
  exact aux h
· -- ⊢ y = 0
  rw [add_comm] at h
  -- h : y ^ 2 + x ^ 2 = 0
  exact aux h
· -- ⊢ x = 0 ∧ y = 0 → x ^ 2 + y ^ 2 = 0
  rintro ⟨rfl, rfl⟩
  -- ⊢ 0 ^ 2 + 0 ^ 2 = 0
  norm_num

-- Lemas usados
-- =====

-- #check (add_comm x y : x + y = y + x)
-- #check (le_add_of_nonneg_right : 0 ≤ y → x ≤ x + y)
-- #check (le_antisymm : x ≤ y → y ≤ x → x = y)
-- #check (pow_eq_zero : ∀ {n : ℕ}, x ^ n = 0 → x = 0)
-- #check (pow_two_nonneg x : 0 ≤ x ^ 2)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

### 3.14. En $\mathbb{R}$ , $x^2 = 1 \rightarrow x = 1 \vee x = -1$

```

-- -----
-- Demostrar que si
--   x^2 = 1
-- entonces
--   x = 1 ∨ x = -1
-- -----

-- Demostración en lenguaje natural
-- =====

-- Usaremos los siguientes lemas
--   (∀ x ∈ ℝ)[x - x = 0]                                     (L1)
--   (∀ x, y ∈ ℝ)[xy = 0 → x = 0 ∨ y = 0]                   (L2)
--   (∀ x, y ∈ ℝ)[x - y = 0 ↔ x = y]                       (L3)
--   (∀ x, y ∈ ℝ)[x + y = 0 → x = -y]                       (L4)
--

```



```
-- Se tiene que
--   (x - 1)(x + 1) = x^2 - 1
--                   = 1 - 1      [por la hipótesis]
--                   = 0          [por L1]
-- y, por el lema L2, se tiene que
--   x - 1 = 0 ∨ x + 1 = 0
-- Acabaremos la demostración por casos.
--
-- Primer caso:
--   x - 1 = 0 ⇒ x = 1           [por L3]
--               ⇒ x = 1 ∨ x = -1
--
-- Segundo caso:
--   x + 1 = 0 ⇒ x = -1          [por L4]
--               ⇒ x = 1 ∨ x = -1
--
-- Demostraciones con Lean4
-- =====
```

```
import Mathlib.Data.Real.Basic
variable (x y : ℝ)
```

```
-- 1ª demostración
-- =====
```

```
example
  (h : x^2 = 1)
  : x = 1 ∨ x = -1 :=
by
  have h1 : (x - 1) * (x + 1) = 0 := by
    calc (x - 1) * (x + 1) = x^2 - 1 := by ring
          _ = 1 - 1      := by rw [h]
          _ = 0          := sub_self 1
  have h2 : x - 1 = 0 ∨ x + 1 = 0 := by
    apply eq_zero_or_eq_zero_of_mul_eq_zero h1
  rcases h2 with h3 | h4
  . -- h3 : x - 1 = 0
    left
    -- ⊢ x = 1
    exact sub_eq_zero.mp h3
  . -- h4 : x + 1 = 0
    right
    -- ⊢ x = -1
    exact eq_neg_of_add_eq_zero_left h4
```

```

-- 2ª demostración
-- =====

example
  (h : x^2 = 1)
  : x = 1 ∨ x = -1 :=
by
  have h1 : (x - 1) * (x + 1) = 0 := by nlinarith
  have h2 : x - 1 = 0 ∨ x + 1 = 0 := by aesop
  rcases h2 with h3 | h4
  . -- h3 : x - 1 = 0
    left
    -- ⊢ x = 1
    linarith
  . -- h4 : x + 1 = 0
    right
    -- ⊢ x = -1
    linarith

-- 3ª demostración
-- =====

example
  (h : x^2 = 1)
  : x = 1 ∨ x = -1 :=
sq_eq_one_iff.mp h

-- 3ª demostración
-- =====

example
  (h : x^2 = 1)
  : x = 1 ∨ x = -1 :=
by aesop

-- Lemas usados
-- =====

-- #check (eq_neg_of_add_eq_zero_left : x + y = 0 → x = -y)
-- #check (eq_zero_or_eq_zero_of_mul_eq_zero : x * y = 0 → x = 0 ∨ y = 0)
-- #check (sq_eq_one_iff : x ^ 2 = 1 ↔ x = 1 ∨ x = -1)
-- #check (sub_eq_zero : x - y = 0 ↔ x = y)
-- #check (sub_self x : x - x = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

**3.15. En  $\mathbb{R}$ ,  $x^2 = y^2 \rightarrow x = y \vee x = -y$** 

```

-- -----
-- Demostrar que si
--    $x^2 = y^2$ 
-- entonces
--    $x = y \vee x = -y$ 
-- -----

-- Usaremos los siguientes lemas
--    $(\forall x \in \mathbb{R})[x - x = 0]$                                 (L1)
--    $(\forall x, y \in \mathbb{R})[xy = 0 \rightarrow x = 0 \vee y = 0]$         (L2)
--    $(\forall x, y \in \mathbb{R})[x - y = 0 \leftrightarrow x = y]$           (L3)
--    $(\forall x, y \in \mathbb{R})[x + y = 0 \rightarrow x = -y]$         (L4)
--
-- Se tiene que
--    $(x - y)(x + y) = x^2 - y^2$ 
--                    $= y^2 - y^2$       [por la hipótesis]
--                    $= 0$               [por L1]
-- y, por el lema L2, se tiene que
--    $x - y = 0 \vee x + y = 0$ 
--
-- Acabaremos la demostración por casos.
--
-- Primer caso:
--    $x - y = 0 \implies x = y$                 [por L3]
--                    $\implies x = y \vee x = -y$ 
--
-- Segundo caso:
--    $x + y = 0 \implies x = -y$              [por L4]
--                    $\implies x = y \vee x = -y$ 
--
-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable (x y : ℝ)

-- 1ª demostración
-- =====

example
  (h : x^2 = y^2)
  : x = y ∨ x = -y :=

```

```

by
  have h1 : (x - y) * (x + y) = 0 := by
    calc (x - y) * (x + y) = x^2 - y^2 := by ring
          _ = y^2 - y^2 := by rw [h]
          _ = 0 := sub_self (y ^ 2)
  have h2 : x - y = 0 ∨ x + y = 0 := by
    apply eq_zero_or_eq_zero_of_mul_eq_zero h1
  rcases h2 with h3 | h4
  . -- h3 : x - y = 0
    left
    -- ⊢ x = y
    exact sub_eq_zero.mp h3
  . -- h4 : x + y = 0
    right
    -- ⊢ x = -y
    exact eq_neg_of_add_eq_zero_left h4

```

-- 2ª demostración

-- =====

example

```

(h : x^2 = y^2)
: x = y ∨ x = -y :=

```

by

```

  have h1 : (x - y) * (x + y) = 0 := by nlinarith
  have h2 : x - y = 0 ∨ x + y = 0 := by aesop
  rcases h2 with h3 | h4
  . -- h3 : x - y = 0
    left
    -- ⊢ x = y
    linarith
  . -- h4 : x + y = 0
    right
    -- ⊢ x = -y
    linarith

```

-- 2ª demostración

-- =====

example

```

(h : x^2 = y^2)
: x = y ∨ x = -y :=
sq_eq_sq_iff_eq_or_eq_neg.mp h

```

-- Lemas usados

```
-- =====
-- #check (eq_neg_of_add_eq_zero_left : x + y = 0 → x = -y)
-- #check (eq_zero_or_eq_zero_of_mul_eq_zero : x * y = 0 → x = 0 ∨ y = 0)
-- #check (sq_eq_sq_iff_eq_or_eq_neg : x ^ 2 = y ^ 2 ↔ x = y ∨ x = -y)
-- #check (sub_eq_zero : x - y = 0 ↔ x = y)
-- #check (sub_self x : x - x = 0)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

### 3.16. En $\mathbb{R}$ , $|a| = |a - b + b|$

```
-- -----
-- Demostrar que
--   |a| = |a - b + b|
-- -----

import Mathlib.Data.Real.Basic
variable (a b : ℝ)

-- 1ª demostración
-- =====

example
  : |a| = |a - b + b| :=
by
  congr
  -- a = a - b + b
  ring

-- Comentario: La táctica congr sustituye una conclusión de la forma
-- A = B por las igualdades de sus subtérminos que no no iguales por
-- definición. Por ejemplo, sustituye la conclusión (x * f y = g w * f z)
-- por las conclusiones (x = g w) y (y = z).

-- 2ª demostración
-- =====

example
  (a b : ℝ)
  : |a| = |a - b + b| :=
by { congr ; ring }
```

```
-- 3ª demostración
-- =====

example
  (a b : ℝ)
  : |a| = |a - b + b| :=
by ring_nf
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

# Capítulo 4

## Propiedades elementales de los anillos

### 4.1. Si $R$ es un anillo y $a \in R$ , entonces $a + 0 = a$

```
-- Demostrar en Lean4 que si R es un anillo, entonces
--    $\forall a : R, a + 0 = a$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades
--    $a + 0 = 0 + a$  [por la conmutativa de la suma]
--    $= a$  [por el axioma del cero por la izquierda]

import Mathlib.Algebra.Ring.Defs

variable {R : Type _} [Ring R]
variable (a : R)

-- Demostraciones con Lean4
-- =====

-- 1ª demostración
example : a + 0 = a :=
calc a + 0
    = 0 + a := by rw [add_comm]
```

```

_ = a      := by rw [zero_add]

```

```

-- 2ª demostración

```

```

example : a + 0 = a :=

```

```

by

```

```

  rw [add_comm]

```

```

  rw [zero_add]

```

```

-- 3ª demostración

```

```

example : a + 0 = a :=

```

```

by rw [add_comm, zero_add]

```

```

-- 4ª demostración

```

```

example : a + 0 = a :=

```

```

by exact add_zero a

```

```

-- 5ª demostración

```

```

example : a + 0 = a :=

```

```

  add_zero a

```

```

-- 5ª demostración

```

```

example : a + 0 = a :=

```

```

by simp

```

```

-- Lemas usados

```

```

-- =====

```

```

variable (a b : R)

```

```

-- #check (add_comm a b : a + b = b + a)

```

```

-- #check (zero_add a : 0 + a = a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.2. Si $R$ es un anillo y $a \in R$ , entonces $a + -a = 0$

```

-- -----
-- Demostrar en Lean4 que si R es un anillo, entonces
--    $\forall a : R, a + -a = 0$ 
-- -----

```



```

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades
--   a + -a = -a + a   [por la conmutativa de la suma]
--           = 0        [por el axioma de inverso por la izquierda]

import Mathlib.Algebra.Ring.Defs

variable {R : Type _} [Ring R]
variable (a : R)

-- 1ª demostración
-- =====

example : a + -a = 0 :=
calc a + -a = -a + a := by rw [add_comm]
    _ = 0      := by rw [add_left_neg]

-- 2ª demostración
-- =====

example : a + -a = 0 :=
by
  rw [add_comm]
  rw [add_left_neg]

-- 3ª demostración
-- =====

example : a + -a = 0 :=
by rw [add_comm, add_left_neg]

-- 4ª demostración
-- =====

example : a + -a = 0 :=
by exact add_neg_self a

-- 5ª demostración
-- =====

example : a + -a = 0 :=
  add_neg_self a

```

```

-- 6ª demostración
-- =====

example : a + -a = 0 :=
by simp

-- Lemas usados
-- =====

-- variable (a b : R)
-- #check (add_comm a b : a + b = b + a)
-- #check (add_left_neg a : -a + a = 0)
-- #check (add_neg_self a : a + -a = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 4.3. Si $R$ es un anillo y $a, b \in R$ , entonces $-a + (a + b) = b$

```

-----
-- Demostrar en Lean4 que si R es un anillo, entonces
--    $\forall a, b : R, -a + (a + b) = b$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades
--    $-a + (a + b) = (-a + a) + b$  [por la asociativa]
--    $= 0 + b$  [por inverso por la izquierda]
--    $= b$  [por cero por la izquierda]

import Mathlib.Algebra.Ring.Defs

variable {R : Type _} [Ring R]
variable (a b : R)

-- Demostraciones con Lean4
-- =====

-- 1ª demostración
example : -a + (a + b) = b :=

```

```

calc -a + (a + b) = (-a + a) + b := by rw [← add_assoc]
      _ = 0 + b           := by rw [add_left_neg]
      _ = b               := by rw [zero_add]

-- 2ª demostración
example : -a + (a + b) = b :=
by
  rw [←add_assoc]
  rw [add_left_neg]
  rw [zero_add]

-- 3ª demostración
example : -a + (a + b) = b :=
by rw [←add_assoc, add_left_neg, zero_add]

-- 4ª demostración
example : -a + (a + b) = b :=
by exact neg_add_cancel_left a b

-- 5ª demostración
example : -a + (a + b) = b :=
  neg_add_cancel_left a b

-- 6ª demostración
example : -a + (a + b) = b :=
by simp

-- Lemas usados
-- =====

-- variable (c : R)
-- #check (add_assoc a b c : (a + b) + c = a + (b + c))
-- #check (add_left_neg a : -a + a = 0)
-- #check (neg_add_cancel_left a b : -a + (a + b) = b)
-- #check (zero_add a : 0 + a = a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

```
-- Demostrar en Lean4 que si R es un anillo, entonces
--       $\forall a, b : R, (a + b) + -b = a$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades
--       $(a + b) + -b = a + (b + -b)$       [por la asociativa]
--               $= a + 0$                     [por suma con opuesto]
--               $= a$                         [por suma con cero]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs

variable {R : Type _} [Ring R]
variable (a b : R)

-- 1ª demostración
example : (a + b) + -b = a :=
calc
  (a + b) + -b = a + (b + -b) := by rw [add_assoc]
    _           = a + 0         := by rw [add_right_neg]
    _           = a             := by rw [add_zero]

-- 2ª demostración
example : (a + b) + -b = a :=
by
  rw [add_assoc]
  rw [add_right_neg]
  rw [add_zero]

-- 3ª demostración
example : (a + b) + -b = a :=
by rw [add_assoc, add_right_neg, add_zero]

-- 4ª demostración
example : (a + b) + -b = a :=
```

```

add_neg_cancel_right a b

-- 5ª demostración
example : (a + b) + -b = a :=
  add_neg_cancel_right _ _

-- 6ª demostración
example : (a + b) + -b = a :=
  by simp

-- Lemas usados
-- =====

-- variable (c : R)
-- #check (add_assoc a b c : (a + b) + c = a + (b + c))
-- #check (add_neg_cancel_right a b : (a + b) + -b = a)
-- #check (add_right_neg a : a + -a = 0)
-- #check (add_zero a : a + 0 = a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.5. Si $R$ es un anillo y $a, b, c \in R$ tales que $a+b=a+c$ , entonces $b=c$

```

-----
-- Demostrar que si  $R$  es un anillo y  $a, b, c \in R$  tales que
--    $a + b = a + c$ 
-- entonces
--    $b = c$ 
-----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Por la siguiente cadena de igualdades
--    $b = 0 + b$            [por suma con cero]
--    $= (-a + a) + b$        [por suma con opuesto]
--    $= -a + (a + b)$        [por asociativa]
--    $= -a + (a + c)$        [por hipótesis]

```

```

--      = (-a + a) + c    [por asociativa]
--      = 0 + c          [por suma con opuesto]
--      = c              [por suma con cero]

-- 2ª demostración en LN
-- =====

-- Por la siguiente cadena de implicaciones
--      a + b = a + c
--      ==> -a + (a + b) = -a + (a + c)    [sumando -a]
--      ==> (-a + a) + b = (-a + a) + c    [por la asociativa]
--      ==> 0 + b = 0 + b                  [suma con opuesto]
--      ==> b = c                          [suma con cero]

-- 3ª demostración en LN
-- =====

-- Por la siguiente cadena de igualdades
--      b = -a + (a + b)
--      = -a + (a + c)    [por la hipótesis]
--      = c

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]
variable {a b c : R}

-- 1ª demostración
example
  (h : a + b = a + c)
  : b = c :=
calc
  b = 0 + b      := by rw [zero_add]
  _ = (-a + a) + b := by rw [add_left_neg]
  _ = -a + (a + b) := by rw [add_assoc]
  _ = -a + (a + c) := by rw [h]
  _ = (-a + a) + c := by rw [←add_assoc]
  _ = 0 + c       := by rw [add_left_neg]
  _ = c           := by rw [zero_add]

-- 2ª demostración

```

```

example
  (h : a + b = a + c)
  : b = c :=
by
  have h1 : -a + (a + b) = -a + (a + c) :=
    congrArg (HAdd.hAdd (-a)) h
  clear h
  rw [← add_assoc] at h1
  rw [add_left_neg] at h1
  rw [zero_add] at h1
  rw [← add_assoc] at h1
  rw [add_left_neg] at h1
  rw [zero_add] at h1
  exact h1

-- 3ª demostración
example
  (h : a + b = a + c)
  : b = c :=
calc
  b = -a + (a + b) := by rw [neg_add_cancel_left a b]
  _ = -a + (a + c) := by rw [h]
  _ = c              := by rw [neg_add_cancel_left]

-- 4ª demostración
example
  (h : a + b = a + c)
  : b = c :=
by
  rw [← neg_add_cancel_left a b]
  rw [h]
  rw [neg_add_cancel_left]

-- 5ª demostración
example
  (h : a + b = a + c)
  : b = c :=
by
  rw [← neg_add_cancel_left a b, h, neg_add_cancel_left]

-- 6ª demostración
example
  (h : a + b = a + c)
  : b = c :=
add_left_cancel h

```

```
-- Lemas usados
-- =====

-- #check (add_assoc a b c : (a + b) + c = a + (b + c))
-- #check (add_left_cancel : a + b = a + c → b = c)
-- #check (add_left_neg a : -a + a = 0)
-- #check (neg_add_cancel_left a b : -a + (a + b) = b)
-- #check (zero_add a : 0 + a = a)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.6. Si $R$ es un anillo y $a, b, c \in R$ tales que $a+b=c+b$ , entonces $a=c$

```
-- -----
-- Demostrar que si  $R$  es un anillo y  $a, b, c \in R$  tales que
--    $a + b = c + b$ 
-- entonces
--    $a = c$ 
-- -----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Por la siguiente cadena de igualdades
--    $a = a + 0$            [por suma con cero]
--    $= a + (b + -b)$        [por suma con opuesto]
--    $= (a + b) + -b$        [por asociativa]
--    $= (c + b) + -b$        [por hipótesis]
--    $= c + (b + -b)$        [por asociativa]
--    $= c + 0$              [por suma con opuesto]
--    $= c$                  [por suma con cero]

-- 2ª demostración en LN
-- =====

-- Por la siguiente cadena de igualdades
--    $a = (a + b) + -b$ 
--    $= (c + b) + -b$        [por hipótesis]
```



```

--      = c

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]
variable {a b c : R}

-- 1ª demostración con Lean4
-- =====

example
  (h : a + b = c + b)
  : a = c :=
calc
  a = a + 0      := by rw [add_zero]
  _ = a + (b + -b) := by rw [add_right_neg]
  _ = (a + b) + -b := by rw [add_assoc]
  _ = (c + b) + -b := by rw [h]
  _ = c + (b + -b) := by rw [← add_assoc]
  _ = c + 0        := by rw [← add_right_neg]
  _ = c            := by rw [add_zero]

-- 2ª demostración con Lean4
-- =====

example
  (h : a + b = c + b)
  : a = c :=
calc
  a = (a + b) + -b := (add_neg_cancel_right a b).symm
  _ = (c + b) + -b := by rw [h]
  _ = c           := add_neg_cancel_right c b

-- 3ª demostración con Lean4
-- =====

example
  (h : a + b = c + b)
  : a = c :=
by
  rw [← add_neg_cancel_right a b]

```

```

rw [h]
rw [add_neg_cancel_right]

-- 4ª demostración con Lean4
-- =====

example
  (h : a + b = c + b)
  : a = c :=
by
  rw [← add_neg_cancel_right a b, h, add_neg_cancel_right]

-- 5ª demostración con Lean4
-- =====

example
  (h : a + b = c + b)
  : a = c :=
add_right_cancel h

-- Lemas usados
-- =====

-- #check (add_assoc a b c : (a + b) + c = a + (b + c))
-- #check (add_neg_cancel_right a b : (a + b) + -b = a)
-- #check (add_right_cancel : a + b = c + b → a = c)
-- #check (add_right_neg a : a + -a = 0)
-- #check (add_zero a : a + 0 = a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.7. Si $R$ es un anillo y $a \in R$ , entonces $a \cdot 0 = 0$

```

-- -----
-- Demostrar que si  $R$  es un anillo y  $a \in R$ , entonces
--    $a \cdot 0 = 0$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Basta aplicar la propiedad cancelativa a
--    $a \cdot 0 + a \cdot 0 = a \cdot 0 + 0$ 

```

```

-- que se demuestra mediante la siguiente cadena de igualdades
--   a.0 + a.0 = a.(0 + 0)    [por la distributiva]
--               = a.0        [por suma con cero]
--               = a.0 + 0     [por suma con cero]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]
variable (a : R)

-- 1ª demostración
-- =====

example : a * 0 = 0 :=
by
  have h : a * 0 + a * 0 = a * 0 + 0 :=
    calc a * 0 + a * 0 = a * (0 + 0) := by rw [mul_add a 0 0]
          _ = a * 0 := by rw [add_zero 0]
          _ = a * 0 + 0 := by rw [add_zero (a * 0)]
  rw [add_left_cancel h]

-- 2ª demostración
-- =====

example : a * 0 = 0 :=
by
  have h : a * 0 + a * 0 = a * 0 + 0 :=
    calc a * 0 + a * 0 = a * (0 + 0) := by rw [← mul_add]
          _ = a * 0 := by rw [add_zero]
          _ = a * 0 + 0 := by rw [add_zero]
  rw [add_left_cancel h]

-- 3ª demostración
-- =====

example : a * 0 = 0 :=
by
  have h : a * 0 + a * 0 = a * 0 + 0 :=
    by rw [← mul_add, add_zero, add_zero]
  rw [add_left_cancel h]

```

```

-- 4ª demostración
-- =====

example : a * 0 = 0 :=
by
  have : a * 0 + a * 0 = a * 0 + 0 :=
    calc a * 0 + a * 0 = a * (0 + 0) := by simp
          _ = a * 0 := by simp
          _ = a * 0 + 0 := by simp
  simp

-- 5ª demostración
-- =====

example : a * 0 = 0 :=
  mul_zero a

-- 6ª demostración
-- =====

example : a * 0 = 0 :=
by simp

-- Lemas usados
-- =====

-- variable (b c : R)
-- #check (add_left_cancel : a + b = a + c → b = c)
-- #check (add_zero a : a + 0 = a)
-- #check (mul_add a b c : a * (b + c) = a * b + a * c)
-- #check (mul_zero a : a * 0 = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.8. Si $R$ es un anillo y $a \in R$ , entonces $0.a = 0$

```

-----
-- Demostrar que si  $R$  es un anillo y  $a \in R$ , entonces
--    $0 * a = 0$ 
-----

-- Demostración en lenguaje natural
-- =====

```

```

-- Basta aplicar la propiedad cancelativa a
--    $0.a + 0.a = 0.a + 0$ 
-- que se demuestra mediante la siguiente cadena de igualdades
--    $0.a + 0.a = (0 + 0).a$  [por la distributiva]
--                $= 0.a$  [por suma con cero]
--                $= 0.a + 0$  [por suma con cero]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]
variable (a : R)

-- 1ª demostración
example : 0 * a = 0 :=
by
  have h : 0 * a + 0 * a = 0 * a + 0 :=
    calc 0 * a + 0 * a = (0 + 0) * a := by rw [add_mul]
        _ = 0 * a := by rw [add_zero]
        _ = 0 * a + 0 := by rw [add_zero]
  rw [add_left_cancel h]

-- 2ª demostración
example : 0 * a = 0 :=
by
  have h : 0 * a + 0 * a = 0 * a + 0 :=
    by rw [←add_mul, add_zero, add_zero]
  rw [add_left_cancel h]

-- 3ª demostración
example : 0 * a = 0 :=
by
  have : 0 * a + 0 * a = 0 * a + 0 :=
    calc 0 * a + 0 * a = (0 + 0) * a := by simp
        _ = 0 * a := by simp
        _ = 0 * a + 0 := by simp
  simp

-- 4ª demostración
example : 0 * a = 0 :=
by

```

```

have : 0 * a + 0 * a = 0 * a + 0 := by simp
simp

-- 5ª demostración
example : 0 * a = 0 :=
by simp

-- 6ª demostración
example : 0 * a = 0 :=
zero_mul a

-- Lemas usados
-- =====

-- variable (b c : R)
-- #check (add_mul a b c : (a + b) * c = a * c + b * c)
-- #check (add_zero a : a + 0 = a)
-- #check (zero_mul a : 0 * a = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.9. Si $R$ es un anillo y $a, b \in R$ tales que $a+b=0$ , entonces $-a=b$

```

-- -----
-- Demostrar que si es un anillo y  $a, b \in R$  tales que
--    $a + b = 0$ 
-- entonces
--    $-a = b$ 
-- -----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- -----

-- Por la siguiente cadena de igualdades
--    $-a = -a + 0$            [por suma cero]
--    $= -a + (a + b)$        [por hipótesis]
--    $= b$                    [por cancelativa]

```

```

-- 2ª demostración en LN
-- -----

-- Sumando -a a ambos lados de la hipótesis, se tiene
--   -a + (a + b) = -a + 0
-- El término de la izquierda se reduce a b (por la cancelativa) y el de
-- la derecha a -a (por la suma con cero). Por tanto, se tiene
--   b = -a
-- Por la simetría de la igualdad, se tiene
--   -a = b

-- Demostraciones con Lean 4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]
variable {a b : R}

-- 1ª demostración (basada en la 1ª en LN)
example
  (h : a + b = 0)
  : -a = b :=
calc
  -a = -a + 0      := by rw [add_zero]
  _  = -a + (a + b) := by rw [h]
  _  = b           := by rw [neg_add_cancel_left]

-- 2ª demostración (basada en la 1ª en LN)
example
  (h : a + b = 0)
  : -a = b :=
calc
  -a = -a + 0      := by simp
  _  = -a + (a + b) := by rw [h]
  _  = b           := by simp

-- 3ª demostración (basada en la 2ª en LN)
example
  (h : a + b = 0)
  : -a = b :=
by
  have h1 : -a + (a + b) = -a + 0 := congrArg (HAdd.hAdd (-a)) h
  have h2 : -a + (a + b) = b := neg_add_cancel_left a b

```

```

have h3 : -a + 0 = -a := add_zero (-a)
rw [h2, h3] at h1
exact h1.symm

-- 4ª demostración
example
  (h : a + b = 0)
  : -a = b :=
neg_eq_iff_add_eq_zero.mpr h

-- Lemas usados
-- =====

-- #check (add_zero a : a + 0 = a)
-- #check (neg_add_cancel_left a b : -a + (a + b) = b)
-- #check (neg_eq_iff_add_eq_zero : -a = b ↔ a + b = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.10. Si $R$ es un anillo y $a, b \in R$ tales que $a+b=0$ , entonces $a=-b$

```

-----
-- Demostrar que si  $R$  es un anillo y  $a, b \in R$  tales que
--    $a + b = 0$ 
-- entonces
--    $a = -b$ 
-----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- -----

-- Por la siguiente cadena de igualdades
--    $a = (a + b) + -b$       [por la conelativa]
--    $= 0 + -b$               [por la hipótesis]
--    $= -b$                   [por la suma con cero]

-- 2ª demostración en LN
-- -----

```



```

-- Sumando -a a ambos lados de la hipótesis, se tiene
-- (a + b) + -b = 0 + -b
-- El término de la izquierda se reduce a a (por la cancelativa) y el de
-- la derecha a -b (por la suma con cero). Por tanto, se tiene
-- a = -b

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]
variable {a b : R}

-- 1ª demostración (basada en la 1ª en LN)
example
  (h : a + b = 0)
  : a = -b :=
calc
  a = (a + b) + -b := by rw [add_neg_cancel_right]
  _ = 0 + -b       := by rw [h]
  _ = -b           := by rw [zero_add]

-- 2ª demostración (basada en la 1ª en LN)
example
  (h : a + b = 0)
  : a = -b :=
calc
  a = (a + b) + -b := by simp
  _ = 0 + -b       := by rw [h]
  _ = -b           := by simp

-- 3ª demostración (basada en la 1ª en LN)
example
  (h : a + b = 0)
  : a = -b :=
by
  have h1 : (a + b) + -b = 0 + -b := by rw [h]
  have h2 : (a + b) + -b = a := add_neg_cancel_right a b
  have h3 : 0 + -b = -b := zero_add (-b)
  rwa [h2, h3] at h1

-- 4ª demostración

```

**example**

```

(h : a + b = 0)
: a = -b :=
add_eq_zero_iff_eq_neg.mp h

-- Lemas usados
-- =====

-- #check (add_eq_zero_iff_eq_neg : a + b = 0 ↔ a = -b)
-- #check (add_neg_cancel_right a b : (a + b) + -b = a)
-- #check (zero_add a : 0 + a = a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.11. Si $R$ es un anillo, entonces $-0 = 0$

```

-- -----
-- Demostrar que si  $R$  es un anillo, entonces
--    $-0 = 0$ 
-- -----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Por la suma con cero se tiene
--    $0 + 0 = 0$ 
-- Aplicándole la propiedad
--    $\forall a b \in R, a + b = 0 \rightarrow -a = b$ 
-- se obtiene
--    $-0 = 0$ 

-- 2ª demostración en LN
-- =====

-- Puesto que
--    $\forall a b \in R, a + b = 0 \rightarrow -a = b$ 
-- basta demostrar que
--    $0 + 0 = 0$ 
-- que es cierta por la suma con cero.

```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]

-- 1ª demostración (basada en la 1ª en LN)
example : (-0 : R) = 0 :=
by
  have h1 : (0 : R) + 0 = 0 := add_zero 0
  show (-0 : R) = 0
  exact neg_eq_of_add_eq_zero_left h1

-- 2ª demostración (basada en la 2ª en LN)
example : (-0 : R) = 0 :=
by
  apply neg_eq_of_add_eq_zero_left
  rw [add_zero]

-- 3ª demostración
example : (-0 : R) = 0 :=
  neg_zero

-- 4ª demostración
example : (-0 : R) = 0 :=
by simp

-- Lemas usados
-- =====

-- variable (a b : R)
-- #check (add_zero a : a + 0 = a)
-- #check (neg_eq_of_add_eq_zero_left : a + b = 0 → -b = a)
-- #check (neg_zero : -0 = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.12. Si $R$ es un anillo y $a \in R$ , entonces $-(-a) = a$

```

-----
-- Demostrar que si  $R$  es un anillo y  $a \in R$ , entonces
--  $-(-a) = a$ 
-----

-- Demostración en lenguaje natural
-- =====

-- Es consecuencia de las siguientes propiedades demostradas en
-- ejercicios anteriores:
--  $\forall a, b \in R, a + b = 0 \rightarrow -a = b$ 
--  $\forall a \in R, -a + a = 0$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic

variable {R : Type _} [Ring R]
variable {a : R}

-- 1ª demostración
example : -(-a) = a :=
by
  have h1 : -a + a = 0 := add_left_neg a
  show -(-a) = a
  exact neg_eq_of_add_eq_zero_right h1

-- 2ª demostración
example : -(-a) = a :=
by
  apply neg_eq_of_add_eq_zero_right
  rw [add_left_neg]

-- 3ª demostración
example : -(-a) = a :=
neg_neg a

-- 4ª demostración
example : -(-a) = a :=

```

```
by simp

-- Lemas usados
-- =====

-- variable (b : R)
-- #check (add_left_neg a : -a + a = 0)
-- #check (neg_eq_of_add_eq_zero_right : a + b = 0 → -a = b)
-- #check (neg_neg a : -(-a) = a)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.13. Si $R$ es un anillo y $a, b \in R$ , entonces $a - b = a + -b$

```
-- -----
-- Demostrar que si  $R$  es un anillo y  $a, b \in R$ , entonces
--  $a - b = a + -b$ 
-- -----
```

```
-- Demostración en lenguaje natural
-- =====
```

```
-- Por la definición de la resta.
```

```
-- Demostración en Lean4
-- =====
```

```
import Mathlib.Algebra.Ring.Defs
```

```
variable {R : Type _} [Ring R]
```

```
variable (a b : R)
```

```
example : a - b = a + -b :=
```

```
-- by exact?
```

```
sub_eq_add_neg a b
```

```
-- Lemas usados
-- =====
```

```
-- #check (sub_eq_add_neg a b : a - b = a + -b)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.14. Si $R$ es un anillo y $a \in R$ , entonces $a - a = 0$

```
-- Demostrar que si R es un anillo y a ∈ R, entonces
--   a - a = 0
-----
```

```
-- Demostración en lenguaje natural
-- =====
```

```
-- Por la siguiente cadena de igualdades:
--   a - a = a + -a   [por definición de resta]
--         = 0        [por suma con opuesto]
```

```
-- Demostraciones con Lean4
-- =====
```

```
import Mathlib.Algebra.Ring.Defs
variable {R : Type _} [Ring R]
variable (a : R)
```

```
-- 1ª demostración
```

```
example : a - a = 0 :=
calc
```

```
  a - a = a + -a := by rw [sub_eq_add_neg a a]
    _ = 0       := by rw [add_right_neg]
```

```
-- 2ª demostración
```

```
example : a - a = 0 :=
sub_self a
```

```
-- 3ª demostración
```

```
example : a - a = 0 :=
by simp
```

```
-- Lemas usados
-- =====
```

```
-- #check (add_right_neg a : a + -a = 0)
-- #check (sub_eq_add_neg a b : a - b = a + -b)
-- #check (sub_self a : a - a = 0)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.15. En los anillos, $1 + 1 = 2$

```

-- -----
-- Demostrar que en los anillos,
--   1 + 1 = 2
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por cálculo.

-- Demostración con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs
import Mathlib.Tactic
variable {R : Type _} [Ring R]

-- Demostraciones con Lean4
-- =====

-- 1ª demostración
example : 1 + 1 = (2 : R) :=
by norm_num

-- 2ª demostración
example : 1 + 1 = (2 : R) :=
one_add_one_eq_two

-- Lemas usados
-- =====

-- #check (one_add_one_eq_two : 1 + 1 = 2)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 4.16. Si $R$ es un anillo y $a \in R$ , entonces $2a = a + a$

```

-----
-- Demostrar que si R es un anillo y a ∈ R, entonces
--   2 * a = a + a
-----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de igualdades
--   2·a = (1 + 1)·a    [por la definición de 2]
--       = 1·a + 1·a    [por la distributiva]
--       = a + a        [por producto con uno]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Ring.Defs

variable {R : Type _} [Ring R]
variable (a : R)

-- 1ª demostración
example : 2 * a = a + a :=
calc
  2 * a = (1 + 1) * a    := by rw [one_add_one_eq_two]
  _     = 1 * a + 1 * a := by rw [add_mul]
  _     = a + a         := by rw [one_mul]

-- 2ª demostración
example : 2 * a = a + a :=
by exact two_mul a

-- Lemas usados
-- =====

-- variable (b c : R)
-- #check (add_mul a b c : (a + b) * c = a * c + b * c)
-- #check (one_add_one_eq_two : (1 : R) + 1 = 2)
-- #check (one_mul a : 1 * a = a)
-- #check (two_mul a : 2 * a = a + a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).



## Capítulo 5

# Propiedades elementales de los grupos

### 5.1. Si $G$ es un grupo y $a \in G$ , entonces $aa^{-1} = 1$

```
-- En Lean4, se declara que G es un grupo mediante la expresión
--   variable {G : Type _} [Group G]
--
-- Como consecuencia, se tiene los siguientes axiomas
--   mul_assoc :    ∀ a b c : G, a * b * c = a * (b * c)
--   one_mul :      ∀ a : G, 1 * a = a
--   mul_left_inv : ∀ a : G, a⁻¹ * a = 1
--
-- Demostrar que si G es un grupo y a ∈ G, entonces
--   a * a⁻¹ = 1
--
-- -----
-- Demostración en lenguaje natural
-- =====
--
-- Por la siguiente cadena de igualdades
--   a · a⁻¹ = 1 · (a · a⁻¹)           [por producto con uno]
--           = (1 · a) · a⁻¹           [por asociativa]
--           = (((a⁻¹)⁻¹ · a⁻¹) · a) · a⁻¹ [por producto con inverso]
--           = ((a⁻¹)⁻¹ · (a⁻¹ · a)) · a⁻¹ [por asociativa]
--           = ((a⁻¹)⁻¹ · 1) · a⁻¹       [por producto con inverso]
--           = (a⁻¹)⁻¹ · (1 · a⁻¹)       [por asociativa]
--           = (a⁻¹)⁻¹ · a⁻¹            [por producto con uno]
```

```

--          = 1                                     [por producto con inverso]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Group.Defs

variable {G : Type _} [Group G]
variable (a b : G)

-- 1ª demostración
example : a * a⁻¹ = 1 :=
calc
  a * a⁻¹ = 1 * (a * a⁻¹)           := by rw [one_mul]
    _      = (1 * a) * a⁻¹           := by rw [mul_assoc]
    _      = (((a⁻¹)⁻¹ * a⁻¹) * a) * a⁻¹ := by rw [mul_left_inv]
    _      = ((a⁻¹)⁻¹ * (a⁻¹ * a)) * a⁻¹ := by rw [← mul_assoc]
    _      = ((a⁻¹)⁻¹ * 1) * a⁻¹       := by rw [mul_left_inv]
    _      = (a⁻¹)⁻¹ * (1 * a⁻¹)       := by rw [mul_assoc]
    _      = (a⁻¹)⁻¹ * a⁻¹           := by rw [one_mul]
    _      = 1                      := by rw [mul_left_inv]

-- 2ª demostración
example : a * a⁻¹ = 1 :=
calc
  a * a⁻¹ = 1 * (a * a⁻¹)           := by simp
    _      = (1 * a) * a⁻¹           := by simp
    _      = (((a⁻¹)⁻¹ * a⁻¹) * a) * a⁻¹ := by simp
    _      = ((a⁻¹)⁻¹ * (a⁻¹ * a)) * a⁻¹ := by simp
    _      = ((a⁻¹)⁻¹ * 1) * a⁻¹       := by simp
    _      = (a⁻¹)⁻¹ * (1 * a⁻¹)       := by simp
    _      = (a⁻¹)⁻¹ * a⁻¹           := by simp
    _      = 1                      := by simp

-- 3ª demostración
example : a * a⁻¹ = 1 :=
by simp

-- 4ª demostración
example : a * a⁻¹ = 1 :=
by exact mul_inv_self a

-- Lemas usados
-- =====

```

```
-- variable (c : G)
-- #check (mul_assoc a b c : (a * b) * c = a * (b * c))
-- #check (mul_inv_self a : a * a⁻¹ = 1)
-- #check (mul_left_inv a : a⁻¹ * a = 1)
-- #check (one_mul a : 1 * a = a)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 5.2. Si $G$ es un grupo y $a \in G$ , entonces $a \cdot 1 = a$

```
-- -----
-- Demostrar que si  $G$  es un grupo y  $a \in G$ , entonces
--    $a * 1 = a$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se tiene por la siguiente cadena de igualdades
--    $a \cdot 1 = a \cdot (a^{-1} \cdot a)$  [por producto con inverso]
--    $= (a \cdot a^{-1}) \cdot a$  [por asociativa]
--    $= 1 \cdot a$  [por producto con inverso]
--    $= a$  [por producto con uno]

-- Demostraciones con Lean4
-- =====
```

```
import Mathlib.Algebra.Group.Defs

variable {G : Type _} [Group G]
variable (a b : G)

-- 1ª demostración
example : a * 1 = a :=
calc
  a * 1 = a * (a⁻¹ * a) := by rw [mul_left_inv]
  _ = (a * a⁻¹) * a := by rw [mul_assoc]
  _ = 1 * a := by rw [mul_right_inv]
  _ = a := by rw [one_mul]

-- 2ª demostración
example : a * 1 = a :=
calc
```

```

a * 1 = a * (a-1 * a) := by simp
_      = (a * a-1) * a := by simp
_      = 1 * a          := by simp
_      = a              := by simp

-- 3ª demostración
example : a * 1 = a :=
by simp

-- 4ª demostración
example : a * 1 = a :=
by exact mul_one a

-- Lemas usados
-- =====

-- variable (c : G)
-- #check (mul_left_inv a : a-1 * a = 1)
-- #check (mul_assoc a b c : (a * b) * c = a * (b * c))
-- #check (mul_right_inv a : a * a-1 = 1)
-- #check (one_mul a : 1 * a = a)
-- #check (mul_one a : a * 1 = a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 5.3. Si $G$ es un grupo y $a, b \in G$ tales que $ab = 1$ entonces $a^{-1} = b$

```

-- -----
-- Demostrar que si  $G$  es un grupo y  $a, b \in G$ , tales que
--    $a * b = 1$ 
-- entonces
--    $a^{-1} = b$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se tiene a partir de la siguiente cadena de igualdades
--    $a^{-1} = a^{-1} * 1$            [por producto por uno]
--    $= a^{-1} * (a * b)$          [por hipótesis]
--    $= (a^{-1} * a) * b$          [por asociativa]

```

```

--      = 1 * b          [por producto con inverso]
--      = b              [por producto por uno]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Group.Defs

variable {G : Type _} [Group G]
variable (a b : G)

-- 1º demostración
example
  (h : a * b = 1)
  : a⁻¹ = b :=
calc
  a⁻¹ = a⁻¹ * 1          := by rw [mul_one]
  _   = a⁻¹ * (a * b)    := by rw [h]
  _   = (a⁻¹ * a) * b    := by rw [mul_assoc]
  _   = 1 * b            := by rw [mul_left_inv]
  _   = b                := by rw [one_mul]

-- 2º demostración
example
  (h : a * b = 1)
  : a⁻¹ = b :=
calc
  a⁻¹ = a⁻¹ * 1          := by simp
  _   = a⁻¹ * (a * b)    := by simp [h]
  _   = (a⁻¹ * a) * b    := by simp
  _   = 1 * b            := by simp
  _   = b                := by simp

-- 3º demostración
example
  (h : a * b = 1)
  : a⁻¹ = b :=
calc
  a⁻¹ = a⁻¹ * (a * b)    := by simp [h]
  _   = b                := by simp

-- 4º demostración
example
  (h : a * b = 1)
  : a⁻¹ = b :=

```

```

by exact inv_eq_of_mul_eq_one_right h

-- Lemas usados
-- =====

-- variable (c : G)
-- #check (mul_assoc a b c : (a * b) * c = a * (b * c))
-- #check (mul_left_inv a : a-1 * a = 1)
-- #check (mul_one a : a * 1 = a)
-- #check (one_mul a : 1 * a = a)
-- #check (inv_eq_of_mul_eq_one_right : a * b = 1 → a-1 = b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 5.4. Si $G$ es un grupo y $a, b \in G$ , entonces $(ab)^{-1} = b^{-1}a^{-1}$

```

-----
-- Demostrar que si  $G$  es un grupo y  $a, b \in G$ , entonces
--  $(a * b)^{-1} = b^{-1} * a^{-1}$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Teniendo en cuenta la propiedad
--  $\forall a, b \in G, ab = 1 \rightarrow a^{-1} = b$ ,
-- basta demostrar que
--  $(a \cdot b) \cdot (b^{-1} \cdot a^{-1}) = 1$ .
-- La identidad anterior se demuestra mediante la siguiente cadena de
-- igualdades
--  $(a \cdot b) \cdot (b^{-1} \cdot a^{-1}) = a \cdot (b \cdot (b^{-1} \cdot a^{-1}))$  [por la asociativa]
--  $= a \cdot ((b \cdot b^{-1}) \cdot a^{-1})$  [por la asociativa]
--  $= a \cdot (1 \cdot a^{-1})$  [por producto con inverso]
--  $= a \cdot a^{-1}$  [por producto con uno]
--  $= 1$  [por producto con inverso]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Group.Defs

```

```

variable {G : Type _} [Group G]
variable (a b : G)

lemma aux : (a * b) * (b⁻¹ * a⁻¹) = 1 :=
calc
  (a * b) * (b⁻¹ * a⁻¹)
    = a * (b * (b⁻¹ * a⁻¹)) := by rw [mul_assoc]
  _ = a * ((b * b⁻¹) * a⁻¹) := by rw [mul_assoc]
  _ = a * (1 * a⁻¹)         := by rw [mul_right_inv]
  _ = a * a⁻¹               := by rw [one_mul]
  _ = 1                     := by rw [mul_right_inv]

-- 1ª demostración
example : (a * b)⁻¹ = b⁻¹ * a⁻¹ :=
by
  have h1 : (a * b) * (b⁻¹ * a⁻¹) = 1 :=
    aux a b
  show (a * b)⁻¹ = b⁻¹ * a⁻¹
  exact inv_eq_of_mul_eq_one_right h1

-- 3ª demostración
example : (a * b)⁻¹ = b⁻¹ * a⁻¹ :=
by
  have h1 : (a * b) * (b⁻¹ * a⁻¹) = 1 :=
    aux a b
  show (a * b)⁻¹ = b⁻¹ * a⁻¹
  simp [h1]

-- 4ª demostración
example : (a * b)⁻¹ = b⁻¹ * a⁻¹ :=
by
  have h1 : (a * b) * (b⁻¹ * a⁻¹) = 1 :=
    aux a b
  simp [h1]

-- 5ª demostración
example : (a * b)⁻¹ = b⁻¹ * a⁻¹ :=
by
  apply inv_eq_of_mul_eq_one_right
  rw [aux]

-- 6ª demostración
example : (a * b)⁻¹ = b⁻¹ * a⁻¹ :=
by exact mul_inv_rev a b

```

```
-- 7ª demostración
example : (a * b)-1 = b-1 * a-1 :=
by simp

-- Lemas usados
-- =====

-- variable (c : G)
-- #check (inv_eq_of_mul_eq_one_right : a * b = 1 → a-1 = b)
-- #check (mul_assoc a b c : (a * b) * c = a * (b * c))
-- #check (mul_inv_rev a b : (a * b)-1 = b-1 * a-1)
-- #check (mul_right_inv a : a * a-1 = 1)
-- #check (one_mul a : 1 * a = a)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).



# Capítulo 6

## Propiedades de orden en los números reales

**6.1. En  $\mathbb{R}$ , si  $a \leq b$ ,  $b < c$ ,  $c \leq d$  y  $d < e$ , entonces  $a < e$**

```
-- -----  
-- Demostrar que si a, b, c, d y e son números reales tales que  
--   a ≤ b,  
--   b < c,  
--   c ≤ d y  
--   d < e,  
-- entonces  
--   a < e.  
-- -----
```

```
-- Demostraciones en lenguaje natural (LN)  
-- =====
```

```
-- 1ª demostración en LN  
-- =====
```

```
-- Por la siguiente cadena de desigualdades  
--   a ≤ b    [por h1]  
--   < c     [por h2]  
--   ≤ d     [por h3]  
--   < e     [por h4]
```

```
-- 2ª demostración en LN  
-- =====
```

```

-- A partir de las hipótesis 1 ( $a \leq b$ ) y 2 ( $b < c$ ) se tiene
--    $a < c$ 
-- que, junto la hipótesis 3 ( $c \leq d$ ) da
--    $a < d$ 
-- que, junto la hipótesis 4 ( $d < e$ ) da
--    $a < e$ .

-- 3ª demostración en LN
-- =====

-- Para demostrar  $a < e$ , por la hipótesis 1 ( $a \leq b$ ) se reduce a probar
--    $b < e$ 
-- que, por la hipótesis 2 ( $b < c$ ), se reduce a
--    $c < e$ 
-- que, por la hipótesis 3 ( $c \leq d$ ), se reduce a
--    $d < e$ 
-- que es cierto, por la hipótesis 4.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (a b c d e : ℝ)

-- 1ª demostración
-- =====

example
  (h1 :  $a \leq b$ )
  (h2 :  $b < c$ )
  (h3 :  $c \leq d$ )
  (h4 :  $d < e$ ) :
   $a < e$  :=
calc
   $a \leq b$  := h1
  _ < c := h2
  _  $\leq d$  := h3
  _ < e := h4

-- 2ª demostración
-- =====

example

```

```

(h1 : a ≤ b)
(h2 : b < c)
(h3 : c ≤ d)
(h4 : d < e) :
a < e :=
by
  have h5 : a < c := lt_of_le_of_lt h1 h2
  have h6 : a < d := lt_of_lt_of_le h5 h3
  show a < e
  exact lt_trans h6 h4

```

```

-- 3ª demostración
-- =====

```

```

example
  (h1 : a ≤ b)
  (h2 : b < c)
  (h3 : c ≤ d)
  (h4 : d < e) :
  a < e :=
by
  apply lt_of_le_of_lt h1
  apply lt_trans h2
  apply lt_of_le_of_lt h3
  exact h4

```

```

-- El desarrollo de la prueba es
--
--   a b c d e : ℝ,
--   h1 : a ≤ b,
--   h2 : b < c,
--   h3 : c ≤ d,
--   h4 : d < e
--   ⊢ a < e
--   apply lt_of_le_of_lt h1,
--   ⊢ b < e
--   apply lt_trans h2,
--   ⊢ c < e
--   apply lt_of_le_of_lt h3,
--   ⊢ d < e
--   exact h4,
--   no goals
--
-- 4ª demostración
-- =====

```

```

example
  (h1 : a ≤ b)
  (h2 : b < c)
  (h3 : c ≤ d)
  (h4 : d < e) :
  a < e :=
by linarith

-- Lemas usados
-- =====

-- #check (lt_of_le_of_lt : a ≤ b → b < c → a < c)
-- #check (lt_of_lt_of_le : a < b → b ≤ c → a < c)
-- #check (lt_trans : a < b → b < c → a < c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 6.2. En $\mathbb{R}$ , si $2a \leq 3b$ , $1 \leq a$ y $d = 2$ , entonces $d + a \leq 5b$

```

-----
-- Demostrar que si a, b y c son números reales tales que
--   2 * a ≤ 3 * b
--   1 ≤ a
--   c = 2
-- entonces
--   c + a ≤ 5 * b
--
-----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de desigualdades
--   c + a = 2 + a      [por la hipótesis 3 (c = 2)]
--           ≤ 2·a + a   [por la hipótesis 2 (1 ≤ a)]
--           = 3·a
--           ≤ 9/2·b     [por la hipótesis 1 (2·a ≤ 3·b)]
--           ≤ 5·b

-- Demostraciones con Lean4
-- =====

```

```

import Mathlib.Data.Real.Basic

variable (a b c : ℝ)

-- 1ª demostración
example
  (h1 : 2 * a ≤ 3 * b)
  (h2 : 1 ≤ a)
  (h3 : c = 2)
  : c + a ≤ 5 * b :=
calc
  c + a = 2 + a      := by rw [h3]
  _ ≤ 2 * a + a := by linarith only [h2]
  _ = 3 * a      := by linarith only []
  _ ≤ 9/2 * b    := by linarith only [h1]
  _ ≤ 5 * b      := by linarith

-- 2ª demostración
example
  (h1 : 2 * a ≤ 3 * b)
  (h2 : 1 ≤ a)
  (h3 : c = 2)
  : c + a ≤ 5 * b :=
by linarith

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

### 6.3. En $\mathbb{R}$ , si $1 \leq a$ y $b \leq d$ , entonces $2 + a + e^b \leq 3a + e^d$

```

-----
-- Sean a, b, y d números reales. Demostrar que si
--   1 ≤ a
--   b ≤ d
-- entonces
--   2 + a + exp b ≤ 3 * a + exp d
-----

-- Demostración en lenguaje natural
-- =====

-- De la primera hipótesis (1 ≤ a), multiplicando por 2, se obtiene

```

```

--       $2 \leq 2a$ 
-- y, sumando a ambos lados, se tiene
--       $2 + a \leq 3a$  (1)
-- De la hipótesis 2 ( $b \leq d$ ) y de la monotonía de la función exponencial
-- se tiene
--       $e^b \leq e^d$  (2)
-- Finalmente, de (1) y (2) se tiene
--       $2 + a + e^b \leq 3a + e^d$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Analysis.SpecialFunctions.Log.Basic

open Real

variable (a b d : ℝ)

-- 1ª demostración
example
  (h1 : 1 ≤ a)
  (h2 : b ≤ d)
  : 2 + a + exp b ≤ 3 * a + exp d :=
by
  have h3 : 2 + a ≤ 3 * a := calc
    2 + a = 2 * 1 + a := by linarith only []
    _ ≤ 2 * a + a := by linarith only [h1]
    _ ≤ 3 * a := by linarith only []
  have h4 : exp b ≤ exp d := by
    linarith only [exp_le_exp.mpr h2]
  show 2 + a + exp b ≤ 3 * a + exp d
  exact add_le_add h3 h4

-- 2ª demostración
example
  (h1 : 1 ≤ a)
  (h2 : b ≤ d)
  : 2 + a + exp b ≤ 3 * a + exp d :=
calc
  2 + a + exp b
    ≤ 3 * a + exp b := by linarith only [h1]
  _ ≤ 3 * a + exp d := by linarith only [exp_le_exp.mpr h2]

-- 3ª demostración
example

```

```

(h1 : 1 ≤ a)
(h2 : b ≤ d)
: 2 + a + exp b ≤ 3 * a + exp d :=
by linarith [exp_le_exp.mpr h2]

-- Lemas usados
-- =====

-- variable (c : ℝ)
-- #check (add_le_add : a ≤ b → c ≤ d → a + c ≤ b + d)
-- #check (exp_le_exp : exp a ≤ exp b ↔ a ≤ b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#).

## 6.4. En $\mathbb{R}$ , si $a \leq b$ y $c < d$ , entonces $a + e^c + f \leq b + e^d + f$

```

-----
-- Demostrar que si a, b, c, d y f son números reales tales que
--   a ≤ b
--   c < d
-- entonces
--   a + e^c + f ≤ b + e^d + f
-----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Aplicando a la hipótesis 3 (c < d) la monotonía de la exponencial, se
-- tiene
--   e^c < e^d
-- que, junto a la hipótesis 1 (a ≤ b) y la monotonía de la suma da
--   a + e^c < b + e^d
-- y, de nuevo por la monotonía de la suma, se tiene
--   a + e^c + f < b + e^d + f

-- 2ª demostración en LN
-- =====

```

```

-- Tenemos que demostrar que
--    $(a + e^c) + f < (b + e^d) + f$ 
-- que, por la monotonía de la suma, se reduce a las siguientes dos
-- desigualdades:
--    $a + e^c < b + e^d$  (1)
--    $f \leq f$  (2)
--
-- La (1), de nuevo por la monotonía de la suma, se reduce a las
-- siguientes dos:
--    $a \leq b$  (1.1)
--    $e^c < e^d$  (1.2)
--
-- La (1.1) se tiene por la hipótesis 1.
--
-- La (1.2) se tiene aplicando la monotonía de la exponencial a la
-- hipótesis 2.
--
-- La (2) se tiene por la propiedad reflexiva.

-- Demostraciones con Lean4
-- =====

import Mathlib.Analysis.SpecialFunctions.Log.Basic
open Real
variable (a b c d f : ℝ)

-- 1ª demostración
example
  (h1 : a ≤ b)
  (h2 : c < d)
  : a + exp c + f < b + exp d + f :=
by
  have h3 : exp c < exp d :=
    exp_lt_exp.mpr h2
  have h4 : a + exp c < b + exp d :=
    add_lt_add_of_le_of_lt h1 h3
  show a + exp c + f < b + exp d + f
  exact add_lt_add_right h4 f

-- 2ª demostración
example
  (h1 : a ≤ b)
  (h2 : c < d)
  : a + exp c + f < b + exp d + f :=
by

```



```

apply add_lt_add_of_lt_of_le
{ apply add_lt_add_of_le_of_lt
  { exact h1 }
  { apply exp_lt_exp.mpr
    exact h2 } }
{ apply le_refl }

-- 3ª demostración
example
  (h1 : a ≤ b)
  (h2 : c < d)
  : a + exp c + f < b + exp d + f :=
by
  apply add_lt_add_of_lt_of_le
  . apply add_lt_add_of_le_of_lt h1
    apply exp_lt_exp.mpr h2
  rfl

-- Lemas usados
-- =====

-- #check (add_lt_add_of_le_of_lt : a ≤ b → c < d → a + c < b + d)
-- #check (add_lt_add_of_lt_of_le : a < b → c ≤ d → a + c < b + d)
-- #check (add_lt_add_right : b < c → ∀ (a : ℝ), b + a < c + a)
-- #check (exp_lt_exp : exp a < exp b ↔ a < b)
-- #check (le_refl a : a ≤ a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.5. En $\mathbb{R}$ , si $d \leq f$ , entonces $c + e^{(a + d)} \leq c + e^{(a + f)}$

```

-----
-- Demostrar que si a, c, d y f son números reales tales que
--   d ≤ f
-- entonces
--   c + exp (a + d) ≤ c + exp (a + f)
-----

-- Demostraciones en lenguaje natural (LN)
-- =====

```

```

-- 1ª demostración en LN
-- =====

-- De la hipótesis, por la monotonía de la suma, se tiene
--    $a + d \leq a + f$ 
-- que, por la monotonía de la exponencial, da
--    $\exp(a + d) \leq \exp(a + f)$ 
-- y, por la monotonía de la suma, se tiene
--    $c + \exp(a + d) \leq c + \exp(a + f)$ 

-- 2ª demostración en LN
-- =====

-- Tenemos que demostrar que
--    $c + \exp(a + d) \leq c + \exp(a + f)$ 
-- Por la monotonía de la suma, se reduce a
--    $\exp(a + d) \leq \exp(a + f)$ 
-- que, por la monotonía de la exponencial, se reduce a
--    $a + d \leq a + f$ 
-- que, por la monotonía de la suma, se reduce a
--    $d \leq f$ 
-- que es la hipótesis.

-- Demostraciones con Lean4
-- =====

import Mathlib.Analysis.SpecialFunctions.Log.Basic
open Real
variable (a c d f : ℝ)

-- 1ª demostración
example
  (h : d ≤ f)
  : c + exp (a + d) ≤ c + exp (a + f) :=
by
  have h1 : a + d ≤ a + f :=
    add_le_add_left h a
  have h2 : exp (a + d) ≤ exp (a + f) :=
    exp_le_exp.mpr h1
  show c + exp (a + d) ≤ c + exp (a + f)
  exact add_le_add_left h2 c

-- 2ª demostración
example
  (h : d ≤ f)

```

```

: c + exp (a + d) ≤ c + exp (a + f) :=
by
  apply add_le_add_left
  apply exp_le_exp.mpr
  apply add_le_add_left
  exact h

-- Lemas usados
-- =====

-- variable (b : ℝ)
-- #check (add_le_add_left : b ≤ c → ∀ (a : ℝ), a + b ≤ a + c)
-- #check (exp_le_exp : exp a ≤ exp b ↔ a ≤ b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.6. En $\mathbb{R}$ , si $a \leq b$ , entonces $\log(1+e^a) \leq \log(1+e^b)$

```

-- -----
-- Demostrar que si a y b son números reales tales que
--   a ≤ b
-- entonces
--   log(1+e^a) ≤ log(1+e^b)
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la monotonía del logaritmo, basta demostrar que
--   0 < 1 + e^a                (1)
--   1 + e^a ≤ 1 + e^b          (2)
--
-- La (1), por la suma de positivos, se reduce a
--   0 < 1                (1.1)
--   0 < e^a              (1.2)
-- La (1.1) es una propiedad de los números naturales y la (1.2) de la
-- función exponencial.
--
-- La (2), por la monotonía de la suma, se reduce a
--   e^a ≤ e^b
-- que, por la monotonía de la exponencial, se reduce a
--   a ≤ b
-- que es la hipótesis.

```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Analysis.SpecialFunctions.Log.Basic
open Real
variable (a b : ℝ)

-- 1ª demostración
example
  (h : a ≤ b)
  : log (1 + exp a) ≤ log (1 + exp b) :=
by
  have h1 : (0 : ℝ) < 1 :=
    zero_lt_one
  have h2 : 0 < exp a :=
    exp_pos a
  have h3 : 0 < 1 + exp a :=
    add_pos h1 h2
  have h4 : exp a ≤ exp b :=
    exp_le_exp.mpr h
  have h5 : 1 + exp a ≤ 1 + exp b :=
    add_le_add_left h4 1
  show log (1 + exp a) ≤ log (1 + exp b)
  exact log_le_log' h3 h5

-- 2ª demostración
example
  (h : a ≤ b)
  : log (1 + exp a) ≤ log (1 + exp b) :=
by
  apply log_le_log'
  { apply add_pos
    { exact zero_lt_one }
    { exact exp_pos a }}
  { apply add_le_add_left
    exact exp_le_exp.mpr h }

-- Lemas usados
-- =====

-- variable (c : ℝ)
-- #check (add_le_add_left : b ≤ c → ∀ (a : ℝ), a + b ≤ a + c)
-- #check (add_pos : 0 < a → 0 < b → 0 < a + b)
-- #check (exp_le_exp : exp a ≤ exp b ↔ a ≤ b)

```

```
-- #check (exp_pos a : 0 < exp a)
-- #check (log_le_log' : 0 < a → a ≤ b → log a ≤ log b)
-- #check (zero_lt_one : 0 < 1)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.7. En $\mathbb{R}$ , si $a \leq b$ , entonces $c - e^b \leq c - e^a$

```
-- -----
-- Sean a, b y c números reales. Demostrar que si
--   a ≤ b
-- entonces
--   c - e^b ≤ c - e^a
-- -----

-- Demostración en lenguaje natural
-- =====

-- Aplicando la monotonía de la exponencial a la hipótesis, se tiene
--   e^a ≤ e^b
-- y, restando de c, se invierte la desigualdad
--   c - e^b ≤ c - e^a

-- Demostraciones con Lean4
-- =====

import Mathlib.Analysis.SpecialFunctions.Log.Basic

open Real

variable (a b c : ℝ)

-- 1ª demostración
example
  (h : a ≤ b)
  : c - exp b ≤ c - exp a :=
by
  have h1 : exp a ≤ exp b :=
    exp_le_exp.mpr h
  show c - exp b ≤ c - exp a
  exact sub_le_sub_left h1 c

-- 2ª demostración
```

```

example
  (h : a ≤ b)
  : c - exp b ≤ c - exp a :=
by
  apply sub_le_sub_left _ c
  apply exp_le_exp.mpr h

-- 3ª demostración
example
  (h : a ≤ b)
  : c - exp b ≤ c - exp a :=
sub_le_sub_left (exp_le_exp.mpr h) c

-- 4ª demostración
example
  (h : a ≤ b)
  : c - exp b ≤ c - exp a :=
by linarith [exp_le_exp.mpr h]

-- Lemas usados
-- =====

-- #check (exp_le_exp : exp a ≤ exp b ↔ a ≤ b)
-- #check (sub_le_sub_left : a ≤ b → ∀ (c : ℝ), c - b ≤ c - a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.8. En $\mathbb{R}$ , $2ab \leq a^2 + b^2$

```

-- Sean a y b números reales. Demostrar que
--   2ab ≤ a² + b²
--
-- Demostración en lenguaje natural
-- =====
--
-- Puesto que los cuadrados son positivos, se tiene
--   (a - b)² ≥ 0
-- Desarrollando el cuadrado, se obtiene
--   a² - 2ab + b² ≥ 0
-- Sumando 2ab a ambos lados, queda
--   a² + b² ≥ 2ab

```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (a b : ℝ)

-- 1ª demostración
example : 2*a*b ≤ a^2 + b^2 :=
by
  have h1 : 0 ≤ (a - b)^2      := sq_nonneg (a - b)
  have h2 : 0 ≤ a^2 - 2*a*b + b^2 := by linarith only [h1]
  show 2*a*b ≤ a^2 + b^2
  linarith

-- 2ª demostración
example : 2*a*b ≤ a^2 + b^2 :=
by
  have h : 0 ≤ a^2 - 2*a*b + b^2
  { calc a^2 - 2*a*b + b^2
        = (a - b)^2                := (sub_sq a b).symm
        ≥ 0                        := sq_nonneg (a - b) }
  calc 2*a*b
    = 2*a*b + 0                    := (add_zero (2*a*b)).symm
    ≤ 2*a*b + (a^2 - 2*a*b + b^2) := add_le_add (le_refl _) h
    = a^2 + b^2                    := by ring

-- 3ª demostración
example : 2*a*b ≤ a^2 + b^2 :=
by
  have h : 0 ≤ a^2 - 2*a*b + b^2
  { calc a^2 - 2*a*b + b^2
        = (a - b)^2                := (sub_sq a b).symm
        ≥ 0                        := sq_nonneg (a - b) }
  linarith only [h]

-- 4ª demostración
example : 2*a*b ≤ a^2 + b^2 :=
-- by apply?
two_mul_le_add_sq a b

-- Lemas usados
-- =====

```

```
-- variable (c : ℝ)
-- #check (add_le_add : a ≤ b → c ≤ d → a + c ≤ b + d)
-- #check (add_zero a : a + 0 = a)
-- #check (sq_nonneg a : 0 ≤ a ^ 2)
-- #check (sub_sq a b : (a - b) ^ 2 = a ^ 2 - 2 * a * b + b ^ 2)
-- #check (two_mul_le_add_sq a b : 2 * a * b ≤ a ^ 2 + b ^ 2)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.9. En $\mathbb{R}$ , $|ab| \leq (a^2 + b^2)/2$

```
-- -----
-- Sean a y b números reales. Demostrar que
--   |a*b| ≤ (a^2 + b^2) / 2
-- -----

-- Demostración en lenguaje natural
-- =====

-- Para demostrar
--   |ab| ≤ (a^2 + b^2) / 2
-- basta demostrar estas dos desigualdades
--   ab ≤ (a^2 + b^2) / 2                                (1)
--   -(ab) ≤ (a^2 + b^2) / 2                             (2)
--
-- Para demostrar (1) basta demostrar que
--   2ab ≤ a^2 + b^2
-- que se prueba como sigue. En primer lugar, como los cuadrados son no
-- negativos, se tiene
--   (a - b)^2 ≥ 0
-- Desarrollando el cuadrado,
--   a^2 - 2ab + b^2 ≥ 0
-- Sumando 2ab,
--   a^2 + b^2 ≥ 2ab
--
-- Para demostrar (2) basta demostrar que
--   -2ab ≤ a^2 + b^2
-- que se prueba como sigue. En primer lugar, como los cuadrados son no
-- negativos, se tiene
--   (a + b)^2 ≥ 0
-- Desarrollando el cuadrado,
--   a^2 + 2ab + b^2 ≥ 0
-- Restando 2ab,
```



```

--  $a^2 + b^2 \geq -2ab$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (a b : ℝ)

-- Lemas auxiliares
-- =====

lemma aux1 : a * b * 2 ≤ a ^ 2 + b ^ 2 := by
  have h : 0 ≤ a ^ 2 - 2 * a * b + b ^ 2
  calc
    a ^ 2 - 2 * a * b + b ^ 2
      = (a - b) ^ 2                := by ring
    _ ≥ 0                          := pow_two_nonneg (a - b)
  linarith only [h]

lemma aux2 : -(a * b) * 2 ≤ a ^ 2 + b ^ 2 := by
  have h : 0 ≤ a ^ 2 + 2 * a * b + b ^ 2
  calc
    a ^ 2 + 2 * a * b + b ^ 2
      = (a + b) ^ 2                := by ring
    _ ≥ 0                          := pow_two_nonneg (a + b)
  linarith only [h]

-- 1ª demostración
-- =====

example : |a * b| ≤ (a ^ 2 + b ^ 2) / 2 := by
  have h : (0 : ℝ) < 2 := by norm_num
  apply abs_le'.mpr
  constructor
  { have h1 : a * b * 2 ≤ a ^ 2 + b ^ 2 := aux1 a b
    show a * b ≤ (a ^ 2 + b ^ 2) / 2
    exact (le_div_iff h).mpr h1 }
  { have h2 : -(a * b) * 2 ≤ a ^ 2 + b ^ 2 := aux2 a b
    show -(a * b) ≤ (a ^ 2 + b ^ 2) / 2
    exact (le_div_iff h).mpr h2 }

-- 2ª demostración
-- =====

```

```

example : |a * b| ≤ (a ^ 2 + b ^ 2) / 2 := by
  have h : (0 : ℝ) < 2 := by norm_num
  apply abs_le'.mpr
  constructor
  { exact (le_div_iff h).mpr (aux1 a b) }
  { exact (le_div_iff h).mpr (aux2 a b) }

-- 3ª demostración
-- =====

example : |a * b| ≤ (a ^ 2 + b ^ 2) / 2 := by
  have h : (0 : ℝ) < 2 := by norm_num
  apply abs_le'.mpr
  constructor
  { rw [le_div_iff h]
    apply aux1 }
  { rw [le_div_iff h]
    apply aux2 }

-- Lemas usados
-- =====

-- variable (c : ℝ)
-- #check (abs_le' : |a| ≤ b ↔ a ≤ b ∧ -a ≤ b)
-- #check (le_div_iff : 0 < c → (a ≤ b / c ↔ a * c ≤ b))
-- #check (pow_two_nonneg a : 0 ≤ a ^ 2)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.10. En $\mathbb{R}$ , $\min(a,b) = \min(b,a)$

```

-- -----
-- Sean a y b números reales. Demostrar que
--   min a b = min b a
-- -----

-- Demostración en lenguaje natural
-- =====

-- Es consecuencia de la siguiente propiedad
--   min(a, b) ≤ min(b, a) (1)
-- En efecto, intercambiando las variables en (1) se obtiene
--   min(b, a) ≤ min(a, b) (2)

```

```

-- Finalmente de (1) y (2) se obtiene
--   min(b, a) = min(a, b)
--
-- Para demostrar (1), se observa que
--   min(a, b) ≤ b
--   min(a, b) ≤ a
-- y, por tanto,
--   min(a, b) = min(b, a)

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (a b : ℝ)

-- Lema auxiliar
-- =====

-- 1ª demostración del lema auxiliar
-- =====

example : min a b ≤ min b a :=
by
  have h1 : min a b ≤ b := min_le_right a b
  have h2 : min a b ≤ a := min_le_left a b
  show min a b ≤ min b a
  exact le_min h1 h2

-- 2ª demostración del lema auxiliar
-- =====

example : min a b ≤ min b a :=
by
  apply le_min
  { apply min_le_right }
  { apply min_le_left }

-- 3ª demostración del lema auxiliar
-- =====

lemma aux : min a b ≤ min b a :=
by exact le_min (min_le_right a b) (min_le_left a b)

-- 1ª demostración

```

```

-- =====

example : min a b = min b a :=
by
  apply le_antisymm
  { exact aux a b }
  { exact aux b a }

-- 2ª demostración
-- =====

example : min a b = min b a :=
le_antisymm (aux a b) (aux b a)

-- 3ª demostración
-- =====

example : min a b = min b a :=
min_comm a b

-- Lemas usados
-- =====

-- variable (c : ℝ)
-- #check (le_antisymm : a ≤ b → b ≤ a → a = b)
-- #check (le_min : c ≤ a → c ≤ b → c ≤ min a b)
-- #check (min_comm a b : min a b = min b a)
-- #check (min_le_left a b : min a b ≤ a)
-- #check (min_le_right a b : min a b ≤ b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.11. En $\mathbb{R}$ , $\max(a,b) = \max(b,a)$

```

-----
-- Sean a y b números reales. Demostrar que
--   max a b = max b a
-----

-- Demostración en lenguaje natural
-- =====

-- Es consecuencia de la siguiente propiedad

```

```

--       $\max(a, b) \leq \max(b, a)$  (1)
-- En efecto, intercambiando las variables en (1) se obtiene
--       $\max(b, a) \leq \max(a, b)$  (2)
-- Finalmente de (1) y (2) se obtiene
--       $\max(b, a) = \max(a, b)$ 
--
-- Para demostrar (1), se observa que
--       $a \leq \max(b, a)$ 
--       $b \leq \max(b, a)$ 
-- y, por tanto,
--       $\max(a, b) \leq \max(b, a)$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (a b : ℝ)

-- Lema auxiliar
-- =====

-- 1ª demostración del lema auxiliar
-- =====

example : max a b ≤ max b a :=
by
  have h1 : a ≤ max b a := le_max_right b a
  have h2 : b ≤ max b a := le_max_left b a
  show max a b ≤ max b a
  exact max_le h1 h2

-- 2ª demostración del lema auxiliar
-- =====

example : max a b ≤ max b a :=
by
  apply max_le
  { apply le_max_right }
  { apply le_max_left }

-- 3ª demostración del lema auxiliar
-- =====

lemma aux : max a b ≤ max b a :=

```

```

by exact max_le (le_max_right b a) (le_max_left b a)

-- 1ª demostración
-- =====

example : max a b = max b a :=
by
  apply le_antisymm
  { exact aux a b }
  { exact aux b a }

-- 2ª demostración
-- =====

example : max a b = max b a :=
le_antisymm (aux a b) (aux b a)

-- 3ª demostración
-- =====

example : max a b = max b a :=
max_comm a b

-- Lemas usados
-- =====

-- variable (c : ℝ)
-- #check (le_antisymm : a ≤ b → b ≤ a → a = b)
-- #check (le_max_left a b : a ≤ max a b)
-- #check (le_max_right a b : b ≤ max a b)
-- #check (max_comm a b : max a b = max b a)
-- #check (max_le : a ≤ c → b ≤ c → max a b ≤ c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.12. En $\mathbb{R}$ , $\min(\min(a,b),c) = \min(a,\min(b,c))$

```

-----
-- Sean a, b y c números reales. Demostrar que
--   min (min a b) c = min a (min b c)
-----

-- Demostración en lenguaje natural

```

```

-- =====

-- Por la propiedad antisimétrica, la igualdad es consecuencia de las
-- siguientes desigualdades
--    $\min(\min(a, b), c) \leq \min(a, \min(b, c))$  (1)
--    $\min(a, \min(b, c)) \leq \min(\min(a, b), c)$  (2)
--
-- La (1) es consecuencia de las siguientes desigualdades
--    $\min(\min(a, b), c) \leq a$  (1a)
--    $\min(\min(a, b), c) \leq b$  (1b)
--    $\min(\min(a, b), c) \leq c$  (1c)
-- En efecto, de (1b) y (1c) se obtiene
--    $\min(\min(a, b), c) \leq \min(b, c)$ 
-- que, junto con (1a) da (1).
--
-- La (2) es consecuencia de las siguientes desigualdades
--    $\min(a, \min(b, c)) \leq a$  (2a)
--    $\min(a, \min(b, c)) \leq b$  (2b)
--    $\min(a, \min(b, c)) \leq c$  (2c)
-- En efecto, de (2a) y (2b) se obtiene
--    $\min(a, \min(b, c)) \leq \min(a, b)$ 
-- que, junto con (2c) da (2).
--
-- La demostración de (1a) es
--    $\min(\min(a, b), c) \leq \min(a, b) \leq a$ 
-- La demostración de (1b) es
--    $\min(\min(a, b), c) \leq \min(a, b) \leq b$ 
-- La demostración de (2b) es
--    $\min(a, \min(b, c)) \leq \min(b, c) \leq b$ 
-- La demostración de (2c) es
--    $\min(a, \min(b, c)) \leq \min(b, c) \leq c$ 
-- La (1c) y (2a) son inmediatas.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable {a b c : ℝ}

-- Lemas auxiliares
-- =====

lemma aux1a : min (min a b) c ≤ a :=
calc min (min a b) c

```

```

    ≤ min a b := by exact min_le_left (min a b) c
  _ ≤ a      := min_le_left a b

lemma aux1b : min (min a b) c ≤ b :=
calc min (min a b) c
    ≤ min a b := by exact min_le_left (min a b) c
  _ ≤ b      := min_le_right a b

lemma aux1c : min (min a b) c ≤ c :=
by exact min_le_right (min a b) c

-- 1ª demostración del lema aux1
lemma aux1 : min (min a b) c ≤ min a (min b c) :=
by
  apply le_min
  { show min (min a b) c ≤ a
    exact aux1a }
  { show min (min a b) c ≤ min b c
    apply le_min
    { show min (min a b) c ≤ b
      exact aux1b }
    { show min (min a b) c ≤ c
      exact aux1c }}

-- 2ª demostración del lema aux1
lemma aux1' : min (min a b) c ≤ min a (min b c) :=
le_min aux1a (le_min aux1b aux1c)

lemma aux2a : min a (min b c) ≤ a :=
by exact min_le_left a (min b c)

lemma aux2b : min a (min b c) ≤ b :=
calc min a (min b c)
    ≤ min b c      := by exact min_le_right a (min b c)
  _ ≤ b            := min_le_left b c

lemma aux2c : min a (min b c) ≤ c :=
calc min a (min b c)
    ≤ min b c      := by exact min_le_right a (min b c)
  _ ≤ c            := min_le_right b c

-- 1ª demostración del lema aux2
lemma aux2 : min a (min b c) ≤ min (min a b) c :=
by
  apply le_min

```



```

{ show min a (min b c) ≤ min a b
  apply le_min
  { show min a (min b c) ≤ a
    exact aux2a }
  { show min a (min b c) ≤ b
    exact aux2b }}
{ show min a (min b c) ≤ c
  exact aux2c }

-- 2ª demostración del lema aux2
lemma aux2' : min a (min b c) ≤ min (min a b) c :=
le_min (le_min aux2a aux2b) aux2c

-- 1ª demostración
-- =====

example :
  min (min a b) c = min a (min b c) :=
by
  apply le_antisymm
  { show min (min a b) c ≤ min a (min b c)
    exact aux1 }
  { show min a (min b c) ≤ min (min a b) c
    exact aux2 }

-- 2ª demostración
-- =====

example : min (min a b) c = min a (min b c) :=
by
  apply le_antisymm
  { exact aux1 }
  { exact aux2 }

-- 3ª demostración
-- =====

example : min (min a b) c = min a (min b c) :=
le_antisymm aux1 aux2

-- 4ª demostración
-- =====

example : min (min a b) c = min a (min b c) :=
min_assoc a b c

```

```
-- Lemas usados
-- =====

-- #check (le_antisymm : a ≤ b → b ≤ a → a = b)
-- #check (le_min : c ≤ a → c ≤ b → c ≤ min a b)
-- #check (min_assoc a b c : min (min a b) c = min a (min b c))
-- #check (min_le_left a b : min a b ≤ a)
-- #check (min_le_right a b : min a b ≤ b)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

### 6.13. En $\mathbb{R}$ , $\min(a,b)+c = \min(a+c,b+c)$

```
-- -----
-- Sean a, b y c números reales. Demostrar que
--   min a b + c = min (a + c) (b + c)
-- -----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Aplicando la propiedad antisimétrica a las siguientes desigualdades
--   min(a, b) + c ≤ min(a + c, b + c)                                     (1)
--   min(a + c, b + c) ≤ min(a, b) + c                                   (2)
--
-- Para demostrar (1) basta demostrar que se verifican las siguientes
-- desigualdades
--   min(a, b) + c ≤ a + c                                               (1a)
--   min(a, b) + c ≤ b + c                                               (1b)
-- que se tienen porque se verifican las siguientes desigualdades
--   min(a, b) ≤ a
--   min(a, b) ≤ b
--
-- Para demostrar (2) basta demostrar que se verifica
--   min(a + c, b + c) - c ≤ min(a, b)
-- que se demuestra usando (1); en efecto,
--   min(a + c, b + c) - c ≤ min(a + c - c, b + c - c)   [por (1)]
--                               = min(a, b)
```

```

-- 2ª demostración en LN
-- =====

-- Por casos según  $a \leq b$ .
--
-- 1º caso: Supongamos que  $a \leq b$ . Entonces,
--    $\min(a, b) + c = a + c$ 
--                    $= \min(a + c, b + c)$ 
--
-- 2º caso: Supongamos que  $a \not\leq b$ . Entonces,
--    $\min(a, b) + c = b + c$ 
--                    $= \min(a + c, b + c)$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable {a b c : ℝ}

-- En las demostraciones se usarán los siguientes lemas auxiliares
--   aux1 :  $\min a b + c \leq \min (a + c) (b + c)$ 
--   aux2 :  $\min (a + c) (b + c) \leq \min a b + c$ 
--   cuyas demostraciones se exponen a continuación.

-- 1ª demostración de aux1
lemma aux1 :
  min a b + c ≤ min (a + c) (b + c) :=
by
  have h1 : min a b ≤ a :=
    min_le_left a b
  have h2 : min a b + c ≤ a + c :=
    add_le_add_right h1 c
  have h3 : min a b ≤ b :=
    min_le_right a b
  have h4 : min a b + c ≤ b + c :=
    add_le_add_right h3 c
  show min a b + c ≤ min (a + c) (b + c)
  exact le_min h2 h4

-- 2ª demostración de aux1
example :
  min a b + c ≤ min (a + c) (b + c) :=
by
  apply le_min

```

```

{ apply add_le_add_right
  exact min_le_left a b }
{ apply add_le_add_right
  exact min_le_right a b }

-- 3ª demostración de aux1
example :
  min a b + c ≤ min (a + c) (b + c) :=
le_min (add_le_add_right (min_le_left a b) c)
      (add_le_add_right (min_le_right a b) c)

-- 1ª demostración de aux2
lemma aux2 :
  min (a + c) (b + c) ≤ min a b + c :=
by
  have h1 : min (a + c) (b + c) + -c ≤ min a b
  { calc min (a + c) (b + c) + -c
        ≤ min (a + c + -c) (b + c + -c) := aux1
        _ = min a b                      := by ring_nf }
  show min (a + c) (b + c) ≤ min a b + c
  exact add_neg_le_iff_le_add.mp h1

-- 1ª demostración del ejercicio
example :
  min a b + c = min (a + c) (b + c) :=
by
  have h1 : min a b + c ≤ min (a + c) (b + c) := aux1
  have h2 : min (a + c) (b + c) ≤ min a b + c := aux2
  show min a b + c = min (a + c) (b + c)
  exact le_antisymm h1 h2

-- 2ª demostración del ejercicio
example :
  min a b + c = min (a + c) (b + c) :=
by
  apply le_antisymm
  { show min a b + c ≤ min (a + c) (b + c)
    exact aux1 }
  { show min (a + c) (b + c) ≤ min a b + c
    exact aux2 }

-- 3ª demostración del ejercicio
example :
  min a b + c = min (a + c) (b + c) :=
by

```

```

apply le_antisymm
{ exact aux1 }
{ exact aux2 }

-- 4ª demostración del ejercicio
example :
  min a b + c = min (a + c) (b + c) :=
le_antisymm aux1 aux2

-- 5ª demostración del ejercicio
example : min a b + c = min (a + c) (b + c) :=
by
  by_cases h : a ≤ b
  { have h1 : a + c ≤ b + c := add_le_add_right h c
    calc min a b + c = a + c := by simp [min_eq_left h]
      _ = min (a + c) (b + c) := by simp [min_eq_left h1]}
  { have h2 : b ≤ a := le_of_not_le h
    have h3 : b + c ≤ a + c := add_le_add_right h2 c
    calc min a b + c = b + c := by simp [min_eq_right h2]
      _ = min (a + c) (b + c) := by simp [min_eq_right h3]}

-- 6ª demostración del ejercicio
example : min a b + c = min (a + c) (b + c) :=
(min_add_add_right a b c).symm

-- Lemas usados
-- =====

-- #check (add_le_add_right : b ≤ c → ∀ (a : ℝ), b + a ≤ c + a)
-- #check (add_neg_le_iff_le_add : a - b ≤ c ↔ a ≤ c + b)
-- #check (le_antisymm : a ≤ b → b ≤ a → a = b)
-- #check (le_min : c ≤ a → c ≤ b → c ≤ min a b)
-- #check (min_add_add_right a b c : min (a + c) (b + c) = min a b + c)
-- #check (min_eq_left : a ≤ b → min a b = a)
-- #check (min_eq_right : b ≤ a → min a b = b)
-- #check (min_le_left a b : min a b ≤ a)
-- #check (min_le_right a b : min a b ≤ b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.14. En $\mathbb{R}$ , $|a| - |b| \leq |a - b|$

```

-- Sean a y b números reales. Demostrar que
--    $|a| - |b| \leq |a - b|$ 
--
-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Por la siguiente cadena de desigualdades
--    $|a| - |b| = |a - b + b| - |b|$ 
--              $\leq (|a - b| + |b|) - |b|$  [por la desigualdad triangular]
--              $= |a - b|$ 

-- 2ª demostración en LN
-- =====

-- Por la desigualdad triangular
--    $|a - b + b| \leq |a - b| + |b|$ 
-- simplificando en la izquierda
--    $|a| \leq |a - b| + |b|$ 
-- y, pasando  $|b|$  a la izquierda
--    $|a| - |b| \leq |a - b|$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (a b : ℝ)

-- 1ª demostración (basada en la 1ª en LN)
example :  $|a| - |b| \leq |a - b|$  :=
calc |a| - |b|
  = |a - b + b| - |b| :=
    congrArg (fun x => |x| - |b|) (sub_add_cancel a b).symm
_ ≤ (|a - b| + |b|) - |b| :=
    sub_le_sub_right (abs_add (a - b) b) (|b|)
_ = |a - b| :=
    add_sub_cancel (|a - b|) (|b|)

```

```

-- 2ª demostración (basada en la 1ª en LN)
example : |a| - |b| ≤ |a - b| :=
calc |a| - |b|
    = |a - b + b| - |b| := by
      rw [sub_add_cancel]
    _ ≤ (|a - b| + |b|) - |b| := by
      apply sub_le_sub_right
      apply abs_add
    _ = |a - b| := by
      rw [add_sub_cancel]

-- 3ª demostración (basada en la 2ª en LN)
example : |a| - |b| ≤ |a - b| :=
by
  have h1 : |a - b + b| ≤ |a - b| + |b| := abs_add (a - b) b
  rw [sub_add_cancel] at h1
  exact abs_sub_abs_le_abs_sub a b

-- 4ª demostración
example : |a| - |b| ≤ |a - b| :=
abs_sub_abs_le_abs_sub a b

-- Lemas usados
-- =====

-- #check (abs_add a b : |a + b| ≤ |a| + |b|)
-- #check (abs_sub_abs_le_abs_sub a b : |a| - |b| ≤ |a - b|)
-- #check (add_sub_cancel a b : a + b - b = a)
-- #check (sub_add_cancel a b : a - b + b = a)
-- #check (sub_le_sub_right : a ≤ b → ∀ (c : ℝ), a - c ≤ b - c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

**6.15. En  $\mathbb{R}$ ,  $\{0 < \varepsilon, \varepsilon \leq 1, |x| < \varepsilon, |y| < \varepsilon\} \vdash |xy| < \varepsilon$**

```

-----
-- Demostrar que para todos los números reales x, y, ε si
--   0 < ε
--   ε ≤ 1
--   |x| < ε
--   |y| < ε

```

```

-- entonces
--    $|x * y| < \varepsilon$ 
--   -----

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas
--   abs_mul      :  $|a * b| = |a| * |b|$ 
--   zero_mul     :  $0 * a = 0$ 
--   abs_nonneg a  :  $0 \leq |a|$ 
--   lt_of_le_of_ne :  $a \leq b \rightarrow a \neq b \rightarrow a < b$ 
--   ne_comm      :  $a \neq b \leftrightarrow b \neq a$ 
--   mul_lt_mul_left  :  $0 < a \rightarrow (a * b < a * c \leftrightarrow b < c)$ 
--   mul_lt_mul_right :  $0 < a \rightarrow (b * a < c * a \leftrightarrow b < c)$ 
--   mul_le_mul_right :  $0 < a \rightarrow (b * a \leq c * a \leftrightarrow b \leq c)$ 
--   one_mul       :  $1 * a = a$ 
--
-- Sean  $x, y, \varepsilon \in \mathbb{R}$  tales que
--    $0 < \varepsilon$                                      (he1)
--    $\varepsilon \leq 1$                                    (he2)
--    $|x| < \varepsilon$                                      (hx)
--    $|y| < \varepsilon$                                      (hy)
-- y tenemos que demostrar que
--    $|x * y| < \varepsilon$ 
-- Lo haremos distinguiendo caso según  $|x| = 0$ .
--
-- 1º caso. Supongamos que
--    $|x| = 0$                                              (1)
-- Entonces,
--    $|x * y| = |x| * |y|$       [por abs_mul]
--              $= 0 * |y|$       [por h1]
--              $= 0$             [por zero_mul]
--              $< \varepsilon$     [por he1]
--
-- 2º caso. Supongamos que
--    $|x| \neq 0$                                              (2)
-- Entonces, por lt_of_le_of_ne, abs_nonneg y ne_comm, se tiene
--    $0 < x$                                              (3)
-- y, por tanto,
--    $|x * y| = |x| * |y|$       [por abs_mul]
--              $< |x| * \varepsilon$  [por mul_lt_mul_left, (3) y (hy)]
--              $< \varepsilon * \varepsilon$  [por mul_lt_mul_right, (he1) y (hx)]
--              $\leq 1 * \varepsilon$  [por mul_le_mul_right, (he1) y (he2)]
--              $= \varepsilon$       [por one_mul]

```



```

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

-- 1ª demostración
-- =====

example :
  ∀ {x y ε : ℝ}, 0 < ε → ε ≤ 1 → |x| < ε → |y| < ε → |x * y| < ε :=
by
  intros x y ε he1 he2 hx hy
  by_cases h : (|x| = 0)
  . -- h : |x| = 0
    show |x * y| < ε
    calc
      |x * y|
      = |x| * |y| := abs_mul x y
      _ = 0 * |y| := by rw [h]
      _ = 0      := zero_mul (abs y)
      _ < ε      := he1
  . -- h : ¬|x| = 0
    have h1 : 0 < |x| := by
      have h2 : 0 ≤ |x| := abs_nonneg x
      show 0 < |x|
    exact lt_of_le_of_ne h2 (ne_comm.mpr h)
    show |x * y| < ε
    calc |x * y|
      = |x| * |y| := abs_mul x y
      _ < |x| * ε := (mul_lt_mul_left h1).mpr hy
      _ < ε * ε   := (mul_lt_mul_right he1).mpr hx
      _ ≤ 1 * ε   := (mul_le_mul_right he1).mpr he2
      _ = ε       := one_mul ε

-- 2ª demostración
-- =====

example :
  ∀ {x y ε : ℝ}, 0 < ε → ε ≤ 1 → |x| < ε → |y| < ε → |x * y| < ε :=
by
  intros x y ε he1 he2 hx hy
  by_cases (|x| = 0)
  . -- h : |x| = 0
    show |x * y| < ε

```

```

calc
  |x * y| = |x| * |y| := by apply abs_mul
  _ = 0 * |y| := by rw [h]
  _ = 0 := by apply zero_mul
  _ < ε := by apply he1
. -- h : ¬|x| = 0
have h1 : 0 < |x| := by
  have h2 : 0 ≤ |x| := by apply abs_nonneg
  exact lt_of_le_of_ne h2 (ne_comm.mpr h)
show |x * y| < ε
calc
  |x * y| = |x| * |y| := by rw [abs_mul]
  _ < |x| * ε := by apply (mul_lt_mul_left h1).mpr hy
  _ < ε * ε := by apply (mul_lt_mul_right he1).mpr hx
  _ ≤ 1 * ε := by apply (mul_le_mul_right he1).mpr he2
  _ = ε := by rw [one_mul]

-- 3ª demostración
-- =====

example :
  ∀ {x y ε : ℝ}, 0 < ε → ε ≤ 1 → |x| < ε → |y| < ε → |x * y| < ε :=
by
  intros x y ε he1 he2 hx hy
  by_cases (|x| = 0)
  . -- h : |x| = 0
    show |x * y| < ε
    calc |x * y| = |x| * |y| := by simp only [abs_mul]
        _ = 0 * |y| := by simp only [h]
        _ = 0 := by simp only [zero_mul]
        _ < ε := by simp only [he1]
  . -- h : ¬|x| = 0
    have h1 : 0 < |x| := by
      have h2 : 0 ≤ |x| := by simp only [abs_nonneg]
      exact lt_of_le_of_ne h2 (ne_comm.mpr h)
    show |x * y| < ε
    calc
      |x * y| = |x| * |y| := by simp [abs_mul]
      _ < |x| * ε := by simp only [mul_lt_mul_left, h1, hy]
      _ < ε * ε := by simp only [mul_lt_mul_right, he1, hx]
      _ ≤ 1 * ε := by simp only [mul_le_mul_right, he1, he2]
      _ = ε := by simp only [one_mul]

-- Lemas usados
-- =====

```

```
-- variable (a b c : ℝ)
-- #check (abs_mul a b : |a * b| = |a| * |b|)
-- #check (abs_nonneg a : 0 ≤ |a|)
-- #check (lt_of_le_of_ne : a ≤ b → a ≠ b → a < b)
-- #check (mul_le_mul_right : 0 < a → (b * a ≤ c * a ↔ b ≤ c))
-- #check (mul_lt_mul_left : 0 < a → (a * b < a * c ↔ b < c))
-- #check (mul_lt_mul_right : 0 < a → (b * a < c * a ↔ b < c))
-- #check (ne_comm : a ≠ b ↔ b ≠ a)
-- #check (one_mul a : 1 * a = a)
-- #check (zero_mul a : 0 * a = 0)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.16. En $\mathbb{R}$ , $a < b \rightarrow \neg(b < a)$

```
-- -----
-- Demostrar que para todo par de numero reales a y b, si a < b entonces
-- no se tiene que b < a.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por hipótesis a < b y tenemos que demostrar que ¬(b < a). Supongamos
-- que b < a. Entonces, por la propiedad transitiva a < a que es una
-- contradicción con la propiedad irreflexiva.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (a b : ℝ)

-- 1ª demostración
example
  (h : a < b)
  : ¬ b < a :=
by
  intro h1
  -- h1 : b < a
  -- ⊢ False
```

```

have : a < a := lt_trans h h1
apply lt_irrefl a this

-- 2ª demostración
example
  (h : a < b)
  : ¬ b < a :=
by
  intro h1
  -- h1 : b < a
  -- ⊢ False
  exact lt_irrefl a (lt_trans h h1)

-- 3ª demostración
example
  (h : a < b)
  : ¬ b < a :=
fun h1 ↦ lt_irrefl a (lt_trans h h1)

-- 4ª demostración
example
  (h : a < b)
  : ¬ b < a :=
lt_asymm h

-- Lemas usados
-- =====

-- variable (c : ℝ)
-- #check (lt_asymm : a < b → ¬b < a)
-- #check (lt_irrefl a : ¬a < a)
-- #check (lt_trans : a < b → b < c → a < c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.17. Hay algún número real entre 2 y 3

```

-----
-- Demostrar que hay algún número real entre 2 y 3.
-----

-- Demostración en lenguaje natural
-- =====

```

```

-- Puesto que  $2 < 5/2 < 3$ , basta elegir  $5/2$ .

-- Demostracione con Lean4
-- =====

import Mathlib.Data.Real.Basic

-- 1ª demostración
example :  $\exists x : \mathbb{R}, 2 < x \wedge x < 3 :=$ 
by
  have h :  $2 < (5 : \mathbb{R}) / 2 \wedge (5 : \mathbb{R}) / 2 < 3 :=$ 
    by norm_num
  show  $\exists x : \mathbb{R}, 2 < x \wedge x < 3$ 
  exact Exists.intro (5 / 2) h

-- 2ª demostración
example :  $\exists x : \mathbb{R}, 2 < x \wedge x < 3 :=$ 
by
  have h :  $2 < (5 : \mathbb{R}) / 2 \wedge (5 : \mathbb{R}) / 2 < 3 :=$ 
    by norm_num
  show  $\exists x : \mathbb{R}, 2 < x \wedge x < 3$ 
  exact ⟨5 / 2, h⟩

-- 3ª demostración
example :  $\exists x : \mathbb{R}, 2 < x \wedge x < 3 :=$ 
by
  use 5 / 2
  norm_num

-- 4ª demostración
example :  $\exists x : \mathbb{R}, 2 < x \wedge x < 3 :=$ 
(5 / 2, by norm_num)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.18. Si $(\forall \varepsilon > 0)[x \leq \varepsilon]$ , entonces $x \leq 0$

```

-- -----
-- Sea  $x$  un número real tal que para todo número positivo  $\varepsilon$ ,  $x \leq \varepsilon$ 
-- Demostrar que  $x \leq 0$ .
-- -----

```

```

-- Demostración en lenguaje natural
-- =====

-- Basta demostrar que  $x \neq 0$ . Para ello, supongamos que  $x > 0$  y vamos a
-- demostrar que
--  $\neg(\forall \varepsilon)[\varepsilon > 0 \rightarrow x \leq \varepsilon]$  (1)
-- que es una contradicción con la hipótesis. Interiorizando la
-- negación, (1) es equivalente a
--  $(\exists \varepsilon)[\varepsilon > 0 \wedge \varepsilon < x]$  (2)
-- Para demostrar (2) se puede elegir  $\varepsilon = x/2$  ya que, como  $x > 0$ , se
-- tiene
--  $0 < x/2 < x$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (x : ℝ)

-- 1ª demostración
-- =====

example
  (h :  $\forall \varepsilon > 0, x \leq \varepsilon$ )
  :  $x \leq 0$  :=
by
  apply le_of_not_gt
  --  $\vdash \neg x > 0$ 
  intro hx0
  --  $hx0 : x > 0$ 
  --  $\vdash \text{False}$ 
  apply absurd h
  --  $\vdash \neg \forall (\varepsilon : \mathbb{R}), \varepsilon > 0 \rightarrow x \leq \varepsilon$ 
  push_neg
  --  $\vdash \exists \varepsilon, \varepsilon > 0 \wedge \varepsilon < x$ 
  use x / 2
  --  $\vdash x / 2 > 0 \wedge x / 2 < x$ 
  constructor
  { show x / 2 > 0
    exact half_pos hx0 }
  { show x / 2 < x
    exact half_lt_self hx0 }

-- 2ª demostración

```

```

-- =====

example
  (x : ℝ)
  (h : ∀ ε > 0, x ≤ ε)
  : x ≤ 0 :=
by
  contrapose! h
  -- ⊢ ∃ ε, ε > 0 ∧ ε < x
  use x / 2
  -- ⊢ x / 2 > 0 ∧ x / 2 < x
  constructor
  { show x / 2 > 0
    exact half_pos h }
  { show x / 2 < x
    exact half_lt_self h }

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- variable (p q : Prop)
-- #check (le_of_not_gt : ¬a > b → a ≤ b)
-- #check (half_lt_self : 0 < a → a / 2 < a)
-- #check (half_pos : 0 < a → 0 < a / 2)
-- #check (absurd : p → ¬p → q)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.19. Si $0 < \epsilon$ , entonces $a > 3\epsilon$ para cualquier número $a$

```

-----
-- Demostrar que si  $0 < \epsilon$ , entonces  $a > 3\epsilon$  para cualquier número  $a$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Basta demostrar una contradicción, ya que de una contradicción se
-- sigue cualquier cosa.
--

```

```

-- La hipótesis es una contradicción con la propiedad irreflexiva de la
-- relación <.

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic

variable (a : ℝ)

-- 1ª demostración
-- =====

example
  (h : 0 < 0)
  : a > 37 :=
by
  ex falso
  -- ⊢ False
  show False
  exact lt_irrefl 0 h

-- 2ª demostración
-- =====

example
  (h : 0 < 0)
  : a > 37 :=
by
  ex falso
  -- ⊢ False
  apply lt_irrefl 0 h

-- 3ª demostración
-- =====

example
  (h : 0 < 0)
  : a > 37 :=
absurd h (lt_irrefl 0)

-- 4ª demostración
-- =====

example

```



```

(h : 0 < 0)
: a > 37 :=
by
  have : ¬ 0 < 0 := lt_irrefl 0
  contradiction

-- 5ª demostración
-- =====

example
  (h : 0 < 0)
  : a > 37 :=
by linarith

-- Lemas usados
-- =====

-- variable (p q : Prop)
-- #check (lt_irrefl a : ¬a < a)
-- #check (absurd : p → ¬p → q)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.20. $\{x \leq y, y \neq x\} \vdash x \leq y \wedge x \neq y$

```

-----
-- Sean x e y dos números tales que
--   x ≤ y
--   ¬ y ≤ x
-- entonces
--   x ≤ y ∧ x ≠ y
-----

-- Demostración en lenguaje natural
-- =====

-- Como la conclusión es una conjunción, tenemos que demostrar sus dos
-- partes. La primera parte (x ≤ y) coincide con la hipótesis. Para
-- demostrar la segunda parte (x ≠ y), supongamos que x = y; entonces
-- y ≤ x en contradicción con la segunda hipótesis.

-- Demostraciones con Lean4
-- =====

```

```

import Mathlib.Data.Real.Basic

variable {x y : ℝ}

-- 1ª demostración
-- =====

example
  (h1 : x ≤ y)
  (h2 : ¬y ≤ x)
  : x ≤ y ∧ x ≠ y :=
by
  constructor
  . -- ⊢ x ≤ y
    exact h1
  . -- ⊢ x ≠ y
    intro h3
    -- h3 : x = y
    -- ⊢ False
    have h4 : y ≤ x := h3.symm.le
    show False
    exact h2 h4

-- 2ª demostración
-- =====

example
  (h1 : x ≤ y)
  (h2 : ¬y ≤ x)
  : x ≤ y ∧ x ≠ y :=
by
  constructor
  . -- ⊢ x ≤ y
    exact h1
  . -- ⊢ x ≠ y
    intro h3
    -- h3 : x = y
    -- ⊢ False
    exact h2 (h3.symm.le)

-- 3ª demostración
-- =====

example

```

```

(h1 : x ≤ y)
(h2 : ¬y ≤ x)
: x ≤ y ∧ x ≠ y :=
⟨h1, fun h3 ↦ h2 (h3.symm.le)⟩

```

```

-- 4ª demostración
-- =====

```

**example**

```

(h1 : x ≤ y)
(h2 : ¬y ≤ x)
: x ≤ y ∧ x ≠ y :=

```

**by**

```

constructor
. -- ⊢ x ≤ y
  exact h1
. -- ⊢ x ≠ y
  intro h3
  -- h3 : x = y
  -- ⊢ False
  apply h2
  -- ⊢ y ≤ x
  rw [h3]

```

```

-- 5ª demostración
-- =====

```

**example**

```

(h1 : x ≤ y)
(h2 : ¬y ≤ x)
: x ≤ y ∧ x ≠ y :=

```

**by**

```

constructor
. -- ⊢ x ≤ y
  exact h1
. -- ⊢ x ≠ y
  intro h3
  -- h3 : x = y
  -- ⊢ False
  exact h2 (by rw [h3])

```

```

-- 6ª demostración
-- =====

```

**example**

```

(h1 : x ≤ y)
(h2 : ¬ y ≤ x)
: x ≤ y ∧ x ≠ y :=
⟨h1, fun h ↦ h2 (by rw [h])⟩

-- 7ª demostración
-- =====

example
  (h1 : x ≤ y)
  (h2 : ¬ y ≤ x)
  : x ≤ y ∧ x ≠ y :=
by
  have h3 : x ≠ y
  . contrapose! h2
    -- ⊢ y ≤ x
    rw [h2]
  exact ⟨h1, h3⟩

-- 8ª demostración
-- =====

example
  (h1 : x ≤ y)
  (h2 : ¬ y ≤ x)
  : x ≤ y ∧ x ≠ y :=
by aesop

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.21. $x \leq y \wedge x \neq y \vdash y \not\leq x$

```

-- -----
-- Demostrar que en los reales, si
--   x ≤ y ∧ x ≠ y
-- entonces
--   ¬ y ≤ x
-- -----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que y ≤ x. Entonces, por la antisimetría y la primera

```

```

-- parte de la hipótesis, se tiene que  $x = y$  que contradice la segunda
-- parte de la hipótesis.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable {x y : ℝ}

-- 1ª demostración
-- =====

example
  (h : x ≤ y ∧ x ≠ y)
  : ¬ y ≤ x :=
by
  intro h1
  cases' h with h2 h3
  -- h2 :  $x \leq y$ 
  -- h3 :  $x \neq y$ 
  have h4 : x = y := le_antisymm h2 h1
  show False
  exact h3 h4

-- 2ª demostración
-- =====

example
  (h : x ≤ y ∧ x ≠ y)
  : ¬ y ≤ x :=
by
  intro h1
  have h4 : x = y := le_antisymm h.1 h1
  show False
  exact h.2 h4

-- 3ª demostración
-- =====

example
  (h : x ≤ y ∧ x ≠ y)
  : ¬ y ≤ x :=
by
  intro h1

```

```

show False
exact h.2 (le_antisymm h.1 h1)

-- 4ª demostración
-- =====

example
  (h : x ≤ y ∧ x ≠ y)
  : ¬ y ≤ x :=
fun h1 → h.2 (le_antisymm h.1 h1)

-- 5ª demostración
-- =====

example
  (h : x ≤ y ∧ x ≠ y)
  : ¬ y ≤ x :=
by
  intro h'
  -- h' : y ≤ x
  -- ⊢ False
  apply h.right
  -- ⊢ x = y
  exact le_antisymm h.left h'

-- 6ª demostración
-- =====

example
  (h : x ≤ y ∧ x ≠ y)
  : ¬ y ≤ x :=
by
  cases' h with h1 h2
  -- h1 : x ≤ y
  -- h2 : x ≠ y
  contrapose! h2
  -- h2 : y ≤ x
  -- ⊢ x = y
  exact le_antisymm h1 h2

-- 7ª demostración
-- =====

example : x ≤ y ∧ x ≠ y → ¬ y ≤ x :=
by

```

```

rintro ⟨h1, h2⟩ h'
-- h1 : x ≤ y
-- h2 : x ≠ y
-- h' : y ≤ x
-- ⊢ False
exact h2 (le_antisymm h1 h')

-- 8ª demostración
-- =====

example : x ≤ y ∧ x ≠ y → ¬ y ≤ x :=
fun ⟨h1, h2⟩ h' ↦ h2 (le_antisymm h1 h')

-- Lemas usados
-- =====

-- #check (le_antisymm : x ≤ y → y ≤ x → x = y)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.22. $(\exists x \in \mathbb{R})[2 < x < 3]$

```

-- -----
-- Demostrar que  $(\exists x \in \mathbb{R})[2 < x < 3]$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Podemos usar el número  $5/2$  y comprobar que  $2 < 5/2 < 3$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

-- 1ª demostración
-- =====

example : ∃ x : ℝ, 2 < x ∧ x < 3 :=
by
  use 5 / 2
  show 2 < 5 / 2 ∧ 5 / 2 < 3

```

```

constructor
. show 2 < 5 / 2
  norm_num
. show 5 / 2 < 3
  norm_num

-- 2ª demostración
-- =====

example :  $\exists x : \mathbb{R}, 2 < x \wedge x < 3 :=$ 
by
  use 5 / 2
  constructor
  . norm_num
  . norm_num

-- 3ª demostración
-- =====

example :  $\exists x : \mathbb{R}, 2 < x \wedge x < 3 :=$ 
by
  use 5 / 2
  constructor <.> norm_num

-- 4ª demostración
-- =====

example :  $\exists x : \mathbb{R}, 2 < x \wedge x < 3 :=$ 
(5/2, by norm_num)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.23. Si $(\exists z \in \mathbb{R})[x < z < y]$ , entonces $x < y$

```

-- -----
-- Demostrar que si  $(\exists z \in \mathbb{R})[x < z < y]$ , entonces  $x < y$ .
-- -----

-- Demostración en lenguaje natural
-- =====

-- Sea z tal que verifica las siguientes relaciones:
--      x < z

```

(1)



```

--      z < y                                     (2)
-- Aplicando la propiedad transitiva a (1) y (2) se tiene que
--      x < y.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable (x y : ℝ)

-- 1ª demostración
-- =====

example : (∃ z : ℝ, x < z ∧ z < y) → x < y :=
by
  rintro ⟨z, h1 : x < z, h2 : z < y⟩
  show x < y
  exact lt_trans h1 h2

-- 2ª demostración
-- =====

example : (∃ z : ℝ, x < z ∧ z < y) → x < y :=
by
  rintro ⟨z, h1, h2⟩
  exact lt_trans h1 h2

-- 3ª demostración
-- =====

example : (∃ z : ℝ, x < z ∧ z < y) → x < y :=
fun ⟨_, h1, h2⟩ ↦ lt_trans h1 h2

-- Lemas usados
-- =====

-- variable (a b c : ℝ)
-- #check (lt_trans : a < b → b < c → a < c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.24. En $\mathbb{R}$ , $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge y \not\leq x$

```

-----
-- Demostrar que, en  $\mathbb{R}$ ,  $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge y \not\leq x$ 
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que
--    $x \leq y$  (1)
--    $x \neq y$  (2)
-- Entonces, se tiene  $x \leq y$  (por (1)) y, para probar  $y \not\leq x$ , supongamos
-- que  $y \leq x$ . Por (1), se tiene que  $x = y$ , en contradicción con (2).

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable (x y : ℝ)

-- 1ª demostración
-- =====

example : x ≤ y ∧ x ≠ y → x ≤ y ∧ ¬ y ≤ x :=
by
  rintro ⟨h1 : x ≤ y, h2 : x ≠ y⟩
  constructor
  . show x ≤ y
    exact h1
  . show ¬ y ≤ x
    rintro h3 : y ≤ x
    -- ⊢ False
    have h4 : x = y := le_antisymm h1 h3
    show False
    exact h2 h4

-- 2ª demostración
-- =====

example : x ≤ y ∧ x ≠ y → x ≤ y ∧ ¬ y ≤ x :=
by
  rintro ⟨h1 : x ≤ y, h2 : x ≠ y⟩
  -- ⊢ x ≤ y ∧ ¬ y ≤ x

```

```

constructor
. show  $x \leq y$ 
  exact h1
. show  $\neg y \leq x$ 
  rintro h3 :  $y \leq x$ 
  --  $\vdash \text{False}$ 
  show False
  exact h2 (le_antisymm h1 h3)

-- 3ª demostración
-- =====

example :  $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge \neg y \leq x :=$ 
by
  rintro ⟨h1 :  $x \leq y$ , h2 :  $x \neq y$ ⟩
  constructor
  . show  $x \leq y$ 
    exact h1
  . show  $\neg y \leq x$ 
    exact fun h3 ↦ h2 (le_antisymm h1 h3)

-- 4ª demostración
-- =====

example :  $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge \neg y \leq x :=$ 
by
  rintro ⟨h1, h2⟩
  exact ⟨h1, fun h3 ↦ h2 (le_antisymm h1 h3)⟩

-- 5ª demostración
-- =====

example :  $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge \neg y \leq x :=$ 
  fun ⟨h1, h2⟩ ↦ ⟨h1, fun h3 ↦ h2 (le_antisymm h1 h3)⟩

-- 6ª demostración
-- =====

example :  $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge \neg y \leq x :=$ 
by
  rintro ⟨h1 :  $x \leq y$ , h2 :  $x \neq y$ ⟩
  use h1
  exact fun h3 ↦ h2 (le_antisymm h1 h3)

-- 7ª demostración

```

```

-- =====
example :  $x \leq y \wedge x \neq y \rightarrow x \leq y \wedge \neg y \leq x :=$ 
by
  rintro ⟨h1, h2⟩
  -- h1 :  $x \leq y$ 
  -- h2 :  $x \neq y$ 
  --  $\vdash x \leq y \wedge \neg y \leq x$ 
  use h1
  --  $\neg y \leq x$ 
  contrapose! h2
  -- h2 :  $y \leq x$ 
  --  $\vdash x = y$ 
  apply le_antisymm h1 h2

-- Lemas usados
-- =====

-- #check (le_antisymm :  $x \leq y \rightarrow y \leq x \rightarrow x = y$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.25. En $\mathbb{R}$ , si $x \leq y$ , entonces $y \not\leq x \leftrightarrow x \neq y$

```

-- =====
-- Sean  $x, y$  números reales tales que  $x \leq y$ . Entonces,  $y \not\leq x \leftrightarrow x \neq y$ .
-- =====

-- Demostración en lenguaje natural
-- =====

-- Para demostrar la equivalencia, demostraremos cada una de las
-- implicaciones.
--
-- Para demostrar la primera, supongamos que  $y \not\leq x$  y que  $x =$ 
--  $y$ . Entonces,  $y \leq x$  que es una contradicción.
--
-- Para demostrar la segunda, supongamos que  $x \neq y$  y que  $y \leq$ 
--  $x$ . Entonces, por la hipótesis y la antisimetría, se tiene que  $x = y$ 
-- lo que es una contradicción.

-- Demostraciones con Lean4
-- =====

```

```

import Mathlib.Data.Real.Basic
variable {x y : ℝ}

-- 1ª demostración
-- =====

example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x ≠ y :=
by
  constructor
  . show ¬y ≤ x → x ≠ y
    { intro h1
      -- h1 : ¬y ≤ x
      -- ⊢ x ≠ y
      intro h2
      -- h2 : x = y
      -- ⊢ False
      have h3 : y ≤ x := by rw [h2]
      show False
      exact h1 h3 }
  . show x ≠ y → ¬y ≤ x
    { intro h1
      -- h1 : x ≠ y
      -- ⊢ ¬y ≤ x
      intro h2
      -- h2 : y ≤ x
      -- ⊢ False
      have h3 : x = y := le_antisymm h h2
      show False
      exact h1 h3 }

-- 2ª demostración
-- =====

example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x ≠ y :=
by
  constructor
  . show ¬y ≤ x → x ≠ y
    { intro h1
      -- h1 : ¬y ≤ x
      -- ⊢ x ≠ y

```

```

    intro h2
    -- h2 : x = y
    -- ⊢ False
    show False
    exact h1 (by rw [h2]) }
. show x ≠ y → ¬y ≤ x
{ intro h1
  -- h1 : x ≠ y
  -- ⊢ ¬y ≤ x
  intro h2
  -- h2 : y ≤ x
  -- ⊢ False
  show False
  exact h1 (le_antisymm h h2) }

-- 3ª demostración
-- =====

example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x ≠ y :=
by
  constructor
  . show ¬y ≤ x → x ≠ y
  { intro h1 h2
    exact h1 (by rw [h2]) }
  . show x ≠ y → ¬y ≤ x
  { intro h1 h2
    exact h1 (le_antisymm h h2) }

-- 4ª demostración
-- =====

example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x ≠ y :=
by
  constructor
  . intro h1 h2
    exact h1 (by rw [h2])
  . intro h1 h2
    exact h1 (le_antisymm h h2)

-- 5ª demostración
-- =====

```

```
example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x = y :=
by
  constructor
  . exact fun h1 h2 => h1 (by rw [h2])
  . exact fun h1 h2 => h1 (le_antisymm h h2)
```

```
-- 6ª demostración
-- =====
```

```
example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x = y :=
  { fun h1 h2 => h1 (by rw [h2]),
    fun h1 h2 => h1 (le_antisymm h h2) }
```

```
-- 7ª demostración
-- =====
```

```
example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x = y :=
by
  constructor
  . show ¬y ≤ x → x = y
    { intro h1
      -- h1 : ¬y ≤ x
      -- ⊢ x ≠ y
      contrapose! h1
      -- h1 : x = y
      -- ⊢ y ≤ x
      calc y = x := h1.symm
          _ ≤ x := by rfl }
  . show x = y → ¬y ≤ x
    { intro h2
      -- h2 : x = y
      -- ⊢ ¬y ≤ x
      contrapose! h2
      -- h2 : y ≤ x
      -- ⊢ x = y
      show x = y
      exact le_antisymm h h2 }
```

```

-- 8ª demostración
-- =====

example
  (h : x ≤ y)
  : ¬y ≤ x ↔ x ≠ y :=
by
  constructor
  · -- ⊢ ¬y ≤ x → x ≠ y
    contrapose!
    -- ⊢ x = y → y ≤ x
    rintro rfl
    -- ⊢ x ≤ x
    rfl
  · -- ⊢ x ≠ y → ¬y ≤ x
    contrapose!
    -- ⊢ y ≤ x → x = y
    exact le_antisymm h

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.26. Si $|x + 3| < 5$ , entonces $-8 < x < 2$

```

-- -----
-- Demostrar que si
--    $|x + 3| < 5$ 
-- entonces
--    $-8 < x < 2$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que
--    $|x + 3| < 5$ 
-- entonces
--    $-5 < x + 3 < 5$ 
-- por tanto
--    $-8 < x < 2$ 

-- Demostraciones con Lean4
-- =====

```



```

import Mathlib.Data.Real.Basic
variable (x y : ℝ)

-- 1ª demostración
-- =====

example
  :  $|x + 3| < 5 \rightarrow -8 < x \wedge x < 2$  :=
by
  rw [abs_lt]
  --  $\vdash -5 < x + 3 \wedge x + 3 < 5 \rightarrow -8 < x \wedge x < 2$ 
  intro h
  --  $h : -5 < x + 3 \wedge x + 3 < 5$ 
  --  $\vdash -8 < x \wedge x < 2$ 
  constructor
  . --  $\vdash -8 < x$ 
    linarith
  . --  $x < 2$ 
    linarith

-- 2ª demostración
-- =====

example
  :  $|x + 3| < 5 \rightarrow -8 < x \wedge x < 2$  :=
by
  rw [abs_lt]
  intro h
  constructor <=> linarith

-- Comentario: La composición (constructor <=> linarith) aplica constructor y a
-- continuación le aplica linarith a cada subobjetivo.

-- 3ª demostración
-- =====

example
  :  $|x + 3| < 5 \rightarrow -8 < x \wedge x < 2$  :=
by
  rw [abs_lt]
  exact fun _ => (by linarith, by linarith)

-- Lemas usados
-- =====

```

```
-- #check (abs_lt: |x| < y ↔ -y < x ∧ x < y)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.27. En $\mathbb{R}$ , $y > x^2 \vdash y > 0 \vee y < -1$

```
-- -----
-- Demostrar que si
--   y > x^2
-- entonces
--   y > 0 ∨ y < -1
-- -----

-- Demostración en lenguaje natural
-- =====

-- Puesto que
--   (∀ x ∈ ℝ)[x² ≥ 0]
-- se tiene que
--   y > x²
--   ≥ 0
-- Por tanto, y > 0 y, al verificar la primera parte de la disyunción, se
-- verifica la disyunción.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {x y : ℝ}

-- 1ª demostración
-- =====

example
  (h : y > x^2)
  : y > 0 ∨ y < -1 :=
by
  have h1 : y > 0 := by
    calc y > x^2 := h
         _ ≥ 0 := pow_two_nonneg x
  show y > 0 ∨ y < -1
exact Or.inl h1
```

```

-- 2ª demostración
-- =====

example
  (h : y > x^2)
  : y > 0 ∨ y < -1 :=
by
  left
  -- ⊢ y > 0
  calc y > x^2 := h
      _ ≥ 0   := pow_two_nonneg x

-- 3ª demostración
-- =====

example
  (h : y > x^2)
  : y > 0 ∨ y < -1 :=
by
  left
  -- ⊢ y > 0
  linarith [pow_two_nonneg x]

-- 4ª demostración
-- =====

example
  (h : y > x^2)
  : y > 0 ∨ y < -1 :=
by { left ; linarith [pow_two_nonneg x] }

-- Lema usado
-- =====

-- #check (pow_two_nonneg x : 0 ≤ x ^ 2)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.28. En $\mathbb{R}$ , $-y > x^2 + 1 \vdash y > 0 \vee y < -1$

```

-- -----
-- Demostrar que si
--   -y > x^2 + 1

```

```

-- entonces
--    $y > 0 \vee y < -1$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Usaremos los siguientes lemas
--    $(\forall b, c \in \mathbb{R})[b \leq c \rightarrow \forall (a : \mathbb{R}), b + a \leq c + a]$           (L1)
--    $(\forall a \in \mathbb{R})[0 \leq a^2]$                                            (L2)
--    $(\forall a \in \mathbb{R})[0 + a = a]$                                          (L3)
--    $(\forall a, b \in \mathbb{R})[a < -b \leftrightarrow b < -a]$                      (L4)

-- Se tiene
--    $-y > x^2 + 1$       [por la hipótesis]
--    $\geq 0 + 1$         [por L1 y L2]
--    $= 1$               [por L3]
-- Por tanto,
--    $-y > 1$ 
-- y, aplicando el lema L4, se tiene
--    $y < -1$ 
-- Como se verifica la segunda parte de la disyunción, se verifica la
-- disyunción.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {x y : ℝ}

-- 1ª demostración
-- =====

example
  (h : -y > x^2 + 1)
  : y > 0 ∨ y < -1 :=
by
  have h1 : -y > 1 := by
    calc -y > x^2 + 1 := by exact h
          _ ≥ 0 + 1   := add_le_add_right (pow_two_nonneg x) 1
          _ = 1       := zero_add 1
  have h2 : y < -1 := lt_neg.mp h1
  show y > 0 ∨ y < -1
  exact Or.inr h2

```

```

-- 2ª demostración
-- =====

example
  (h : -y > x^2 + 1)
  : y > 0 ∨ y < -1 :=
by
  have h1 : -y > 1 := by linarith [pow_two_nonneg x]
  have h2: y < -1 := lt_neg.mp h1
  show y > 0 ∨ y < -1
  exact Or.inr h2

-- 3ª demostración
-- =====

example
  (h : -y > x^2 + 1)
  : y > 0 ∨ y < -1 :=
by
  have h1: y < -1 := by linarith [pow_two_nonneg x]
  show y > 0 ∨ y < -1
  exact Or.inr h1

-- 4ª demostración
-- =====

example
  (h : -y > x^2 + 1)
  : y > 0 ∨ y < -1 :=
by
  right
  -- ⊢ y < -1
  linarith [pow_two_nonneg x]

-- 5ª demostración
-- =====

example
  (h : -y > x^2 + 1)
  : y > 0 ∨ y < -1 :=
by { right ; linarith [pow_two_nonneg x] }

-- Lemas usados
-- =====

```

```
-- variable (a b c : ℝ)
-- #check (add_le_add_right : b ≤ c → ∀ (a : ℝ), b + a ≤ c + a)
-- #check (lt_neg : a < -b ↔ b < -a)
-- #check (pow_two_nonneg a : 0 ≤ a ^ 2)
-- #check (zero_add a : 0 + a = a)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.29. En $\mathbb{R}$ , si $x < |y|$ , entonces $x < y$ ó $x < -y$

```
-- -----
-- Demostrar que para todo par de números reales x e y, si x < |y|,
-- entonces x < y ó x < -y.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se demostrará por casos según y ≥ 0.
--
-- Primer caso: Supongamos que y ≥ 0. Entonces, |y| = y y, por tanto,
-- x < y.
--
-- Segundo caso: Supongamos que y < 0. Entonces, |y| = -y y, por tanto,
-- x < -y.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {x y : ℝ}

-- 1ª demostración
-- =====

example : x < |y| → x < y ∨ x < -y :=
by
  intro h1
  -- h1 : x < |y|
  -- ⊢ x < y ∨ x < -y
  cases' le_or_gt 0 y with h2 h3
  . -- h2 : 0 ≤ y
    left
```

```

--  $\vdash x < y$ 
rwa [abs_of_nonneg h2] at h1
. --  $h3 : 0 > y$ 
right
--  $\vdash x < -y$ 
rwa [abs_of_neg h3] at h1

-- 2ª demostración
-- =====

example :  $x < |y| \rightarrow x < y \vee x < -y :=$ 
lt_abs.mp

-- Lemas usados
-- =====

-- #check (le_or_gt x y :  $x \leq y \vee x > y$ )
-- #check (abs_of_nonneg :  $0 \leq x \rightarrow \text{abs } x = x$ )
-- #check (abs_of_neg :  $x < 0 \rightarrow \text{abs } x = -x$ )
-- #check (lt_abs :  $x < |y| \leftrightarrow x < y \vee x < -y$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.30. En $\mathbb{R}$ , $x \leq |x|$

```

-----
-- Demostrar que en  $\mathbb{R}$ ,
--    $x \leq |x|$ 
-----

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas
--    $(\forall x \in \mathbb{R})[0 \leq x \rightarrow |x| = x]$  (L1)
--    $(\forall x, y \in \mathbb{R})[x < y \rightarrow x \leq y]$  (L2)
--    $(\forall x \in \mathbb{R})[x \leq 0 \rightarrow x \leq -x]$  (L3)
--    $(\forall x \in \mathbb{R})[x < 0 \rightarrow |x| = -x]$  (L4)
--
-- Se demostrará por casos según  $x \geq 0$ :
--
-- Primer caso: Supongamos que  $x \geq 0$ . Entonces,
--    $x \leq x$ 

```

```

--      = |x|      [por L1]
--
-- Segundo caso: Supongamos que  $x < 0$ . Entonces, por el L2, se tiene
--       $x \leq 0$                                      (1)
-- Por tanto,
--       $x \leq -x$       [por L3 y (1)]
--      = |x|      [por L4]

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {x : ℝ}

-- 1ª demostración
-- =====

example : x ≤ |x| :=
by
  cases' le_or_gt 0 x with h1 h2
  . -- h1 : 0 ≤ x
    show x ≤ |x|
    calc x ≤ x := le_refl x
         _ = |x| := (abs_of_nonneg h1).symm
  . -- h2 : 0 > x
    have h3 : x ≤ 0 := le_of_lt h2
    show x ≤ |x|
    calc x ≤ -x := le_neg_self_iff.mpr h3
         _ = |x| := (abs_of_neg h2).symm

-- 2ª demostración
-- =====

example : x ≤ |x| :=
by
  cases' le_or_gt 0 x with h1 h2
  . -- h1 : 0 ≤ x
    rw [abs_of_nonneg h1]
  . -- h2 : 0 > x
    rw [abs_of_neg h2]
    -- ⊢ x ≤ -x
    apply Left.self_le_neg
    -- ⊢ x ≤ 0
    exact le_of_lt h2

```



```

-- 3ª demostración
-- =====

example : x ≤ |x| :=
by
  rcases (le_or_gt 0 x) with h1 | h2
  . -- h1 : 0 ≤ x
    rw [abs_of_nonneg h1]
  . -- h2 : 0 > x
    rw [abs_of_neg h2]
  linarith

-- 4ª demostración
-- =====

example : x ≤ |x| :=
  le_abs_self x

-- Lemas usados
-- =====

-- variable (y : ℝ)
-- #check (Left.self_le_neg : x ≤ 0 → x ≤ -x)
-- #check (abs_of_neg : x < 0 → |x| = -x)
-- #check (abs_of_nonneg : 0 ≤ x → |x| = x)
-- #check (le_abs_self x : x ≤ |x|)
-- #check (le_neg_self_iff : x ≤ -x ↔ x ≤ 0)
-- #check (le_of_lt : x < y → x ≤ y)
-- #check (le_or_gt x y : x ≤ y ∨ x > y)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.31. En $\mathbb{R}$ , $-x \leq |x|$

```

-----
-- Demostrar que
--   -x ≤ |x|
--
-----

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas

```

```

--       $(\forall x \in \mathbb{R})[0 \leq x \rightarrow -x \leq x]$                                 (L1)
--       $(\forall x \in \mathbb{R})[0 \leq x \rightarrow |x| = x]$                         (L2)
--       $(\forall x \in \mathbb{R})[x \leq x]$                                           (L3)
--       $(\forall x \in \mathbb{R})[x < 0 \rightarrow |x| = -x]$                         (L4)
--
-- Se demostrará por casos según  $x \geq 0$ :
--
-- Primer caso: Supongamos que  $x \geq 0$ . Entonces,
--       $-x \leq x$  [por L1]
--       $= |x|$  [por L2]
--
-- Segundo caso: Supongamos que  $x < 0$ . Entonces,
--       $-x \leq -x$  [por L3]
--       $= |x|$  [por L4]

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {x : ℝ}

-- 1ª demostración
-- =====

example :  $-x \leq |x|$  :=
by
  cases' (le_or_gt 0 x) with h1 h2
  . -- h1 :  $0 \leq x$ 
    show  $-x \leq |x|$ 
    calc  $-x \leq x$  := by exact neg_le_self h1
         _ = |x| := (abs_of_nonneg h1).symm
  . -- h2 :  $0 > x$ 
    show  $-x \leq |x|$ 
    calc  $-x \leq -x$  := by exact le_refl (-x)
         _ = |x| := (abs_of_neg h2).symm

-- 2ª demostración
-- =====

example :  $-x \leq |x|$  :=
by
  cases' (le_or_gt 0 x) with h1 h2
  . -- h1 :  $0 \leq x$ 
    rw [abs_of_nonneg h1]
    --  $\vdash -x \leq x$ 

```

```

    exact neg_le_self h1
  . -- h2 : 0 > x
    rw [abs_of_neg h2]

-- 3ª demostración
-- =====

example : -x ≤ |x| :=
by
  rcases (le_or_gt 0 x) with h1 | h2
  . -- h1 : 0 ≤ x
    rw [abs_of_nonneg h1]
    -- ⊢ -x ≤ x
    linarith
  . -- h2 : 0 > x
    rw [abs_of_neg h2]

-- 4ª demostración
-- =====

example : -x ≤ |x| :=
  neg_le_abs_self x

-- Lemas usados
-- =====

-- variable (y : ℝ)
-- #check (abs_of_neg : x < 0 → |x| = -x)
-- #check (abs_of_nonneg : 0 ≤ x → |x| = x)
-- #check (le_or_gt x y : x ≤ y ∨ x > y)
-- #check (le_refl x : x ≤ x)
-- #check (neg_le_abs_self x : -x ≤ |x|)
-- #check (neg_le_self : 0 ≤ x → -x ≤ x)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.32. En $\mathbb{R}$ , $|x + y| \leq |x| + |y|$

```

-- -----
-- Demostrar que
--   |x + y| ≤ |x| + |y|
-- -----

```

```

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas
--       $(\forall x \in \mathbb{R})[0 \leq x \rightarrow |x| = x]$  (L1)
--       $(\forall a, b, c, d \in \mathbb{R})[a \leq b \wedge c \leq d \rightarrow a + c \leq b + d]$  (L2)
--       $(\forall x \in \mathbb{R})[x \leq |x|]$  (L3)
--       $(\forall x \in \mathbb{R})[x < 0 \rightarrow |x| = -x]$  (L4)
--       $(\forall x, y \in \mathbb{R})[-(x + y) = -x + -y]$  (L5)
--       $(\forall x \in \mathbb{R})[-x \leq |x|]$  (L6)
--
-- Se demostrará por casos según  $x + y \geq 0$ :
--
-- Primer caso: Supongamos que  $x + y \geq 0$ . Entonces,
--       $|x + y| = x + y$  [por L1]
--       $\leq |x| + |y|$  [por L2 y L3]
--
-- Segundo caso: Supongamos que  $x + y < 0$ . Entonces,
--       $|x + y| = -(x + y)$  [por L4]
--       $= -x + -y$  [por L5]
--       $\leq |x| + |y|$  [por L2 y L6]

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {x y : ℝ}

-- 1ª demostración
-- =====

example :  $|x + y| \leq |x| + |y|$  :=
by
  rcases le_or_gt 0 (x + y) with h1 | h2
  · -- h1 :  $0 \leq x + y$ 
    show  $|x + y| \leq |x| + |y|$ 
    calc  $|x + y| = x + y$  := by exact abs_of_nonneg h1
          $\leq |x| + |y|$  := add_le_add (le_abs_self x) (le_abs_self y)
  · -- h2 :  $0 > x + y$ 
    show  $|x + y| \leq |x| + |y|$ 
    calc  $|x + y| = -(x + y)$  := by exact abs_of_neg h2
          $= -x + -y$  := by exact neg_add x y
          $\leq |x| + |y|$  := add_le_add (neg_le_abs_self x) (neg_le_abs_self y)

-- 2ª demostración

```

```

-- =====
example : |x + y| ≤ |x| + |y| := by
  rcases le_or_gt 0 (x + y) with h1 | h2
  · -- h1 : 0 ≤ x + y
    rw [abs_of_nonneg h1]
    -- ⊢ x + y ≤ |x| + |y|
    exact add_le_add (le_abs_self x) (le_abs_self y)
  · -- h2 : 0 > x + y
    rw [abs_of_neg h2]
    -- ⊢ -(x + y) ≤ |x| + |y|
    calc -(x + y) = -x + -y      := by exact neg_add x y
           _ ≤ |x| + |y|      := add_le_add (neg_le_abs_self x) (neg_le_abs_self y)

-- 2ª demostración
-- =====
example : |x + y| ≤ |x| + |y| := by
  rcases le_or_gt 0 (x + y) with h1 | h2
  · -- h1 : 0 ≤ x + y
    rw [abs_of_nonneg h1]
    -- ⊢ x + y ≤ |x| + |y|
    linarith [le_abs_self x, le_abs_self y]
  · -- h2 : 0 > x + y
    rw [abs_of_neg h2]
    -- ⊢ -(x + y) ≤ |x| + |y|
    linarith [neg_le_abs_self x, neg_le_abs_self y]

-- 3ª demostración
-- =====
example : |x + y| ≤ |x| + |y| :=
  abs_add x y

-- Lemas usados
-- =====

-- variable (a b c d : ℝ)
-- #check (abs_add x y : |x + y| ≤ |x| + |y|)
-- #check (abs_of_neg : x < 0 → |x| = -x)
-- #check (abs_of_nonneg : 0 ≤ x → |x| = x)
-- #check (add_le_add : a ≤ b → c ≤ d → a + c ≤ b + d)
-- #check (le_abs_self a : a ≤ |a|)
-- #check (le_or_gt x y : x ≤ y ∨ x > y)
-- #check (neg_add x y : -(x + y) = -x + -y)

```

```
-- #check (neg_le_abs_self x : -x ≤ |x|)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

### 6.33. En $\mathbb{R}$ , si $x \neq 0$ entonces $x < 0$ ó $x > 0$

```
-- -----
-- Ejercicio. Sea x un número real. Demostrar que si
--   x ≠ 0
-- entonces
--   x < 0 ∨ x > 0
-- -----

-- Demostración en lenguaje natural
-- =====

-- Usando el siguiente lema
--   (∀ x y ∈ ℝ)[x < y ∨ x = y ∨ y < x]
-- se demuestra distinguiendo tres casos.
--
-- Caso 1: Supongamos que x < 0. Entonces, se verifica la disyunción ya
-- que se verifica su primera parte.
--
-- Caso 2: Supongamos que x = 0. Entonces, se tiene una contradicción
-- con la hipótesis.
--
-- Caso 3: Supongamos que x > 0. Entonces, se verifica la disyunción ya
-- que se verifica su segunda parte.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {x : ℝ}

-- 1ª demostración
-- =====

example
  (h : x ≠ 0)
  : x < 0 ∨ x > 0 :=
by
  rcases lt_trichotomy x 0 with hx1 | hx2 | hx3
```

```

. -- hx1 : x < 0
  left
  -- ⊢ x < 0
  exact hx1
. -- hx2 : x = 0
  contradiction
. -- hx3 : 0 < x
  right
  -- ⊢ x > 0
  exact hx3

-- 2ª demostración
-- =====

example
  (h : x ≠ 0)
  : x < 0 ∨ x > 0 :=
Ne.lt_or_lt h

-- 3ª demostración
-- =====

example
  (h : x ≠ 0)
  : x < 0 ∨ x > 0 :=
by aesop

-- Lemas usados
-- =====

-- variable (y : ℝ)
-- #check (lt_trichotomy x y : x < y ∨ x = y ∨ y < x)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.34. Si $(\exists x, y \in \mathbb{R})[z = x^2 + y^2 \vee z = x^2 + y^2 + 1]$ , entonces $z \geq 0$

```

-- -----
-- Demostrar que si
--   ∃ x y, z = x^2 + y^2 ∨ z = x^2 + y^2 + 1
-- entonces

```

```

--       $z \geq 0$ 
--      -----

--  Demostración en lenguaje natural
--  =====

--  Usaremos los siguientes lemas
--       $(\forall x \in \mathbb{R})[x^2 \geq 0]$                                 (L1)
--       $(\forall x, y \in \mathbb{R})[x \geq 0 \rightarrow y \geq 0 \rightarrow x + y \geq 0]$  (L2)
--       $1 \geq 0$                                                 (L3)

--  Sean  $a$  y  $b$  tales que
--       $z = a^2 + b^2 \vee z = a^2 + b^2 + 1$ 
--  Entonces, por L1, se tiene que
--       $a^2 \geq 0$                                                 (1)
--       $b^2 \geq 0$                                                 (2)
--
--  En el primer caso,  $z = a^2 + b^2$  y se tiene que  $z \geq 0$  por el lema L2
--  aplicado a (1) y (2).
--
--  En el segundo caso,  $z = a^2 + b^2 + 1$  y se tiene que  $z \geq 0$  por el lema L2
--  aplicado a (1), (2) y L3.

--  Demostraciones con Lean4
--  =====

import Mathlib.Data.Real.Basic
import Mathlib.Tactic
variable {z : ℝ}

--  1ª demostración
--  =====

example
  (h :  $\exists x y, z = x^2 + y^2 \vee z = x^2 + y^2 + 1$ )
  :  $z \geq 0$  :=
by
  rcases h with ⟨a, b, h1⟩
  -- a b : ℝ
  -- h1 :  $z = a^2 + b^2 \vee z = a^2 + b^2 + 1$ 
  have h2 :  $a^2 \geq 0$  := pow_two_nonneg a
  have h3 :  $b^2 \geq 0$  := pow_two_nonneg b
  have h4 :  $a^2 + b^2 \geq 0$  := add_nonneg h2 h3
  rcases h1 with h5 | h6
  . -- h5 :  $z = a^2 + b^2$ 

```



```

show z ≥ 0
calc z = a ^ 2 + b ^ 2 := h5
_ ≥ 0 := add_nonneg h2 h3
. -- h6 : z = a ^ 2 + b ^ 2 + 1
show z ≥ 0
calc z = (a ^ 2 + b ^ 2) + 1 := h6
_ ≥ 0 := add_nonneg h4 zero_le_one

-- 2ª demostración
-- =====

example
(h : ∃ x y, z = x^2 + y^2 ∨ z = x^2 + y^2 + 1)
: z ≥ 0 :=
by
rcases h with ⟨a, b, h1 | h2⟩
. -- h1 : z = a ^ 2 + b ^ 2
have h1a : a ^ 2 ≥ 0 := pow_two_nonneg a
have h1b : b ^ 2 ≥ 0 := pow_two_nonneg b
show z ≥ 0
calc z = a ^ 2 + b ^ 2 := h1
_ ≥ 0 := add_nonneg h1a h1b
. -- h2 : z = a ^ 2 + b ^ 2 + 1
have h2a : a ^ 2 ≥ 0 := pow_two_nonneg a
have h2b : b ^ 2 ≥ 0 := pow_two_nonneg b
have h2c : a ^ 2 + b ^ 2 ≥ 0 := add_nonneg h2a h2b
show z ≥ 0
calc z = (a ^ 2 + b ^ 2) + 1 := h2
_ ≥ 0 := add_nonneg h2c zero_le_one

-- 3ª demostración
-- =====

example
(h : ∃ x y, z = x^2 + y^2 ∨ z = x^2 + y^2 + 1)
: z ≥ 0 :=
by
rcases h with ⟨a, b, h1 | h2⟩
. -- h1 : z = a ^ 2 + b ^ 2
rw [h1]
-- ⊢ a ^ 2 + b ^ 2 ≥ 0
apply add_nonneg
. -- ⊢ 0 ≤ a ^ 2
apply pow_two_nonneg
. -- ⊢ 0 ≤ b ^ 2

```

```

    apply pow_two_nonneg
  . -- h2 : z = a ^ 2 + b ^ 2 + 1
  rw [h2]
  --  $\vdash a^2 + b^2 + 1 \geq 0$ 
  apply add_nonneg
  . --  $\vdash 0 \leq a^2 + b^2$ 
  apply add_nonneg
  . --  $\vdash 0 \leq a^2$ 
  apply pow_two_nonneg
  . --  $\vdash 0 \leq b^2$ 
  apply pow_two_nonneg
  . --  $\vdash 0 \leq 1$ 
  exact zero_le_one

-- 4ª demostración
-- =====

example
  (h :  $\exists x y, z = x^2 + y^2 \vee z = x^2 + y^2 + 1$ )
  :  $z \geq 0$  :=
by
  rcases h with ⟨a, b, rfl | rfl⟩
  . --  $\vdash a^2 + b^2 \geq 0$ 
  apply add_nonneg
  . --  $\vdash 0 \leq a^2$ 
  apply pow_two_nonneg
  . --  $\vdash 0 \leq b^2$ 
  apply pow_two_nonneg
  . --  $\vdash a^2 + b^2 + 1 \geq 0$ 
  apply add_nonneg
  . --  $\vdash 0 \leq a^2 + b^2$ 
  apply add_nonneg
  . --  $\vdash 0 \leq a^2$ 
  apply pow_two_nonneg
  . --  $\vdash 0 \leq b^2$ 
  apply pow_two_nonneg
  . --  $\vdash 0 \leq 1$ 
  exact zero_le_one

-- 5ª demostración
-- =====

example
  (h :  $\exists x y, z = x^2 + y^2 \vee z = x^2 + y^2 + 1$ )
  :  $z \geq 0$  :=

```

```

by
  rcases h with ⟨a, b, rfl | rfl⟩
  . --  $\vdash a^2 + b^2 \geq 0$ 
    nlinarith
  . --  $\vdash a^2 + b^2 + 1 \geq 0$ 
    nlinarith

-- 6ª demostración
-- =====

example
  (h :  $\exists x y, z = x^2 + y^2 \vee z = x^2 + y^2 + 1$ )
  :  $z \geq 0$  :=
by rcases h with ⟨a, b, rfl | rfl⟩ <|> nlinarith

-- Lemas usados
-- =====

-- variable (x y :  $\mathbb{R}$ )
-- #check (add_nonneg :  $0 \leq x \rightarrow 0 \leq y \rightarrow 0 \leq x + y$ )
-- #check (pow_two_nonneg x :  $0 \leq x^2$ )
-- #check (zero_le_one :  $0 \leq 1$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 6.35. En $\mathbb{R}$ , si $1 < a$ , entonces $a < aa$

```

-- -----
-- Demostrar, para todo  $a \in \mathbb{R}$ , si
--    $1 < a$ 
-- entonces
--    $a < a * a$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas
-- L1:  $0 < 1$ 
-- L2:  $(\forall a \in \mathbb{R}[1a = a])$ 
-- L3:  $(\forall a, b, c \in \mathbb{R})[0 < a \rightarrow (ba < ca \leftrightarrow b < c)]$ 
--
-- En primer lugar, tenemos que

```

```

--      0 < a                                     (1)
-- ya que
--      0 < 1      [por L1]
--      < a      [por la hipótesis]
-- Entonces,
--      a = 1a     [por L2]
--      < aa      [por L3, (1) y la hipótesis]

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {a : ℝ}

-- 1ª demostración
-- =====

example
  (h : 1 < a)
  : a < a * a :=
by
  have h1 : 0 < a := calc
    0 < 1 := zero_lt_one
    _ < a := h
  show a < a * a
  calc a = 1 * a := (one_mul a).symm
    _ < a * a := (mul_lt_mul_right h1).mpr h

-- Comentarios: La táctica (convert e) genera nuevos subobjetivos cuya
-- conclusiones son las diferencias entre el tipo de e y la conclusión.

-- 2ª demostración
-- =====

example
  (h : 1 < a)
  : a < a * a :=
by
  convert (mul_lt_mul_right _).2 h
  . -- ⊢ a = 1 * a
    rw [one_mul]
  . -- ⊢ 0 < a
    exact lt_trans zero_lt_one h

-- Lemas usados

```

```
-- =====  
  
-- variables (a b c : ℝ)  
-- #check (mul_lt_mul_right : 0 < a → (b * a < c * a ↔ b < c))  
-- #check (one_mul a : 1 * a = a)  
-- #check (lt_trans : a < b → b < c → a < c)  
-- #check (zero_lt_one : 0 < 1)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)



# Capítulo 7

## Divisibilidad

### 7.1. Si $x, y, z \in \mathbb{N}$ , entonces $x \mid yxz$

```
-- Demostrar que si  $x, y, z \in \mathbb{N}$ , entonces
--  $x \mid y * x * z$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la transitividad de la divisibilidad aplicada a las relaciones
--  $x \mid yx$ 
--  $yx \mid yxz$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable (x y z :  $\mathbb{N}$ )

-- 1ª demostración
-- =====

example :  $x \mid y * x * z$  :=
by
  have h1 :  $x \mid y * x$  :=
    dvd_mul_left x y
  have h2 :  $(y * x) \mid (y * x * z)$  :=
    dvd_mul_right (y * x) z
  show  $x \mid y * x * z$ 
```

```

exact dvd_trans h1 h2

-- 2ª demostración
-- =====

example : x | y * x * z :=
dvd_trans (dvd_mul_left x y) (dvd_mul_right (y * x) z)

-- 3ª demostración
-- =====

example : x | y * x * z :=
by
  apply dvd_mul_of_dvd_left
  apply dvd_mul_left

-- Lemas usados
-- =====

-- #check (dvd_mul_left x y : x | y * x)
-- #check (dvd_mul_right x y : x | x * y)
-- #check (dvd_trans : x | y → y | z → x | z)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.2. Si $x$ divide a $w$ , también divide a $y(xz)+x^2+w^2$

```

-- -----
-- Demostrar que si
--   x | w
-- entonces
--   x | y * (x * z) + x^2 + w^2
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la divisibilidad de la suma basta probar que
--   x | yxz                                     (1)
--   x | x^2                                     (2)
--   x | w^2                                     (3)
--
-- Para demostrar (1), por la divisibilidad del producto se tiene

```



```

--      x | xz
-- y, de nuevo por la divisibilidad del producto,
--      x | y(xz).
--
-- La propiedad (2) se tiene por la definición de cuadrado y la
-- divisibilidad del producto.
--
-- La propiedad (3) se tiene por la definición de cuadrado, la hipótesis
-- y la divisibilidad del producto.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable (w x y z : ℕ)

-- 1ª demostración
example
  (h : x | w)
  : x | y * (x * z) + x^2 + w^2 :=
by
  have h1 : x | x * z :=
    dvd_mul_right x z
  have h2 : x | y * (x * z) :=
    dvd_mul_of_dvd_right h1 y
  have h3 : x | x^2 := by
    apply dvd_mul_left
  have h4 : x | w * w :=
    dvd_mul_of_dvd_left h w
  have h5 : x | w^2 := by
    rwa [← pow_two w] at h4
  have h6 : x | y * (x * z) + x^2 :=
    dvd_add h2 h3
  show x | y * (x * z) + x^2 + w^2
  exact dvd_add h6 h5

-- 2ª demostración
example
  (h : x | w)
  : x | y * (x * z) + x^2 + w^2 :=
by
  apply dvd_add
  { apply dvd_add
    { apply dvd_mul_of_dvd_right
      apply dvd_mul_right }

```

```

    { rw [pow_two]
      apply dvd_mul_right }}
  { rw [pow_two]
    apply dvd_mul_of_dvd_left h }

-- 3ª demostración
example
  (h : x | w)
  : x | y * (x * z) + x^2 + w^2 :=
by
  repeat' apply dvd_add
  { apply dvd_mul_of_dvd_right
    apply dvd_mul_right }
  { rw [pow_two]
    apply dvd_mul_right }
  { rw [pow_two]
    apply dvd_mul_of_dvd_left h }

-- Lemas usados
-- =====

-- #check (dvd_add : x | y → x | z → x | y + z)
-- #check (dvd_mul_left x y : x | y * x)
-- #check (dvd_mul_right x y : x | x * y)
-- #check (dvd_mul_of_dvd_left : x | y → ∀ (c : ℕ), x | y * c)
-- #check (dvd_mul_of_dvd_right : x | y → ∀ (c : ℕ), x | c * y)
-- #check (pow_two x : x ^ 2 = x * x)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.3. Transitividad de la divisibilidad

```

-----
-- Demostrar que la relación de divisibilidad es transitiva.
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que  $a \mid b$  y  $b \mid c$ . Entonces, existen  $d$  y  $e$  tales que
--    $b = ad$  (1)
--    $c = be$  (2)
-- Por tanto,

```

```

--      c = be      [por (2)]
--      = (ad)e     [por (1)]
--      = a(de)
-- Por consiguiente, a | c.

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic

variable {a b c : ℕ}

-- 1ª demostración
example
  (divab : a | b)
  (divbc : b | c) :
  a | c :=
by
  rcases divab with ⟨d, beq : b = a * d⟩
  rcases divbc with ⟨e, ceq : c = b * e⟩
  have h1 : c = a * (d * e) :=
    calc c = b * e      := ceq
          _ = (a * d) * e := congrArg (. * e) beq
          _ = a * (d * e) := mul_assoc a d e
  show a | c
  exact Dvd.intro (d * e) h1.symm

-- 2ª demostración
example
  (divab : a | b)
  (divbc : b | c) :
  a | c :=
by
  rcases divab with ⟨d, beq : b = a * d⟩
  rcases divbc with ⟨e, ceq : c = b * e⟩
  use (d * e)
  -- ⊢ c = a * (d * e)
  rw [ceq, beq]
  -- ⊢ (a * d) * e = a * (d * e)
  exact mul_assoc a d e

-- 3ª demostración
example
  (divab : a | b)
  (divbc : b | c) :

```

```

a | c :=
by
  rcases divbc with ⟨e, rfl⟩
  --  $\vdash a \mid b * e$ 
  rcases divab with ⟨d, rfl⟩
  --  $\vdash a \mid a * d * e$ 
  use (d * e)
  --  $\vdash a * d * e = a * (d * e)$ 
  ring

-- 4ª demostración
example
  (divab : a | b)
  (divbc : b | c) :
  a | c :=
by
  cases' divab with d beq
  --  $d : \mathbb{N}$ 
  --  $beq : b = a * d$ 
  cases' divbc with e ceq
  --  $e : \mathbb{N}$ 
  --  $ceq : c = b * e$ 
  rw [ceq, beq]
  --  $\vdash a \mid a * d * e$ 
  use (d * e)
  --  $\vdash (a * d) * e = a * (d * e)$ 
  exact mul_assoc a d e

-- 5ª demostración
example
  (divab : a | b)
  (divbc : b | c) :
  a | c :=
by exact dvd_trans divab divbc

-- Lemas usados
-- =====

-- #check (mul_assoc a b c : (a * b) * c = a * (b * c))
-- #check (Dvd.intro c : a * c = b → a | b)
-- #check (dvd_trans : a | b → b | c → a | c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.4. Si $a$ divide a $b$ y a $c$ , entonces divide a $b+c$

```

-----
-- Demostrar que si  $a$  es un divisor de  $b$  y de  $c$ , tambien lo es de  $b + c$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Puesto que  $a$  divide a  $b$  y a  $c$ , existen  $d$  y  $e$  tales que
--       $b = ad$                                      (1)
--       $c = ae$                                      (2)
-- Por tanto,
--       $b + c = ad + c$       [por (1)]
--               $= ad + ae$     [por (2)]
--               $= a(d + e)$   [por la distributiva]
-- Por consiguiente,  $a$  divide a  $b + c$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic

variable {a b c : ℕ}

-- 1ª demostración
example
  (h1 : a ∣ b)
  (h2 : a ∣ c)
  : a ∣ (b + c) :=
by
  rcases h1 with ⟨d, beq : b = a * d⟩
  rcases h2 with ⟨e, ceq : c = a * e⟩
  have h1 : b + c = a * (d + e) :=
    calc b + c
      = (a * d) + c      := congrArg (. + c) beq
      _ = (a * d) + (a * e) := congrArg ((a * d) + .) ceq
      _ = a * (d + e)      := by rw [← mul_add]
  show a ∣ (b + c)
  exact Dvd.intro (d + e) h1.symm

-- 2ª demostración
example
  (h1 : a ∣ b)

```

```

(h2 : a | c)
: a | (b + c) :=
by
  rcases h1 with ⟨d, beq : b = a * d⟩
  rcases h2 with ⟨e, ceq : c = a * e⟩
  have h1 : b + c = a * (d + e) := by linarith
  show a | (b + c)
  exact Dvd.intro (d + e) h1.symm

-- 3ª demostración
example
  (h1 : a | b)
  (h2 : a | c)
  : a | (b + c) :=
by
  rcases h1 with ⟨d, beq : b = a * d⟩
  rcases h2 with ⟨e, ceq : c = a * e⟩
  show a | (b + c)
  exact Dvd.intro (d + e) (by linarith)

-- 4ª demostración
example
  (h1 : a | b)
  (h2 : a | c)
  : a | (b + c) :=
by
  cases' h1 with d beq
  -- d : ℕ
  -- beq : b = a * d
  cases' h2 with e ceq
  -- e : ℕ
  -- ceq : c = a * e
  rw [ceq, beq]
  -- ⊢ a | a * d + a * e
  use (d + e)
  -- ⊢ a * d + a * e = a * (d + e)
  ring

-- 5ª demostración
example
  (h1 : a | b)
  (h2 : a | c)
  : a | (b + c) :=
by
  rcases h1 with ⟨d, rfl⟩

```

```

--  $\vdash a \mid a * d + c$ 
rcases h2 with (e, rfl)
--  $\vdash a \mid a * d + a * e$ 
use (d + e)
--  $\vdash a * d + a * e = a * (d + e)$ 
ring

-- 6ª demostración
example
  (h1 :  $a \mid b$ )
  (h2 :  $a \mid c$ )
  :  $a \mid (b + c) :=$ 
dvd_add h1 h2

-- Lemas usados
-- =====

-- #check (Dvd.intro c :  $a * c = b \rightarrow a \mid b$ )
-- #check (dvd_add :  $a \mid b \rightarrow a \mid c \rightarrow a \mid (b + c)$ )
-- #check (mul_add a b c :  $a * (b + c) = a * b + a * c$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.5. Conmutatividad del máximo común divisor

```

-----
-- Demostrar que si  $m$  y  $n$  son números naturales, entonces
--  $\gcd m n = \gcd n m$ 
-----

-- Demostración en lenguaje natural
-- =====

-- Es consecuencia del siguiente lema auxiliar
--  $(\forall x, y \in \mathbb{N})[\gcd(x, y) \mid \gcd(y, x)]$  (1)
-- En efecto, sustituyendo en (1)  $x$  por  $m$  e  $y$  por  $n$ , se tiene
--  $\gcd(m, n) \mid \gcd(n, m)$  (2)
-- y sustituyendo en (1)  $x$  por  $n$  e  $y$  por  $m$ , se tiene
--  $\gcd(n, m) \mid \gcd(m, n)$  (3)
-- Finalmente, aplicando la propiedad antisimétrica de la divisibilidad
-- a (2) y (3), se tiene

```

```

--      gcd(m, n) = gcd(n, m)
--
-- Para demostrar (1), por la definición del máximo común divisor, basta
-- demostrar las siguientes relaciones
--      gcd(m, n) | n
--      gcd(m, n) | m
-- y ambas se tienen por la definición del máximo común divisor.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable (k m n : ℕ)

open Nat

-- 1ª demostración del lema auxiliar
lemma aux : gcd m n | gcd n m :=
by
  have h1 : gcd m n | n :=
    gcd_dvd_right m n
  have h2 : gcd m n | m :=
    gcd_dvd_left m n
  show gcd m n | gcd n m
  exact dvd_gcd h1 h2

-- 2ª demostración del lema auxiliar
example : gcd m n | gcd n m :=
dvd_gcd (gcd_dvd_right m n) (gcd_dvd_left m n)

-- 1ª demostración
example : gcd m n = gcd n m :=
by
  have h1 : gcd m n | gcd n m := aux m n
  have h2 : gcd n m | gcd m n := aux n m
  show gcd m n = gcd n m
  exact _root_.dvd_antisymm h1 h2

-- 2ª demostración
example : gcd m n = gcd n m :=
by
  apply _root_.dvd_antisymm
  { exact aux m n }
  { exact aux n m }

```



```

-- 3ª demostración
example : gcd m n = gcd n m :=
  _root_.dvd_antisymm (aux m n) (aux n m)

-- 4ª demostración
example : gcd m n = gcd n m :=
  -- by apply?
  gcd_comm m n

-- Lemas usados
-- =====

-- #check (_root_.dvd_antisymm : m | n → n | m → m = n)
-- #check (dvd_gcd : k | m → k | n → k | gcd m n)
-- #check (gcd_comm m n : gcd m n = gcd n m)
-- #check (gcd_dvd_left m n : gcd m n | m)
-- #check (gcd_dvd_right m n : gcd m n | n)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.6. Si $(m \mid n \wedge m \neq n)$ , entonces $(m \mid n \wedge \neg(n \mid m))$

```

-----
-- Sean m y n números naturales. Demostrar que si
--   m | n ∧ m ≠ n
-- entonces
--   m | n ∧ ¬(n | m)
-----

-- Demostración en lenguaje natural
-- =====

-- La primera parte de la conclusión coincide con la primera de la
-- hipótesis. Nos queda demostrar la segunda parte; es decir, que
-- ¬(n | m). Para ello, supongamos que n | m. Entonces, por la propiedad
-- antisimétrica de la divisibilidad y la primera parte de la hipótesis,
-- se tiene que m = n en contradicción con la segunda parte de la
-- hipótesis.

-- Demostraciones con Lean4
-- =====

```

```

import Mathlib.Data.Nat.GCD.Basic

variable {m n : ℕ}

-- 1ª demostración
-- =====

example
  (h : m | n ∧ m ≠ n)
  : m | n ∧ ¬ n | m :=
by
  constructor
  . show m | n
    exact h.left
  . show ¬n | m
    { intro (h1 : n | m)
      have h2 : m = n := dvd_antisymm h.left h1
      show False
      exact h.right h2 }

-- 2ª demostración
-- =====

example
  (h : m | n ∧ m ≠ n)
  : m | n ∧ ¬ n | m :=
by
  constructor
  . exact h.left
  . intro (h1 : n | m)
    exact h.right (dvd_antisymm h.left h1)

-- 3ª demostración
-- =====

example
  (h : m | n ∧ m ≠ n)
  : m | n ∧ ¬ n | m :=
(h.left, fun h1 ↦ h.right (dvd_antisymm h.left h1))

-- 4ª demostración
-- =====

example

```

```

(h : m | n ∧ m ≠ n)
: m | n ∧ ¬ n | m :=
by
  cases' h with h1 h2
  -- h1 : m | n
  -- h2 : m ≠ n
  constructor
  . -- ⊢ m | n
    exact h1
  . -- ⊢ ¬n | m
    contrapose! h2
    -- h2 : n | m
    -- ⊢ m = n
    apply dvd_antisymm h1 h2

-- 5ª demostración
-- =====

example
  (h : m | n ∧ m ≠ n)
  : m | n ∧ ¬ n | m :=
by
  rcases h with ⟨h1 : m | n, h2 : m ≠ n⟩
  constructor
  . -- ⊢ m | n
    exact h1
  . -- ⊢ ¬n | m
    contrapose! h2
    -- h2 : n | m
    -- ⊢ m = n
    apply dvd_antisymm h1 h2

-- Lemas usados
-- =====

-- #check (dvd_antisymm : m | n → n | m → m = n)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.7. Existen números primos $m$ y $n$ tales que $4 < m < n < 10$

```

-----
-- Demostrar que existen números primos  $m$  y  $n$  tales que
--  $4 < m < n < 10$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Basta considerar los números 5 y 7, ya que son primos y
--  $4 < 5 < 7 < 10$ .

-- Demostración con Lean4
-- =====

import Mathlib.Data.Nat.Prime
import Mathlib.Tactic

example :
   $\exists m\ n : \mathbb{N}, 4 < m \wedge m < n \wedge n < 10 \wedge \text{Nat.Prime } m \wedge \text{Nat.Prime } n :=$ 
by
  use 5, 7
  norm_num

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.8. 3 divide al máximo común divisor de 6 y 15

```

-----
-- Demostrar que 3 divide al máximo común divisor de 6 y 15.
-----

-- Demostración en lenguaje natural
-- =====

-- Se usará el siguiente lema
--  $(\forall k, m, n \in \mathbb{N})[k \mid \text{gcd } m\ n \leftrightarrow k \mid m \wedge k \mid n]$ 
--

```

```

-- Por el lema,
--   3 | gcd 6 15
-- se reduce a
--   3 | 6 ∧ 3 | 15
-- que se verifican fácilmente.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
import Mathlib.Data.Nat.GCD.Basic

open Nat

-- 1ª demostración
-- =====

example : 3 | gcd 6 15 :=
by
  rw [dvd_gcd_iff]
  -- ⊢ 3 | 6 ∧ 3 | 15
  constructor
  . -- 3 | 6
    norm_num
  . -- ⊢ 3 | 15
    norm_num

-- 2ª demostración
-- =====

example : 3 | gcd 6 15 :=
by
  rw [dvd_gcd_iff]
  -- ⊢ 3 | 6 ∧ 3 | 15
  constructor <|> norm_num

-- Lemas usados
-- =====

-- variable (k m n : ℕ)
-- #check (dvd_gcd_iff : k | gcd m n ↔ k | m ∧ k | n)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.9. Si $m$ divide a $n$ o a $k$ , entonces $m$ divide a $nk$

```

-----
-- Demostrar que si  $m$  divide a  $n$  o a  $k$ , entonces  $m$  divide a  $nk$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Se demuestra por casos.
--
-- Caso 1: Supongamos que  $m \mid n$ . Entonces, existe un  $a \in \mathbb{N}$  tal que
--    $n = ma$ 
-- Por tanto,
--    $nk = (ma)k$ 
--    $= m(ak)$ 
-- que es divisible por  $m$ .
--
-- Caso 2: Supongamos que  $m \mid k$ . Entonces, existe un  $b \in \mathbb{N}$  tal que
--    $k = mb$ 
-- Por tanto,
--    $nk = n(mb)$ 
--    $= m(nb)$ 
-- que es divisible por  $m$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable {m n k : ℕ}

-- 1ª demostración
-- =====

example
  (h : m ∣ n ∨ m ∣ k)
  : m ∣ n * k :=
by
  rcases h with h1 | h2
  . -- h1 : m ∣ n
    rcases h1 with ⟨a, ha⟩
    -- a : ℕ
    -- ha : n = m * a

```

```

rw [ha]
--  $\vdash m \mid (m * a) * k$ 
rw [mul_assoc]
--  $\vdash m \mid m * (a * k)$ 
exact dvd_mul_right m (a * k)
. --  $h2 : m \mid k$ 
rcases h2 with (b, hb)
--  $b : \mathbb{N}$ 
--  $hb : k = m * b$ 
rw [hb]
--  $\vdash m \mid n * (m * b)$ 
rw [mul_comm]
--  $\vdash m \mid (m * b) * n$ 
rw [mul_assoc]
--  $\vdash m \mid m * (b * n)$ 
exact dvd_mul_right m (b * n)

```

-- 2ª demostración

-- =====

example

```

(h : m  $\mid$  n  $\vee$  m  $\mid$  k)
: m  $\mid$  n * k :=

```

by

```

rcases h with h1  $\mid$  h2
. --  $h1 : m \mid n$ 
rcases h1 with (a, ha)
--  $a : \mathbb{N}$ 
--  $ha : n = m * a$ 
rw [ha, mul_assoc]
--  $\vdash m \mid m * (a * k)$ 
exact dvd_mul_right m (a * k)
. --  $h2 : m \mid k$ 
rcases h2 with (b, hb)
--  $b : \mathbb{N}$ 
--  $hb : k = m * b$ 
rw [hb, mul_comm, mul_assoc]
--  $\vdash m \mid m * (b * n)$ 
exact dvd_mul_right m (b * n)

```

-- 3ª demostración

-- =====

example

```

(h : m  $\mid$  n  $\vee$  m  $\mid$  k)

```

```

: m | n * k :=
by
  rcases h with ⟨a, rfl⟩ | ⟨b, rfl⟩
  . -- a : ℕ
    -- ⊢ m | (m * a) * k
    rw [mul_assoc]
    -- ⊢ m | m * (a * k)
    exact dvd_mul_right m (a * k)
  . -- ⊢ m | n * (m * b)
    rw [mul_comm, mul_assoc]
    -- ⊢ m | m * (b * n)
    exact dvd_mul_right m (b * n)

-- 4ª demostración
-- =====

example
  (h : m | n ∨ m | k)
  : m | n * k :=
by
  rcases h with h1 | h2
  . -- h1 : m | n
    exact dvd_mul_of_dvd_left h1 k
  . -- h2 : m | k
    exact dvd_mul_of_dvd_right h2 n

-- Lemas usados
-- =====

-- #check (dvd_mul_of_dvd_left : m | n → ∀ (c : ℕ), m | n * c)
-- #check (dvd_mul_of_dvd_right : m | n → ∀ (c : ℕ), m | c * n)
-- #check (dvd_mul_right m n : m | m * n)
-- #check (mul_assoc m n k : m * n * k = m * (n * k))
-- #check (mul_comm m n : m * n = n * m)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.10. Existen infinitos números primos

```

-----
-- Demostrar que hay infinitos números primos.
-----

```



```

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas de los números naturales
--    $n \neq 1 \rightarrow$  el menor factor primo de  $n$  es primo           (L1)
--    $n! > 0$                                                          (L2)
--    $0 < k \rightarrow n < k + n$                                      (L3)
--    $k < n \rightarrow n \neq k$                                          (L4)
--    $k \not\leq n \rightarrow k \leq n$                                      (L5)
--    $0 < k \rightarrow k \leq n \rightarrow k \mid n!$                  (L6)
--    $0 < \text{minFac}(n)$                                              (L7)
--    $k \mid m \rightarrow (k \mid n \leftrightarrow k \mid m + n)$    (L8)
--    $\text{minFac}(n) \mid n$                                              (L9)
--    $\text{Prime}(n) \rightarrow \neg n \mid 1$                                    (L10)
--
-- Sea  $p$  el menor factor primo de  $n! + 1$ . Tenemos que demostrar que  $n \leq$ 
--  $p$  y que  $p$  es primo.
--
-- Para demostrar que  $p$  es primo, por el lema L1, basta demostrar que
--    $n! + 1 \neq 1$ 
-- Su demostración es
--    $n! > 0$  [por L2]
--    $\implies n! + 1 > 1$  [por L3]
--    $\implies n! + 1 \neq 1$  [por L4]
--
-- Para demostrar  $n \leq p$ , por el lema L5, basta demostrar que
--    $n \not\leq p$ 
-- Su demostración es
--    $n \geq p$ 
--    $\implies p \mid n!$  [por L6 y L7]
--    $\implies p \mid 1$  [por L8 y  $(p \mid n! + 1)$  por L9]
--    $\implies \text{Falso}$  [por L10 y  $p$  es primo]

-- Demostración con Lean4
-- =====

import Mathlib.Tactic
import Mathlib.Data.Nat.Prime
open Nat

-- 1ª demostración
-- =====

example
  (n : ℕ) :

```

```

    ∃ p, n ≤ p ∧ Nat.Prime p :=
by
  let p := minFac (n ! + 1)
  have h1 : Nat.Prime p := by
    apply minFac_prime
    -- ⊢ n ! + 1 ≠ 1
    have h3 : n ! > 0      := factorial_pos n
    have h4 : n ! + 1 > 1 := Nat.lt_add_of_pos_left h3
    show n ! + 1 ≠ 1
    exact Nat.ne_of_gt h4
  use p
  constructor
  . -- ⊢ n ≤ p
    apply le_of_not_ge
    -- ⊢ ¬n ≥ p
    intro h5
    -- h5 : n ≥ p
    -- ⊢ False
    have h6 : p | n ! := dvd_factorial (minFac_pos _) h5
    have h7 : p | 1   := (Nat.dvd_add_iff_right h6).mpr (minFac_dvd _)
    show False
    exact (Nat.Prime.not_dvd_one h1) h7
  . -- ⊢ Nat.Prime p
    exact h1
done

-- 2ª demostración
-- =====

example
  (n : ℕ) :
  ∃ p, n ≤ p ∧ Nat.Prime p :=
exists_infinite_primes n

-- Lemas usados
-- =====

-- variable (k m n : ℕ)
-- #check (Nat.Prime.not_dvd_one : Nat.Prime n → ¬n | 1)
-- #check (Nat.dvd_add_iff_right : k | m → (k | n ↔ k | m + n))
-- #check (Nat.dvd_one : n | 1 ↔ n = 1)
-- #check (Nat.lt_add_of_pos_left : 0 < k → n < k + n)
-- #check (Nat.ne_of_gt : k < n → n ≠ k)
-- #check (dvd_factorial : 0 < k → k ≤ n → k | n !)
-- #check (factorial_pos n : n ! > 0)

```

```
-- #check (le_of_not_ge : ¬k ≥ n → k ≤ n)
-- #check (minFac_dvd n : minFac n | n)
-- #check (minFac_pos n : 0 < minFac n)
-- #check (minFac_prime : n ≠ 1 → Nat.Prime (minFac n))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.11. Si $n^2$ es par, entonces $n$ es par

```
-- -----
-- Demostrar que si  $n^2$  es par, entonces  $n$  es par.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se usara el siguiente lema: "Si  $p$  es primo, entonces
--    $(\forall a, b \in \mathbb{N})[p \mid ab \leftrightarrow p \mid a \vee p \mid b]$ .
--
-- Si  $n^2$  es par, entonces 2 divide a  $n \cdot n$  y, por el lema, 2 divide a  $n$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
open Nat
variable (n : ℕ)

-- 1ª demostración
-- =====

example
  (h : 2 | n ^ 2)
  : 2 | n :=
by
  rw [pow_two] at h
  -- h : 2 | n * n
  have h1 : Nat.Prime 2 := prime_two
  have h2 : 2 | n ∨ 2 | n := (Prime.dvd_mul h1).mp h
  rcases h2 with h3 | h4
  · -- h3 : 2 | n
    exact h3
  · -- h4 : 2 | n
```

```

    exact h4
  done

-- 2ª demostración
-- =====

example
  (h : 2 | n ^ 2)
  : 2 | n :=
by
  rw [pow_two] at h
  -- h : 2 | n * n
  have h2 : 2 | n v 2 | n := (Prime.dvd_mul prime_two).mp h
  rcases h2 with h3 | h4
  · exact h3
  · exact h4
done

-- 3ª demostración
-- =====

example
  (h : 2 | n ^ 2)
  : 2 | n :=
by
  rw [pow_two] at h
  -- h : 2 | n * n
  have h2 : 2 | n v 2 | n := (Prime.dvd_mul prime_two).mp h
  tauto
done

-- Lemas usados
-- =====

-- variable (p a b : ℕ)
-- #check (prime_two : Nat.Prime 2)
-- #check (Prime.dvd_mul : Nat.Prime p → (p | a * b ↔ p | a v p | b))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 7.12. La raíz cuadrada de 2 es irracional

```

-----
-- Demostrar que la raíz cuadrada de 2 es irracional; es decir, que no
-- existen  $m, n \in \mathbb{N}$  tales que  $m$  y  $n$  son coprimos (es decir, que no
-- factores comunes distintos de uno) y  $m^2 = 2n^2$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Usaremos el lema del ejercicio anterior:
--  $(\forall n \in \mathbb{N})[2 \mid n^2 \rightarrow 2 \mid n]$ 
--
-- Supongamos que existen  $m, n \in \mathbb{N}$  tales que  $m$  y  $n$  son coprimos y
--  $m^2 = 2n^2$  y tenemos que demostrar una contradicción. Puesto que 2 no
-- divide a 1, para tener la contradicción basta demostrar que 2 divide
-- a 1 y (puesto que  $m$  y  $n$  son coprimos), para ello es suficiente
-- demostrar que 2 divide al máximo común divisor de  $m$  y  $n$ . En
-- definitiva, basta demostrar que 2 divide a  $m$  y a  $n$ .
--
-- La demostración de que 2 divide a  $m$  es
--  $m^2 = 2n^2 \implies 2 \mid m^2$ 
--  $\implies 2 \mid m$  [por el lema]
--
-- Para demostrar que 2 divide a  $n$ , observamos que, puesto que 2 divide
-- a  $m$ , existe un  $k \in \mathbb{N}$  tal que  $m = 2k$ . Sustituyendo en
--  $m^2 = 2n^2$ 
-- se tiene
--  $(2k)^2 = 2n^2$ 
-- Simplificando, queda
--  $2k = n^2$ 
-- Por tanto, 2 divide a  $n^2$  y, por el lema, 2 divide a  $n$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
import Mathlib.Data.Nat.Prime
import Std.Data.Nat.Gcd
open Nat
variable {m n : ℕ}

lemma par_si_cuadrado_par

```

```

(h : 2 | n ^ 2)
: 2 | n :=
by
  rw [pow_two] at h
  -- h : 2 | n * n
  have h2 : 2 | n v 2 | n := (Prime.dvd_mul prime_two).mp h
  tauto

example : ¬∃ m n, coprime m n ∧ m ^ 2 = 2 * n ^ 2 :=
by
  rintro ⟨m, n, ⟨h1, h2⟩⟩
  -- m n : ℕ
  -- h1 : coprime m n
  -- h2 : m ^ 2 = 2 * n ^ 2
  -- ⊢ False
  have h3 : ¬(2 | 1) := by norm_num
  have h4 : 2 | 1 := by
    have h5 : Nat.gcd m n = 1 := h1
    rw [← h5]
    -- ⊢ 2 | Nat.gcd m n
  have h6 : 2 | m := by
    apply par_si_cuadrado_par
    -- ⊢ 2 | m ^ 2
    rw [h2]
    -- ⊢ 2 | 2 * n ^ 2
    exact Nat.dvd_mul_right 2 (n ^ 2)
  have h7 : 2 | n := by
    have h8 : ∃ k, m = 2 * k := h6
    rcases h8 with ⟨k, h9⟩
    -- k : ℕ
    -- h9 : m = 2 * k
    have h10 : 2 * k ^ 2 = n ^ 2 := by
      have h10a : 2 * (2 * k ^ 2) = 2 * n ^ 2 := calc
        2 * (2 * k ^ 2) = (2 * k) ^ 2 := by nlinarith
        _ = m ^ 2 := by rw [← h9]
        _ = 2 * n ^ 2 := h2
      show 2 * k ^ 2 = n ^ 2
    exact (mul_right_inj' (by norm_num : 2 ≠ 0)).mp h10a
  have h11 : 2 | n ^ 2 := by
    rw [← h10]
    -- ⊢ 2 | 2 * k ^ 2
    exact Nat.dvd_mul_right 2 (k ^ 2)
  show 2 | n
  exact par_si_cuadrado_par h11
  show 2 | Nat.gcd m n

```

```
exact Nat.dvd_gcd h6 h7
show False
exact h3 h4

-- Lemas usados
-- =====

-- variable (p k : ℕ)
-- #check (pow_two n : n ^ 2 = n * n)
-- #check (Prime.dvd_mul : Nat.Prime p → (p | m * n ↔ p | m ∨ p | n))
-- #check (prime_two : Nat.Prime 2)
-- #check (Nat.dvd_gcd : k | m → k | n → k | Nat.gcd m n)
-- #check (Nat.dvd_mul_right m n : m | m * n)
-- #check (mul_right_inj' : k ≠ 0 → (k * m = k * n ↔ m = n))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)





# Capítulo 8

## Retículos

### 8.1. En los retículos, $x \sqcap y = y \sqcap x$

```
-- -----  
-- Demostrar que en los retículos se verifica que  
--       $x \sqcap y = y \sqcap x$   
-- -----  
  
-- Demostración en lenguaje natural  
-- =====  
  
-- Es consecuencia del siguiente lema auxiliar  
--       $(\forall a, b)[a \sqcap b \leq b \sqcap a]$  (1)  
-- En efecto, sustituyendo en (1)  $a$  por  $x$  y  $b$  por  $y$ , se tiene  
--       $x \sqcap y \leq y \sqcap x$  (2)  
-- y sustituyendo en (1)  $a$  por  $y$  y  $b$  por  $x$ , se tiene  
--       $y \sqcap x \leq x \sqcap y$  (3)  
-- Finalmente, aplicando la propiedad antisimétrica de la divisibilidad  
-- a (2) y (3), se tiene  
--       $x \sqcap y = y \sqcap x$   
--  
-- Para demostrar (1), por la definición del ínfimo, basta demostrar  
-- las siguientes relaciones  
--       $y \sqcap x \leq x$   
--       $y \sqcap x \leq y$   
-- y ambas se tienen por la definición del ínfimo.  
  
-- Demostraciones con Lean4  
-- =====  
  
import Mathlib.Order.Lattice
```

```

variable {α : Type _} [Lattice α]
variable (x y z : α)

-- 1ª demostración del lema auxiliar
lemma aux : x ⊓ y ≤ y ⊓ x :=
by
  have h1 : x ⊓ y ≤ y :=
    inf_le_right
  have h2 : x ⊓ y ≤ x :=
    inf_le_left
  show x ⊓ y ≤ y ⊓ x
  exact le_inf h1 h2

-- 2ª demostración del lema auxiliar
example : x ⊓ y ≤ y ⊓ x :=
by
  apply le_inf
  { apply inf_le_right }
  { apply inf_le_left }

-- 3ª demostración del lema auxiliar
example : x ⊓ y ≤ y ⊓ x :=
le_inf inf_le_right inf_le_left

-- 1ª demostración
example : x ⊓ y = y ⊓ x :=
by
  have h1 : x ⊓ y ≤ y ⊓ x :=
    aux x y
  have h2 : y ⊓ x ≤ x ⊓ y :=
    aux y x
  show x ⊓ y = y ⊓ x
  exact le_antisymm h1 h2

-- 2ª demostración
example : x ⊓ y = y ⊓ x :=
by
  apply le_antisymm
  { apply aux }
  { apply aux }

-- 3ª demostración
example : x ⊓ y = y ⊓ x :=
le_antisymm (aux x y) (aux y x)

```

```

-- 4ª demostración
example : x ⊓ y = y ⊓ x :=
by apply le_antisymm; simp ; simp

-- 5ª demostración
example : x ⊓ y = y ⊓ x :=
-- by apply?
inf_comm

-- Lemas usados
-- =====

-- #check (inf_comm : x ⊓ y = y ⊓ x)
-- #check (inf_le_left : x ⊓ y ≤ x)
-- #check (inf_le_right : x ⊓ y ≤ y)
-- #check (le_antisymm : x ≤ y → y ≤ x → x = y)
-- #check (le_inf : z ≤ x → z ≤ y → z ≤ x ⊓ y)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 8.2. En los retículos, $x \sqcup y = y \sqcup x$

```

-----
-- Demostrar que en los retículos se verifica que
--   x ⊔ y = y ⊔ x
-- para todo x e y en el retículo.
-----

-- Demostración en lenguaje natural
-- =====

-- Es consecuencia del siguiente lema auxiliar
--   (∀ a, b)[a ⊔ b ≤ b ⊔ a]                                     (1)
-- En efecto, sustituyendo en (1) a por x y b por y, se tiene
--   x ⊔ y ≤ y ⊔ x                                             (2)
-- y sustituyendo en (1) a por y y b por x, se tiene
--   y ⊔ x ≤ x ⊔ y                                             (3)
-- Finalmente, aplicando la propiedad antisimétrica de la divisibilidad
-- a (2) y (3), se tiene
--   x ⊔ y = y ⊔ x
--
-- Para demostrar (1), por la definición del supremo, basta demostrar
-- las siguientes relaciones

```

```

--       $x \leq y \sqcup x$ 
--       $y \leq y \sqcup x$ 
-- y ambas se tienen por la definición del supremo.

-- Demostraciones con Lean4
-- =====

import Mathlib.Order.Lattice
variable {α : Type _} [Lattice α]
variable (x y z : α)

-- 1ª demostración del lema auxiliar
lemma aux :  $x \sqcup y \leq y \sqcup x$  :=
by
  have h1 :  $x \leq y \sqcup x$  :=
    le_sup_right
  have h2 :  $y \leq y \sqcup x$  :=
    le_sup_left
  show  $x \sqcup y \leq y \sqcup x$ 
  exact sup_le h1 h2

-- 2ª demostración del lema auxiliar
example :  $x \sqcup y \leq y \sqcup x$  :=
by
  apply sup_le
  { apply le_sup_right }
  { apply le_sup_left }

-- 3ª demostración del lema auxiliar
example :  $x \sqcup y \leq y \sqcup x$  :=
sup_le le_sup_right le_sup_left

-- 1ª demostración
example :  $x \sqcup y = y \sqcup x$  :=
by
  have h1 :  $x \sqcup y \leq y \sqcup x$  :=
    aux x y
  have h2 :  $y \sqcup x \leq x \sqcup y$  :=
    aux y x
  show  $x \sqcup y = y \sqcup x$ 
  exact le_antisymm h1 h2

-- 2ª demostración
example :  $x \sqcup y = y \sqcup x$  :=
by

```

```

apply le_antisymm
{ apply aux }
{ apply aux }

-- 3ª demostración
example : x ⊔ y = y ⊔ x :=
le_antisymm (aux x y) (aux y x)

-- 4ª demostración
example : x ⊔ y = y ⊔ x :=
by apply le_antisymm; simp ; simp

-- 5ª demostración
example : x ⊔ y = y ⊔ x :=
-- by apply?
sup_comm

-- Lemas usados
-- =====

-- #check (le_antisymm : x ≤ y → y ≤ x → x = y)
-- #check (le_sup_left : x ≤ x ⊔ y)
-- #check (le_sup_right : y ≤ x ⊔ y)
-- #check (sup_comm : x ⊔ y = y ⊔ x)
-- #check (sup_le : x ≤ z → y ≤ z → x ⊔ y ≤ z)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 8.3. En los retículos, $(x \sqcap y) \sqcap z = x \sqcap (y \sqcap z)$

```

-----
-- Demostrar que en los retículos se verifica que
--      (x ⊓ y) ⊓ z = x ⊓ (y ⊓ z)
-----

-- Demostración en lenguaje natural
-- =====

-- En la demostración se usarán los siguientes lemas
--   le_antisymm : x ≤ y → y ≤ x → x = y
--   le_inf      : z ≤ x → z ≤ y → z ≤ x ⊓ y
--   inf_le_left : x ⊓ y ≤ x
--   inf_le_right : x ⊓ y ≤ y)

```

```

--
-- Por le_antisym, es suficiente demostrar las siguientes relaciones:
--    $(x \sqcap y) \sqcap z \leq x \sqcap (y \sqcap z)$  (1)
--    $x \sqcap (y \sqcap z) \leq (x \sqcap y) \sqcap z$  (2)
--
-- Para demostrar (1), por le_inf, basta probar que
--    $(x \sqcap y) \sqcap z \leq x$  (1a)
--    $(x \sqcap y) \sqcap z \leq y \sqcap z$  (1b)
--
-- La (1a) se demuestra por la siguiente cadena de desigualdades
--    $(x \sqcap y) \sqcap z \leq x \sqcap y$  [por inf_le_left]
--    $\leq x$  [por inf_le_left]
--
-- Para demostrar (1b), por le_inf, basta probar que
--    $(x \sqcap y) \sqcap z \leq y$  (1b1)
--    $(x \sqcap y) \sqcap z \leq z$  (1b2)
--
-- La (1b1) se demuestra por la siguiente cadena de desigualdades
--    $(x \sqcap y) \sqcap z \leq x \sqcap y$  [por inf_le_left]
--    $\leq y$  [por inf_le_right]
--
-- La (1b2) se tiene por inf_le_right.
--
-- Para demostrar (2), por le_inf, basta probar que
--    $x \sqcap (y \sqcap z) \leq x \sqcap y$  (2a)
--    $x \sqcap (y \sqcap z) \leq z$  (2b)
--
-- Para demostrar (2a), por le_inf, basta probar que
--    $x \sqcap (y \sqcap z) \leq x$  (2a1)
--    $x \sqcap (y \sqcap z) \leq y$  (2a2)
--
-- La (2a1) se tiene por inf_le_left.
--
-- La (2a2) se demuestra por la siguiente cadena de desigualdades
--    $x \sqcap (y \sqcap z) \leq y \sqcap z$  [por inf_le_right]
--    $\leq y$  [por inf_le_left]
--
-- La (2b) se demuestra por la siguiente cadena de desigualdades
--    $x \sqcap (y \sqcap z) \leq y \sqcap z$  [por inf_le_right]
--    $\leq z$  [por inf_le_right]
--
-- Demostraciones con Lean4
-- =====
import Mathlib.Order.Lattice

```

```

variable {α : Type _} [Lattice α]
variable (x y z : α)

-- 1ª demostración
-- =====

example : (x ⊓ y) ⊓ z = x ⊓ (y ⊓ z) :=
by
  have h1 : (x ⊓ y) ⊓ z ≤ x ⊓ (y ⊓ z) := by
    { have h1a : (x ⊓ y) ⊓ z ≤ x := calc
      (x ⊓ y) ⊓ z ≤ x ⊓ y := by exact inf_le_left
      _ ≤ x := by exact inf_le_left
    have h1b : (x ⊓ y) ⊓ z ≤ y ⊓ z := by
      { have h1b1 : (x ⊓ y) ⊓ z ≤ y := calc
        (x ⊓ y) ⊓ z ≤ x ⊓ y := by exact inf_le_left
        _ ≤ y := by exact inf_le_right
      have h1b2 : (x ⊓ y) ⊓ z ≤ z :=
        inf_le_right
      show (x ⊓ y) ⊓ z ≤ y ⊓ z
      exact le_inf h1b1 h1b2 }
    show (x ⊓ y) ⊓ z ≤ x ⊓ (y ⊓ z)
    exact le_inf h1a h1b }
  have h2 : x ⊓ (y ⊓ z) ≤ (x ⊓ y) ⊓ z := by
    { have h2a : x ⊓ (y ⊓ z) ≤ x ⊓ y := by
      { have h2a1 : x ⊓ (y ⊓ z) ≤ x :=
        inf_le_left
      have h2a2 : x ⊓ (y ⊓ z) ≤ y := calc
        x ⊓ (y ⊓ z) ≤ y ⊓ z := by exact inf_le_right
        _ ≤ y := by exact inf_le_left
      show x ⊓ (y ⊓ z) ≤ x ⊓ y
      exact le_inf h2a1 h2a2 }
    have h2b : x ⊓ (y ⊓ z) ≤ z := by calc
      x ⊓ (y ⊓ z) ≤ y ⊓ z := by exact inf_le_right
      _ ≤ z := by exact inf_le_right
    show x ⊓ (y ⊓ z) ≤ (x ⊓ y) ⊓ z
    exact le_inf h2a h2b }
  show (x ⊓ y) ⊓ z = x ⊓ (y ⊓ z)
  exact le_antisymm h1 h2

-- 2ª demostración
-- =====

example : x ⊓ y ⊓ z = x ⊓ (y ⊓ z) := by
  apply le_antisymm
  · apply le_inf

```

```

    · apply le_trans
      apply inf_le_left
      apply inf_le_left
    · apply le_inf
      · apply le_trans
        apply inf_le_left
        apply inf_le_right
      · apply inf_le_right
  · apply le_inf
    · apply le_inf
      · apply inf_le_left
    · apply le_trans
      apply inf_le_right
      apply inf_le_left
  · apply le_trans
    apply inf_le_right
    apply inf_le_right

-- 3ª demostración
-- =====

example : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z) :=
by
  apply le_antisymm
  · apply le_inf
    · apply inf_le_of_left_le inf_le_left
    · apply le_inf (inf_le_of_left_le inf_le_right) inf_le_right
  · apply le_inf
    · apply le_inf inf_le_left (inf_le_of_right_le inf_le_left)
    · apply inf_le_of_right_le inf_le_right

-- 4ª demostración
-- =====

example : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z) :=
le_antisymm
  (le_inf
    (inf_le_of_left_le inf_le_left)
    (le_inf (inf_le_of_left_le inf_le_right) inf_le_right))
  (le_inf
    (le_inf inf_le_left (inf_le_of_right_le inf_le_left))
    (inf_le_of_right_le inf_le_right))

-- 5ª demostración
-- =====

```



```

example : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z) :=
-- by apply?
inf_assoc

-- Lemas usados
-- =====

-- #check (inf_assoc : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z))
-- #check (inf_le_left : x ⊔ y ≤ x)
-- #check (inf_le_of_left_le : x ≤ z → x ⊔ y ≤ z)
-- #check (inf_le_of_right_le : y ≤ z → x ⊔ y ≤ z)
-- #check (inf_le_right : x ⊔ y ≤ y)
-- #check (le_antisymm : x ≤ y → y ≤ x → x = y)
-- #check (le_inf : z ≤ x → z ≤ y → z ≤ x ⊔ y)
-- #check (le_trans : x ≤ y → y ≤ z → x ≤ z)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 8.4. En los retículos, $(x \sqcup y) \sqcup z = x \sqcup (y \sqcup z)$

```

-----
-- Demostrar que en los retículos se verifica que
--   (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z)
-----

-- Demostración en lenguaje natural
-- =====

-- En la demostración se usarán los siguientes lemas
--   le_antisymm : x ≤ y → y ≤ x → x = y
--   le_sup_left  : x ≤ x ⊔ y
--   le_sup_right : y ≤ x ⊔ y
--   sup_le       : x ≤ z → y ≤ z → x ⊔ y ≤ z
--
-- Por le_antisymm, basta demostrar las siguientes relaciones:
--   (x ⊔ y) ⊔ z ≤ x ⊔ (y ⊔ z)                                (1)
--   x ⊔ (y ⊔ z) ≤ (x ⊔ y) ⊔ z                                (2)
--
-- Para demostrar (1), por sup_le, basta probar
--   x ⊔ y ≤ x ⊔ (y ⊔ z)                                       (1a)
--   z ≤ x ⊔ (y ⊔ z)                                           (1b)
--

```

```

-- Para demostrar (1a), por sup_le, basta probar
--    $x \leq x \sqcup (y \sqcup z)$  (1a1)
--    $y \leq x \sqcup (y \sqcup z)$  (1a2)
--
-- La (1a1) se tiene por le_sup_left.
--
-- La (1a2) se tiene por la siguiente cadena de desigualdades:
--    $y \leq y \sqcup z$  [por le_sup_left]
--    $\leq x \sqcup (y \sqcup z)$  [por le_sup_right]
--
-- La (1b) se tiene por la siguiente cadena de desigualdades
--    $z \leq y \sqcup z$  [por le_sup_right]
--    $\leq x \sqcup (y \sqcup z)$  [por le_sup_right]
--
-- Para demostrar (2), por sup_le, basta probar
--    $x \leq (x \sqcup y) \sqcup z$  (2a)
--    $y \sqcup z \leq (x \sqcup y) \sqcup z$  (2b)
--
-- La (2a) se demuestra por la siguiente cadena de desigualdades:
--    $x \leq x \sqcup y$  [por le_sup_left]
--    $\leq (x \sqcup y) \sqcup z$  [por le_sup_left]
--
-- Para demostrar (2b), por sup_le, basta probar
--    $y \leq (x \sqcup y) \sqcup z$  (2b1)
--    $z \leq (x \sqcup y) \sqcup z$  (2b2)
--
-- La (2b1) se demuestra por la siguiente cadena de desigualdades:
--    $y \leq x \sqcup y$  [por le_sup_right]
--    $\leq (x \sqcup y) \sqcup z$  [por le_sup_left]
--
-- La (2b2) se tiene por le_sup_right.

-- Demostraciones con Lean 4
-- =====

import Mathlib.Order.Lattice

variable {α : Type _} [Lattice α]
variable (x y z : α)

-- 1ª demostración
-- =====

example : (x  $\sqcup$  y)  $\sqcup$  z = x  $\sqcup$  (y  $\sqcup$  z) :=
by

```

```

have h1 : (x  $\sqcup$  y)  $\sqcup$  z  $\leq$  x  $\sqcup$  (y  $\sqcup$  z) := by
{ have h1a : x  $\sqcup$  y  $\leq$  x  $\sqcup$  (y  $\sqcup$  z) := by
  { have h1a1 : x  $\leq$  x  $\sqcup$  (y  $\sqcup$  z) := by exact le_sup_left
    have h1a2 : y  $\leq$  x  $\sqcup$  (y  $\sqcup$  z) := calc
      y  $\leq$  y  $\sqcup$  z := by exact le_sup_left
      _  $\leq$  x  $\sqcup$  (y  $\sqcup$  z) := by exact le_sup_right
    show x  $\sqcup$  y  $\leq$  x  $\sqcup$  (y  $\sqcup$  z)
    exact sup_le h1a1 h1a2 }
  have h1b : z  $\leq$  x  $\sqcup$  (y  $\sqcup$  z) := calc
    z  $\leq$  y  $\sqcup$  z := by exact le_sup_right
    _  $\leq$  x  $\sqcup$  (y  $\sqcup$  z) := by exact le_sup_right
  show (x  $\sqcup$  y)  $\sqcup$  z  $\leq$  x  $\sqcup$  (y  $\sqcup$  z)
  exact sup_le h1a h1b }
have h2 : x  $\sqcup$  (y  $\sqcup$  z)  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z := by
{ have h2a : x  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z := calc
  x  $\leq$  x  $\sqcup$  y := by exact le_sup_left
  _  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z := by exact le_sup_left
  have h2b : y  $\sqcup$  z  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z := by
  { have h2b1 : y  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z := calc
    y  $\leq$  x  $\sqcup$  y := by exact le_sup_right
    _  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z := by exact le_sup_left
  have h2b2 : z  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z := by
    exact le_sup_right
  show y  $\sqcup$  z  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z
  exact sup_le h2b1 h2b2 }
  show x  $\sqcup$  (y  $\sqcup$  z)  $\leq$  (x  $\sqcup$  y)  $\sqcup$  z
  exact sup_le h2a h2b }
show (x  $\sqcup$  y)  $\sqcup$  z = x  $\sqcup$  (y  $\sqcup$  z)
exact le_antisymm h1 h2

-- 2ª demostración
-- =====

example : x  $\sqcup$  y  $\sqcup$  z = x  $\sqcup$  (y  $\sqcup$  z) :=
by
  apply le_antisymm
  · -- (x  $\sqcup$  y)  $\sqcup$  z  $\leq$  x  $\sqcup$  (y  $\sqcup$  z)
    apply sup_le
    · -- x  $\sqcup$  y  $\leq$  x  $\sqcup$  (y  $\sqcup$  z)
      apply sup_le
      · -- x  $\leq$  x  $\sqcup$  (y  $\sqcup$  z)
        apply le_sup_left
      · -- y  $\leq$  x  $\sqcup$  (y  $\sqcup$  z)
        apply le_trans
        · -- y  $\leq$  y  $\sqcup$  z

```

```

    apply @le_sup_left _ _ y z
  · --  $y \sqcup z \leq x \sqcup (y \sqcup z)$ 
    apply le_sup_right
  · --  $z \leq x \sqcup (y \sqcup z)$ 
    apply le_trans
  · --  $z \leq x \sqcup (y \sqcup z)$ 
    apply @le_sup_right _ _ y z
  · --  $y \sqcup z \leq x \sqcup (y \sqcup z)$ 
    apply le_sup_right
  · --  $x \sqcup (y \sqcup z) \leq (x \sqcup y) \sqcup z$ 
    apply sup_le
  · --  $x \leq (x \sqcup y) \sqcup z$ 
    apply le_trans
  · --  $x \leq x \sqcup y$ 
    apply @le_sup_left _ _ x y
  · --  $x \sqcup y \leq (x \sqcup y) \sqcup z$ 
    apply le_sup_left
  · --  $y \sqcup z \leq (x \sqcup y) \sqcup z$ 
    apply sup_le
  · --  $y \leq (x \sqcup y) \sqcup z$ 
    apply le_trans
  · --  $y \leq x \sqcup y$ 
    apply @le_sup_right _ _ x y
  · --  $x \sqcup y \leq (x \sqcup y) \sqcup z$ 
    apply le_sup_left
  · --  $z \leq (x \sqcup y) \sqcup z$ 
    apply le_sup_right

-- 3ª demostración
-- =====

example :  $x \sqcup y \sqcup z = x \sqcup (y \sqcup z) :=$ 
by
  apply le_antisymm
  · apply sup_le
    · apply sup_le
      · apply le_sup_left
      · apply le_trans
        · apply @le_sup_left _ _ y z
        · apply le_sup_right
    · apply le_trans
      · apply @le_sup_right _ _ y z
      · apply le_sup_right
  · apply sup_le
    · apply le_trans

```

```

    . apply @le_sup_left _ _ x y
    . apply le_sup_left
  . apply sup_le
  . apply le_trans
    . apply @le_sup_right _ _ x y
    . apply le_sup_left
    . apply le_sup_right

-- 4ª demostración
-- =====

example : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z) :=
by
  apply le_antisymm
  . -- (x ⊔ y) ⊔ z ≤ x ⊔ (y ⊔ z)
    apply sup_le
    . -- x ⊔ y ≤ x ⊔ (y ⊔ z)
      apply sup_le le_sup_left (le_sup_of_le_right le_sup_left)
    . -- z ≤ x ⊔ (y ⊔ z)
      apply le_sup_of_le_right le_sup_right
  . -- x ⊔ (y ⊔ z) ≤ (x ⊔ y) ⊔ z
    apply sup_le
    . -- x ≤ (x ⊔ y) ⊔ z
      apply le_sup_of_le_left le_sup_left
    . -- y ⊔ z ≤ (x ⊔ y) ⊔ z
      apply sup_le (le_sup_of_le_left le_sup_right) le_sup_right

-- 5ª demostración
-- =====

example : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z) :=
by
  apply le_antisymm
  . apply sup_le
    . apply sup_le le_sup_left (le_sup_of_le_right le_sup_left)
    . apply le_sup_of_le_right le_sup_right
  . apply sup_le
    . apply le_sup_of_le_left le_sup_left
    . apply sup_le (le_sup_of_le_left le_sup_right) le_sup_right

-- 6ª demostración
-- =====

example : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z) :=
le_antisymm

```

```

(sup_le
  (sup_le le_sup_left (le_sup_of_le_right le_sup_left))
  (le_sup_of_le_right le_sup_right))
(sup_le
  (le_sup_of_le_left le_sup_left)
  (sup_le (le_sup_of_le_left le_sup_right) le_sup_right))

-- 7ª demostración
-- =====

example : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z) :=
-- by apply?
sup_assoc

-- Lemas usados
-- =====

-- #check (le_antisymm : x ≤ y → y ≤ x → x = y)
-- #check (le_sup_left : x ≤ x ⊔ y)
-- #check (le_sup_of_le_left : z ≤ x → z ≤ x ⊔ y)
-- #check (le_sup_of_le_right : z ≤ y → z ≤ x ⊔ y)
-- #check (le_sup_right : y ≤ x ⊔ y)
-- #check (le_trans : x ≤ y → y ≤ z → x ≤ z)
-- #check (sup_assoc : (x ⊔ y) ⊔ z = x ⊔ (y ⊔ z))
-- #check (sup_le : x ≤ z → y ≤ z → x ⊔ y ≤ z)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 8.5. En los retículos, $x \sqcap (x \sqcup y) = x$

```

-----
-- Demostrar que en los retículos se verifica que
--   x ⊓ (x ⊔ y) = x
-- -----

-- Demostración en lenguaje natural
-- =====

-- En la demostración se usarán los siguientes lemas
--   le_antisymm : x ≤ y → y ≤ x → x = y
--   inf_le_left : x ⊓ y ≤ x
--   le_inf      : z ≤ x → z ≤ y → z ≤ x ⊓ y
--   le_refl     : x ≤ x

```

```

--      le_sup_left  :  $x \leq x \sqcup y$ 
--
-- Por le_antisymm, basta demostrar las siguientes relaciones:
--       $x \sqcap (x \sqcup y) \leq x$                                 (1)
--       $x \leq x \sqcap (x \sqcup y)$                             (2)
--
-- La (1) se tiene por inf_le_left.
--
-- Para demostrar la (2), por le_inf, basta probar las relaciones:
--       $x \leq x$                                               (2a)
--       $x \leq x \sqcup y$                                     (2b)
--
-- La (2a) se tiene por le_refl.
--
-- La (2b) se tiene por le_sup_left

-- Demostraciones con Lean4
-- =====

import Mathlib.Order.Lattice
variable {α : Type _} [Lattice α]
variable (x y : α)

-- 1ª demostración
-- =====

example :  $x \sqcap (x \sqcup y) = x$  :=
by
  have h1 :  $x \sqcap (x \sqcup y) \leq x$  := inf_le_left
  have h2 :  $x \leq x \sqcap (x \sqcup y)$ 
  { have h2a :  $x \leq x$  := le_refl
    have h2b :  $x \leq x \sqcup y$  := le_sup_left
    show  $x \leq x \sqcap (x \sqcup y)$ 
    exact le_inf h2a h2b }
  show  $x \sqcap (x \sqcup y) = x$ 
  exact le_antisymm h1 h2

-- 2ª demostración
-- =====

example :  $x \sqcap (x \sqcup y) = x$  :=
by
  have h1 :  $x \sqcap (x \sqcup y) \leq x$  := by simp
  have h2 :  $x \leq x \sqcap (x \sqcup y)$  := by simp
  show  $x \sqcap (x \sqcup y) = x$ 

```

```

exact le_antisymm h1 h2

-- 3ª demostración
-- =====

example : x  $\sqcap$  (x  $\sqcup$  y) = x :=
by
  apply le_antisymm
  . -- x  $\sqcap$  (x  $\sqcup$  y)  $\leq$  x
    apply inf_le_left
  . -- x  $\leq$  x  $\sqcap$  (x  $\sqcup$  y)
    apply le_inf
  . -- x  $\leq$  x
    apply le_refl
  . -- x  $\leq$  x  $\sqcup$  y
    apply le_sup_left

-- 4ª demostración
-- =====

example : x  $\sqcap$  (x  $\sqcup$  y) = x :=
le_antisymm inf_le_left (le_inf le_refl le_sup_left)

-- 5ª demostración
-- =====

example : x  $\sqcap$  (x  $\sqcup$  y) = x :=
-- by apply?
inf_sup_self

-- 6ª demostración
-- =====

example : x  $\sqcap$  (x  $\sqcup$  y) = x :=
by simp

-- Lemas usados
-- =====

-- variable (z :  $\alpha$ )
-- #check (inf_le_left : x  $\sqcap$  y  $\leq$  x)
-- #check (inf_sup_self : x  $\sqcap$  (x  $\sqcup$  y) = x)
-- #check (le_antisymm : x  $\leq$  y  $\rightarrow$  y  $\leq$  x  $\rightarrow$  x = y)
-- #check (le_inf : z  $\leq$  x  $\rightarrow$  z  $\leq$  y  $\rightarrow$  z  $\leq$  x  $\sqcap$  y)
-- #check (le_refl : x  $\leq$  x)

```



```
-- #check (le_sup_left : x ≤ x ∪ y)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 8.6. En los retículos, $x \sqcup (x \sqcap y) = x$

```
-----
-- Demostrar que en los retículos se verifica que
--   x ∪ (x ∩ y) = x
-- -----

-- Demostración en lenguaje natural
-- =====

-- En la demostración se usarán los siguientes lemas
--   le_antisymm : x ≤ y → y ≤ x → x = y
--   inf_le_left  : x ∩ y ≤ x
--   le_refl      : x ≤ x
--   le_sup_left  : x ≤ x ∪ y
--   sup_le       : x ≤ z → y ≤ z → x ∪ y ≤ z
--
-- Por le_antisymm, basta demostrar las siguientes relaciones:
--   x ∪ (x ∩ y) ≤ x                                     (1)
--   x ≤ x ∪ (x ∩ y)   [que se tiene por le_sup_left]
--
-- Para demostrar (1), por sup_le, basta probar las relaciones:
--   x ≤ x           [que se tiene por le_refl]
--   x ∩ y ≤ x        [que se tiene por inf_le_left]
--
-- Demostraciones con Lean4
-- =====

import Mathlib.Order.Lattice
variable {α : Type _} [Lattice α]--
variable (x y : α)

-- 1ª demostración
-- =====

example : x ∪ (x ∩ y) = x :=
by
  have h1 : x ∪ (x ∩ y) ≤ x
  { have h1a : x ≤ x := le_refl
```

```

    have h1b :  $x \sqcap y \leq x$  := inf_le_left
    show  $x \sqcup (x \sqcap y) \leq x$ 
    exact sup_le h1a h1b }
  have h2 :  $x \leq x \sqcup (x \sqcap y)$  := le_sup_left
  show  $x \sqcup (x \sqcap y) = x$ 
  exact le_antisymm h1 h2

-- 2ª demostración
-- =====

example :  $x \sqcup (x \sqcap y) = x$  :=
by
  have h1 :  $x \sqcup (x \sqcap y) \leq x$  := by simp
  have h2 :  $x \leq x \sqcup (x \sqcap y)$  := by simp
  show  $x \sqcup (x \sqcap y) = x$ 
  exact le_antisymm h1 h2

-- 3ª demostración
-- =====

example :  $x \sqcup (x \sqcap y) = x$  :=
by
  apply le_antisymm
  . --  $x \sqcup (x \sqcap y) \leq x$ 
  apply sup_le
  . --  $x \leq x$ 
  apply le_refl
  . --  $x \sqcap y \leq x$ 
  apply inf_le_left
  . --  $x \leq x \sqcup (x \sqcap y)$ 
  apply le_sup_left

-- 4ª demostración
-- =====

example :  $x \sqcup (x \sqcap y) = x$  :=
-- by apply?
sup_inf_self

-- 5ª demostración
-- =====

example :  $x \sqcup (x \sqcap y) = x$  :=
by simp

```

```
-- Lemas usados
-- =====

-- variable (z :  $\alpha$ )
-- #check (le_rfl :  $x \leq x$ )
-- #check (inf_le_left :  $x \sqcap y \leq x$ )
-- #check (sup_le :  $x \leq z \rightarrow y \leq z \rightarrow x \sqcup y \leq z$ )
-- #check (le_sup_left :  $x \leq x \sqcup y$ )
-- #check (le_antisymm :  $x \leq y \rightarrow y \leq x \rightarrow x = y$ )
-- #check (sup_inf_self :  $x \sqcup (x \sqcap y) = x$ )
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 8.7. En los retículos, una distributiva del ínfimo implica la otra

```
-- -----
-- Demostrar que si  $\alpha$  es un retículo tal que
--    $\forall x y z : \alpha, x \sqcap (y \sqcup z) = (x \sqcap y) \sqcup (x \sqcap z)$ 
-- entonces
--    $(a \sqcup b) \sqcap c = (a \sqcap c) \sqcup (b \sqcap c)$ 
-- para todos los elementos de  $\alpha$ .
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se demuestra por la siguiente cadena de igualdades
--    $(a \sqcup b) \sqcap c = c \sqcap (a \sqcup b)$            [por conmutatividad de  $\sqcap$ ]
--    $= (c \sqcap a) \sqcup (c \sqcap b)$            [por la hipótesis]
--    $= (a \sqcap c) \sqcup (c \sqcap b)$            [por conmutatividad de  $\sqcap$ ]
--    $= (a \sqcap c) \sqcup (b \sqcap c)$            [por conmutatividad de  $\sqcap$ ]

-- Demostraciones con Lean4
-- =====

import Mathlib.Order.Lattice
variable { $\alpha$  : Type _} [Lattice  $\alpha$ ]
variable (a b c :  $\alpha$ )

-- 1ª demostración
example
```

```

(h : ∀ x y z : α, x ⊓ (y ⊔ z) = (x ⊓ y) ⊔ (x ⊓ z))
: (a ⊔ b) ⊓ c = (a ⊓ c) ⊔ (b ⊓ c) :=
calc
  (a ⊔ b) ⊓ c = c ⊓ (a ⊔ b)          := by rw [inf_comm]
    _ = (c ⊓ a) ⊔ (c ⊓ b) := by rw [h]
    _ = (a ⊓ c) ⊔ (c ⊓ b) := by rw [@inf_comm _ _ c a]
    _ = (a ⊓ c) ⊔ (b ⊓ c) := by rw [@inf_comm _ _ c b]

-- 2ª demostración
example
  (h : ∀ x y z : α, x ⊓ (y ⊔ z) = (x ⊓ y) ⊔ (x ⊓ z))
  : (a ⊔ b) ⊓ c = (a ⊓ c) ⊔ (b ⊓ c) :=
by simp [h, inf_comm]

-- Lemas usados
-- =====

-- #check (inf_comm : a ⊓ b = b ⊓ a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 8.8. En los retículos, una distributiva del supremos implica la otra

```

-- -----
-- Demostrar que si α es un retículo tal que
--   ∀ x y z : α, x ⊔ (y ⊓ z) = (x ⊔ y) ⊓ (x ⊔ z)
-- entonces
--   (a ⊓ b) ⊔ c = (a ⊔ c) ⊓ (b ⊔ c)
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se demuestra por la siguiente cadena de igualdades
--   (a ⊓ b) ⊔ c = c ⊔ (a ⊓ b)          [por la conmutatividad de ⊔]
--               = (c ⊔ a) ⊓ (c ⊔ b)    [por la hipótesis]
--               = (a ⊔ c) ⊓ (c ⊔ b)    [por la conmutatividad de ⊔]
--               = (a ⊔ c) ⊓ (b ⊔ c)    [por la conmutatividad de ⊔]

-- Demostraciones con Lean4
-- =====

```

```

import Mathlib.Order.Lattice
variable {α : Type _} [Lattice α]
variable (a b c : α)

-- 1ª demostración
example
  (h : ∀ x y z : α, x ⊔ (y ⊓ z) = (x ⊔ y) ⊓ (x ⊔ z))
  : (a ⊓ b) ⊔ c = (a ⊔ c) ⊓ (b ⊔ c) :=
calc
  (a ⊓ b) ⊔ c = c ⊔ (a ⊓ b)           := by rw [sup_comm]
  _           = (c ⊔ a) ⊓ (c ⊔ b)      := by rw [h]
  _           = (a ⊔ c) ⊓ (c ⊔ b)      := by rw [@sup_comm _ _ c a]
  _           = (a ⊔ c) ⊓ (b ⊔ c)      := by rw [@sup_comm _ _ c b]

-- 2ª demostración
example
  (h : ∀ x y z : α, x ⊔ (y ⊓ z) = (x ⊔ y) ⊓ (x ⊔ z))
  : (a ⊓ b) ⊔ c = (a ⊔ c) ⊓ (b ⊔ c) :=
by simp [h, sup_comm]

-- Lemas usados
-- =====

-- #check (sup_comm : a ⊔ b = b ⊔ a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)



# Capítulo 9

## Relaciones de orden

### 9.1. En los órdenes parciales, $a < b \leftrightarrow a \leq b \wedge a \neq b$

```
-- -----  
-- Demostrar que en un orden parcial  
--    $a < b \leftrightarrow a \leq b \wedge a \neq b$   
-- -----  
  
-- Demostración en lenguaje natural  
-- =====  
  
-- Usaremos los siguientes lemas  
--    $(\forall a, b)[a < b \leftrightarrow a \leq b \wedge b \neq a]$  (L1)  
--    $(\forall a, b)[a \leq b \rightarrow b \leq a \rightarrow a = b]$  (L2)  
--  
-- Por el lema L1, lo que tenemos que demostrar es  
--    $a \leq b \wedge b \neq a \leftrightarrow a < b$   
-- Lo haremos demostrando las dos implicaciones.  
--  
-- ( $\Rightarrow$ ) Supongamos que  $a \leq b$  y  $b \neq a$ . Tenemos que demostrar que  
--  $a < b$ . Lo haremos por reducción al absurdo. Para ello, supongamos que  
--  $a = b$ . Entonces,  $b \leq a$  que contradice a  $b \neq a$ .  
--  
-- ( $\Leftarrow$ ) Supongamos que  $a < b$  y  $a \neq b$ . Tenemos que demostrar que  
--  $a \leq b$ . Lo haremos por reducción al absurdo. Para ello, supongamos que  
--  $b \leq a$ . Entonces, junto con  $a < b$ , se tiene que  $a = b$  que es una  
-- contradicción con  $a \neq b$ .  
  
-- Demostraciones con Lean4
```

```

-- =====

import Mathlib.Tactic

variable {α : Type _} [PartialOrder α]
variable (a b : α)

-- 1ª demostración
-- =====

example : a < b ↔ a ≤ b ∧ a ≠ b :=
by
  rw [lt_iff_le_not_le]
  -- ⊢ a ≤ b ∧ ¬b ≤ a ↔ a ≤ b ∧ a ≠ b
  constructor
  . -- ⊢ a ≤ b ∧ ¬b ≤ a → a ≤ b ∧ a ≠ b
    rintro ⟨h1 : a ≤ b, h2 : ¬b ≤ a⟩
    -- ⊢ a ≤ b ∧ a ≠ b
    constructor
    . -- ⊢ a ≤ b
      exact h1
    . -- ⊢ a ≠ b
      rintro (h3 : a = b)
      -- ⊢ False
      have h4 : b = a := h3.symm
      have h5 : b ≤ a := le_of_eq h4
      show False
      exact h2 h5
  . -- ⊢ a ≤ b ∧ a ≠ b → a ≤ b ∧ ¬b ≤ a
    rintro ⟨h5 : a ≤ b, h6 : a ≠ b⟩
    -- ⊢ a ≤ b ∧ ¬b ≤ a
    constructor
    . -- ⊢ a ≤ b
      exact h5
    . -- ⊢ ¬b ≤ a
      rintro (h7 : b ≤ a)
      have h8 : a = b := le_antisymm h5 h7
      show False
      exact h6 h8

-- 2ª demostración
-- =====

example : a < b ↔ a ≤ b ∧ a ≠ b :=
by

```



```

rw [lt_iff_le_not_le]
--  $\vdash a \leq b \wedge \neg b \leq a \leftrightarrow a \leq b \wedge a \neq b$ 
constructor
. --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow a \leq b \wedge a \neq b$ 
  rintro ⟨h1 :  $a \leq b$ , h2 :  $\neg b \leq a$ ⟩
  --  $\vdash a \leq b \wedge a \neq b$ 
  constructor
  . --  $\vdash a \leq b$ 
    exact h1
  . --  $\vdash a \neq b$ 
    rintro (h3 :  $a = b$ )
    --  $\vdash \text{False}$ 
    exact h2 (le_of_eq h3.symm)
. --  $\vdash a \leq b \wedge a \neq b \rightarrow a \leq b \wedge \neg b \leq a$ 
  rintro ⟨h4 :  $a \leq b$ , h5 :  $a \neq b$ ⟩
  --  $\vdash a \leq b \wedge \neg b \leq a$ 
  constructor
  . --  $\vdash a \leq b$ 
    exact h4
  . --  $\vdash \neg b \leq a$ 
    rintro (h6 :  $b \leq a$ )
    exact h5 (le_antisymm h4 h6)

-- 3ª demostración
-- =====

example :  $a < b \leftrightarrow a \leq b \wedge a \neq b :=$ 
by
  rw [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \leftrightarrow a \leq b \wedge a \neq b$ 
  constructor
  . --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow a \leq b \wedge a \neq b$ 
    rintro ⟨h1 :  $a \leq b$ , h2 :  $\neg b \leq a$ ⟩
    --  $\vdash a \leq b \wedge a \neq b$ 
    constructor
    . --  $\vdash a \leq b$ 
      exact h1
    . --  $\vdash a \neq b$ 
      exact fun h3 ↦ h2 (le_of_eq h3.symm)
  . --  $\vdash a \leq b \wedge a \neq b \rightarrow a \leq b \wedge \neg b \leq a$ 
    rintro ⟨h4 :  $a \leq b$ , h5 :  $a \neq b$ ⟩
    --  $\vdash a \leq b \wedge \neg b \leq a$ 
    constructor
    . --  $\vdash a \leq b$ 
      exact h4

```

```

. --  $\vdash \neg b \leq a$ 
  exact fun h6  $\mapsto$  h5 (le_antisymm h4 h6)

-- 4ª demostración
-- =====

example : a < b  $\leftrightarrow$  a  $\leq$  b  $\wedge$  a  $\neq$  b :=
by
  rw [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \leftrightarrow a \leq b \wedge a \neq b$ 
  constructor
  . --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow a \leq b \wedge a \neq b$ 
    rintro ⟨h1 : a  $\leq$  b, h2 :  $\neg b \leq a$ ⟩
    --  $\vdash a \leq b \wedge a \neq b$ 
    exact ⟨h1, fun h3  $\mapsto$  h2 (le_of_eq h3.symm)⟩
  . --  $\vdash a \leq b \wedge a \neq b \rightarrow a \leq b \wedge \neg b \leq a$ 
    rintro ⟨h4 : a  $\leq$  b, h5 : a  $\neq$  b⟩
    --  $\vdash a \leq b \wedge \neg b \leq a$ 
    exact ⟨h4, fun h6  $\mapsto$  h5 (le_antisymm h4 h6)⟩

-- 5ª demostración
-- =====

example : a < b  $\leftrightarrow$  a  $\leq$  b  $\wedge$  a  $\neq$  b :=
by
  rw [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \leftrightarrow a \leq b \wedge a \neq b$ 
  constructor
  . --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow a \leq b \wedge a \neq b$ 
    exact fun ⟨h1, h2⟩  $\mapsto$  ⟨h1, fun h3  $\mapsto$  h2 (le_of_eq h3.symm)⟩
  . --  $\vdash a \leq b \wedge a \neq b \rightarrow a \leq b \wedge \neg b \leq a$ 
    exact fun ⟨h4, h5⟩  $\mapsto$  ⟨h4, fun h6  $\mapsto$  h5 (le_antisymm h4 h6)⟩

-- 6ª demostración
-- =====

example : a < b  $\leftrightarrow$  a  $\leq$  b  $\wedge$  a  $\neq$  b :=
by
  rw [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \leftrightarrow a \leq b \wedge a \neq b$ 
  exact ⟨fun ⟨h1, h2⟩  $\mapsto$  ⟨h1, fun h3  $\mapsto$  h2 (le_of_eq h3.symm)⟩,
        fun ⟨h4, h5⟩  $\mapsto$  ⟨h4, fun h6  $\mapsto$  h5 (le_antisymm h4 h6)⟩⟩

-- 7ª demostración
-- =====

```

```

example : a < b ↔ a ≤ b ∧ a ≠ b :=
by
  constructor
  . -- ⊢ a < b → a ≤ b ∧ a ≠ b
    intro h
    -- h : a < b
    -- ⊢ a ≤ b ∧ a ≠ b
    constructor
    . -- ⊢ a ≤ b
      exact le_of_lt h
    . -- ⊢ a ≠ b
      exact ne_of_lt h
  . -- ⊢ a ≤ b ∧ a ≠ b → a < b
    rintro ⟨h1, h2⟩
    -- h1 : a ≤ b
    -- h2 : a ≠ b
    -- ⊢ a < b
    exact lt_of_le_of_ne h1 h2

-- 8ª demostración
-- =====

example : a < b ↔ a ≤ b ∧ a ≠ b :=
  ⟨fun h ↦ ⟨le_of_lt h, ne_of_lt h⟩,
   fun ⟨h1, h2⟩ ↦ lt_of_le_of_ne h1 h2⟩

-- 9ª demostración
-- =====

example : a < b ↔ a ≤ b ∧ a ≠ b :=
  lt_iff_le_and_ne

-- Lemas usados
-- =====

-- #check (le_antisymm : a ≤ b → b ≤ a → a = b)
-- #check (le_of_eq : a = b → a ≤ b)
-- #check (lt_iff_le_and_ne : a < b ↔ a ≤ b ∧ a ≠ b)
-- #check (lt_iff_le_not_le : a < b ↔ a ≤ b ∧ ¬b ≤ a)
-- #check (lt_of_le_of_ne : a ≤ b → a ≠ b → a < b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 9.2. Si $\leq$ es un preorden, entonces $<$ es irreflexiva

```

-----
-- Demostrar que si  $\leq$  es un preorden, entonces  $<$  es irreflexiva.
-----

-- Demostración en lenguaje natural
-- =====

-- Se usará la siguiente propiedad de los preórdenes
--  $(\forall a, b)[a < b \leftrightarrow a \leq b \wedge b \neq a]$ 
-- Con dicha propiedad, lo que tenemos que demostrar se transforma en
--  $\neg(a \leq a \wedge a \neq a)$ 
-- Para demostrarla, supongamos que
--  $a \leq a \wedge a \neq a$ 
-- lo que es una contradicción.

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable {α : Type _} [Preorder α]
variable (a : α)

-- 1ª demostración
-- =====

example : ¬a < a :=
by
  rw [lt_iff_le_not_le]
  --  $\vdash \neg(a \leq a \wedge \neg a \leq a)$ 
  rintro ⟨h1, h2⟩
  --  $h1 : a \leq a$ 
  --  $h2 : \neg a \leq a$ 
  --  $\vdash \text{False}$ 
  exact h2 h1

-- 2ª demostración
-- =====

example : ¬a < a :=
  irrefl a

```

```
-- Lemas usados
-- =====

-- variable (b :  $\alpha$ )
-- #check (lt_iff_le_not_le :  $a < b \leftrightarrow a \leq b \wedge \neg b \leq a$ )
-- #check (irrefl a :  $\neg a < a$ )
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 9.3. Si $\leq$ es un preorden, entonces $<$ es transitiva

```
-- -----
-- Demostrar que si  $\leq$  es un preorden, entonces  $<$  es transitiva.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se usará la siguiente propiedad de los preórdenes
-- ( $\forall a, b$ ) [ $a < b \leftrightarrow a \leq b \wedge b \not\leq a$ ]
-- Con dicha propiedad, lo que tenemos que demostrar se transforma en
--  $a \leq b \wedge b \not\leq a \rightarrow b \leq c \wedge c \not\leq b \rightarrow a \leq c \wedge c \not\leq a$ 
-- Para demostrarla, supongamos que
--  $a \leq b$  (1)
--  $b \not\leq a$  (2)
--  $b \leq c$  (3)
--  $c \not\leq b$  (4)
-- y tenemos que demostrar las siguientes relaciones
--  $a \leq c$  (5)
--  $c \not\leq a$  (6)
--
-- La (5) se tiene aplicando la propiedad transitiva a (1) y (3).
--
-- Para demostrar la (6), supongamos que
--  $c \leq a$  (7)
-- entonces, junto a la (1), por la propiedad transitiva se tiene
--  $c \leq b$ 
-- que es una contradicción con la (4).

-- Demostraciones con Lean4
-- =====
```

```

import Mathlib.Tactic
variable {α : Type _} [Preorder α]
variable (a b c : α)

-- 1ª demostración
-- =====

example : a < b → b < c → a < c :=
by
  simp only [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow b \leq c \wedge \neg c \leq b \rightarrow a \leq c \wedge \neg c \leq a$ 
  rintro ⟨h1 : a ≤ b, _h2 : ¬b ≤ a⟩ ⟨h3 : b ≤ c, h4 : ¬c ≤ b⟩
  --  $\vdash a \leq c \wedge \neg c \leq a$ 
  constructor
  . --  $\vdash a \leq c$ 
    exact le_trans h1 h3
  . --  $\vdash \neg c \leq a$ 
    contrapose! h4
    --  $h4 : c \leq a$ 
    --  $\vdash c \leq b$ 
    exact le_trans h4 h1

-- 2ª demostración
-- =====

example : a < b → b < c → a < c :=
by
  simp only [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow b \leq c \wedge \neg c \leq b \rightarrow a \leq c \wedge \neg c \leq a$ 
  rintro ⟨h1 : a ≤ b, _h2 : ¬b ≤ a⟩ ⟨h3 : b ≤ c, h4 : ¬c ≤ b⟩
  --  $\vdash a \leq c \wedge \neg c \leq a$ 
  constructor
  . --  $\vdash a \leq c$ 
    exact le_trans h1 h3
  . --  $\vdash \neg c \leq a$ 
    rintro (h5 : c ≤ a)
    --  $\vdash \text{False}$ 
    have h6 : c ≤ b := le_trans h5 h1
    show False
    exact h4 h6

-- 3ª demostración
-- =====

```

```

example : a < b → b < c → a < c :=
by
  simp only [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow b \leq c \wedge \neg c \leq b \rightarrow a \leq c \wedge \neg c \leq a$ 
  rintro ⟨h1 : a ≤ b, _h2 : ¬b ≤ a⟩ ⟨h3 : b ≤ c, h4 : ¬c ≤ b⟩
  --  $\vdash a \leq c \wedge \neg c \leq a$ 
  constructor
  . --  $\vdash a \leq c$ 
    exact le_trans h1 h3
  . --  $\vdash \neg c \leq a$ 
    exact fun h5 ↦ h4 (le_trans h5 h1)

-- 4ª demostración
-- =====

example : a < b → b < c → a < c :=
by
  simp only [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow b \leq c \wedge \neg c \leq b \rightarrow a \leq c \wedge \neg c \leq a$ 
  rintro ⟨h1 : a ≤ b, _h2 : ¬b ≤ a⟩ ⟨h3 : b ≤ c, h4 : ¬c ≤ b⟩
  --  $\vdash a \leq c \wedge \neg c \leq a$ 
  exact ⟨le_trans h1 h3, fun h5 ↦ h4 (le_trans h5 h1)⟩

-- 5ª demostración
-- =====

example : a < b → b < c → a < c :=
by
  simp only [lt_iff_le_not_le]
  --  $\vdash a \leq b \wedge \neg b \leq a \rightarrow b \leq c \wedge \neg c \leq b \rightarrow a \leq c \wedge \neg c \leq a$ 
  exact fun ⟨h1, _h2⟩ ⟨h3, h4⟩ ↦ ⟨le_trans h1 h3,
    fun h5 ↦ h4 (le_trans h5 h1)⟩

-- 6ª demostración
-- =====

example : a < b → b < c → a < c :=
  lt_trans

-- Lemas usados
-- =====

-- #check (lt_iff_le_not_le : a < b ↔ a ≤ b ∧ ¬b ≤ a)
-- #check (le_trans : a ≤ b → b ≤ c → a ≤ c)
-- #check (lt_trans : a < b → b < c → a < c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)



# Capítulo 10

## Anillos ordenados

### 10.1. En los anillos ordenados, $a \leq b \rightarrow 0 \leq b - a$

```
-- -----  
-- Demostrar que en los anillos ordenados se verifica que  
--    $a \leq b \rightarrow 0 \leq b - a$   
-- -----  
  
-- Demostración en lenguaje natural  
-- =====  
  
-- Se usarán los siguientes lemas:  
--   sub_self      :  $a - a = 0$   
--   sub_le_sub_right :  $a \leq b \rightarrow \forall (c : R), a - c \leq b - c$   
--  
-- Supongamos que  
--    $a \leq b$  (1)  
-- La demostración se tiene por la siguiente cadena de desigualdades:  
--    $0 = a - a$  [por sub_self]  
--    $\leq b - a$  [por (1) y sub_le_sub_right]  
  
-- Demostraciones con Lean4  
-- =====  
  
import Mathlib.Algebra.Order.Ring.Defs  
variable {R : Type _} [StrictOrderedRing R]  
variable (a b c : R)  
  
-- 1ª demostración
```

```

example : a ≤ b → 0 ≤ b - a :=
by
  intro h
  calc
    0 = a - a := (sub_self a).symm
    _ ≤ b - a := sub_le_sub_right h a

-- 2ª demostración
example : a ≤ b → 0 ≤ b - a :=
sub_nonneg.mpr

-- 3ª demostración
example : a ≤ b → 0 ≤ b - a :=
by simp

-- Lemas usados
-- =====

-- #check (sub_le_sub_right : a ≤ b → ∀ (c : R), a - c ≤ b - c)
-- #check (sub_nonneg : 0 ≤ a - b ↔ b ≤ a)
-- #check (sub_self a : a - a = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 10.2. En los anillos ordenados, $0 \leq b - a \rightarrow a \leq b$

```

-----
-- Demostrar que en los anillos ordenados
--   0 ≤ b - a → a ≤ b
-----

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas:
--   zero_add a : 0 + a = a
--   add_le_add_right : b ≤ c → ∀ (a : R), b + a ≤ c + a
--   sub_add_cancel a b : a - b + b = -a
-- Supongamos que
--   0 ≤ b - a
-- La demostración se tiene por la siguiente cadena de desigualdades:
--
-- (1)

```

```

--      a = 0 + a          [por zero_add]
--      ≤ (b - a) + a      [por (1) y add_le_add_right]
--      = b                [por sub_add_cancel]

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Order.Ring.Defs
variable {R : Type _} [StrictOrderedRing R]
variable (a b c : R)

-- 1ª demostración
-- =====

example : 0 ≤ b - a → a ≤ b :=
by
  intro h
  calc
    a = 0 + a          := (zero_add a).symm
    _ ≤ (b - a) + a    := add_le_add_right h a
    _ = b              := sub_add_cancel b a

-- 2ª demostración
-- =====

example : 0 ≤ b - a → a ≤ b :=
-- by apply?
sub_nonneg.mp

-- 3ª demostración
-- =====

example : 0 ≤ b - a → a ≤ b :=
by simp

-- Lemas usados
-- =====

-- #check (zero_add a : 0 + a = a)
-- #check (add_le_add_right : b ≤ c → ∀ (a : R), b + a ≤ c + a)
-- #check (sub_add_cancel a b : a - b + b = a)
-- #check (sub_nonneg : 0 ≤ a - b ↔ b ≤ a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

### 10.3. En los anillos ordenados, $\{a \leq b, 0 \leq c\} \vdash ac \leq bc$

```

-----
-- Demostrar que, en los anillos ordenados, si
--    $a \leq b$ 
--    $0 \leq c$ 
-- entonces
--    $a * c \leq b * c$ 
-----

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas:
--   sub_nonneg           :  $0 \leq a - b \leftrightarrow b \leq a$ )
--   mul_nonneg           :  $0 \leq a \rightarrow 0 \leq b \rightarrow 0 \leq a * b$ )
--   sub_mul a b c       :  $(a - b) * c = a * c - b * c$ )
--
-- Supongamos que
--    $a \leq b$                                      (1)
--    $0 \leq c$ 
-- De (1), por sub_nonneg, se tiene
--    $0 \leq b - a$ 
-- y con (2), por mul_nonneg, se tiene
--    $0 \leq (b - a) * c$ 
-- que, por sub_mul, da
--    $0 \leq b * c - a * c$ 
-- y, aplicándole sub_nonneg, se tiene
--    $a * c \leq b * c$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Algebra.Order.Ring.Defs
variable {R : Type _} [StrictOrderedRing R]
variable (a b c : R)

-- 1ª demostración
-- =====

example
  (h1 : a ≤ b)
  (h2 : 0 ≤ c)

```

```

: a * c ≤ b * c :=
by
  have h3 : 0 ≤ b - a :=
    sub_nonneg.mpr h1
  have h4 : 0 ≤ b * c - a * c := calc
    0 ≤ (b - a) * c := mul_nonneg h3 h2
    _ = b * c - a * c := sub_mul b a c
  show a * c ≤ b * c
  exact sub_nonneg.mp h4

-- 2ª demostración
-- =====

example
  (h1 : a ≤ b)
  (h2 : 0 ≤ c)
  : a * c ≤ b * c :=
by
  have h3 : 0 ≤ b - a := sub_nonneg.mpr h1
  have h4 : 0 ≤ (b - a) * c := mul_nonneg h3 h2
  -- h4 : 0 ≤ b * c - a * c
  rw [sub_mul] at h4
  -- a * c ≤ b * c
  exact sub_nonneg.mp h4

-- 3ª demostración
-- =====

example
  (h1 : a ≤ b)
  (h2 : 0 ≤ c)
  : a * c ≤ b * c :=
by
  -- 0 ≤ b * c - a * c
  apply sub_nonneg.mp
  -- 0 ≤ (b - a) * c
  rw [← sub_mul]
  apply mul_nonneg
  . -- 0 ≤ b - a
    exact sub_nonneg.mpr h1
  . -- 0 ≤ c
    exact h2

-- 4ª demostración
-- =====

```

```

example
  (h1 : a ≤ b)
  (h2 : 0 ≤ c)
  : a * c ≤ b * c :=
by
  apply sub_nonneg.mp
  rw [← sub_mul]
  apply mul_nonneg (sub_nonneg.mpr h1) h2

-- 5ª demostración
example
  (h1 : a ≤ b)
  (h2 : 0 ≤ c)
  : a * c ≤ b * c :=
-- by apply?
mul_le_mul_of_nonneg_right h1 h2

-- Lemas usados
-- =====

-- #check (mul_le_mul_of_nonneg_right : a ≤ b → 0 ≤ c → a * c ≤ b * c)
-- #check (mul_nonneg : 0 ≤ a → 0 ≤ b → 0 ≤ a * b)
-- #check (sub_mul a b c : (a - b) * c = a * c - b * c)
-- #check (sub_nonneg : 0 ≤ a - b ↔ b ≤ a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

# Capítulo 11

## Espacios métricos

### 11.1. En los espacios métricos, $\text{dist}(x,y) \geq 0$

```
-- -----
-- Ejercicio. Demostrar que en los espacios métricos
--    $0 \leq \text{dist } x \ y$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se usarán los siguientes lemas:
--   dist_comm x y           : dist x y = dist y x
--   dist_self x             : dist x x = 0
--   dist_triangle x y z     : dist x z ≤ dist x y + dist y z
--   mul_two a               : a * 2 = a + a
--   nonneg_of_mul_nonneg_left :  $0 \leq a * b \rightarrow 0 < b \rightarrow 0 \leq a$ 
--   zero_lt_two             :  $0 < 2$ 
--
-- Por nonneg_of_mul_nonneg_left es suficiente demostrar las siguientes
-- desigualdades:
--    $0 \leq \text{dist } x \ y * 2$                                 (1)
--    $0 < 2$                                                     (2)
--
-- La (1) se demuestra por las siguiente cadena de desigualdades:
--   0 = dist x x           [por dist_self]
--     ≤ dist x y + dist y x [por dist_triangle]
--     = dist x y + dist x y [por dist_comm]
--     = dist x y * 2        [por mul_two]
--
-- La (2) se tiene por zero_lt_two.
```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Topology.MetricSpace.Basic
variable {X : Type _} [MetricSpace X]
variable (x y : X)

-- 1ª demostración
example : 0 ≤ dist x y :=
by
  have h1 : 0 ≤ dist x y * 2 := calc
    0 = dist x x                := (dist_self x).symm
    _ ≤ dist x y + dist y x := dist_triangle x y x
    _ = dist x y + dist x y := by rw [dist_comm x y]
    _ = dist x y * 2         := (mul_two (dist x y)).symm
  show 0 ≤ dist x y
  exact nonneg_of_mul_nonneg_left h1 zero_lt_two

-- 2ª demostración
example : 0 ≤ dist x y :=
by
  apply nonneg_of_mul_nonneg_left
  . -- 0 ≤ dist x y * 2
    calc 0 = dist x x                := by simp only [dist_self]
          _ ≤ dist x y + dist y x := by simp only [dist_triangle]
          _ = dist x y + dist x y := by simp only [dist_comm]
          _ = dist x y * 2         := by simp only [mul_two]
  . -- 0 < 2
    exact zero_lt_two

-- 3ª demostración
example : 0 ≤ dist x y :=
by
  have : 0 ≤ dist x y + dist y x := by
    rw [← dist_self x]
    apply dist_triangle
    linarith [dist_comm x y]

-- 3ª demostración
example : 0 ≤ dist x y :=
-- by apply?
dist_nonneg

-- Lemas usados

```



```
-- =====  
  
-- variable (a b : ℝ)  
-- variable (z : X)  
-- #check (dist_comm x y : dist x y = dist y x)  
-- #check (dist_nonneg : 0 ≤ dist x y)  
-- #check (dist_self x : dist x x = 0)  
-- #check (dist_triangle x y z : dist x z ≤ dist x y + dist y z)  
-- #check (mul_two a : a * 2 = a + a)  
-- #check (nonneg_of_mul_nonneg_left : 0 ≤ a * b → 0 < b → 0 ≤ a)  
-- #check (zero_lt_two : 0 < 2)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)



# Capítulo 12

## Funciones reales

### 12.1. La suma de una cota superior de $f$ y una cota superior de $g$ es una cota superior de $f+g$

```
-- -----  
-- Demostrar que la suma de una cota superior de  $f$  y una cota superior  
-- de  $g$  es una cota superior de  $f + g$ .  
-- -----  
  
-- Demostración en lenguaje natural  
-- =====  
  
-- Se usará el siguiente lema  
--   add_le_add :  $a \leq b \rightarrow c \leq d \rightarrow a + c \leq b + d$   
--  
-- Por la definición de cota superior, hay que demostrar que  
--    $(\forall x \in \mathbb{R}) [f(x) + g(x) \leq a + b]$  (1)  
-- Para ello, sea  $x \in \mathbb{R}$ . Puesto que  $a$  es una cota superior de  $f$ , se  
-- tiene que  
--    $f(x) \leq a$  (2)  
-- y, puesto que  $b$  es una cota superior de  $g$ , se tiene que  
--    $g(x) \leq b$  (3)  
-- De (2) y (3), por add_le_add, se tiene que  
--    $f(x) + g(x) \leq a + b$   
-- que es lo que había que demostrar.  
  
-- Demostraciones con Lean4  
-- =====
```

```

import Mathlib.Data.Real.Basic

-- (CotaSuperior f a) se verifica si a es una cota superior de f.
def CotaSuperior (f : ℝ → ℝ) (a : ℝ) : Prop :=
  ∀ x, f x ≤ a

variable {f g : ℝ → ℝ}
variable {a b : ℝ}

-- 1ª demostración
-- =====

example
  (hfa : CotaSuperior f a)
  (hgb : CotaSuperior g b)
  : CotaSuperior (f + g) (a + b) :=
by
  have h1 : ∀ x, (f + g) x ≤ a + b := by
  { intro x
    have h2 : f x ≤ a := hfa x
    have h3 : g x ≤ b := hgb x
    show (f + g) x ≤ a + b
    exact add_le_add h2 h3 }
  show CotaSuperior (f + g) (a + b)
  exact h1

-- 2ª demostración
-- =====

example
  (hfa : CotaSuperior f a)
  (hgb : CotaSuperior g b)
  : CotaSuperior (f + g) (a + b) :=
by
  have h1 : ∀ x, (f + g) x ≤ a + b := by
  { intro x
    show (f + g) x ≤ a + b
    exact add_le_add (hfa x) (hgb x) }
  show CotaSuperior (f + g) (a + b)
  exact h1

-- 3ª demostración
-- =====

example

```

```

(hfa : CotaSuperior f a)
(hgb : CotaSuperior g b)
: CotaSuperior (f + g) (a + b) :=
by
  intro x
  dsimp
  apply add_le_add
  . apply hfa
  . apply hgb

-- 4ª demostración
-- =====

theorem sumaCotaSup
  (hfa : CotaSuperior f a)
  (hgb : CotaSuperior g b)
  : CotaSuperior (f + g) (a + b) :=
λ x ↦ add_le_add (hfa x) (hgb x)

-- Lemas usados
-- =====

-- variable (c d : ℝ)
-- #check (add_le_add : a ≤ b → c ≤ d → a + c ≤ b + d)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.2. La suma de una cota inferior de $f$ y una cota inferior de $g$ es una cota inferior de $f+g$

```

-- -----
-- Demostrar que la suma de una cota inferior de  $f$  y una cota inferior
-- de  $g$  es una cota inferior de  $f + g$ .
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se usará el siguiente lema
--   add_le_add : a ≤ b → c ≤ d → a + c ≤ b + d
--

```

```

-- Por la definición de cota inferior, hay que demostrar que
--    $(\forall x \in \mathbb{R}) [a + b \leq f(x) + g(x)]$  (1)
-- Para ello, sea  $x \in \mathbb{R}$ . Puesto que  $a$  es una cota inferior de  $f$ , se
-- tiene que
--    $a \leq f(x)$  (2)
-- y, puesto que  $b$  es una cota inferior de  $g$ , se tiene que
--    $b \leq g(x)$  (3)
-- De (2) y (3), por add_le_add, se tiene que
--    $a + b \leq f(x) + g(x)$ 
-- que es lo que había que demostrar.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

-- (CotaInferior f a) expresa que  $a$  es una cota inferior de  $f$ .
def CotaInferior (f :  $\mathbb{R} \rightarrow \mathbb{R}$ ) (a :  $\mathbb{R}$ ) : Prop :=
   $\forall x, a \leq f\ x$ 

variable {f g :  $\mathbb{R} \rightarrow \mathbb{R}$ }
variable {a b :  $\mathbb{R}$ }

-- 1ª demostración
example
  (hfa : CotaInferior f a)
  (hgb : CotaInferior g b)
  : CotaInferior (f + g) (a + b) :=
by
  have h1 :  $\forall x, a + b \leq f\ x + g\ x$ 
  { intro x
    have h1a :  $a \leq f\ x := hfa\ x$ 
    have h1b :  $b \leq g\ x := hgb\ x$ 
    show  $a + b \leq f\ x + g\ x$ 
    exact add_le_add h1a h1b }
  show CotaInferior (f + g) (a + b)
  exact h1

-- 2ª demostración
example
  (hfa : CotaInferior f a)
  (hgb : CotaInferior g b)
  : CotaInferior (f + g) (a + b) :=
by
  have h1 :  $\forall x, a + b \leq f\ x + g\ x$ 

```

```

{ intro x
  show  $a + b \leq f\ x + g\ x$ 
  exact add_le_add (hfa x) (hgb x) }
show CotaInferior (f + g) (a + b)
exact h1

-- 3ª demostración
example
  (hfa : CotaInferior f a)
  (hgb : CotaInferior g b)
  : CotaInferior (f + g) (a + b) :=
by
  intro x
  dsimp
  apply add_le_add
  . apply hfa
  . apply hgb

-- 4ª demostración
theorem sumaCotaInf
  (hfa : CotaInferior f a)
  (hgb : CotaInferior g b)
  : CotaInferior (f + g) (a + b) :=
λ x ↦ add_le_add (hfa x) (hgb x)

-- Lemas usados
-- =====

-- variable (c d : ℝ)
-- #check (add_le_add :  $a \leq b \rightarrow c \leq d \rightarrow a + c \leq b + d$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.3. El producto de funciones no negativas es no negativo

```

-----
-- Demostrar que el producto de dos funciones no negativas es no
-- negativa.
-----

-- Demostración en lenguaje natural

```

```

-- =====

-- Se usará el siguiente lema
--   mul_nonneg :  $0 \leq a \rightarrow 0 \leq b \rightarrow 0 \leq a * b$ 
--
-- Hay que demostrar que
--    $(\forall x \in \mathbb{R}) [0 \leq f(x) * g(x)]$  (1)
-- Para ello, sea  $x \in \mathbb{R}$ . Puesto que  $f$  es no negativa, se tiene que
--    $0 \leq f(x)$  (2)
-- y, puesto que  $g$  es no negativa, se tiene que
--    $0 \leq g(x)$  (3)
-- De (2) y (3), por mul_nonneg, se tiene que
--    $0 \leq f(x) * g(x)$ 
-- que es lo que había que demostrar.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

-- (CotaInferior f a) expresa que a es una cota inferior de f.
def CotaInferior (f :  $\mathbb{R} \rightarrow \mathbb{R}$ ) (a :  $\mathbb{R}$ ) : Prop :=
   $\forall x, a \leq f x$ 

variable (f g :  $\mathbb{R} \rightarrow \mathbb{R}$ )

-- 1ª demostración
example
  (nnf : CotaInferior f 0)
  (nng : CotaInferior g 0)
  : CotaInferior (f * g) 0 :=
by
  have h1 :  $\forall x, 0 \leq f x * g x$ 
  { intro x
    have h2 :  $0 \leq f x := nnf x$ 
    have h3 :  $0 \leq g x := nng x$ 
    show  $0 \leq f x * g x$ 
    exact mul_nonneg h2 h3 }
  show CotaInferior (f * g) 0
  exact h1

-- 2ª demostración
example
  (nnf : CotaInferior f 0)
  (nng : CotaInferior g 0)

```



```

: CotaInferior (f * g) 0 :=
by
  have h1 :  $\forall x, 0 \leq f\ x * g\ x$ 
  { intro x
    show  $0 \leq f\ x * g\ x$ 
    exact mul_nonneg (nnf x) (nng x) }
  show CotaInferior (f * g) 0
  exact h1

-- 3ª demostración
example
  (nnf : CotaInferior f 0)
  (nng : CotaInferior g 0)
  : CotaInferior (f * g) 0 :=
by
  intro x
  dsimp
  apply mul_nonneg
  . apply nnf
  . apply nng

-- 4ª demostración
example
  (nnf : CotaInferior f 0)
  (nng : CotaInferior g 0)
  : CotaInferior (f * g) 0 :=
λ x ↦ mul_nonneg (nnf x) (nng x)

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (mul_nonneg :  $0 \leq a \rightarrow 0 \leq b \rightarrow 0 \leq a * b$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.4. Si $a$ es una cota superior no negativa de $f$ y $b$ es una cota superior de la función no negativa $g$ , entonces $ab$ es una cota superior de $fg$

```

-----
-- Demostrar que si  $a$  es una cota superior de  $f$ ,  $b$  es una cota superior
-- de  $g$ ,  $a$  es no negativa y  $g$  es no negativa, entonces  $ab$  es una cota
-- superior de  $fg$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Se usará el siguiente lema
--   mul_le_mul :  $a \leq b \rightarrow c \leq d \rightarrow 0 \leq c \rightarrow 0 \leq b \rightarrow a * c \leq b * d$ 
--
-- Hay que demostrar que
--    $(\forall x \in \mathbb{R}) [f\ x * g\ x \leq a * b]$  (1)
-- Para ello, sea  $x \in \mathbb{R}$ . Puesto que  $a$  es una cota superior de  $f$ , se tiene que
--    $f(x) \leq a$  (2)
-- puesto que  $b$  es una cota superior de  $g$ , se tiene que
--    $g(x) \leq b$  (3)
-- puesto que  $g$  es no negativa, se tiene que
--    $0 \leq g(x)$  (4)
-- y, puesto que  $a$  es no negativa, se tiene que
--    $0 \leq a$  (5)
-- De (2), (3), (4) y (5), por mul_le_mul, se tiene que
--    $f\ x * g\ x \leq a * b$ 
-- que es lo que había que demostrar.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

-- (CotaSuperior f a) se verifica si  $a$  es una cota superior de  $f$ .
def CotaSuperior (f :  $\mathbb{R} \rightarrow \mathbb{R}$ ) (a :  $\mathbb{R}$ ) : Prop :=
   $\forall x, f\ x \leq a$ 

```

12.4. Si  $a$  es una cota superior no negativa de  $f$  y  $b$  es es una cota superior de la función no negativa  $g$ , entonces  $ab$  es una cota superior de  $fg$  235

-- (CotaInferior f a) expresa que a es una cota inferior de f.

```
def CotaInferior (f : ℝ → ℝ) (a : ℝ) : Prop :=
  ∀ x, a ≤ f x
```

```
variable (f g : ℝ → ℝ)
```

```
variable (a b : ℝ)
```

-- 1ª demostración

example

```
(hfa : CotaSuperior f a)
(hgb : CotaSuperior g b)
(nng : CotaInferior g 0)
(nna : 0 ≤ a)
: CotaSuperior (f * g) (a * b) :=
```

by

```
have h1 : ∀ x, f x * g x ≤ a * b
{ intro x
  have h2 : f x ≤ a := hfa x
  have h3 : g x ≤ b := hgb x
  have h4 : 0 ≤ g x := nng x
  show f x * g x ≤ a * b
    exact mul_le_mul h2 h3 h4 nna }
show CotaSuperior (f * g) (a * b)
exact h1
```

-- 2ª demostración

example

```
(hfa : CotaSuperior f a)
(hgb : CotaSuperior g b)
(nng : CotaInferior g 0)
(nna : 0 ≤ a)
: CotaSuperior (f * g) (a * b) :=
```

by

```
intro x
dsimp
apply mul_le_mul
. apply hfa
. apply hgb
. apply nng
. apply nna
```

-- 3ª demostración

example

```
(hfa : CotaSuperior f a)
(hgb : CotaSuperior g b)
```

```

(nng : CotaInferior g 0)
(nna : 0 ≤ a)
: CotaSuperior (f * g) (a * b) :=
by
  intro x
  have h1:= hfa x
  have h2:= hgb x
  have h3:= nng x
  exact mul_le_mul h1 h2 h3 nna

-- 4ª demostración
example
  (hfa : CotaSuperior f a)
  (hgb : CotaSuperior g b)
  (nng : CotaInferior g 0)
  (nna : 0 ≤ a)
  : CotaSuperior (f * g) (a * b) :=
by
  intro x
  specialize hfa x
  specialize hgb x
  specialize nng x
  exact mul_le_mul hfa hgb nng nna

-- 5ª demostración
example
  (hfa : CotaSuperior f a)
  (hgb : CotaSuperior g b)
  (nng : CotaInferior g 0)
  (nna : 0 ≤ a)
  : CotaSuperior (f * g) (a * b) :=
λ x ↦ mul_le_mul (hfa x) (hgb x) (nng x) nna

-- Lemas usados
-- =====

-- variable (c d : ℝ)
-- #check (mul_le_mul : a ≤ b → c ≤ d → 0 ≤ c → 0 ≤ b → a * c ≤ b * d)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.5. La suma de dos funciones acotadas superiormente también lo está

```

-----
-- Demostrar que la suma de dos funciones acotadas superiormente también
-- lo está.
-----

-- Demostración en lenguaje natural
-- =====

-- Del ejercicio "La suma de una cota superior de f y una cota superior
-- de g es una cota superior de f+g" (que se encuentra en
-- https://bit.ly/3QauluK ) usaremos la definición de cota superior
-- (CotaSuperior) y el lema sumaCotaSup.
--
-- Puesto que f está acotada superiormente, tiene una cota superior. Sea
-- a una de dichas cotas. Análogamente, puesto que g está acotada
-- superiormente, tiene una cota superior. Sea b una de dichas
-- cotas. Por el lema sumaCotaSup, a+b es una cota superior de f+g. or
-- consiguiente, f+g está acotada superiormente.

-- Demostraciones con Lean4
-- =====

import src.Suma_de_cotas_superiores

variable {f g : ℝ → ℝ}

-- (acotadaSup f) afirma que f tiene cota superior.
def acotadaSup (f : ℝ → ℝ) :=
  ∃ a, CotaSuperior f a

-- 1ª demostración
example
  (hf : acotadaSup f)
  (hg : acotadaSup g)
  : acotadaSup (f + g) :=
by
  cases' hf with a ha
  -- a : ℝ
  -- ha : CotaSuperior f a
  cases' hg with b hb
  -- b : ℝ

```

```

-- hb : CotaSuperior g b
have h1 : CotaSuperior (f + g) (a + b) :=
  sumaCotaSup ha hb
have h2 :  $\exists$  z, CotaSuperior (f+g) z :=
  Exists.intro (a + b) h1
show acotadaSup (f + g)
exact h2

-- 2ª demostración
example
  (hf : acotadaSup f)
  (hg : acotadaSup g)
  : acotadaSup (f + g) :=
by
  cases' hf with a ha
  -- a :  $\mathbb{R}$ 
  -- ha : FnUb f a
  cases' hg with b hb
  -- b :  $\mathbb{R}$ 
  -- hb : FnUb g b
  use a + b
  apply sumaCotaSup ha hb

-- 4ª demostración
example
  (hf : acotadaSup f)
  (hg : acotadaSup g)
  : acotadaSup (f + g) :=
by
  rcases hf with ⟨a, ha⟩
  rcases hg with ⟨b, hb⟩
  exact ⟨a + b, sumaCotaSup ha hb⟩

-- 5ª demostración
example :
  acotadaSup f → acotadaSup g → acotadaSup (f + g) :=
by
  rintro ⟨a, ha⟩ ⟨b, hb⟩
  exact ⟨a + b, sumaCotaSup ha hb⟩

-- 6ª demostración
example :
  acotadaSup f → acotadaSup g → acotadaSup (f + g) :=
fun ⟨a, ha⟩ ⟨b, hb⟩ ↦ ⟨a + b, sumaCotaSup ha hb⟩

```

## 12.6. La suma de dos funciones acotadas inferiormente también lo está 239

```
-- Lemas usados
-- =====

-- #check (sumaCotaSup : CotaSuperior f a → CotaSuperior g b → CotaSuperior (f + g) (a + b))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.6. La suma de dos funciones acotadas inferiormente también lo está

```
-----
-- Demostrar que la suma de dos funciones acotadas inferiormente también
-- lo está.
-----

-- Demostración en lenguaje natural
-- =====

-- Del ejercicio "La suma de una cota inferior de f y una cota inferior
-- de g es una cota inferior de f+g" usaremos la definición de cota
-- inferior (CotaInferior) y el lema sumaCotaInf.
--
-- Puesto que f está acotada inferiormente, tiene una cota inferior. Sea
-- a una de dichas cotas. Análogamente, puesto que g está acotada
-- inferiormente, tiene una cota inferior. Sea b una de dichas
-- cotas. Por el lema FnLb_add, a+b es una cota inferior de f+g. Por
-- consiguiente, f+g está acotada inferiormente.

-- Demostraciones con Lean4
-- =====

import src.Suma_de_cotas_inferiores
variable {f g : ℝ → ℝ}

-- (acotadaInf f) afirma que f tiene cota inferior.
def acotadaInf (f : ℝ → ℝ) :=
  ∃ a, CotaInferior f a

-- 1ª demostración
example
  (hf : acotadaInf f)
  (hg : acotadaInf g)
```

```

: acotadaInf (f + g) :=
by
  cases' hf with a ha
  -- a : ℝ
  -- ha : CotaInferior f a
  cases' hg with b hb
  -- b : ℝ
  -- hb : CotaInferior g b
  have h1 : CotaInferior (f + g) (a + b) := sumaCotaInf ha hb
  have h2 : ∃ z, CotaInferior (f + g) z :=
    Exists.intro (a + b) h1
  show acotadaInf (f + g)
  exact h2

-- 2ª demostración
example
  (hf : acotadaInf f)
  (hg : acotadaInf g)
  : acotadaInf (f + g) :=
by
  cases' hf with a ha
  -- a : ℝ
  -- ha : FnLb f a
  cases' hg with b hgb
  -- b : ℝ
  -- hgb : FnLb g b
  use a + b
  -- ⊢ FnLb (f + g) (a + b)
  apply sumaCotaInf ha hgb

-- 3ª demostración
example
  (hf : acotadaInf f)
  (hg : acotadaInf g)
  : acotadaInf (f + g) :=
by
  rcases hf with ⟨a, ha⟩
  -- a : ℝ
  -- ha : FnLb f a
  rcases hg with ⟨b, hb⟩
  -- b : ℝ
  -- hb : FnLb g b
  exact ⟨a + b, sumaCotaInf ha hb⟩

-- 4ª demostración

```



```
example :
  acotadaInf f → acotadaInf g → acotadaInf (f + g) :=
by
  rintro ⟨a, ha⟩ ⟨b, hb⟩
  -- a : ℝ
  -- ha : FnLb f a
  -- b : ℝ
  -- hb : FnLb g b
  exact ⟨a + b, sumaCotaInf ha hb⟩

-- 5ª demostración
example :
  acotadaInf f → acotadaInf g → acotadaInf (f + g) :=
fun ⟨a, ha⟩ ⟨b, hb⟩ ↦ ⟨a + b, sumaCotaInf ha hb⟩

-- Lemas usados
-- =====

-- #check (sumaCotaInf : FnLb f a → FnLb g b → FnLb (f + g) (a + b))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.7. Si $a$ es una cota superior de $f$ y $c \geq 0$ , entonces $ca$ es una cota superior de $cf$

```
-----
-- Demostrar que si a es una cota superior de f y c ≥ 0,
-- entonces c * a es una cota superior de c * f.
-----

-- Demostración en lenguaje natural
-- =====

-- Se usará el lema
--   {b ≤ c, 0 ≤ a} ⊢ ab ≤ ac                                     (L1)
--
-- Tenemos que demostrar que
--   (∀ y ∈ ℝ) cf(y) ≤ ca.
-- Sea y ∈ ℝ. Puesto que a es una cota de f, se tiene que
--   f(y) ≤ a
-- que, junto con c ≥ 0, por el lema L1 nos da
--   cf(y) ≤ ca
```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

-- (CotaSuperior f a) se verifica si a es una cota superior de f.
def CotaSuperior (f : ℝ → ℝ) (a : ℝ) : Prop :=
  ∀ x, f x ≤ a

variable {f : ℝ → ℝ}
variable {c : ℝ}

-- Demostraciones con Lean4
-- =====

-- 1ª demostración
example
  (hfa : CotaSuperior f a)
  (h : c ≥ 0)
  : CotaSuperior (fun x ↦ c * f x) (c * a) :=
by
  intro y
  -- y : ℝ
  -- ⊢ (fun x => c * f x) y ≤ c * a
  have ha : f y ≤ a := hfa y
  calc (fun x => c * f x) y
    = c * f y := by rfl
    _ ≤ c * a := mul_le_mul_of_nonneg_left ha h

-- 2ª demostración
example
  (hfa : CotaSuperior f a)
  (h : c ≥ 0)
  : CotaSuperior (fun x ↦ c * f x) (c * a) :=
by
  intro y
  calc (fun x => c * f x) y
    = c * f y := by rfl
    _ ≤ c * a := mul_le_mul_of_nonneg_left (hfa y) h

-- 3ª demostración
example
  (hfa : CotaSuperior f a)
  (h : c ≥ 0)

```

## 12.8. Si $c \geq 0$ y $f$ está acotada superiormente, entonces $c \cdot f$ también lo está 243

```

: CotaSuperior (fun x ↦ c * f x) (c * a) :=
by
  intro y
  show (fun x => c * f x) y ≤ c * a
  exact mul_le_mul_of_nonneg_left (hfa y) h

-- 4ª demostración
Lemma CotaSuperior_mul
  (hfa : CotaSuperior f a)
  (h : c ≥ 0)
  : CotaSuperior (fun x ↦ c * f x) (c * a) :=
fun y ↦ mul_le_mul_of_nonneg_left (hfa y) h

-- Lemas usados
-- =====

-- variable (c : ℝ)
-- #check (mul_le_mul_of_nonneg_left : b ≤ c → 0 ≤ a → a * b ≤ a * c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.8. Si $c \geq 0$ y $f$ está acotada superiormente, entonces $c \cdot f$ también lo está

```

-- -----
-- Demostrar que si  $c \geq 0$  y  $f$  está acotada superiormente, entonces  $c * f$ 
-- también lo está.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Usaremos el siguiente lema:
--   CotaSuperior_mul : CotaSuperior f a → c ≥ 0 → CotaSuperior (fun x ↦ c * f x) (c * a)
--
-- Puesto que  $f$  está acotada superiormente, tiene una cota superior. Sea
--  $a$  una de dichas cotas. Entonces, por el lema CotaSuperior_mul,  $ca$  es una cota
-- superior de  $cf$ . Por consiguiente,  $cf$  está acotada superiormente.

-- Demostraciones con Lean4
-- =====

```

```

import src.Cota_superior_de_producto_por_escalor

variable {f : ℝ → ℝ}
variable {c : ℝ}

-- (acotadaSup f) afirma que f tiene cota superior.
def acotadaSup (f : ℝ → ℝ) :=
  ∃ a, CotaSuperior f a

-- 1ª demostración
example
  (hf : acotadaSup f)
  (hc : c ≥ 0)
  : acotadaSup (fun x ↦ c * f x) :=
by
  cases' hf with a ha
  -- a : ℝ
  -- ha : CotaSuperior f a
  have h1 : CotaSuperior (fun x ↦ c * f x) (c * a) :=
    CotaSuperior_mul ha hc
  have h2 : ∃ z, ∀ x, (fun x ↦ c * f x) x ≤ z :=
    Exists.intro (c * a) h1
  show acotadaSup (fun x ↦ c * f x)
  exact h2

-- 2ª demostración
example
  (hf : acotadaSup f)
  (hc : c ≥ 0)
  : acotadaSup (fun x ↦ c * f x) :=
by
  cases' hf with a ha
  -- a : ℝ
  -- ha : CotaSuperior f a
  use c * a
  -- ⊢ CotaSuperior (fun x => c * f x) (c * a)
  apply CotaSuperior_mul ha hc

-- 3ª demostración
example
  (hf : acotadaSup f)
  (hc : c ≥ 0)
  : acotadaSup (fun x ↦ c * f x) :=
by
  rcases hf with ⟨a, ha⟩

```

```
-- a : ℝ
-- ha : CotaSuperior f a
exact ⟨c * a, CotaSuperior_mul ha hc⟩

-- 4ª demostración
example
  (hc : c ≥ 0)
  : acotadaSup f → acotadaSup (fun x ↦ c * f x) :=
by
  rintro ⟨a, ha⟩
  -- a : ℝ
  -- ha : CotaSuperior f a
  exact ⟨c * a, CotaSuperior_mul ha hc⟩

-- 5ª demostración
example
  (hc : c ≥ 0)
  : acotadaSup f → acotadaSup (fun x ↦ c * f x) :=
fun ⟨a, ha⟩ ↦ ⟨c * a, CotaSuperior_mul ha hc⟩

-- Lemas usados
-- =====

-- #check (CotaSuperior_mul : CotaSuperior f a → c ≥ 0 → CotaSuperior (fun x ↦ c * f x))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.9. Si para cada $a$ existe un $x$ tal que $f(x) > a$ , entonces $f$ no tiene cota superior

```
-----
-- Demostrar que si  $f$  es una función de  $\mathbb{R}$  en  $\mathbb{R}$  tal que para cada  $a$ ,
-- existe un  $x$  tal que  $f x > a$ , entonces  $f$  no tiene cota superior.
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que  $f$  tiene cota superior. Sea  $b$  una de dichas cotas
-- superiores. Por la hipótesis, existe un  $x$  tal que  $f(x) > b$ . Además,
-- como  $b$  es una cota superior de  $f$ ,  $f(x) \leq b$  que contradice la
-- desigualdad anterior.
```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

def CotaSuperior (f : ℝ → ℝ) (a : ℝ) : Prop :=
  ∀ x, f x ≤ a

def acotadaSup (f : ℝ → ℝ) : Prop :=
  ∃ a, CotaSuperior f a

variable (f : ℝ → ℝ)

-- 1ª demostración
example
  (h : ∀ a, ∃ x, f x > a)
  : ¬ acotadaSup f :=
by
  intros hf
  -- hf : acotadaSup f
  -- ⊢ False
  cases' hf with b hb
  -- b : ℝ
  -- hb : CotaSuperior f b
  cases' h b with x hx
  -- x : ℝ
  -- hx : f x > b
  have : f x ≤ b := hb x
  linarith

-- 2ª demostración
theorem sinCotaSup
  (h : ∀ a, ∃ x, f x > a)
  : ¬ acotadaSup f :=
by
  intros hf
  -- hf : acotadaSup f
  -- ⊢ False
  rcases hf with ⟨b, hb : CotaSuperior f b⟩
  rcases h b with ⟨x, hx : f x > b⟩
  have : f x ≤ b := hb x
  linarith

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.10. Si para cada $a$ existe un $x$ tal que $f(x) < a$ , entonces $f$ no tiene cota inferior

```

-----
-- Demostrar que si  $f$  es una función de  $\mathbb{R}$  en  $\mathbb{R}$  tal que para cada  $a$ ,
-- existe un  $x$  tal que  $f x < a$ , entonces  $f$  no tiene cota inferior.
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que  $f$  tiene cota inferior. Sea  $b$  una de dichas cotas
-- inferiores. Por la hipótesis, existe un  $x$  tal que  $f(x) < b$ . Además,
-- como  $b$  es una cota inferior de  $f$ ,  $b \leq f(x)$  que contradice la
-- desigualdad anterior.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

def CotaInferior (f :  $\mathbb{R} \rightarrow \mathbb{R}$ ) (a :  $\mathbb{R}$ ) : Prop :=
   $\forall x, a \leq f x$ 

def acotadaInf (f :  $\mathbb{R} \rightarrow \mathbb{R}$ ) : Prop :=
   $\exists a, \text{CotaInferior } f a$ 

variable (f :  $\mathbb{R} \rightarrow \mathbb{R}$ )

-- 1ª demostración
example
  (h :  $\forall a, \exists x, f x < a$ )
  :  $\neg \text{acotadaInf } f :=$ 
by
  intros hf
  -- hf : acotadaInf f
  --  $\vdash \text{False}$ 
  cases' hf with b hb
  -- b :  $\mathbb{R}$ 
  -- hb : CotaInferior f b
  cases' h b with x hx
  -- x :  $\mathbb{R}$ 
  -- hx :  $f x < b$ 
  have :  $b \leq f x := hb x$ 

```

```

linarith

-- 2ª demostración
example
  (h :  $\forall a, \exists x, f\ x < a$ )
  :  $\neg$  acotadaInf f :=
by
  intros hf
  -- hf : acotadaInf f
  --  $\vdash$  False
  rcases hf with ⟨b, hb : CotaInferior f b⟩
  rcases h b with ⟨x, hx : f x < b⟩
  have : b  $\leq$  f x := hb x
  linarith

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.11. La función identidad no está acotada superiormente

```

-----
-- Demostrar que la función identidad no está acotada superiormente.
-----

-- Demostración en lenguaje natural
-- =====

-- Usamos el lema de ejercicio anterior (que afirma que si para cada a,
-- existe un x tal que f x > a, entonces f no tiene cota superior) basta
-- demostrar que
--    $(\forall a \in \mathbb{R})(\exists x \in \mathbb{R}) [x > a]$ 
-- Sea a  $\in \mathbb{R}$ . Entonces a + 1 > a y, por tanto,  $(\exists x \in \mathbb{R}) [x > a]$ .

-- Demostraciones con Lean4
-- =====

import src.Funcion_no_acotada_superiormente

-- 1ª demostración
example :  $\neg$  acotadaSup (fun x ↦ x) :=
by
  apply sinCotaSup

```



```

--  $\vdash \forall (a : \mathbb{R}), \exists x, x > a$ 
intro a
--  $a : \mathbb{R}$ 
--  $\vdash \exists x, x > a$ 
use a + 1
--  $\vdash a + 1 > a$ 
linarith

-- 2ª demostración
example :  $\neg$  acotadaSup (fun x  $\mapsto$  x) :=
by
  apply sinCotaSup
  --  $\vdash \forall (a : \mathbb{R}), \exists x, x > a$ 
  intro a
  --  $a : \mathbb{R}$ 
  --  $\vdash \exists x, x > a$ 
  exact (a + 1, by linarith)

-- 3ª demostración
example :  $\neg$  acotadaSup (fun x  $\mapsto$  x) :=
by
  apply sinCotaSup
  --  $\vdash \forall (a : \mathbb{R}), \exists x, x > a$ 
  exact fun a  $\mapsto$  (a + 1, by linarith)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.12. Si $f$ no está acotada superiormente, entonces $(\forall a)(\exists x)[f(x) > a]$

```

-----
-- Sea  $f$  una función de  $\mathbb{R}$  en  $\mathbb{R}$ . Demostrar que si  $f$  no está acotada
-- superiormente, entonces  $(\forall a)(\exists x)[f(x) > a]$ .
-----

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Usaremos los siguientes lemas

```

```

--       $\neg(\exists x)P(x) \rightarrow (\forall x)\neg P(x)$                                 (L1)
--       $\neg a > b \rightarrow a \leq b$                                        (L2)
--
-- Sea  $a \in \mathbb{R}$ . Tenemos que demostrar que
--       $(\exists x)[f(x) > a]$ 
-- Lo haremos por reducción al absurdo. Para ello, suponemos que
--       $\neg(\exists x)[f(x) > a]$                                              (1)
-- y tenemos que obtener una contradicción. Aplicando L1 a (1) se tiene
--       $(\forall x)[\neg f(x) > a]$ 
-- y, aplicando L2, se tiene
--       $(\forall x)[f(x) \leq a]$ 
-- Lo que significa que  $a$  es una cota superior de  $f$  y, por tanto  $f$  está
-- acotada superiormente, en contradicción con la hipótesis.

-- 2ª demostración en LN
-- =====

-- Por la contrarecíproca, se supone que
--       $\neg(\forall a)(\exists x)[f(x) > a]$                                        (1)
-- y tenemos que demostrar que  $f$  está acotada superiormente.
--
-- Interiorizando la negación en (1) y simplificando, se tiene que
--       $(\exists a)(\forall x)[f(x) \leq a]$ 
-- que es lo que teníamos que demostrar.

-- Demostraciones con Lean 4
-- =====

import Mathlib.Data.Real.Basic

def CotaSuperior (f : ℝ → ℝ) (a : ℝ) : Prop :=
  ∀ x, f x ≤ a

def acotadaSup (f : ℝ → ℝ) :=
  ∃ a, CotaSuperior f a

variable (f : ℝ → ℝ)

-- 1ª demostración
-- =====

example
  (h : ¬acotadaSup f)
  : ∀ a, ∃ x, f x > a :=
by

```

```

intro a
-- a : ℝ
-- ⊢ ∃ x, f x > a
by_contra h1
-- h1 : ¬∃ x, f x > a
-- ⊢ False
have h2 : ∀ x, ¬ f x > a :=
  forall_not_of_not_exists h1
have h3 : ∀ x, f x ≤ a := by
  intro x
  have h3a : ¬ f x > a := h2 x
  show f x ≤ a
  exact le_of_not_gt h3a
have h4 : CotaSuperior f a := h3
have h5 : ∃ b, CotaSuperior f b := ⟨a, h4⟩
have h6 : acotadaSup f := h5
show False
exact h h6

-- 2ª demostración
-- =====

example
  (h : ¬acotadaSup f)
  : ∀ a, ∃ x, f x > a :=
by
  intro a
  -- a : ℝ
  -- ⊢ ∃ x, f x > a
  by_contra h1
  -- h1 : ¬∃ x, f x > a
  -- ⊢ False
  apply h
  -- ⊢ acotadaSup f
  use a
  -- ⊢ CotaSuperior f a
  intro x
  -- x : ℝ
  -- ⊢ f x ≤ a
  apply le_of_not_gt
  -- ⊢ ¬f x > a
  intro h2
  -- h2 : f x > a
  -- ⊢ False
  apply h1

```

```

--  $\vdash \exists x, f\ x > a$ 
use x
--  $\vdash f\ x > a$ 
exact h2

-- 3ª demostración
-- =====

example
  (h :  $\neg$ acotadaSup f)
  :  $\forall a, \exists x, f\ x > a :=$ 
by
  unfold acotadaSup at h
  --  $h : \neg \exists a, CotaSuperior\ f\ a$ 
  unfold CotaSuperior at h
  --  $h : \neg \exists a, \forall (x : \mathbb{R}), f\ x \leq a$ 
  push_neg at h
  --  $\forall (a : \mathbb{R}), \exists x, f\ x > a$ 
  exact h

-- 4ª demostración
-- =====

example
  (h :  $\neg$ acotadaSup f)
  :  $\forall a, \exists x, f\ x > a :=$ 
by
  simp only [acotadaSup, CotaSuperior] at h
  --  $h : \neg \exists a, \forall (x : \mathbb{R}), f\ x \leq a$ 
  push_neg at h
  --  $\forall (a : \mathbb{R}), \exists x, f\ x > a$ 
  exact h

-- 5ª demostración
-- =====

example
  (h :  $\neg$ acotadaSup f) :
   $\forall a, \exists x, f\ x > a :=$ 
by
  contrapose h
  --  $h : \neg \forall (a : \mathbb{R}), \exists x, f\ x > a$ 
  --  $\vdash \neg \neg$ acotadaSup f
  push_neg at *
  --  $h : \exists a, \forall (x : \mathbb{R}), f\ x \leq a$ 

```

```

--  $\vdash \text{acotadaSup } f$ 
exact h

-- 6ª demostración
-- =====

example
  (h :  $\neg \text{acotadaSup } f$ ) :
   $\forall a, \exists x, f\ x > a :=$ 
by
  contrapose! h
  --  $h : \exists a, \forall (x : \mathbb{R}), f\ x \leq a$ 
  --  $\vdash \text{acotadaSup } f$ 
  exact h

-- Lemas usados
-- =====

-- variable  $\{\alpha : \text{Type } \_ \}$ 
-- variable  $(P : \alpha \rightarrow \text{Prop})$ 
-- #check (forall_not_of_not_exists :  $(\neg \exists x, P\ x) \rightarrow \forall x, \neg P\ x$ )
--
-- variable  $(a\ b : \mathbb{R})$ 
-- #check (le_of_not_gt :  $\neg a > b \rightarrow a \leq b$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.13. Si $\neg(\forall a)(\exists x)[f(x) > a]$ , entonces $f$ está acotada superiormente

```

-----
-- Demostrar que si  $\neg(\forall a)(\exists x)[f(x) > a]$ , entonces  $f$  está acotada
-- superiormente.
-----

-- Demostración en lenguaje natural
-- =====

-- Tenemos que demostrar que  $f$  es acotada superiormente; es decir, que
--  $(\exists a)(\forall x)[f(x) \leq a]$ 
-- que es exactamente la fórmula obtenida interiorizando la negación en
-- la hipótesis.

```

```

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

def CotaSuperior (f : ℝ → ℝ) (a : ℝ) : Prop :=
  ∀ x, f x ≤ a

def acotadaSup (f : ℝ → ℝ) :=
  ∃ a, CotaSuperior f a

variable (f : ℝ → ℝ)

-- 1ª demostración
-- =====

example
  (h : ¬∀ a, ∃ x, f x > a)
  : acotadaSup f :=
by
  unfold acotadaSup
  -- ⊢ ∃ a, CotaSuperior f a
  unfold CotaSuperior
  -- ⊢ ∃ a, ∀ (x : ℝ), f x ≤ a
  push_neg at h
  -- h : ∃ a, ∀ (x : ℝ), f x ≤ a
  exact h

-- 2ª demostración
-- =====

example
  (h : ¬∀ a, ∃ x, f x > a)
  : acotadaSup f :=
by
  unfold acotadaSup CotaSuperior
  -- ⊢ ∃ a, ∀ (x : ℝ), f x ≤ a
  push_neg at h
  -- h : ∃ a, ∀ (x : ℝ), f x ≤ a
  exact h

-- 3ª demostración
-- =====

```

```

example
  (h :  $\neg \forall a, \exists x, f\ x > a$ )
  : acotadaSup f :=
by
  push_neg at h
  --  $h : \exists a, \forall (x : \mathbb{R}), f\ x \leq a$ 
  exact h

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.14. Suma de funciones monótonas

```

-- -----
-- Demostrar que la suma de dos funciones monótonas es monótona.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se usará el siguiente lema:
--   add_le_add :  $a \leq b \rightarrow c \leq d \rightarrow a + c \leq b + d$ 
--
-- Supongamos que  $f$  y  $g$  son monótonas y tenemos que demostrar que  $f+g$ 
-- también lo es; que
--    $\forall a\ b, a \leq b \rightarrow (f + g)(a) \leq (f + g)(b)$ 
-- Sean  $a, b \in \mathbb{R}$  tales que
--    $a \leq b$  (1)
-- Entonces, por ser  $f$  y  $g$  monótonas se tiene
--    $f(a) \leq f(b)$  (2)
--    $g(a) \leq g(b)$  (3)
-- Entonces,
--    $(f + g)(a) = f(a) + g(a)$ 
--                $\leq f(b) + g(b)$  [por add_le_add, (2) y (3)]
--                $= (f + g)(b)$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (f g :  $\mathbb{R} \rightarrow \mathbb{R}$ )

-- 1ª demostración

```

```

example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f + g) :=
by
  have h1 :  $\forall a b, a \leq b \rightarrow (f + g) a \leq (f + g) b$ 
  { intros a b hab
    have h2 :  $f a \leq f b := mf hab$ 
    have h3 :  $g a \leq g b := mg hab$ 
    calc (f + g) a
      = f a + g a := rfl
      _  $\leq f b + g b := add\_le\_add h2 h3$ 
      _ = (f + g) b := rfl }
  show Monotone (f + g)
  exact h1

-- 2ª demostración
example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f + g) :=
by
  have h1 :  $\forall a b, a \leq b \rightarrow (f + g) a \leq (f + g) b$ 
  { intros a b hab
    calc (f + g) a
      = f a + g a := rfl
      _  $\leq f b + g b := add\_le\_add (mf hab) (mg hab)$ 
      _ = (f + g) b := rfl }
  show Monotone (f + g)
  exact h1

-- 3ª demostración
example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f + g) :=
by
  have h1 :  $\forall a b, a \leq b \rightarrow (f + g) a \leq (f + g) b$ 
  { intros a b hab
    show (f + g) a  $\leq (f + g) b$ 
    exact add_le_add (mf hab) (mg hab) }
  show Monotone (f + g)
  exact h1

-- 4ª demostración

```



```

example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f + g) :=
by
  -- a b : ℝ
  -- hab : a ≤ b
  intros a b hab
  apply add_le_add
  . -- f a ≤ f b
    apply mf hab
  . -- g a ≤ g b
    apply mg hab

-- 5ª demostración
example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f + g) :=
λ _ _ hab ↦ add_le_add (mf hab) (mg hab)

-- Lemas usados
-- =====

-- variable (a b c d : ℝ)
-- #check (add_le_add : a ≤ b → c ≤ d → a + c ≤ b + d)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.15. Si $c$ es no negativo y $f$ es monótona, entonces $cf$ es monótona.

```

-----
-- Demostrar que si c es no negativo y f es monótona, entonces cf es
-- monótona.
-----

-- Demostración en lenguaje natural
-- =====

-- Se usará el Lema
-- mul_le_mul_of_nonneg_left : b ≤ c → 0 ≤ a → a * b ≤ a * c

```

```

--
-- Tenemos que demostrar que
--    $(\forall a, b \in \mathbb{R}) [a \leq b \rightarrow (cf)(a) \leq (cf)(b)]$ 
-- Sean  $a, b \in \mathbb{R}$  tales que  $a \leq b$ . Puesto que  $f$  es monótona, se tiene
--    $f(a) \leq f(b)$ .
-- y, junto con la hipótesis de que  $c$  es no negativo, usando el lema
-- mul_le_mul_of_nonneg_left, se tiene que
--    $cf(a) \leq cf(b)$ 
-- que es lo que había que demostrar.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (f : ℝ → ℝ)
variable {c : ℝ}

-- 1ª demostración
example
  (mf : Monotone f)
  (nnc : 0 ≤ c)
  : Monotone (fun x ↦ c * f x) :=
by
  have h1 : ∀ a b, a ≤ b → (fun x ↦ c * f x) a ≤ (fun x ↦ c * f x) b
  { intros a b hab
    have h2 : f a ≤ f b := mf hab
    show (fun x ↦ c * f x) a ≤ (fun x ↦ c * f x) b
      exact mul_le_mul_of_nonneg_left h2 nnc }
  show Monotone (fun x ↦ c * f x)
  exact h1

-- 2ª demostración
example
  (mf : Monotone f)
  (nnc : 0 ≤ c)
  : Monotone (fun x ↦ c * f x) :=
by
  -- a b : ℝ
  -- hab : a ≤ b
  intros a b hab
  -- (fun x => c * f x) a ≤ (fun x => c * f x) b
  apply mul_le_mul_of_nonneg_left
  . -- f a ≤ f b
    apply mf hab

```

```

. --  $0 \leq c$ 
  apply nnc

-- 3ª demostración
example (mf : Monotone f) (nnc :  $0 \leq c$ ) :
  Monotone (fun x ↦ c * f x) :=
λ _ _ hab ↦ mul_le_mul_of_nonneg_left (mf hab) nnc

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (mul_le_mul_of_nonneg_left :  $b \leq c \rightarrow 0 \leq a \rightarrow a * b \leq a * c$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.16. La composición de dos funciones monótonas es monótona

```

-- -----
-- Demostrar que la composición de dos funciones monótonas es monótona.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Sean  $f$  y  $g$  dos funciones monótonas de  $\mathbb{R}$  en  $\mathbb{R}$ . Tenemos que demostrar
-- que  $f \circ g$  es monótona; es decir, que
--  $(\forall a, b \in \mathbb{R}) [a \leq b \rightarrow (f \circ g)(a) \leq (f \circ g)(b)]$ 
-- Sean  $a, b \in \mathbb{R}$  tales que  $a \leq b$ . Por ser  $g$  monótona, se tiene
--  $g(a) \leq g(b)$ 
-- y, por ser  $f$  monótona, se tiene
--  $f(g(a)) \leq f(g(b))$ 
-- Finalmente, por la definición de composición,
--  $(f \circ g)(a) \leq (f \circ g)(b)$ 
-- que es lo que había que demostrar.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

```

```

variable (f g : ℝ → ℝ)

-- 1ª demostración
example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f ∘ g) :=
by
  have h1 : ∀ a b, a ≤ b → (f ∘ g) a ≤ (f ∘ g) b
  { intros a b hab
    have h1 : g a ≤ g b := mg hab
    show (f ∘ g) a ≤ (f ∘ g) b
    exact mf h1 }
  show Monotone (f ∘ g)
  exact h1

-- 2ª demostración
example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f ∘ g) :=
by
  have h1 : ∀ a b, a ≤ b → (f ∘ g) a ≤ (f ∘ g) b
  { intros a b hab
    show (f ∘ g) a ≤ (f ∘ g) b
    exact mf (mg hab) }
  show Monotone (f ∘ g)
  exact h1

-- 3ª demostración
example
  (mf : Monotone f)
  (mg : Monotone g)
  : Monotone (f ∘ g) :=
by
  -- a b : ℝ
  -- hab : a ≤ b
  intros a b hab
  -- (f ∘ g) a ≤ (f ∘ g) b
  apply mf
  -- g a ≤ g b
  apply mg
  -- a ≤ b
  apply hab

```

```
-- 4ª demostración
example (mf : Monotone f) (mg : Monotone g) :
  Monotone (f ∘ g) :=
λ _ _ hab ↦ mf (mg hab)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.17. Si $f$ es monótona y $f(a) < f(b)$ , entonces $a < b$

```
-- -----
-- Demostrar que si f es monótona y f(a) < f(b), entonces a < b
-- -----

-- Demostración en lenguaje natural
-- =====

-- Usaremos los lemas
--   a ≠ b → a < b                                     (L1)
--   a ≥ b → a ≠ b                                     (L2)
--
-- Usando el lema L1, basta demostrar que a ≠ b. Lo haremos por
-- reducción al absurdo. Para ello, supongamos que a ≥ b. Como f es
-- monótona, se tiene que f(a) ≥ f(b) y, aplicando el lema L2,
-- f(a) ≠ f(b), que contradice a la hipótesis.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable (f : ℝ → ℝ)
variable (a b : ℝ)

-- 1ª demostración
-- =====

example
  (h1 : Monotone f)
  (h2 : f a < f b)
  : a < b :=
by
  apply lt_of_not_ge
```

```

--  $\vdash \neg a \geq b$ 
intro h3
--  $h3 : a \geq b$ 
--  $\vdash \text{False}$ 
have h4 :  $f\ a \geq f\ b := h1\ h3$ 
have h5 :  $\neg f\ a < f\ b := \text{not\_lt\_of\_ge}\ h4$ 
exact h5 h2

-- 2ª demostración
-- =====

example
  (h1 : Monotone f)
  (h2 :  $f\ a < f\ b$ )
  :  $a < b :=$ 
by
  apply lt_of_not_ge
  --  $\vdash \neg a \geq b$ 
  intro h3
  --  $h3 : a \geq b$ 
  --  $\vdash \text{False}$ 
  have h5 :  $\neg f\ a < f\ b := \text{not\_lt\_of\_ge}\ (h1\ h3)$ 
  exact h5 h2

-- 3ª demostración
-- =====

example
  (h1 : Monotone f)
  (h2 :  $f\ a < f\ b$ )
  :  $a < b :=$ 
by
  apply lt_of_not_ge
  --  $\vdash \neg a \geq b$ 
  intro h3
  --  $h3 : a \geq b$ 
  --  $\vdash \text{False}$ 
  exact (not_lt_of_ge (h1 h3)) h2

-- 4ª demostración
-- =====

example
  (h1 : Monotone f)
  (h2 :  $f\ a < f\ b$ )

```

12.18. Si  $a, b \in \mathbb{R}$  tales que  $a \leq b$  y  $f(b) < f(a)$ , entonces  $f$  no es monótona. 263

```
: a < b :=
by
  apply lt_of_not_ge
  --  $\vdash \neg a \geq b$ 
  exact fun h3 ↦ (not_lt_of_ge (h1 h3)) h2

-- 5ª demostración
-- =====

example
  (h1 : Monotone f)
  (h2 : f a < f b)
  : a < b :=
lt_of_not_ge (fun h3 ↦ (not_lt_of_ge (h1 h3)) h2)

-- Lemas usados
-- =====

-- #check (lt_of_not_ge :  $\neg a \geq b \rightarrow a < b$ )
-- #check (not_lt_of_ge :  $a \geq b \rightarrow \neg a < b$ )
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.18. Si $a, b \in \mathbb{R}$ tales que $a \leq b$ y $f(b) < f(a)$ , entonces $f$ no es monótona

```
-- -----
-- Demostrar que si  $a, b \in \mathbb{R}$  tales que  $(a \leq b)$  y  $(f\ b < f\ a)$ , entonces  $f$ 
-- no es monótona.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Usaremos el lema
--  $a \geq b \rightarrow a \neq b$  (L1)
--
-- Lo demostraremos por reducción al absurdo. Para ello, supongamos que
--  $f$  es monótona. Entonces, como  $a \leq b$ , se tiene  $f(a) \leq f(b)$  y, por el
-- lema L1,  $f\ b \neq f\ a$ , en contradicción con la hipótesis.

-- Demostraciones con Lean4
```

```

-- =====

import Mathlib.Data.Real.Basic
variable (f : ℝ → ℝ)
variable (a b : ℝ)

-- 1ª demostración
-- =====

example
  (h1 : a ≤ b)
  (h2 : f b < f a)
  : ¬ Monotone f :=
by
  intro h3
  -- h3 : Monotone f
  -- ⊢ False
  have h4 : f a ≤ f b := h3 h1
  have h5 : ¬(f b < f a) := not_lt_of_ge h4
  exact h5 h2

-- 2ª demostración
-- =====

example
  (h1 : a ≤ b)
  (h2 : f b < f a)
  : ¬ Monotone f :=
by
  intro h3
  -- h3 : Monotone f
  -- ⊢ False
  have h5 : ¬(f b < f a) := not_lt_of_ge (h3 h1)
  exact h5 h2

-- 3ª demostración
-- =====

example
  (h1 : a ≤ b)
  (h2 : f b < f a)
  : ¬ Monotone f :=
by
  intro h3
  -- h3 : Monotone f

```



```

-- ⊢ False
exact (not_lt_of_ge (h3 h1)) h2

-- 4ª demostración
-- =====

example
  (h1 : a ≤ b)
  (h2 : f b < f a)
  : ¬ Monotone f :=
fun h3 ↦ (not_lt_of_ge (h3 h1)) h2

-- Lemas usados
-- =====

-- #check (not_lt_of_ge : a ≥ b → ¬a < b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.19. No para toda $f : \mathbb{R} \rightarrow \mathbb{R}$ monótona, $(\forall a, b)[f(a) \leq f(b) \rightarrow a \leq b]$

```

-----
-- Demostrar que no para toda  $f : \mathbb{R} \rightarrow \mathbb{R}$  monótona,
--  $(\forall a, b)[f(a) \leq f(b) \rightarrow a \leq b]$ 
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que
--  $(\forall f)[f \text{ es monótona} \rightarrow (\forall a, b)[f(a) \leq f(b) \rightarrow a \leq b]]$  (1)
-- Sea  $f : \mathbb{R} \rightarrow \mathbb{R}$  la función constante igual a cero (es decir,
--  $(\forall x \in \mathbb{R})[f(x) = 0]$ 
-- Entonces,  $f$  es monótona y  $f(1) \leq f(0)$  (ya que
--  $f(1) = 0 \leq 0 = f(0)$ ). Luego, por (1),  $1 \leq 0$  que es una
-- contradicción.

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

```

```

-- 1ª demostración
-- =====

example :
  ¬∀ {f : ℝ → ℝ}, Monotone f → ∀ {a b}, f a ≤ f b → a ≤ b :=
by
  intro h1
  -- h1 : ∀ {f : ℝ → ℝ}, Monotone f → ∀ {a b : ℝ}, f a ≤ f b → a ≤ b
  -- ⊢ False
  let f := fun _ : ℝ ↦ (0 : ℝ)
  have h2 : Monotone f := monotone_const
  have h3 : f 1 ≤ f 0 := le_refl 0
  have h4 : 1 ≤ 0 := h1 h2 h3
  linarith

-- Lemas usados
-- =====

-- variable (a c : ℝ)
-- #check (le_refl a : a ≤ a)
-- #check (monotone_const : Monotone fun _ : ℝ ↦ c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.20. Si $f$ no es monótona, entonces $\exists x \exists y [x \leq y \wedge f(y) < f(x)]$

```

-----
-- Demostrar que si  $f$  no es monótona, entonces existen  $x, y$  tales que
--  $x \leq y$  y  $f(y) < f(x)$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Usaremos los siguientes lemas.
--   ¬(∀x)P(x) ↔ (∃ x)¬P(x)                                     (L1)
--   ¬(p → q) ↔ p ∧ ¬q                                         (L2)
--   (∀a, b ∈ ℝ)[¬b ≤ a → a < b]                               (L3)
--
-- Por la definición de función monótona,
--   ¬(∀x)(∀y)[x ≤ y → f(x) ≤ f(y)]

```

```

-- Aplicando L1 se tiene
--    $(\exists x) \neg (\forall y) [x \leq y \rightarrow f(x) \leq f(y)]$ 
-- Sea  $a$  tal que
--    $\neg (\forall y) [a \leq y \rightarrow f(a) \leq f(y)]$ 
-- Aplicando L1 se tiene
--    $(\exists y) \neg [a \leq y \rightarrow f(a) \leq f(y)]$ 
-- Sea  $b$  tal que
--    $\neg [a \leq b \rightarrow f(a) \leq f(b)]$ 
-- Aplicando L2 se tiene que
--    $a \leq b \wedge \neg (f(a) \leq f(b))$ 
-- Aplicando L3 se tiene que
--    $a \leq b \wedge f(b) < f(a)$ 
-- Por tanto,
--    $(\exists x, y) [x \leq y \wedge f(y) < f(x)]$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable (f : ℝ → ℝ)

-- 1ª demostración
-- =====

example
  (h : ¬Monotone f)
  : ∃ x y, x ≤ y ∧ f y < f x :=
by
  have h1 : ¬∀ x y, x ≤ y → f x ≤ f y := h
  have h2 : ∃ x, ¬(∀ y, x ≤ y → f x ≤ f y) := not_forall.mp h1
  rcases h2 with ⟨a, ha : ¬∀ y, a ≤ y → f a ≤ f y⟩
  have h3 : ∃ y, ¬(a ≤ y → f a ≤ f y) := not_forall.mp ha
  rcases h3 with ⟨b, hb : ¬(a ≤ b → f a ≤ f b)⟩
  have h4 : a ≤ b ∧ ¬(f a ≤ f b) := not_imp.mp hb
  have h5 : a ≤ b ∧ f b < f a := ⟨h4.1, lt_of_not_le h4.2⟩
  use a, b
  -- ⊢ a ≤ b ∧ f b < f a
  exact h5

-- 2ª demostración
-- =====

example
  (h : ¬Monotone f)
  : ∃ x y, x ≤ y ∧ f y < f x :=

```

```

by
  simp only [Monotone] at h
  -- h :  $\neg \forall [a\ b : \mathbb{R}], a \leq b \rightarrow f\ a \leq f\ b$ 
  push_neg at h
  -- h :  $\text{Exists fun } [a] \Rightarrow \text{Exists fun } [b] \Rightarrow a \leq b \wedge f\ b < f\ a$ 
  exact h

-- Lemas usados
-- =====

-- variable { $\alpha : \text{Type } \_$ }
-- variable (P :  $\alpha \rightarrow \text{Prop}$ )
-- variable (p q : Prop)
-- variable (a b :  $\mathbb{R}$ )
-- #check (not_forall :  $(\neg \forall x, P\ x) \leftrightarrow \exists x, \neg P\ x$ )
-- #check (not_imp :  $\neg(p \rightarrow q) \leftrightarrow p \wedge \neg q$ )
-- #check (lt_of_not_le :  $\neg b \leq a \rightarrow a < b$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.21. $f: \mathbb{R} \rightarrow \mathbb{R}$ no es monótona syss $(\exists x, y)(x \leq y \wedge f(x) > f(y))$

```

-- -----
-- Demostrar que  $f : \mathbb{R} \rightarrow \mathbb{R}$  no es monótona syss existen  $x$  e  $y$  tales
-- que  $x \leq y$  y  $f(x) > f(y)$ .
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por la siguiente cadena de equivalencias:
--    $f$  es no monótona  $\leftrightarrow \neg(\forall x\ y)[x \leq y \rightarrow f(x) \leq f(y)]$ 
--    $\leftrightarrow (\exists x\ y)[x \leq y \wedge f(x) > f(y)]$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {f :  $\mathbb{R} \rightarrow \mathbb{R}$ }

-- 1ª demostración

```

```

-- =====

example :
  ¬Monotone f ↔ ∃ x y, x ≤ y ∧ f x > f y :=
calc
  ¬Monotone f
    ↔ ¬∀ x y, x ≤ y → f x ≤ f y := by rw [Monotone]
  _ ↔ ∃ x y, x ≤ y ∧ f y < f x := by simp_all only [not_forall, not_le, exists_prop]
  _ ↔ ∃ x y, x ≤ y ∧ f x > f y := by rfl

-- 2ª demostración
-- =====

example :
  ¬Monotone f ↔ ∃ x y, x ≤ y ∧ f x > f y :=
calc
  ¬Monotone f
    ↔ ¬∀ x y, x ≤ y → f x ≤ f y := by rw [Monotone]
  _ ↔ ∃ x y, x ≤ y ∧ f x > f y := by aesop

-- 3ª demostración
-- =====

example :
  ¬Monotone f ↔ ∃ x y, x ≤ y ∧ f x > f y :=
by
  rw [Monotone]
  -- ⊢ (¬∀ [a b : ℝ], a ≤ b → f a ≤ f b) ↔ ∃ x y, x ≤ y ∧ f x > f y
  push_neg
  -- ⊢ (Exists fun [a] => Exists fun [b] => a ≤ b ∧ f b < f a) ↔ ∃ x y, x ≤ y ∧ f x > f y
  rfl

-- 4ª demostración
-- =====

lemma not_Monotone_iff :
  ¬Monotone f ↔ ∃ x y, x ≤ y ∧ f x > f y :=
by
  rw [Monotone]
  -- ⊢ (¬∀ [a b : ℝ], a ≤ b → f a ≤ f b) ↔ ∃ x y, x ≤ y ∧ f x > f y
  aesop

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.22. La función $x \mapsto -x$ no es monótona creciente

```

-----
-- Demostrar que la función opuesta no es monótona.
-----

-- Demostración en lenguaje natural
-- =====

-- Usando el lema del ejercicio anterior que afirma que una función f no
-- es monótona syss existen x e y tales que  $x \leq y$  y  $f(x) > f(y)$ , basta
-- demostrar que
--  $(\exists x y)[x \leq y \wedge -x > -y]$ 
-- Basta elegir 2 y 3 ya que
--  $2 \leq 3 \wedge -2 > -3$ 

-- Demostración con Lean4
-- =====

import Mathlib.Data.Real.Basic
import src.CNS_de_no_monotona

example : ¬Monotone fun x : ℝ ↦ -x :=
by
  apply not_Monotone_iff.mpr
  --  $\vdash \exists x y, x \leq y \wedge -x > -y$ 
  use 2, 3
  --  $\vdash 2 \leq 3 \wedge -2 > -3$ 
  norm_num

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.23. La suma de dos funciones pares es par

```

-----
-- Demostrar que la suma de dos funciones pares es par.
-----

-- Demostración en lenguaje natural
-- =====

```

```
-- Supongamos que f y g son funciones pares. Tenemos que demostrar que
-- f+g es par; es decir, que
--    $(\forall x \in \mathbb{R}) (f + g)(x) = (f + g)(-x)$ 
-- Sea  $x \in \mathbb{R}$ . Entonces,
--    $(f + g) x = f x + g x$ 
--                $= f (-x) + g x$       [porque f es par]
--                $= f (-x) + g (-x)$  [porque g es par]
--                $= (f + g) (-x)$ 
```

```
-- Demostraciones con Lean4
```

```
-- =====
```

```
import Mathlib.Data.Real.Basic
```

```
variable (f g : ℝ → ℝ)
```

```
-- (esPar f) expresa que f es par.
```

```
def esPar (f : ℝ → ℝ) : Prop :=
```

```
   $\forall x, f x = f (-x)$ 
```

```
-- 1ª demostración
```

```
-- =====
```

```
example
```

```
  (h1 : esPar f)
```

```
  (h2 : esPar g)
```

```
  : esPar (f + g) :=
```

```
by
```

```
  intro x
```

```
  have h1 : f x = f (-x) := h1 x
```

```
  have h2 : g x = g (-x) := h2 x
```

```
  calc (f + g) x
```

```
    = f x + g x      := rfl
```

```
    _ = f (-x) + g x := congrArg (. + g x) h1
```

```
    _ = f (-x) + g (-x) := congrArg (f (-x) + .) h2
```

```
    _ = (f + g) (-x) := rfl
```

```
-- 2ª demostración
```

```
-- =====
```

```
example
```

```
  (h1 : esPar f)
```

```
  (h2 : esPar g)
```

```
  : esPar (f + g) :=
```

```
by
```

```

intro x
calc (f + g) x
  = f x + g x      := rfl
  _ = f (-x) + g x  := congrArg (. + g x) (h1 x)
  _ = f (-x) + g (-x) := congrArg (f (-x) + .) (h2 x)
  _ = (f + g) (-x)  := rfl

-- 3ª demostración
-- =====

example
  (h1 : esPar f)
  (h2 : esPar g)
  : esPar (f + g) :=
by
  intro x
  calc (f + g) x
    = f x + g x      := rfl
    _ = f (-x) + g (-x) := by rw [h1, h2]
    _ = (f + g) (-x)  := rfl

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.24. El producto de dos funciones impares es par

```

-- -----
-- Demostrar que el producto de dos funciones impares es par.
-- -----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que f y g son funciones impares. Tenemos que demostrar que
-- f·g es par; es decir, que
--    $(\forall x \in \mathbb{R}) (f \cdot g)(x) = (f \cdot g)(-x)$ 
-- Sea  $x \in \mathbb{R}$ . Entonces,
--    $(f \cdot g) x = f(x)g(x)$ 
--    $= (-f(-x))g(x)$            [porque f es impar]
--    $= (-f(-x))(-g(-x))$        [porque g es impar]
--    $= f(-x)g(-x)$ 
--    $= (f \cdot g)(-x)$ 

```



```

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (f g : ℝ → ℝ)

-- (esPar f) expresa que f es par.
def esPar (f : ℝ → ℝ) : Prop :=
  ∀ x, f x = f (-x)

-- (esImpar f) expresa que f es impar.
def esImpar (f : ℝ → ℝ) : Prop :=
  ∀ x, f x = - f (-x)

-- 1ª demostración
example
  (h1 : esImpar f)
  (h2 : esImpar g)
  : esPar (f * g) :=
by
  intro x
  have h1 : f x = -f (-x) := h1 x
  have h2 : g x = -g (-x) := h2 x
  calc (f * g) x
    = f x * g x                := rfl
    _ = (-f (-x)) * g x        := congrArg (. * g x) h1
    _ = (-f (-x)) * (-g (-x)) := congrArg ((-f (-x)) * .) h2
    _ = f (-x) * g (-x)        := neg_mul_neg (f (-x)) (g (-x))
    _ = (f * g) (-x)           := rfl

-- 2ª demostración
example
  (h1 : esImpar f)
  (h2 : esImpar g)
  : esPar (f * g) :=
by
  intro x
  calc (f * g) x
    = f x * g x                := rfl
    _ = (-f (-x)) * g x        := congrArg (. * g x) (h1 x)
    _ = (-f (-x)) * (-g (-x)) := congrArg ((-f (-x)) * .) (h2 x)
    _ = f (-x) * g (-x)        := neg_mul_neg (f (-x)) (g (-x))
    _ = (f * g) (-x)           := rfl

```

```

-- 3ª demostración
example
  (h1 : esImpar f)
  (h2 : esImpar g)
  : esPar (f * g) :=
by
  intro x
  calc (f * g) x
    = f x * g x           := rfl
  _ = -f (-x) * -g (-x) := by rw [h1, h2]
  _ = f (-x) * g (-x)    := by rw [neg_mul_neg]
  _ = (f * g) (-x)       := rfl

-- 4ª demostración
example
  (h1 : esImpar f)
  (h2 : esImpar g)
  : esPar (f * g) :=
by
  intro x
  calc (f * g) x
    = f x * g x           := rfl
  _ = f (-x) * g (-x) := by rw [h1, h2, neg_mul_neg]
  _ = (f * g) (-x)       := rfl

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (neg_mul_neg a b : -a * -b = a * b)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.25. El producto de una función par por una impar es impar

```

-----
-- Demostrar que el producto de una función par por una impar es impar.
-----

-- Demostración en lenguaje natural
-- =====

```

```

-- Supongamos que f es una función par y g lo es impar. Tenemos que
-- demostrar que f·g es imppar; es decir, que
--    $(\forall x \in \mathbb{R}) (f \cdot g)(x) = -(f \cdot g)(-x)$ 
-- Sea  $x \in \mathbb{R}$ . Entonces,
--    $(f \cdot g) x = f(x)g(x)$ 
--                $= f(-x)g(x)$            [porque f es par]
--                $= f(-x)(-g(-x))$        [porque g es impar]
--                $= -f(-x)g(-x)$ 
--                $= -(f \cdot g)(-x)$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic

variable (f g : ℝ → ℝ)

-- (esPar f) expresa que f es par.
def esPar (f : ℝ → ℝ) : Prop :=
  ∀ x, f x = f (-x)

-- (esImpar f) expresa que f es impar.
def esImpar (f : ℝ → ℝ) : Prop :=
  ∀ x, f x = - f (-x)

-- 1ª demostración
example
  (h1 : esPar f)
  (h2 : esImpar g)
  : esImpar (f * g) :=
by
  intro x
  have h1 : f x = f (-x) := h1 x
  have h2 : g x = -g (-x) := h2 x
  calc (f * g) x
    = f x * g x                := rfl
    _ = (f (-x)) * g x          := congrArg (. * g x) h1
    _ = (f (-x)) * (-g (-x))    := congrArg (f (-x) * .) h2
    _ = -(f (-x) * g (-x))      := mul_neg (f (-x)) (g (-x))
    _ = -(f * g) (-x)           := rfl

-- 2ª demostración
example
  (h1 : esPar f)

```

```

(h2 : esImpar g)
: esImpar (f * g) :=
by
  intro x
  calc (f * g) x
    = f x * g x           := rfl
  _ = f (-x) * -g (-x)    := by rw [h1, h2]
  _ = -(f (-x) * g (-x)) := by rw [mul_neg]
  _ = -(f * g) (-x)      := rfl

-- 3ª demostración
example
  (h1 : esPar f)
  (h2 : esImpar g)
  : esImpar (f * g) :=
by
  intro x
  calc (f * g) x
    = f x * g x           := rfl
  _ = -(f (-x) * g (-x)) := by rw [h1, h2, mul_neg]
  _ = -((f * g) (-x))    := rfl

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (mul_neg a b : a * -b = -(a * b))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.26. Si $f$ es par y $g$ es impar, entonces $(f \circ g)$ es par

```

-----
-- Demostrar que si  $f$  es par y  $g$  es impar, entonces  $f \circ g$  es par.
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que  $f$  es una función par y  $g$  lo es impar. Tenemos que
-- demostrar que  $(f \circ g)$  es par; es decir, que

```

```
--       $(\forall x \in \mathbb{R}) (f \circ g)(x) = (f \circ g)(-x)$ 
-- Sea  $x \in \mathbb{R}$ . Entonces,
--       $(f \circ g)(x) = f(g(x))$ 
--                       $= f(-g(-x))$       [porque  $g$  es impar]
--                       $= f(g(-x))$       [porque  $f$  es par]
--                       $= (f \circ g)(-x)$ 
```

```
-- Demostraciones con Lean4
```

```
-- =====
```

```
import Mathlib.Data.Real.Basic
```

```
variable (f g :  $\mathbb{R} \rightarrow \mathbb{R}$ )
```

```
-- (esPar f) expresa que  $f$  es par.
```

```
def esPar (f :  $\mathbb{R} \rightarrow \mathbb{R}$ ) : Prop :=
   $\forall x, f\ x = f\ (-x)$ 
```

```
-- (esImpar f) expresa que  $f$  es impar.
```

```
def esImpar (f :  $\mathbb{R} \rightarrow \mathbb{R}$ ) : Prop :=
   $\forall x, f\ x = -\ f\ (-x)$ 
```

```
-- 1ª demostración
```

```
example
```

```
(h1 : esPar f)
(h2 : esImpar g)
: esPar (f  $\circ$  g) :=
```

```
by
```

```
intro x
calc (f  $\circ$  g) x
    = f (g x)           := rfl
    _ = f (-g (-x))    := congr_arg f (h2 x)
    _ = f (g (-x))     := (h1 (g (-x))).symm
    _ = (f  $\circ$  g) (-x) := rfl
```

```
-- 2ª demostración
```

```
example
```

```
(h1 : esPar f)
(h2 : esImpar g)
: esPar (f  $\circ$  g) :=
```

```
by
```

```
intro x
calc (f  $\circ$  g) x
    = f (g x)           := rfl
    _ = f (-g (-x))    := by rw [h2]
```

```

_ = f (g (-x))    := by rw [← h1]
_ = (f ∘ g) (-x) := rfl

-- 3ª demostración
example
  (h1 : esPar f)
  (h2 : esImpar g)
  : esPar (f ∘ g) :=
by
  intro x
  calc (f ∘ g) x
    = f (g x)      := rfl
    _ = f (g (-x)) := by rw [h2, ← h1]

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.27. Para cualquier conjunto $s$ , $s \subseteq s$

```

-- -----
-- Demostrar que para cualquier conjunto  $s$ ,  $s \subseteq s$ .
-- -----

-- Demostración en lenguaje natural
-- =====

-- Tenemos que demostrar que
--    $(\forall x) [x \in s \rightarrow x \in s]$ 
-- Sea  $x$  tal que
--    $x \in s$  (1)
-- Entonces, por (1), se tiene que
--    $x \in s$ 
-- que es lo que teníamos que demostrar.

-- Demostraciones con Lean 4
-- =====

import Mathlib.Tactic

variable {α : Type _}
variable (s : Set α)

-- 1ª demostración
example : s ⊆ s :=

```

12.28. Las funciones  $f(x,y) = (x + y)^2$  y  $g(x,y) = x^2 + 2xy + y^2$  son iguales 379

```
by
  intro x xs
  exact xs

-- 2ª demostración
example : s ⊆ s :=
  fun (x : α) (xs : x ∈ s) ↦ xs

-- 3ª demostración
example : s ⊆ s :=
  fun _ xs ↦ xs

-- 4ª demostración
example : s ⊆ s :=
  -- by exact?
  rfl.subset

-- 5ª demostración
example : s ⊆ s :=
by rfl
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 12.28. Las funciones $f(x,y) = (x + y)^2$ y $g(x,y) = x^2 + 2xy + y^2$ son iguales

```
-----
-- Demostrar que
--   (fun x y : ℝ ↦ (x + y)^2) = (fun x y : ℝ ↦ x^2 + 2*x*y + y^2)
-----

import Mathlib.Data.Real.Basic

-- 1ª demostración
-- =====

example : (fun x y : ℝ ↦ (x + y)^2) = (fun x y : ℝ ↦ x^2 + 2*x*y + y^2) :=
by
  ext u v
  -- u v : ℝ
  -- ⊢ (u + v) ^ 2 = u ^ 2 + 2 * u * v + v ^ 2
  ring
```

```
-- Comentario: La táctica ext transforma las conclusiones de la forma
-- (fun x ↦ f x) = (fun x ↦ g x) en f x = g x.

-- 2ª demostración
-- =====

example : (fun x y : ℝ ↦ (x + y)^2) = (fun x y : ℝ ↦ x^2 + 2*x*y + y^2) :=
by { ext ; ring }
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)



# Capítulo 13

## Teoría de conjuntos

### 13.1. Si $r \subseteq s$ y $s \subseteq t$ , entonces $r \subseteq t$

```
-- Demostrar que si  $r \subseteq s$  y  $s \subseteq t$ , entonces  $r \subseteq t$ .
```

```
-- Demostración en lenguaje natural (LN)
```

```
-- =====
```

```
-- 1ª demostración en LN
```

```
-- -----
```

```
-- Tenemos que demostrar que
```

```
--  $(\forall x) [x \in r \rightarrow x \in t]$ 
```

```
-- Sea  $x$  tal que
```

```
--  $x \in r$ .
```

```
-- Puesto que  $r \subseteq s$ , se tiene que
```

```
--  $x \in s$ 
```

```
-- y, puesto que  $s \subseteq t$ , se tiene que
```

```
--  $x \in t$ 
```

```
-- que es lo que teníamos que demostrar.
```

```
-- 2ª demostración en LN
```

```
-- -----
```

```
-- Tenemos que demostrar que
```

```
--  $(\forall x) [x \in r \rightarrow x \in t]$ 
```

```
-- Sea  $x$  tal que
```

```
--  $x \in r$ 
```

```
-- Tenemos que demostrar que
```

```

--       $x \in t$ 
--      que, puesto que  $s \subseteq t$ , se reduce a
--       $x \in s$ 
--      que, puesto que  $r \subseteq s$ , se reduce a
--       $x \in r$ 
--      que es lo que hemos supuesto.

```

```

-- Demostraciones con Lean4
-- =====

```

```

import Mathlib.Tactic

```

```

open Set

```

```

variable {α : Type _}
variable (r s t : Set α)

```

```

-- 1ª demostración

```

```

example

```

```

  (rs :  $r \subseteq s$ )

```

```

  (st :  $s \subseteq t$ )

```

```

  :  $r \subseteq t$  :=

```

```

by

```

```

  intros x xr

```

```

  --  $xr : x \in r$ 

```

```

  have xs :  $x \in s$  := rs xr

```

```

  show  $x \in t$ 

```

```

  exact st xs

```

```

-- 2ª demostración

```

```

example

```

```

  (rs :  $r \subseteq s$ )

```

```

  (st :  $s \subseteq t$ )

```

```

  :  $r \subseteq t$  :=

```

```

by

```

```

  intros x xr

```

```

  --  $x : \alpha$ 

```

```

  --  $xr : x \in r$ 

```

```

  apply st

```

```

  --  $\vdash x \in s$ 

```

```

  apply rs

```

```

  --  $\vdash x \in r$ 

```

```

  exact xr

```

```

-- 3ª demostración

```

```

example
  (rs : r ⊆ s)
  (st : s ⊆ t)
  : r ⊆ t :=
fun _ xr ↦ st (rs xr)

-- 4ª demostración
example
  (rs : r ⊆ s)
  (st : s ⊆ t)
  : r ⊆ t :=
-- by exact?
Subset.trans rs st

-- 5ª demostración
example
  (rs : r ⊆ s)
  (st : s ⊆ t)
  : r ⊆ t :=
by tauto

-- Lemas usados
-- =====

-- #check (Subset.trans : r ⊆ s → s ⊆ t → r ⊆ t)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.2. Si $a$ es una cota superior de $s$ y $a \leq b$ , entonces $b$ es una cota superior de $s$

```

-- -----
-- Demostrar que si  $a$  es una cota superior de  $s$  y  $a \leq b$ , entonces  $b$  es
-- una cota superior de  $s$ .
-- -----

```

```

import Mathlib.Tactic

variable {α : Type _} [PartialOrder α]
variable (s : Set α)
variable (a b : α)

```

```

-- (CotaSupConj s a) afirma que a es una cota superior del conjunto s.
def CotaSupConj (s : Set  $\alpha$ ) (a :  $\alpha$ ) :=
   $\forall$  {x}, x  $\in$  s  $\rightarrow$  x  $\leq$  a

-- Demostración en lenguaje natural
-- =====

-- Tenemos que demostrar que
--   ( $\forall$  x) [x  $\in$  s  $\rightarrow$  x  $\leq$  b]
-- Sea x tal que x  $\in$  s. Entonces,
--   x  $\leq$  a [porque a es una cota superior de s]
--    $\leq$  b
-- Por tanto, x  $\leq$  b.

-- 1ª demostración
example
  (h1 : CotaSupConj s a)
  (h2 : a  $\leq$  b)
  : CotaSupConj s b :=
by
  intro x (xs : x  $\in$  s)
  have h3 : x  $\leq$  a := h1 xs
  show x  $\leq$  b
  exact le_trans h3 h2

-- 2ª demostración
example
  (h1 : CotaSupConj s a)
  (h2 : a  $\leq$  b)
  : CotaSupConj s b :=
by
  intro x (xs : x  $\in$  s)
  calc x  $\leq$  a := h1 xs
      _  $\leq$  b := h2
-

-- Lemas usados
-- =====

-- variable (c :  $\alpha$ )
-- #check (le_trans : a  $\leq$  b  $\rightarrow$  b  $\leq$  c  $\rightarrow$  a  $\leq$  c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.3. La función $(x \mapsto x + c)$ es inyectiva

```

-----
-- Demostrar que, para todo  $c$  la función
--    $f(x) = x + c$ 
-- es inyectiva
-----

-- Demostración en lenguaje natural
-- =====

-- Se usará el lema
--    $(\forall a, b, c) [a + b = c + b \rightarrow a = c]$                                 (L1)
-- Hay que demostrar que
--    $(\forall x_1 x_2) [f(x_1) = f(x_2) \rightarrow x_1 = x_2]$ 
-- Sean  $x_1, x_2$  tales que  $f(x_1) = f(x_2)$ . Entonces,
--    $x_1 + c = x_2 + c$ 
-- y, por L1,  $x_1 = x_2$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
open Function

variable {c : ℝ}

-- 1ª demostración
example : Injective ((. + c)) :=
by
  intro (x1 : ℝ) (x2 : ℝ) (h1 : x1 + c = x2 + c)
  show x1 = x2
  exact add_right_cancel h1

-- 2ª demostración
example : Injective ((. + c)) :=
by
  intro x1 x2 h1
  show x1 = x2
  exact add_right_cancel h1

-- 3ª demostración
example : Injective ((. + c)) :=
fun _ _ h ↦ add_right_cancel h

```

```
-- Lemas usados
-- =====

-- variable {a b : ℝ}
-- #check (add_right_cancel : a + b = c + b → a = c)
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.4. Si $c \neq 0$ , entonces la función $(x \mapsto cx)$ es inyectiva

```
-- -----
-- Ejercicio 3. Demostrar que para todo  $c$  distinto de cero la función
--    $f(x) = c * x$ 
-- es inyectiva
-- -----

-- Demostración en lenguaje natural
-- =====

-- Se usará el lema
--    $(\forall a, b, c) [a \neq 0 \rightarrow (a * b = a * c \leftrightarrow b = c)]$           (L1)
-- Hay que demostrar que
--    $(\forall x_1, x_2) [f(x_1) = f(x_2) \rightarrow x_1 = x_2]$ 
-- Sean  $x_1, x_2$  tales que  $f(x_1) = f(x_2)$ . Entonces,
--    $cx_1 = cx_2$ 
-- y, por L1 y puesto que  $c \neq 0$ , se tiene que
--    $x_1 = x_2$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
open Function
variable {c : ℝ}

-- 1ª demostración
example
  (h : c ≠ 0)
  : Injective ((c * .)) :=
by
  intro (x1 : ℝ) (x2 : ℝ) (h1 : c * x1 = c * x2)
```

```

show x1 = x2
exact (mul_right_inj' h).mp h1

-- 2ª demostración
example
  (h : c ≠ 0)
  : Injective ((c * .)) :=
fun _ _ h1 ↦ mul_left_cancel₀ h h1

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (mul_right_inj' : a ≠ 0 → (a * b = a * c ↔ b = c))
-- #check (mul_left_cancel₀ : a ≠ 0 → a * b = a * c → b = c)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.5. La composición de funciones inyectivas es inyectiva

```

-- Demostraciones en lenguaje natural (LN)
-- =====

-- 1ª demostración en LN
-- =====

-- Tenemos que demostrar que
--    $(\forall x, y) [(g \circ f)(x) = (g \circ f)(y) \rightarrow x = y]$ 
-- Sean  $x, y$  tales que
--    $(g \circ f)(x) = (g \circ f)(y)$ 
-- Entonces, por la definición de la composición,
--    $g(f(x)) = g(f(y))$ 
--  $y$ , ser  $g$  inyectiva,
--    $f(x) = f(y)$ 
--  $y$ , ser  $f$  inyectiva,
--    $x = y$ 

-- 2ª demostración en LN
-- =====

-- Tenemos que demostrar que

```

```

--       $(\forall x, y) [(g \circ f)(x) = (g \circ f)(y) \rightarrow x = y]$ 
-- Sean  $x, y$  tales que
--       $(g \circ f)(x) = (g \circ f)(y)$  (1)
-- y tenemos que demostrar que
--       $x = y$  (2)
-- El objetivo (2), usando que  $f$  es inyectiva, se reduce a
--       $f(x) = f(y)$ 
-- que, usando que  $g$  es inyectiva, se reduce a
--       $g(f(x)) = g(f(y))$ 
-- que, por la definición de la composición, coincide con (1).

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic

open Function

variable {α : Type _} {β : Type _} {γ : Type _}
variable {f : α → β} {g : β → γ}

-- 1ª demostración (basada en la 1ª en LN)
example
  (hg : Injective g)
  (hf : Injective f) :
  Injective (g ∘ f) :=
by
  intro (x : α) (y : α) (h1: (g ∘ f) x = (g ∘ f) y)
  have h2: g (f x) = g (f y) := h1
  have h3: f x = f y := hg h2
  show x = y
  exact hf h3

-- 2ª demostración
example
  (hg : Injective g)
  (hf : Injective f) :
  Injective (g ∘ f) :=
by
  intro (x : α) (y : α) (h1: (g ∘ f) x = (g ∘ f) y)
  have h2: f x = f y := hg h1
  show x = y
  exact hf h2

-- 3ª demostración

```



```

example
  (hg : Injective g)
  (hf : Injective f) :
  Injective (g ∘ f) :=
by
  intro x y h
  exact hf (hg h)

-- 4ª demostración
example
  (hg : Injective g)
  (hf : Injective f) :
  Injective (g ∘ f) :=
fun _ _ h ↦ hf (hg h)

-- 5ª demostración (basada en la 2ª en LN)
example
  (hg : Injective g)
  (hf : Injective f) :
  Injective (g ∘ f) :=
by
  intros x y h
  -- x y : α
  -- h : (g ∘ f) x = (g ∘ f) y
  apply hf
  -- ⊢ f x = f y
  apply hg
  -- ⊢ g (f x) = g (f y)
  apply h

-- 6ª demostración
example
  (hg : Injective g)
  (hf : Injective f) :
  Injective (g ∘ f) :=
-- by exact?
Injective.comp hg hf

-- 7ª demostración
example
  (hg : Injective g)
  (hf : Injective f) :
  Injective (g ∘ f) :=
by tauto

```

```
-- Lemas usados
-- =====

-- #check (Injective.comp : Injective g → Injective f → Injective (g ∘ f))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.6. La función $(x \mapsto x + c)$ es suprayectiva

```
-- -----
-- Demostrar que para todo número real  $c$ , la función
--  $f(x) = x + c$ 
-- es suprayectiva.
-- -----
```

```
-- Demostración en lenguaje natural
-- =====
```

```
-- Tenemos que demostrar que
--  $(\forall x \in \mathbb{R})(\exists y \in \mathbb{R})[y+c = x]$ 
-- Sea  $x \in \mathbb{R}$ . Entonces,  $y = x-c \in \mathbb{R}$  y
--  $y + c = (x - c) + c$ 
--  $= x$ 
```

```
-- Demostraciones con Lean4
-- =====
```

```
import Mathlib.Data.Real.Basic
```

```
variable {c : ℝ}
```

```
open Function
```

```
-- 1ª demostración
```

```
example : Surjective (fun x ↦ x + c) :=
by
```

```
  intro x
```

```
  --  $x : \mathbb{R}$ 
```

```
  --  $\vdash \exists a, (fun x \Rightarrow x + c) a = x$ 
```

```
  use x - c
```

```
  --  $\vdash (fun x \Rightarrow x + c) (x - c) = x$ 
```

```
  dsimp
```

```
  --  $\vdash (x - c) + c = x$ 
```

```

exact sub_add_cancel x c

-- 2ª demostración
example : Surjective (fun x ↦ x + c) :=
by
  intro x
  -- x : ℝ
  -- ⊢ ∃ a, (fun x => x + c) a = x
  use x - c
  -- ⊢ (fun x => x + c) (x - c) = x
  change (x - c) + c = x
  -- ⊢ (x - c) + c = x
  exact sub_add_cancel x c

-- 3ª demostración
example : Surjective (fun x ↦ x + c) :=
by
  intro x
  -- x : ℝ
  -- ⊢ ∃ a, (fun x => x + c) a = x
  use x - c
  -- ⊢ (fun x => x + c) (x - c) = x
  exact sub_add_cancel x c

-- 4ª demostración
example : Surjective (fun x ↦ x + c) :=
fun x ↦ ⟨x - c, sub_add_cancel x c⟩

-- 5ª demostración
example : Surjective (fun x ↦ x + c) :=
fun x ↦ ⟨x - c, by ring⟩

-- 6ª demostración
example : Surjective (fun x ↦ x + c) :=
add_right_surjective c

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (sub_add_cancel a b : (a - b) + b = a)
-- #check (add_right_surjective c : Surjective (fun x ↦ x + c))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.7. Si $c \neq 0$ , entonces la función $(x \mapsto cx)$ es suprayectiva

```
-- Demostrar que si c es un número real no nulo, entonces la función
--   f(x) = c * x
-- es suprayectiva.
```

```
-- Demostración en lenguaje natural
```

```
-- =====

-- Hay que demostrar que
--    $(\forall x \in \mathbb{R})(\exists y \in \mathbb{R})[cy = x]$ 
-- Sea  $x \in \mathbb{R}$ . Entonces,  $y = x/c \in \mathbb{R}$  y
--    $cy = c(x/c)$ 
--    $= x$ 
```

```
-- Demostraciones con Lean4
```

```
-- =====
```

```
import Mathlib.Data.Real.Basic
```

```
variable {c : ℝ}
```

```
open Function
```

```
-- 1ª demostración
```

```
example
```

```
(h : c ≠ 0)
```

```
: Surjective (fun x ↦ c * x) :=
```

```
by
```

```
intro x
```

```
-- x : ℝ
```

```
--  $\vdash \exists a, (fun x \Rightarrow c * x) a = x$ 
```

```
use (x / c)
```

```
--  $\vdash (fun x \Rightarrow c * x) (x / c) = x$ 
```

```
dsimp
```

```
--  $\vdash c * (x / c) = x$ 
```

```
rw [mul_comm]
```

```
--  $\vdash (x / c) * c = x$ 
```

```
exact div_mul_cancel x h
```

```
-- 2ª demostración
```

```
example
```

```
(h : c ≠ 0)
```

```

: Surjective (fun x ↦ c * x) :=
by
  intro x
  -- x : ℝ
  -- ⊢ ∃ a, (fun x => c * x) a = x
  use (x / c)
  -- ⊢ (fun x => c * x) (x / c) = x
  exact mul_div_cancel' x h

-- 3ª demostración
example
  (h : c ≠ 0)
  : Surjective (fun x ↦ c * x) :=
fun x ↦ ⟨x / c, mul_div_cancel' x h⟩

-- 4ª demostración
example
  (h : c ≠ 0)
  : Surjective (fun x ↦ c * x) :=
mul_left_surjective₀ h

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (div_mul_cancel a : b ≠ 0 → (a / b) * b = a)
-- #check (mul_comm a b : a * b = b * a)
-- #check (mul_div_cancel' a : b ≠ 0 → b * (a / b) = a)
-- #check (mul_left_surjective₀ : c ≠ 0 → Surjective (fun x ↦ c * x))

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.8. Si $c \neq 0$ , entonces la función $(x \mapsto cx + d)$ es suprayectiva

```

-----
-- Demostrar que si c es un número real no nulo, entonces la función
--   f(x) = c * x + d
-- es suprayectiva.
-----

-- Demostración en lenguaje natural

```

```

-- =====

-- Hay que demostrar que
--  $(\forall x \in \mathbb{R})(\exists y \in \mathbb{R})[cy+d = x]$ 
-- Sea  $x \in \mathbb{R}$ . Entonces,  $y = (x-d)/c \in \mathbb{R}$  y
--  $cy = c((x-d)/c)+d$ 
--  $= (x-d)+d$ 
--  $= x$ 

-- Demostraciones con Lean4
-- =====

import Mathlib.Data.Real.Basic
variable {c d : ℝ}
open Function

-- 1ª demostración
-- =====

example
  (h : c ≠ 0)
  : Surjective (fun x ↦ c * x + d) :=
by
  intro x
  -- x : ℝ
  --  $\vdash \exists a, (fun x \Rightarrow c * x + d) a = x$ 
  use ((x - d) / c)
  --  $\vdash (fun x \Rightarrow c * x + d) ((x - d) / c) = x$ 
  dsimp
  --  $\vdash c * ((x - d) / c) + d = x$ 
  show c * ((x - d) / c) + d = x
  calc c * ((x - d) / c) + d
      = (x - d) + d := congrArg (. + d) (mul_div_cancel' (x - d) h)
      _ = x         := sub_add_cancel x d

-- 2ª demostración
-- =====

example
  (h : c ≠ 0)
  : Surjective (fun x ↦ c * x + d) :=
by
  intro x
  -- x : ℝ
  --  $\vdash \exists a, (fun x \Rightarrow c * x + d) a = x$ 

```

```

use ((x - d) / c)
--  $\vdash (\text{fun } x \Rightarrow c * x + d) ((x - d) / c) = x$ 
dsimp
--  $\vdash c * ((x - d) / c) + d = x$ 
simp [mul_div_cancel', h]

-- 3ª demostración
-- =====

example
  (h : c ≠ 0)
  : Surjective (fun x ↦ c * x + d) :=
by
  intro x
  --  $x : \mathbb{R}$ 
  --  $\vdash \exists a, (\text{fun } x \Rightarrow c * x + d) a = x$ 
  use ((x - d) / c)
  --  $\vdash (\text{fun } x \Rightarrow c * x + d) ((x - d) / c) = x$ 
  simp [mul_div_cancel', h]

-- 4ª demostración
-- =====

example
  (h : c ≠ 0)
  : Surjective (fun x ↦ c * x + d) :=
fun x ↦ ((x - d) / c, by simp [mul_div_cancel', h])

-- Lemas usados
-- =====

-- variable (a b : ℝ)
-- #check (mul_div_cancel' a : b ≠ 0 → b * (a / b) = a)
-- #check (sub_add_cancel a b : a - b + b = a)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.9. Si $f: \mathbb{R} \rightarrow \mathbb{R}$ es suprayectiva, entonces $\exists x \in \mathbb{R}$ tal que $f(x)^2 = 9$

```

-----
-- Demostrar que si  $f$  es una función suprayectiva de  $\mathbb{R}$  en  $\mathbb{R}$ ,
-- entonces existe un  $x$  tal que  $(f\ x)^2 = 9$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Al ser  $f$  suprayectiva, existe un  $y$  tal que  $f(y) = 3$ . Por tanto,
--  $f(y)^2 = 9$ .

-- Demostración con Lean9
-- =====

import Mathlib.Data.Real.Basic

open Function

example
  {f : ℝ → ℝ}
  (h : Surjective f)
  : ∃ x, (f x)^2 = 9 :=
by
  cases' h 3 with y hy
  -- y : ℝ
  -- hy : f y = 3
  use y
  -- ⊢ f y ^ 2 = 9
  rw [hy]
  -- ⊢ 3 ^ 2 = 9
  norm_num

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 13.10. La composición de funciones suprayectivas es suprayectiva

```

-----
-- Demostrar que la composición de funciones suprayectivas es
-- suprayectiva.
-----

-- Demostración en lenguaje natural

```



```

-- =====

-- Supongamos que  $f : A \rightarrow B$  y  $g : B \rightarrow C$  son suprayectivas. Tenemos que
-- demostrar que
--  $(\forall z \in C)(\exists x \in A)[g(f(x)) = z]$ 
-- Sea  $z \in C$ . Por ser  $g$  suprayectiva, existe un  $y \in B$  tal que
--  $g(y) = z$  (1)
-- Por ser  $f$  suprayectiva, existe un  $x \in A$  tal que
--  $f(x) = y$  (2)
-- Por tanto,
--  $g(f(x)) = g(y)$  [por (2)]
--  $= z$  [por (1)]

-- Demostraciones con lean4
-- =====

import Mathlib.Tactic
open Function
variable {α : Type _} {β : Type _} {γ : Type _}
variable {f : α → β} {g : β → γ}

-- 1ª demostración
example
  (hg : Surjective g)
  (hf : Surjective f)
  : Surjective (g ∘ f) :=
by
  intro z
  -- z : γ
  -- ⊢ ∃ a, (g ∘ f) a = z
  cases' hg z with y hy
  -- y : β
  -- hy : g y = z
  cases' hf y with x hx
  -- x : α
  -- hx : f x = y
  use x
  -- ⊢ (g ∘ f) x = z
  dsimp
  -- ⊢ g (f x) = z
  rw [hx]
  -- ⊢ g y = z
  exact hy

-- 2ª demostración

```

```

example
  (hg : Surjective g)
  (hf : Surjective f)
  : Surjective (g ∘ f) :=
by
  intro z
  -- z : γ
  -- ⊢ ∃ a, (g ∘ f) a = z
  cases' hg z with y hy
  -- y : β
  -- hy : g y = z
  cases' hf y with x hx
  -- x : α
  -- hx : f x = y
  use x
  -- ⊢ (g ∘ f) x = z
  dsimp
  -- ⊢ g (f x) = z
  rw [hx, hy]

-- 3ª demostración
example
  (hg : Surjective g)
  (hf : Surjective f)
  : Surjective (g ∘ f) :=
by
  intro z
  -- z : γ
  -- ⊢ ∃ a, (g ∘ f) a = z
  cases' hg z with y hy
  -- y : β
  -- hy : g y = z
  cases' hf y with x hx
  -- x : α
  -- hx : f x = y
  exact ⟨x, by dsimp ; rw [hx, hy]⟩

-- 4ª demostración
example
  (hg : Surjective g)
  (hf : Surjective f)
  : Surjective (g ∘ f) :=
by
  intro z
  -- z : γ

```

```
--  $\vdash \exists a, (g \circ f) a = z$ 
rcases hg z with ⟨y, hy : g y = z⟩
rcases hf y with ⟨x, hx : f x = y⟩
exact ⟨x, by dsimp ; rw [hx, hy]⟩

-- 5ª demostración
example
  (hg : Surjective g)
  (hf : Surjective f)
  : Surjective (g ∘ f) :=
  Surjective.comp hg hf

-- Lemas usados
-- =====

-- #check (Surjective.comp : Surjective g → Surjective f → Surjective (g ∘ f))
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)



# Capítulo 14

## Lógica

### 14.1. Si $\neg(\exists x)P(x)$ , entonces $(\forall x)\neg P(x)$

```
-- Demostrar que si  $\neg(\exists x)P(x)$ , entonces  $(\forall x)\neg P(x)$ .  
-- -----  
  
-- Demostración en lenguaje natural  
-- =====  
  
-- Sea y un elemento cualquiera. Tenemos que demostrar  $\neg P(y)$ . Para ello,  
-- supongamos que  $P(y)$ . Entonces,  $(\exists x)P(x)$  que es una contradicción con  
-- la hipótesis,  
  
-- Demostraciones con Lean4  
-- =====  
  
import Mathlib.Tactic  
variable {α : Type _}  
variable (P : α → Prop)  
  
-- 1ª demostración  
-- =====  
  
example  
  (h : ¬ ∃ x, P x)  
  : ∀ x, ¬ P x :=  
by  
  intros y h1  
  -- y : α  
  -- h1 : P x
```

```

--  $\vdash \text{False}$ 
apply h
--  $\vdash \exists x, P\ x$ 
existsi y
--  $\vdash P\ y$ 
exact h1

-- 2ª demostración
-- =====

example
  (h :  $\neg \exists x, P\ x$ )
  :  $\forall x, \neg P\ x :=$ 
by
  intros y h1
  --  $y : \alpha$ 
  --  $h1 : P\ x$ 
  --  $\vdash \text{False}$ 
  apply h
  --  $\vdash \exists x, P\ x$ 
  use y
  --  $\vdash P\ y$ 
  exact h1

-- 3ª demostración
-- =====

example
  (h :  $\neg \exists x, P\ x$ )
  :  $\forall x, \neg P\ x :=$ 
by
  intros y h1
  --  $y : \alpha$ 
  --  $h1 : P\ x$ 
  --  $\vdash \text{False}$ 
  apply h
  --  $\vdash \exists x, P\ x$ 
  exact ⟨y, h1⟩

-- 4ª demostración
-- =====

example
  (h :  $\neg \exists x, P\ x$ )
  :  $\forall x, \neg P\ x :=$ 

```

```

by
  intros y h1
  -- y :  $\alpha$ 
  -- h1 :  $P\ x$ 
  --  $\vdash \text{False}$ 
  exact h ⟨y, h1⟩

-- 5ª demostración
-- =====

example
  (h :  $\neg \exists x, P\ x$ )
  :  $\forall x, \neg P\ x :=$ 
fun y h1 ↦ h ⟨y, h1⟩

-- 6ª demostración
-- =====

example
  (h :  $\neg \exists x, P\ x$ )
  :  $\forall x, \neg P\ x :=$ 
by
  push_neg at h
  exact h

-- 7ª demostración
-- =====

example
  (h :  $\neg \exists x, P\ x$ )
  :  $\forall x, \neg P\ x :=$ 
not_exists.mp h

-- 8ª demostración
-- =====

example
  (h :  $\neg \exists x, P\ x$ )
  :  $\forall x, \neg P\ x :=$ 
by aesop

-- Lemas usados
-- =====

-- #check (not_exists :  $(\neg \exists x, P\ x) \leftrightarrow \forall (x : \alpha), \neg P\ x$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 14.2. Si $(\forall x)\neg P(x)$ , entonces $\neg(\exists x)P(x)$

```

-- -----
-- Demostrar que si  $(\forall x)\neg P(x)$ , entonces  $\neg(\exists x)P(x)$ .
-- -----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que  $(\exists x)P(x)$ . Sea  $y$  tal que  $P(y)$ . Puesto que  $(\forall x)\neg P(x)$ , se
-- tiene que  $\neg P(y)$  que es una contradicción con  $P(y)$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable {α : Type _}
variable (P : α → Prop)

-- 1ª demostración
-- =====

example
  (h : ∀ x, ¬ P x)
  : ¬ ∃ x, P x :=
by
  intro h1
  -- h1 : ∃ x, P x
  -- ⊢ False
  rcases h1 with ⟨y, hy : P y⟩
  have h2 : ¬P y := h y
  exact h2 hy

-- 2ª demostración
-- =====

example
  (h : ∀ x, ¬ P x)
  : ¬ ∃ x, P x :=
by
  intro h1

```



```

-- h1 :  $\exists x, P x$ 
--  $\vdash \text{False}$ 
rcases h1 with ⟨y, hy : P y⟩
exact (h y) hy

-- 3ª demostración
-- =====

example
  (h :  $\forall x, \neg P x$ )
  :  $\neg \exists x, P x :=$ 
by
  rintro ⟨y, hy : P y⟩
  exact (h y) hy

-- 4ª demostración
-- =====

example
  (h :  $\forall x, \neg P x$ )
  :  $\neg \exists x, P x :=$ 
fun ⟨y, hy⟩ ↦ (h y) hy

-- 5ª demostración
-- =====

example
  (h :  $\forall x, \neg P x$ )
  :  $\neg \exists x, P x :=$ 
not_exists_of_forall_not h

-- 6ª demostración
-- =====

example
  (h :  $\forall x, \neg P x$ )
  :  $\neg \exists x, P x :=$ 
by aesop

-- Lemas usados
-- =====

-- variable (q : Prop)
-- #check (not_exists_of_forall_not :  $(\forall x, P x \rightarrow q) \rightarrow (\exists x, P x) \rightarrow q$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

### 14.3. Si $\neg(\forall x)P(x)$ , entonces $(\exists x)\neg P(x)$

```

-- -----
-- Demostrar que si  $\neg(\forall x)P(x)$ , entonces  $(\exists x)\neg P(x)$ .
-- -----

-- Demostración en lenguaje natural
-- =====

-- Por reducción al absurdo, supongamos que  $\neg(\exists x)\neg P(x)$ . Para obtener una
-- contradicción, demostraremos la negación de la hipótesis; es decir,
-- que  $(\forall x)P(x)$ . Para ello, sea  $y$  un elemento cualquiera y tenemos que
-- demostrar  $P(y)$ . De nuevo, lo haremos por reducción al absurdo: Para
-- ello, supongamos que  $\neg P(y)$ . Entonces, se tiene que  $(\exists x)\neg P(x)$  en
-- contradicción con nuestro primer supuesto de  $\neg(\exists x)\neg P(x)$ .

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable {α : Type _}
variable (P : α → Prop)

-- 1ª demostración
-- =====

example
  (h : ¬ ∀ x, P x)
  : ∃ x, ¬ P x :=
by
  by_contra h1
  -- h1 : ¬∃ x, ¬P x
  -- ⊢ False
  apply h
  -- ⊢ ∀ (x : α), P x
  intro y
  -- y : α
  -- ⊢ P y
  show P y
  by_contra h2
  -- h2 : ¬P y

```

```

--  $\vdash \text{False}$ 
exact h1 ⟨y, h2⟩

-- 2ª demostración
-- =====

example
  (h :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
not_forall.mp h

-- 3ª demostración
-- =====

example
  (h :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by aesop

-- Lemas usados
-- =====

-- #check (not_forall :  $(\neg \forall x, P x) \leftrightarrow \exists x, \neg P x$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 14.4. Si $(\exists x)\neg P(x)$ , entonces $\neg(\forall x)P(x)$

```

-----
-- Demostrar que si  $(\exists x)\neg P(x)$ , entonces  $\neg(\forall x)P(x)$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos que  $(\forall x)P(x)$  y tenemos que demostrar una
-- contradicción. Por hipótesis,  $(\exists x)\neg P(x)$ . Sea y tal que
--  $\neg P(y)$ . Entonces, como  $(\forall x)P(x)$ , se tiene que  $P(y)$  que es una
-- contradicción con  $\neg P(y)$ .

-- Demostraciones con Lean4
-- =====

```

```

import Mathlib.Tactic
variable {α : Type _}
variable (P : α → Prop)

-- 1ª demostración
-- =====

example
  (h : ∃ x, ¬ P x)
  : ¬ ∀ x, P x :=
by
  intro h1
  -- h1 : ∀ (x : α), P x
  -- ⊢ False
  cases' h with y hy
  -- y : α
  -- hy : ¬P y
  apply hy
  -- ⊢ P y
  exact (h1 y)

-- 2ª demostración
-- =====

example
  (h : ∃ x, ¬ P x)
  : ¬ ∀ x, P x :=
by
  intro h1
  -- h1 : ∀ (x : α), P x
  -- ⊢ False
  rcases h with ⟨y, hy : ¬P y⟩
  apply hy
  -- ⊢ P y
  exact (h1 y)

-- 3ª demostración
-- =====

example
  (h : ∃ x, ¬ P x)
  : ¬ ∀ x, P x :=
by
  intro h1
  -- h1 : ∀ (x : α), P x

```

```

--  $\vdash \text{False}$ 
rcases h with ⟨y, hy :  $\neg P\ y$ ⟩
exact hy (h1 y)

-- 4ª demostración
-- =====

example
  (h :  $\exists x, \neg P\ x$ )
  :  $\neg \forall x, P\ x :=$ 
not_forall.mpr h

-- 5ª demostración
-- =====

example
  (h :  $\exists x, \neg P\ x$ )
  :  $\neg \forall x, P\ x :=$ 
not_forall_of_exists_not h

-- 6ª demostración
-- =====

example
  (h :  $\exists x, \neg P\ x$ )
  :  $\neg \forall x, P\ x :=$ 
by aesop

-- Lemas usados
-- =====

-- #check (not_forall : ( $\neg \forall x, P\ x$ )  $\leftrightarrow \exists x, \neg P\ x$ )
-- #check (not_forall_of_exists_not : ( $\exists x, \neg P\ x$ )  $\rightarrow \neg \forall x, P\ x$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 14.5. $\neg\neg P \rightarrow P$

```

-----
-- Demostrar que  $\neg\neg P \rightarrow P$ .
-----

-- Demostración en lenguaje natural

```

```

-- =====

-- Por reducción al absurdo. Supongamos  $\neg P$ . Entonces, tenemos una
-- contradicción con la hipótesis ( $\neg\neg P$ ).

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable (P : Prop)

-- 1ª demostración
-- =====

example
  (h :  $\neg\neg P$ )
  : P :=
by
  by_contra h1
  -- h1 :  $\neg P$ 
  --  $\vdash \text{False}$ 
  exact (h h1)

-- 2ª demostración
-- =====

example
  (h :  $\neg\neg P$ )
  : P :=
by_contra (fun h1 ↦ h h1)

-- 3ª demostración
-- =====

example
  (h :  $\neg\neg P$ )
  : P :=
-- not_not.mp h
of_not_not h

-- 4ª demostración
-- =====

example
  (h :  $\neg\neg P$ )

```

```

: P :=
by tauto

-- Lemas usados
-- =====

-- #check (of_not_not :  $\neg\neg P \rightarrow P$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 14.6. $P \rightarrow \neg\neg P$

```

-----
-- Demostrar que  $P \rightarrow \neg\neg P$ .
-----

-- Demostración en lenguaje natural
-- =====

-- Supongamos  $\neg P$ . Entonces, tenemos una contradicción con la hipótesis
-- ( $P$ ).

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable (P : Prop)

-- 1ª demostración
-- =====

example
  (h : P)
  :  $\neg\neg P$  :=
by
  intro h1
  -- h1 :  $\neg P$ 
  --  $\vdash \text{False}$ 
  exact (h1 h)

-- 2ª demostración
-- =====

```

```

example
  (h : P)
  :  $\neg\neg P$  :=
fun h1 ↦ h1 h

-- 3ª demostración
-- =====

example
  (h : P)
  :  $\neg\neg P$  :=
not_not_intro h

-- 4ª demostración
-- =====

example
  (h : P)
  :  $\neg\neg P$  :=
by tauto

-- Lemas usados
-- =====

-- #check (not_not_intro : P →  $\neg\neg P$ )

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)

## 14.7. $(P \rightarrow Q) \leftrightarrow \neg P \vee Q$

```

-- -----
-- Demostrar que
--    $(P \rightarrow Q) \leftrightarrow \neg P \vee Q$ 
-- -----

-- Demostración en lenguaje natural
-- =====

-- Demostraremos cada una de las implicaciones.
--
-- ( $\Rightarrow$ ) Supongamos que  $P \rightarrow Q$ . Distinguimos dos subcasos según el valor de
--  $P$ .
--

```



```

-- Primer subcaso: suponemos P. Entonces. tenemos Q (por  $P \rightarrow Q$ ) y. por
-- tanto,  $\neg P \vee Q$ .
--
-- Segundo subcaso: suponemos  $\neg P$ . Entonces. tenemos  $\neg P \vee Q$ .
--
-- ( $\Leftarrow$ ) Supongamos que  $\neg P \vee Q$  y P y tenemos que demostrar
-- Q. Distinguimos dos subcasos según  $\neg P \vee Q$ .
--
-- Primer subcaso: Suponemos  $\neg P$ . Entonces tenemos una contradicción con
-- P.
--
-- Segundo subcaso: Suponemos Q, que es lo que tenemos que demostrar.

-- Demostraciones con Lean4
-- =====

import Mathlib.Tactic
variable (P Q : Prop)

-- 1ª demostración
-- =====

example
  : (P → Q) ↔ ¬P ∨ Q :=
by
  constructor
  . --  $\vdash (P \rightarrow Q) \rightarrow \neg P \vee Q$ 
    intro h1
    --  $h1 : P \rightarrow Q$ 
    --  $\vdash \neg P \vee Q$ 
    by_cases h2 : P
    . --  $h2 : P$ 
      right
      --  $\vdash Q$ 
      apply h1
      --  $\vdash P$ 
      exact h2
    . --  $h2 : \neg P$ 
      left
      --  $\vdash \neg P$ 
      exact h2
  . --  $\vdash \neg P \vee Q \rightarrow P \rightarrow Q$ 
    intros h3 h4
    --  $h3 : \neg P \vee Q$ 
    --  $h4 : P$ 

```

```

--  $\vdash Q$ 
rcases h3 with h3a | h3b
. --  $h : \neg P$ 
  exact absurd h4 h3a
. --  $h : Q$ 
  exact h3b
done

```

```

-- 2ª demostración
-- =====

```

```

example
  :  $(P \rightarrow Q) \leftrightarrow \neg P \vee Q :=$ 
by
  constructor
  . --  $\vdash (P \rightarrow Q) \rightarrow \neg P \vee Q$ 
    intro h1
    --  $h1 : P \rightarrow Q$ 
    --  $\vdash \neg P \vee Q$ 
    by_cases h2: P
    . --  $h2 : P$ 
      right
      --  $\vdash Q$ 
      exact h1 h2
    . --  $h2 : \neg P$ 
      left
      --  $\vdash \neg P$ 
      exact h2
  . --  $\vdash \neg P \vee Q \rightarrow P \rightarrow Q$ 
    intros h3 h4
    --  $h3 : \neg P \vee Q$ 
    --  $h4 : P$ 
    --  $\vdash Q$ 
    cases h3
    . --  $h : \neg P$ 
      contradiction
    . --  $h : Q$ 
      assumption
done

```

```

-- 3ª demostración
-- =====

```

```

example
  (P Q : Prop)

```

```
    : (P → Q) ↔ ¬P ∨ Q :=
imp_iff_not_or

-- 4ª demostración
-- =====

example
  (P Q : Prop)
  : (P → Q) ↔ ¬P ∨ Q :=
by tauto
```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)



# Capítulo 15

## Límites de sucesiones

### 15.1. La sucesión constante $s_n = c$ converge a $c$

```
-- -----  
-- Demostrar que, para todo  $a \in \mathbb{R}$ , la sucesión constante  
--  $s(n) = a$   
-- converge a  $a$ .  
-- -----  
  
-- Demostración en lenguaje natural  
-- =====  
  
-- Tenemos que demostrar que para cada  $\varepsilon \in \mathbb{R}$  tal que  $\varepsilon > 0$ , existe un  
--  $N \in \mathbb{N}$ , tal que  $(\forall n \in \mathbb{N})[n \geq N \rightarrow |s(n) - a| < \varepsilon]$ . Basta tomar  $N$  como  
--  $0$ , ya que para todo  $n \geq N$  se tiene  
--  $|s(n) - a| = |a - a|$   
--  $= |0|$   
--  $= 0$   
--  $< \varepsilon$   
  
-- Demostraciones con Lean4  
-- =====  
  
import Mathlib.Data.Real.Basic  
  
def limite (s :  $\mathbb{N} \rightarrow \mathbb{R}$ ) (a :  $\mathbb{R}$ ) :=  
   $\forall \varepsilon > 0, \exists N, \forall n \geq N, |s\ n - a| < \varepsilon$   
  
-- 1ª demostración
```

```

-- =====

example : limite (fun _ : ℕ → c) c :=
by
  intros ε hε
  -- ε : ℝ
  -- hε : ε > 0
  -- ⊢ ∃ N, ∀ (n : ℕ), n ≥ N → |(fun _ => c) n - c| < ε
  use 0
  -- ⊢ ∀ (n : ℕ), n ≥ 0 → |(fun _ => c) n - c| < ε
  intros n _hn
  -- n : ℕ
  -- hn : n ≥ 0
  -- ⊢ |(fun _ => c) n - c| < ε
  show |(fun _ => c) n - c| < ε
  calc |(fun _ => c) n - c| = |c - c| := by dsimp
                                _ = |0| := by {congr ; exact sub_self c}
                                _ = 0 := abs_zero
                                _ < ε := hε

-- 2ª demostración
-- =====

example : limite (fun _ : ℕ → c) c :=
by
  intros ε hε
  -- ε : ℝ
  -- hε : ε > 0
  -- ⊢ ∃ N, ∀ (n : ℕ), n ≥ N → |(fun _ => c) n - c| < ε
  use 0
  -- ⊢ ∀ (n : ℕ), n ≥ 0 → |(fun _ => c) n - c| < ε
  intros n _hn
  -- n : ℕ
  -- hn : n ≥ 0
  -- ⊢ |(fun _ => c) n - c| < ε
  show |(fun _ => c) n - c| < ε
  calc |(fun _ => c) n - c| = 0 := by simp
                                _ < ε := hε

-- 3ª demostración
-- =====

example : limite (fun _ : ℕ → c) c :=
by
  intros ε hε

```

```

--  $\varepsilon : \mathbb{R}$ 
--  $h\varepsilon : \varepsilon > 0$ 
--  $\vdash \exists N, \forall (n : \mathbb{N}), n \geq N \rightarrow |(fun \_ \Rightarrow c) n - c| < \varepsilon$ 
aesop

-- 4ª demostración
-- =====

example : limite (fun _ :  $\mathbb{N} \rightarrow c$ ) c :=
by
  intros  $\varepsilon$  h $\varepsilon$ 
  --  $\varepsilon : \mathbb{R}$ 
  --  $h\varepsilon : \varepsilon > 0$ 
  --  $\vdash \exists N, \forall (n : \mathbb{N}), n \geq N \rightarrow |(fun \_ \Rightarrow c) n - c| < \varepsilon$ 
  aesop

-- 5ª demostración
-- =====

example : limite (fun _ :  $\mathbb{N} \rightarrow c$ ) c :=
  fun  $\varepsilon$  h $\varepsilon$   $\rightarrow$  by aesop

-- Lemas usados
-- =====

-- #check (sub_self a : a - a = 0)

```

Se puede interactuar con las pruebas anteriores en [Lean 4 Web](#)





# Bibliografía

- [1] J. A. Alonso. [Lean para matemáticos](#) <sup>1</sup>, 2021.
- [2] J. A. Alonso. [Matemáticas en Lean](#) <sup>2</sup>, 2021.
- [3] J. A. Alonso. [DAO \(Demostración Asistida por Ordenador\) con Lean](#) <sup>3</sup>, 2021.
- [4] J. A. Alonso. [Calculus \(Vol. 1: Demostraciones con Isabelle/HOL y Lean3\)](#) <sup>4</sup>, 2021.
- [5] J. Avigad, L. de Moura, and S. Kong. [Theorem Proving in Lean4](#) <sup>5</sup>, 2021.
- [6] J. Avigad, G. Ebner, and S. Ullrich. [The Lean4 Manual](#) <sup>6</sup>, 2021.
- [7] J. Avigad, M. J. H. Heule, and W. Nawrocki. [Logic and mechanized reasoning](#) <sup>7</sup>, 2023.
- [8] J. Avigad, R. Y. Lewis, and F. van Doorn. [Logic and proof](#) <sup>8</sup>, 2021.
- [9] J. Avigad and P. Massot. [Mathematics in Lean](#) <sup>9</sup>, 2023.
- [10] A. Baanen, A. Bentkamp, J. Blanchette, J. Hölzl, and J. Limperg. [The Hitchhiker's Guide to Logical Verification](#) <sup>10</sup>, 2020.

---

<sup>1</sup>[https://github.com/jaalonso/Lean\\_para\\_matematicos](https://github.com/jaalonso/Lean_para_matematicos)

<sup>2</sup>[https://github.com/jaalonso/Matematicas\\_en\\_Lean](https://github.com/jaalonso/Matematicas_en_Lean)

<sup>3</sup>[https://raw.githubusercontent.com/jaalonso/DAO\\_con\\_Lean/master/DAO\\_con\\_Lean.pdf](https://raw.githubusercontent.com/jaalonso/DAO_con_Lean/master/DAO_con_Lean.pdf)

<sup>4</sup><https://raw.githubusercontent.com/jaalonso/Calculus/master/Calculus.pdf>

<sup>5</sup>[https://leanprover.github.io/theorem\\_proving\\_in\\_lean4/](https://leanprover.github.io/theorem_proving_in_lean4/)

<sup>6</sup><https://leanprover.github.io/lean4/doc/>

<sup>7</sup>[https://avigad.github.io/lamr/logic\\_and\\_mechanized\\_reasoning.pdf](https://avigad.github.io/lamr/logic_and_mechanized_reasoning.pdf)

<sup>8</sup>[https://leanprover.github.io/logic\\_and\\_proof/logic\\_and\\_proof.pdf](https://leanprover.github.io/logic_and_proof/logic_and_proof.pdf)

<sup>9</sup>[https://leanprover-community.github.io/mathematics\\_in\\_lean/](https://leanprover-community.github.io/mathematics_in_lean/)

<sup>10</sup>[https://raw.githubusercontent.com/blanchette/logical\\_verification\\_2020/master/hitchhikers\\_guide.pdf](https://raw.githubusercontent.com/blanchette/logical_verification_2020/master/hitchhikers_guide.pdf)

- [11] M. Ballard. [Transition to advanced mathematics \(Thinking and communicating like a mathematician\)](#) <sup>11</sup>.
- [12] K. Buzzard. [Sets and logic \(in Lean\)](#) <sup>12</sup>.
- [13] K. Buzzard. [Functions and relations \(in Lean\)](#) <sup>13</sup>.
- [14] K. Buzzard. [Course on formalising mathematics](#) <sup>14</sup>, 2021.
- [15] K. Buzzard. [Course on formalising mathematics](#) <sup>15</sup>, 2023.
- [16] K. Buzzard and M. Pedramfar. [The Natural Number Game, version 1.3.3](#) <sup>16</sup>.
- [17] D. T. Christiansen. [Functional programming in Lean](#) <sup>17</sup>, 2023.
- [18] M. Community. [Undergraduate mathematics in mathlib](#) <sup>18</sup>.
- [19] M. Dvořák. [Lean 4 Cheatsheet](#) <sup>19</sup>.
- [20] S. Hazratpour. [Introduction to proofs](#) <sup>20</sup>, 2022.
- [21] S. Hazratpour. [Introduction to proofs with Lean proof assistant](#) <sup>21</sup>, 2022.
- [22] R. Lewis. [Formal proof and verification, 2022](#) <sup>22</sup>, 2022.
- [23] R. Lewis. [Discrete structures and probability](#) <sup>23</sup>, 2023.
- [24] C. Löh. [Exploring formalisation \(A primer in human-readable mathematics in Lean 3 with examples from simplicial topology\)](#) <sup>24</sup>, 2022.
- [25] H. Macbeth. [The mechanics of proof](#) <sup>25</sup>, 2023.

---

<sup>11</sup><https://300.f22.matthewrobertballard.com/>

<sup>12</sup>[https://www.ma.imperial.ac.uk/~buzzard/M4000x\\_html/M40001/M40001\\_C1.html](https://www.ma.imperial.ac.uk/~buzzard/M4000x_html/M40001/M40001_C1.html)

<sup>13</sup>[https://www.ma.imperial.ac.uk/~buzzard/M4000x\\_html/M40001/M40001\\_C2.html](https://www.ma.imperial.ac.uk/~buzzard/M4000x_html/M40001/M40001_C2.html)

<sup>14</sup><https://github.com/ImperialCollegeLondon/formalising-mathematics>

<sup>15</sup><https://github.com/ImperialCollegeLondon/formalising-mathematics-2023>

<sup>16</sup>[https://www.ma.imperial.ac.uk/~buzzard/xena/natural\\_number\\_game/](https://www.ma.imperial.ac.uk/~buzzard/xena/natural_number_game/)

<sup>17</sup>[https://leanprover.github.io/functional\\_programming\\_in\\_lean/](https://leanprover.github.io/functional_programming_in_lean/)

<sup>18</sup><https://leanprover-community.github.io/undergrad.html>

<sup>19</sup><https://raw.githubusercontent.com/madvorak/lean4-cheatsheet/main/lean-tactics.pdf>

<sup>20</sup><https://introproofs.github.io/s22/>

<sup>21</sup><https://sinhp.github.io/teaching/2022-introduction-to-proofs-with-Lean>

<sup>22</sup><https://github.com/BrownCS1951x/fpv2022>

<sup>23</sup><https://github.com/brown-cs22/CS22-Lean-2023>

<sup>24</sup><https://loeh.app.uni-regensburg.de/mapa/main.pdf>

<sup>25</sup><https://hrmacbeth.github.io/math2001/index.html>

- [26] P. Massot. [Introduction aux mathématiques formalisées](#) <sup>26</sup>.
- [27] F. L. Roux. [Code Lean contenant les preuves d'un cours standard sur les espaces métriques](#) <sup>27</sup>, 2020.
- [28] W. Schulze. [Learning LeanProver](#) <sup>28</sup>.
- [29] Varios. [LFTCM 2020: Lean for the Curious Mathematician 2020](#) <sup>29</sup>.
- [30] D. J. Velleman. [How to prove it with Lean](#) <sup>30</sup>.

---

<sup>26</sup><https://www.imo.universite-paris-saclay.fr/~pmassot/enseignement/math114/>

<sup>27</sup>[https://github.com/FredericLeRoux/LEAN\\_ESPACES\\_METRIQUES](https://github.com/FredericLeRoux/LEAN_ESPACES_METRIQUES)

<sup>28</sup>[https://youtube.com/playlist?list=PLYwF9EIrl42RFQgbmcR\\_LSCWRIx2WKbXs](https://youtube.com/playlist?list=PLYwF9EIrl42RFQgbmcR_LSCWRIx2WKbXs)

<sup>29</sup><https://leanprover-community.github.io/lftcm2020/schedule.html>

<sup>30</sup><https://djvelleman.github.io/HTPIwL/>



# Lemas usados

```
import Mathlib.Algebra.Group.Basic
import Mathlib.Algebra.Order.Ring.Defs
import Mathlib.Algebra.Ring.Defs
import Mathlib.Analysis.SpecialFunctions.Log.Basic
import Mathlib.Data.Real.Basic
import Mathlib.Order.Lattice
import Mathlib.Topology.MetricSpace.Basic

-- Números naturales
-- =====

section naturales
variable (x y z k m n : ℕ)
open Nat
#check (_root_.dvd_antisymm : m | n → n | m → m = n)
#check (dvd_add : x | y → x | z → x | y + z)
#check (dvd_factorial : 0 < k → k ≤ n → k | n !)
#check (dvd_gcd : k | m → k | n → k | gcd m n)
#check (dvd_mul_left x y : x | y * x)
#check (dvd_mul_of_dvd_left : x | y → ∀ (c : ℕ), x | y * c)
#check (dvd_mul_of_dvd_right : x | y → ∀ (c : ℕ), x | c * y)
#check (dvd_mul_right x y : x | x * y)
#check (dvd_trans : x | y → y | z → x | z)
#check (Dvd.intro k : m * k = n → m | n)
#check (factorial_pos n : n ! > 0)
#check (gcd_comm m n : gcd m n = gcd n m)
#check (gcd_dvd_left m n : gcd m n | m)
#check (gcd_dvd_right m n : gcd m n | n)
#check (minFac_dvd n : minFac n | n)
#check (minFac_pos n : 0 < minFac n)
#check (minFac_prime : n ≠ 1 → Nat.Prime (minFac n))
#check (Nat.dvd_add_iff_right : k | m → (k | n ↔ k | m + n))
#check (Nat.dvd_one : n | 1 ↔ n = 1)
#check (Nat.lt_add_of_pos_left : 0 < k → n < k + n)
```

```

#check (Nat.ne_of_gt : k < n → n ≠ k)
#check (Nat.Prime.not_dvd_one : Nat.Prime n → ¬n | 1)
end naturales

-- Números reales
-- =====

section reales
open Real
variable (a b c d x y : ℝ)
#check (Left.self_le_neg : x ≤ 0 → x ≤ -x)
#check (abs_add a b : |a + b| ≤ |a| + |b|)
#check (abs_le' : |a| ≤ b ↔ a ≤ b ∧ -a ≤ b)
#check (abs_lt : |a| < b ↔ -b < a ∧ a < b)
#check (abs_mul a b : |a * b| = |a| * |b|)
#check (abs_nonneg a : 0 ≤ |a|)
#check (abs_of_neg : x < 0 → |x| = -x)
#check (abs_of_nonneg : 0 ≤ x → |x| = x)
#check (abs_sub_abs_le_abs_sub a b : |a| - |b| ≤ |a - b|)
#check (add_le_add : a ≤ b → c ≤ d → a + c ≤ b + d)
#check (add_le_add_left : b ≤ c → ∀ (a : ℝ), a + b ≤ a + c)
#check (add_le_add_right : b ≤ c → ∀ (a : ℝ), b + a ≤ c + a)
#check (add_lt_add_of_le_of_lt : a ≤ b → c < d → a + c < b + d)
#check (add_lt_add_of_lt_of_le : a < b → c ≤ d → a + c < b + d)
#check (add_lt_add_right : b < c → ∀ (a : ℝ), b + a < c + a)
#check (add_neg_le_iff_le_add : a - b ≤ c ↔ a ≤ c + b)
#check (add_pos : 0 < a → 0 < b → 0 < a + b)
#check (add_sub_cancel a b : a + b - b = a)
#check (div_mul_cancel a : b ≠ 0 → (a / b) * b = a)
#check (eq_neg_of_add_eq_zero_left : x + y = 0 → x = -y)
#check (eq_zero_or_eq_zero_of_mul_eq_zero : x * y = 0 → x = 0 ∨ y = 0)
#check (exp_le_exp : exp a ≤ exp b ↔ a ≤ b)
#check (exp_lt_exp : exp a < exp b ↔ a < b)
#check (exp_pos a : 0 < exp a)
#check (half_lt_self : 0 < a → a / 2 < a)
#check (half_pos : 0 < a → 0 < a / 2)
#check (le_abs_self x : x ≤ |x|)
#check (le_add_of_nonneg_right : 0 ≤ b → a ≤ a + b)
#check (le_antisymm : a ≤ b → b ≤ a → a = b)
#check (le_div_iff : 0 < c → (a ≤ b / c ↔ a * c ≤ b))
#check (le_max_left a b : a ≤ max a b)
#check (le_max_right a b : b ≤ max a b)
#check (le_min : c ≤ a → c ≤ b → c ≤ min a b)
#check (le_neg_self_iff : x ≤ -x ↔ x ≤ 0)
#check (le_of_eq : a = b → a ≤ b)

```

```

#check (le_of_lt : x < y → x ≤ y)
#check (le_of_not_ge : ¬x ≥ y → x ≤ y)
#check (le_of_not_gt : ¬a > b → a ≤ b)
#check (le_or_gt x y : x ≤ y ∨ x > y)
#check (le_refl a : a ≤ a)
#check (log_le_log' : 0 < a → a ≤ b → log a ≤ log b)
#check (lt_abs : x < |y| ↔ x < y ∨ x < -y)
#check (lt_asymm : a < b → ¬b < a)
#check (lt_iff_le_and_ne : a < b ↔ a ≤ b ∧ a ≠ b)
#check (lt_iff_le_not_le : a < b ↔ a ≤ b ∧ ¬b ≤ a)
#check (lt_irrefl a : ¬a < a)
#check (lt_neg : a < -b ↔ b < -a)
#check (lt_of_le_of_ne : a ≤ b → a ≠ b → a < b)
#check (lt_of_lt_of_le : a < b → b ≤ c → a < c)
#check (lt_of_le_of_lt : a ≤ b → b < c → a < c)
#check (lt_of_le_of_ne : a ≤ b → a ≠ b → a < b)
#check (lt_of_not_ge : ¬a ≥ b → a < b)
#check (lt_of_not_le : ¬b ≤ a → a < b)
#check (lt_trans : a < b → b < c → a < c)
#check (lt_trichotomy a b : a < b ∨ a = b ∨ b < a)
#check (max_comm a b : max a b = max b a)
#check (max_le : a ≤ c → b ≤ c → max a b ≤ c)
#check (min_add_add_right a b c : min (a + c) (b + c) = min a b + c)
#check (min_assoc a b c : min (min a b) c = min a (min b c))
#check (min_comm a b : min a b = min b a)
#check (min_eq_left : a ≤ b → min a b = a)
#check (min_eq_right : b ≤ a → min a b = b)
#check (min_le_left a b : min a b ≤ a)
#check (min_le_right a b : min a b ≤ b)
#check (mul_comm a b : a * b = b * a)
#check (mul_div_cancel' a : b ≠ 0 → b * (a / b) = a)
#check (mul_le_mul : a ≤ b → c ≤ d → 0 ≤ c → 0 ≤ b → a * c ≤ b * d)
#check (mul_le_mul_right : 0 < a → (b * a ≤ c * a ↔ b ≤ c))
#check (mul_left_cancel : a ≠ 0 → a * b = a * c → b = c)
#check (mul_lt_mul_left : 0 < a → (a * b < a * c ↔ b < c))
#check (mul_lt_mul_right : 0 < a → (b * a < c * a ↔ b < c))
#check (mul_neg a b : a * -b = -(a * b))
#check (mul_right_inj' : a ≠ 0 → (a * b = a * c ↔ b = c))
#check (mul_sub a b c : a * (b - c) = a * b - a * c)
#check (mul_two a : a * 2 = a + a)
#check (ne_comm : a ≠ b ↔ b ≠ a)
#check (neg_add x y : -(x + y) = -x + -y)
#check (neg_add_self a : -a + a = 0)
#check (neg_le_abs_self x : -x ≤ |x|)
#check (neg_mul_neg a b : -a * -b = a * b)

```

```

#check (nonneg_of_mul_nonneg_left :  $0 \leq a * b \rightarrow 0 < b \rightarrow 0 \leq a$ )
#check (not_lt_of_ge :  $a \geq b \rightarrow \neg a < b$ )
#check (pow_eq_zero :  $\forall \{n : \mathbb{N}\}, a^n = 0 \rightarrow a = 0$ )
#check (pow_two a :  $a^2 = a * a$ )
#check (pow_two_nonneg a :  $0 \leq a^2$ )
#check (sq_eq_one_iff :  $x^2 = 1 \leftrightarrow x = 1 \vee x = -1$ )
#check (sq_eq_sq_iff_eq_or_eq_neg :  $x^2 = y^2 \leftrightarrow x = y \vee x = -y$ )
#check (sq_nonneg a :  $0 \leq a^2$ )
#check (sub_add_cancel a b :  $a - b + b = a$ )
#check (sub_eq_zero :  $x - y = 0 \leftrightarrow x = y$ )
#check (sub_le_sub_left :  $a \leq b \rightarrow \forall (c : \mathbb{R}), c - b \leq c - a$ )
#check (sub_le_sub_right :  $a \leq b \rightarrow \forall (c : \mathbb{R}), a - c \leq b - c$ )
#check (sub_sq a b :  $(a - b)^2 = a^2 - 2 * a * b + b^2$ )
#check (two_mul a :  $2 * a = a + a$ )
#check (two_mul_le_add_sq a b :  $2 * a * b \leq a^2 + b^2$ )
#check (zero_lt_one :  $0 < 1$ )
#check (zero_lt_two :  $0 < 2$ )
end reales

-- Anillos
-- =====

section anillos
variable {R : Type _} [Ring R]
variable (a b c : R)
#check (add_assoc a b c :  $(a + b) + c = a + (b + c)$ )
#check (add_comm a b :  $a + b = b + a$ )
#check (add_eq_zero_iff_eq_neg :  $a + b = 0 \leftrightarrow a = -b$ )
#check (add_left_cancel :  $a + b = a + c \rightarrow b = c$ )
#check (add_left_neg a :  $-a + a = 0$ )
#check (add_mul a b c :  $(a + b) * c = a * c + b * c$ )
#check (add_neg_cancel_right a b :  $(a + b) + -b = a$ )
#check (add_neg_self a :  $a + -a = 0$ )
#check (add_right_cancel :  $a + b = c + b \rightarrow a = c$ )
#check (add_right_neg a :  $a + -a = 0$ )
#check (add_zero a :  $a + 0 = a$ )
#check (mul_add a b c :  $a * (b + c) = a * b + a * c$ )
#check (mul_zero a :  $a * 0 = 0$ )
#check (neg_add_cancel_left a b :  $-a + (a + b) = b$ )
#check (neg_eq_iff_add_eq_zero :  $-a = b \leftrightarrow a + b = 0$ )
#check (neg_eq_of_add_eq_zero_left :  $a + b = 0 \rightarrow -b = a$ )
#check (neg_eq_of_add_eq_zero_right :  $a + b = 0 \rightarrow -a = b$ )
#check (neg_neg a :  $-(-a) = a$ )
#check (neg_zero :  $-0 = 0$ )
#check (one_add_one_eq_two :  $(1 : R) + 1 = 2$ )

```



```

#check (sub_add_cancel a b : a - b + b = a)
#check (sub_eq_add_neg a b : a - b = a + -b)
#check (sub_mul a b c : (a - b) * c = a * c - b * c)
#check (sub_self a : a - a = 0)
#check (two_mul a : 2 * a = a + a)
#check (zero_add a : 0 + a = a)
#check (zero_mul a : 0 * a = 0)
end anillos

-- Grupos
-- =====

section grupos
variable {G : Type _} [Group G]
variable (a b c : G)
#check (inv_eq_of_mul_eq_one_right : a * b = 1 → a-1 = b)
#check (mul_assoc a b c : (a * b) * c = a * (b * c))
#check (mul_inv_self a : a * a-1 = 1)
#check (mul_inv_rev a b : (a * b)-1 = b-1 * a-1)
#check (mul_left_inv a : a-1 * a = 1)
#check (mul_one a : a * 1 = a)
#check (mul_right_inv a : a * a-1 = 1)
#check (one_mul a : 1 * a = a)
end grupos

-- Retículos
-- =====

section reticulos
variable {α : Type _} [Lattice α]
variable (x y z : α)
#check (inf_assoc : (x ⊓ y) ⊓ z = x ⊓ (y ⊓ z))
#check (inf_comm : x ⊓ y = y ⊓ x)
#check (inf_le_left : x ⊓ y ≤ x)
#check (inf_le_of_left_le : x ≤ z → x ⊓ y ≤ z)
#check (inf_le_of_right_le : y ≤ z → x ⊓ y ≤ z)
#check (inf_le_right : x ⊓ y ≤ y)
#check (inf_sup_self : x ⊓ (x ⊔ y) = x)
#check (le_antisymm : x ≤ y → y ≤ x → x = y)
#check (le_inf : z ≤ x → z ≤ y → z ≤ x ⊓ y)
#check (le_rfl : x ≤ x)
#check (le_sup_left : x ≤ x ⊔ y)
#check (le_sup_of_le_left : z ≤ x → z ≤ x ⊔ y)
#check (le_sup_of_le_right : z ≤ y → z ≤ x ⊔ y)
#check (le_sup_right : y ≤ x ⊔ y)

```

```

#check (le_trans :  $x \leq y \rightarrow y \leq z \rightarrow x \leq z$ )
#check (sup_assoc :  $(x \sqcup y) \sqcup z = x \sqcup (y \sqcup z)$ )
#check (sup_comm :  $x \sqcup y = y \sqcup x$ )
#check (sup_inf_self :  $x \sqcup (x \sqcap y) = x$ )
#check (sup_le :  $x \leq z \rightarrow y \leq z \rightarrow x \sqcup y \leq z$ )
end reticulos

-- AnillosOrdenados
-- =====

section AnillosOrdenados
variable {R : Type _} [StrictOrderedRing R]
variable (a b c : R)
#check (add_le_add_right :  $b \leq c \rightarrow \forall (a : R), b + a \leq c + a$ )
#check (mul_le_mul_of_nonneg_left :  $b \leq c \rightarrow 0 \leq a \rightarrow a * b \leq a * c$ )
#check (mul_le_mul_of_nonneg_right :  $a \leq b \rightarrow 0 \leq c \rightarrow a * c \leq b * c$ )
#check (mul_nonneg :  $0 \leq a \rightarrow 0 \leq b \rightarrow 0 \leq a * b$ )
#check (sub_le_sub_right :  $a \leq b \rightarrow \forall (c : R), a - c \leq b - c$ )
#check (sub_nonneg :  $0 \leq a - b \leftrightarrow b \leq a$ )
end AnillosOrdenados

-- Espacios métricos
-- =====

section EspacioMetrico
variable {X : Type _} [MetricSpace X]
variable (x y z : X)
#check (dist_comm x y :  $\text{dist } x \ y = \text{dist } y \ x$ )
#check (dist_nonneg :  $0 \leq \text{dist } x \ y$ )
#check (dist_self x :  $\text{dist } x \ x = 0$ )
#check (dist_triangle x y z :  $\text{dist } x \ z \leq \text{dist } x \ y + \text{dist } y \ z$ )
end EspacioMetrico

-- Conjuntos
-- =====

section Conjuntos
open Set
variable {α : Type _}
variable (r s t : Set α)
#check (Subset.trans :  $r \subseteq s \rightarrow s \subseteq t \rightarrow r \subseteq t$ )
end Conjuntos

-- Órdenes parciales
-- =====

```

```

section OrdenParcial
variable {α : Type _} [PartialOrder α]
variable (a b c : α)
#check (irrefl a : ¬a < a)
#check (le_trans : a ≤ b → b ≤ c → a ≤ c)
#check (lt_trans : a < b → b < c → a < c)
#check (monotone_const : Monotone fun _ : ℝ → c)
end OrdenParcial

-- Funciones
-- =====

section Funciones
open Function
variable {α : Type _} {β : Type _} {γ : Type _}
variable {f : α → β} {g : β → γ}
variable (c : ℝ)
#check (Injective.comp : Injective g → Injective f → Injective (g ∘ f))
#check (Surjective.comp : Surjective g → Surjective f → Surjective (g ∘ f))
#check (add_right_surjective c : Surjective (fun x ↦ x + c))
#check (mul_left_surjective₀ : c ≠ 0 → Surjective (fun x ↦ c * x))
end Funciones

-- Lógica
-- =====

section Logica
variable (p q : Prop)
variable {α : Type _}
variable (P : α → Prop)
#check (absurd : p → ¬p → q)
#check (forall_not_of_not_exists : (¬∃ x, P x) → ∀ x, ¬P x)
#check (not_exists : (¬∃ x, P x) ↔ ∀ (x : α), ¬P x)
#check (not_exists_of_forall_not : (∀ x, P x → q) → (∃ x, P x) → q)
#check (not_imp : ¬(p → q) ↔ p ∧ ¬q)
#check (not_forall : (¬∀ x, P x) ↔ ∃ x, ¬P x)
#check (not_forall_of_exists_not : (∃ x, ¬P x) → ¬∀ x, P x)
#check (not_not_intro : p → ¬¬p)
#check (of_not_not : ¬¬p → p)
#check (Or.inl : p → p ∨ q)
#check (Or.inr : q → p ∨ q)
end Logica

```